
Changelog for Elixir v1.17
    

This release includes type inference of patterns to provide warnings for an initial set of constructs (binaries, maps, and atoms) within the same function. It also includes a new Duration data type to interact with Calendar types, support for Erlang/OTP 27, and many other improvements.

  
    
  
  Warnings from gradual set-theoretic types


This release introduces gradual set-theoretic types to infer types from patterns and use them to type check programs, enabling the Elixir compiler to find faults and bugs in codebases without requiring changes to existing software. The underlying principles, theory, and roadmap of our work have been outlined in "The Design Principles of the Elixir Type System" by Giuseppe Castagna, Guillaume Duboc, José Valim.
At the moment, Elixir developers will interact with set-theoretic types only through warnings found by the type system. The current implementation models all data types in the language:
	binary(), integer(), float(), pid(), port(), reference() - these types are indivisible. This means both 1 and 13 get the same integer() type.

	atom() - it represents all atoms and it is divisible. For instance, the atom :foo and :hello_world are also valid (distinct) types.

	map() and structs - maps can be "closed" or "open". Closed maps only allow the specified keys, such as %{key: atom(), value: integer()}. Open maps support any other keys in addition to the ones listed and their definition starts with ..., such as %{..., key: atom(), value: integer()}. Structs are closed maps with the __struct__ key.

	tuple(), list(), and function() - currently they are modelled as indivisible types. The next Elixir versions will also introduce fine-grained support to them.


We focused on atoms and maps on this initial release as they are respectively the simplest and the most complex types representations, so we can stress the performance of the type system and quality of error messages. Modelling these types will also provide the most immediate benefits to Elixir developers. Assuming there is a variable named user, holding a %User{} struct with an address field, Elixir v1.17 will emit the following warnings at compile-time:
	Pattern matching against a map or a struct that does not have the given key, such as %{adress: ...} = user (notice address vs adress)

	Accessing a key on a map or a struct that does not have the given key, such as user.adress

	Invoking a function on non-modules, such as user.address()

	Capturing a function on non-modules, such as &user.address/0

	Attempting to invoke to call an anonymous function without an actual function, such as user.()

	Performing structural comparisons with structs, such as my_date < ~D[2010-04-17]

	Performing structural comparisons between non-overlapping types, such as integer >= string

	Building and pattern matching on binaries without the relevant specifiers, such as <<name>> (this warns because by default it expects an integer, it should have been <<name::binary>> instead)

	Attempting to rescue an undefined exception or a struct that is not an exception

	Accessing a field that is not defined in a rescued exception


These new warnings help Elixir developers find bugs earlier and give more confidence when refactoring code, especially around maps and structs. While some of these warnings were emitted in the past, they were discovered using syntax analysis. The new warnings are more reliable, precise, and with better error messages. Keep in mind, however, that the Elixir typechecker only infers types from patterns within the same function at the moment. Analysis from guards and across function boundaries will be added in future relases. For more details, see our new reference document on gradual set-theoretic types.
The type system was made possible thanks to a partnership between CNRS and Remote. The development work is currently sponsored by Fresha, Starfish*, and Dashbit.

  
    
  
  Erlang/OTP support


This release adds support for Erlang/OTP 27 and drops support for Erlang/OTP 24. We recommend Elixir developers to migrate to Erlang/OTP 26 or later, especially on Windows. Support for WERL (a graphical user interface for the Erlang terminal on Windows) will be removed in Elixir v1.18.

  
    
  
  Adding Duration and shift/2 functions


Elixir introduces the Duration data type and APIs to shift dates, times, and date times by a given duration, considering different calendars and time zones.
iex> Date.shift(~D[2016-01-31], month: 2)
~D[2016-03-31]
Note the operation is called shift (instead of add) since working with durations does not obey properties such as associativity. For instance, adding one month and then one month does not give the same result as adding two months:
iex> ~D[2016-01-31] |> Date.shift(month: 1) |> Date.shift(month: 1)
~D[2016-03-29]
Still, durations are essential for building intervals, recurring events, and modelling scheduling complexities found in the world around us. For DateTimes, Elixir will correctly deal with time zone changes (such as Daylight Saving Time), but provisions are also available in case you want to surface conflicts (for example, you shifted to a wall clock that does not exist, because the clock has been moved forward by one hour). See DateTime.shift/2 for examples.
Finally, a new Kernel.to_timeout/1 function has been added, which helps developers normalize durations and integers to a timeout used by Process APIs. For example, to send a message after one hour, one can now write:
Process.send_after(pid, :wake_up, to_timeout(hour: 1))

  
    
  
  v1.17.3 (2024-09-18)



  
    
  
  1. Bug fixes


Elixir
	[Duration] Fix parsing of fractional durations with non-positive seconds
	[Kernel] Do not attempt to group module warnings when they have a large context

IEx
	[IEx.Helpers] Properly reconsolidate protocols on recompile

Mix
	[mix compile.elixir] Do not verify modules twice
	[mix xref] Respect the --label option on stats and cycles


  
    
  
  v1.17.2 (2024-07-06)



  
    
  
  1. Bug fixes


Logger
	[Logger.Translator] Fix logger crash when :gen_statem's format_status/2 returns non-tuple

Mix
	[mix deps.get] Fix regression when fetching a git repository with a :ref
	[mix release] Validate RELEASE_MODE and set ERRORLEVEL on .bat scripts
	[mix release] Fix invalid example in code comment inside the generated vm.args.eex


  
    
  
  v1.17.1 (2024-06-18)



  
    
  
  1. Enhancements


Mix
	[mix compile.elixir] Do not run fixpoint computation on runtime dependencies. This should considerably improve compilation times for large projects when changing only one or two files


  
    
  
  2. Bug fixes


EEx
	[EEx] Do not warn for assignment with blocks in EEx

Elixir
	[Kernel] Fix bug when using pinned variables inside with's else patterns
	[Kernel] Fix Dialyzer error when with else clause is calling a no_return function

ExUnit
	[ExUnit] Do not alternative sync/async suites on --repeat-until-failure


  
    
  
  v1.17.0 (2024-06-12)



  
    
  
  1. Enhancements


Elixir
	[Access] Add Access.find/1 that mirrors Enum.find/2
	[Code] Support cursor inside fn/rescue/catch/else/after inside Code.Fragment.container_cursor_to_quoted/2
	[Date] Add Date.shift/2 to shift dates with duration and calendar-specific semantics
	[Date] Allow Date to accept years outside of -9999..9999 range
	[DateTime] Add DateTime.shift/2 to shift datetimes with duration and calendar-specific semantics
	[Duration] Add a new Duration data type
	[GenServer] Add GenServer.format_status/1 callback
	[Kernel] Add Kernel.get_in/1 with safe nil-handling for access and structs
	[Kernel] Add Kernel.is_non_struct_map/1 guard
	[Kernel] Add Kernel.to_timeout/1
	[Kernel] Emit warnings for undefined functions from modules defined within the same context as the caller code
	[Kernel] Support integers in uppercase sigils
	[Keyword] Add Keyword.intersect/2-3 to mirror the Map API
	[Macro] Add Macro.Env.define_alias/4, Macro.Env.define_import/4, Macro.Env.define_require/4, Macro.Env.expand_alias/4, Macro.Env.expand_import/5, and Macro.Env.expand_require/6 to aid the implementation of language servers and embedded languages
	[NaiveDateTime] Add NaiveDateTime.shift/2 to shift naive datetimes with duration and calendar-specific semantics
	[Process] Add Process.set_label/1
	[String] Add String.byte_slice/3 to slice a string to a maximum number of bytes while keeping it UTF-8 encoded
	[System] Support use_stdio: false in System.cmd/3 and System.shell/2
	[Time] Add Time.shift/2 to shift times with duration and calendar-specific semantics

ExUnit
	[ExUnit] Propagate the test process itself as a caller in start_supervised
	[ExUnit] Include max cases in ExUnit reports

IEx
	[IEx.Helpers] Warns if recompile was called and the current working directory changed
	[IEx.Helpers] Add c/0 as an alias to continue/0
	[IEx.Pry] Add IEx.Pry.annotate_quoted/3 to annotate a quoted expression with pry breakpoints

Logger
	[Logger] Format :gen_statem reports using Elixir data structures
	[Logger] Include process label in logger events

Mix
	[mix deps] Add :depth option to Mix.SCM.Git, thus supporting shallow clones of Git dependencies
	[mix deps] Warn if :optional is used in combination with :in_umbrella
	[mix deps.get] Do not add optional dependency requirements if its parent dep was skipped
	[mix deps.tree] Add --umbrella-only to mix deps.tree
	[mix profile.tprof] Add a new profiler, available on Erlang/OTP 27+, which can measure count, time, and heap usage
	[mix test] Add mix test --breakpoints that sets up a breakpoint before each test that will run
	[mix test] Add mix test --repeat-until-failure to rerun tests until a failure occurs
	[mix test] Add mix test --slowest-modules to print slowest modules based on all of the tests they hold
	[mix test] Generate cover HTML files in parallel


  
    
  
  2. Bug fixes


Elixir
	[bin/elixir.bat] Improve handling of quotes and exclamation marks in flags
	[Code] Address a bug where AST nodes for (a -> b) were not wrapped as part of the literal encoder
	[Kernel] Resolve inconsistencies of how .. and ... are handled at the AST level
	[Kernel] Fix parsing precedence of ambiguous operators followed by containers
	[Kernel] Do not expand code in quote bind_quoted: ... twice
	[Kernel] Respect :line property when :file is given as option to quote
	[Kernel] Do not crash on Macro.escape/2 when passing a quote triplet without valid meta
	[Kernel] Avoid double tracing events when capturing a function
	[Kernel] Fix a bug where captured arguments would conflict when a capture included a macro that also used captures
	[Module] Return default value in Module.get_attribute/3 for persisted attributes which have not yet been written to
	[String] Properly handle transpositions in jaro_distance. This will correct the distance result in certain cases

IEx
	[IEx.Helpers] Update the history size whenever history is pruned

Mix
	[mix deps] Fix error message for diverged SCM definition in sibling


  
    
  
  3. Soft deprecations (no warnings emitted)


Elixir
	[GenServer] Deprecate GenServer.format_status/2 callback to align with Erlang/OTP 25+

Mix
	[mix profile.cprof] Deprecated in favor of the new mix profile.tprof
	[mix profile.eprof] Deprecated in favor of the new mix profile.tprof


  
    
  
  4. Hard deprecations


Elixir
	[IO] Passing :all to IO.read/2 and IO.binread/2 is deprecated, pass :eof instead
	[Kernel] Single-quote charlists are deprecated, use ~c instead
	[Kernel] Deprecate escaping closing delimiter in uppercase sigils
	[Range] left..right without explicit steps inside patterns and guards is deprecated, write left..right//step instead
	[Range] Decreasing ranges, such as 10..1 without an explicit step is deprecated, write 10..1//-1 instead

ExUnit
	[ExUnit.Case] register_test/4 is deprecated in favor of register_test/6 for performance reasons


  
    
  
  v1.16


The CHANGELOG for v1.16 releases can be found in the v1.16 branch.


  

    
Introduction
    

Welcome!
This guide will teach you about Elixir fundamentals - the language syntax, how to define modules, the common data structures in the language, and more. This chapter will focus on ensuring that Elixir is installed and that you can successfully run Elixir's Interactive Shell, called IEx.
Let's get started.

  
    
  
  Installation


If you haven't yet installed Elixir, visit our installation page. Once you are done, you can run elixir --version to get the current Elixir version. The requirements for this guide are:
	Elixir 1.15.0 onwards
	Erlang/OTP 26 onwards

If you are looking for other resources for learning Elixir, you can also consult the learning page of the official website.

  
    
  
  Interactive mode


When you install Elixir, you will have three new command line executables: iex, elixir and elixirc.
For now, let's start by running iex (or iex.bat if you are on Windows PowerShell, where iex is a PowerShell command) which stands for Interactive Elixir. In interactive mode, we can type any Elixir expression and get its result. Let's warm up with some basic expressions.
Open up iex and type the following expressions:
Erlang/OTP 26 [64-bit] [smp:2:2] [...]

Interactive Elixir - press Ctrl+C to exit
iex(1)> 40 + 2
42
iex(2)> "hello" <> " world"
"hello world"
Please note that some details like version numbers may differ a bit in your session, that's not important. By executing the code above, you should evaluate expressions and see their results. To exit iex press Ctrl+C twice.
It seems we are ready to go! We will use the interactive shell quite a lot in the next chapters to get a bit more familiar with the language constructs and basic types, starting in the next chapter.
Note: if you are on Windows and running on an Erlang/OTP version earlier than 26, you can also try iex --werl (iex.bat --werl on PowerShell) which may provide a better experience depending on which console you are using.


  
    
  
  Running scripts


After getting familiar with the basics of the language you may want to try writing simple programs. This can be accomplished by putting the following Elixir code into a file:
IO.puts("Hello world from Elixir")
Save it as simple.exs and execute it with elixir:
$ elixir simple.exs
Hello world from Elixir

Later on we will learn how to compile Elixir code and how to create and work within Elixir projects using the Mix build tool. For now, let's move on to learn the basic data types in the language.


  

    
Basic types
    

In this chapter we will learn more about Elixir basic types: integers, floats, booleans, atoms, and strings. Other data types, such as lists and tuples, will be explored in the next chapter.
iex> 1          # integer
iex> 0x1F       # integer
iex> 1.0        # float
iex> true       # boolean
iex> :atom      # atom / symbol
iex> "elixir"   # string
iex> [1, 2, 3]  # list
iex> {1, 2, 3}  # tuple

  
    
  
  Basic arithmetic


Open up iex and type the following expressions:
iex> 1 + 2
3
iex> 5 * 5
25
iex> 10 / 2
5.0
Notice that 10 / 2 returned a float 5.0 instead of an integer 5. This is expected. In Elixir, the operator / always returns a float. If you want to do integer division or get the division remainder, you can invoke the div and rem functions:
iex> div(10, 2)
5
iex> div 10, 2
5
iex> rem 10, 3
1
Notice that Elixir allows you to drop the parentheses when invoking functions that expect one or more arguments. This feature gives a cleaner syntax when writing declarations and control-flow constructs. However, Elixir developers generally prefer to use parentheses.
Elixir also supports shortcut notations for entering binary, octal, and hexadecimal numbers:
iex> 0b1010
10
iex> 0o777
511
iex> 0x1F
31
Float numbers require a dot followed by at least one digit and also support e for scientific notation:
iex> 1.0
1.0
iex> 1.0e-10
1.0e-10
Floats in Elixir are 64-bit precision.
You can invoke the round function to get the closest integer to a given float, or the trunc function to get the integer part of a float.
iex> round(3.58)
4
iex> trunc(3.58)
3
Finally, we work with different data types, we will learn Elixir provides several predicate functions to check for the type of a value. For example, the is_integer can be used to check if a value is an integer or not:
iex> is_integer(1)
true
iex> is_integer(2.0)
false
You can also use is_float or is_number to check, respectively, if an argument is a float, or either an integer or float.

  
    
  
  Identifying functions and documentation


Before we move on to the next data type, let's talk about how Elixir identifies functions.
Functions in Elixir are identified by both their name and their arity. The arity of a function describes the number of arguments that the function takes. From this point on we will use both the function name and its arity to describe functions throughout the documentation. trunc/1 identifies the function which is named trunc and takes 1 argument, whereas trunc/2 identifies a different (nonexistent) function with the same name but with an arity of 2.
We can also use this syntax to access documentation. The Elixir shell defines the h function, which you can use to access documentation for any function. For example, typing h trunc/1 is going to print the documentation for the trunc/1 function:
iex> h trunc/1
                             def trunc()

Returns the integer part of number.
h trunc/1 works because it is defined in the Kernel module. All functions in the Kernel module are automatically imported into our namespace. Most often you will also include the module name when looking up the documentation for a given function:
iex> h Kernel.trunc/1
                             def trunc()

Returns the integer part of number.
You can use the module+function to lookup for anything, including operators (try h Kernel.+/2). Invoking h without arguments displays the documentation for IEx.Helpers, which is where h and other functionality is defined.

  
    
  
  Booleans and nil


Elixir supports true and false as booleans:
iex> true
true
iex> true == false
false
Elixir also provides three boolean operators: or/2, and/2, and not/1. These operators are strict in the sense that they expect something that evaluates to a boolean (true or false) as their first argument:
iex> true and true
true
iex> false or is_boolean(true)
true
Providing a non-boolean will raise an exception:
iex> 1 and true
** (BadBooleanError) expected a boolean on left-side of "and", got: 1
or and and are short-circuit operators. They only execute the right side if the left side is not enough to determine the result:
iex> false and raise("This error will never be raised")
false
iex> true or raise("This error will never be raised")
true
Elixir also provides the concept of nil, to indicate the absence of a value, and a set of logical operators that also manipulate nil: ||/2, 
  
    
    Lists and tuples - Elixir v1.17.3
    
    

    


  
  

    
Lists and tuples
    

In this chapter we will learn two of the most used collection data-types in Elixir: lists and tuples.

  
    
  
  (Linked) Lists


Elixir uses square brackets to specify a list of values. Values can be of any type:
iex> [1, 2, true, 3]
[1, 2, true, 3]
iex> length([1, 2, 3])
3
Two lists can be concatenated or subtracted using the ++/2 and --/2 operators respectively:
iex> [1, 2, 3] ++ [4, 5, 6]
[1, 2, 3, 4, 5, 6]
iex> [1, true, 2, false, 3, true] -- [true, false]
[1, 2, 3, true]
List operators never modify the existing list. Concatenating to or removing elements from a list returns a new list. We say that Elixir data structures are immutable. One advantage of immutability is that it leads to clearer code. You can freely pass the data around with the guarantee no one will mutate it in memory - only transform it.
Throughout the tutorial, we will talk a lot about the head and tail of a list. The head is the first element of a list and the tail is the remainder of the list. They can be retrieved with the functions hd/1 and tl/1. Let's assign a list to a variable and retrieve its head and tail:
iex> list = [1, 2, 3]
iex> hd(list)
1
iex> tl(list)
[2, 3]
Getting the head or the tail of an empty list throws an error:
iex> hd([])
** (ArgumentError) argument error
Sometimes you will create a list and it will return a quoted value preceded by ~c. For example:
iex> [11, 12, 13]
~c"\v\f\r"
iex> [104, 101, 108, 108, 111]
~c"hello"
When Elixir sees a list of printable ASCII numbers, Elixir will print that as a charlist (literally a list of characters). Charlists are quite common when interfacing with existing Erlang code. Whenever you see a value in IEx and you are not quite sure what it is, you can use the i/1 to retrieve information about it:
iex> i ~c"hello"
Term
  i ~c"hello"
Data type
  List
Description
  ...
Raw representation
  [104, 101, 108, 108, 111]
Reference modules
  List
Implemented protocols
  ...
We will talk more about charlists in the "Binaries, strings, and charlists" chapter.
Single-quoted strings
In Elixir, you can also use 'hello' to build charlists, but this notation has been soft-deprecated in Elixir v1.15 and will emit warnings in future versions. Prefer to write ~c"hello" instead.


  
    
  
  Tuples


Elixir uses curly brackets to define tuples. Like lists, tuples can hold any value:
iex> {:ok, "hello"}
{:ok, "hello"}
iex> tuple_size({:ok, "hello"})
2
Tuples store elements contiguously in memory. This means accessing a tuple element by index or getting the tuple size is a fast operation. Indexes start from zero:
iex> tuple = {:ok, "hello"}
{:ok, "hello"}
iex> elem(tuple, 1)
"hello"
iex> tuple_size(tuple)
2
It is also possible to put an element at a particular index in a tuple with put_elem/3:
iex> tuple = {:ok, "hello"}
{:ok, "hello"}
iex> put_elem(tuple, 1, "world")
{:ok, "world"}
iex> tuple
{:ok, "hello"}
Notice that put_elem/3 returned a new tuple. The original tuple stored in the tuple variable was not modified. Like lists, tuples are also immutable. Every operation on a tuple returns a new tuple, it never changes the given one.

  
    
  
  Lists or tuples?


What is the difference between lists and tuples?
Lists are stored in memory as linked lists, meaning that each element in a list holds its value and points to the following element until the end of the list is reached. This means accessing the length of a list is a linear operation: we need to traverse the whole list in order to figure out its size.
Similarly, the performance of list concatenation depends on the length of the left-hand list:
iex> list = [1, 2, 3]
[1, 2, 3]

# This is fast as we only need to traverse `[0]` to prepend to `list`
iex> [0] ++ list
[0, 1, 2, 3]

# This is slow as we need to traverse `list` to append 4
iex> list ++ [4]
[1, 2, 3, 4]
Tuples, on the other hand, are stored contiguously in memory. This means getting the tuple size or accessing an element by index is fast. On the other hand, updating or adding elements to tuples is expensive because it requires creating a new tuple in memory:
iex> tuple = {:a, :b, :c, :d}
{:a, :b, :c, :d}
iex> put_elem(tuple, 2, :e)
{:a, :b, :e, :d}
Note, however, the elements themselves are not copied. When you update a tuple, all entries are shared between the old and the new tuple, except for the entry that has been replaced. This rule applies to most data structures in Elixir. This reduces the amount of memory allocation the language needs to perform and is only possible thanks to the immutable semantics of the language.
Those performance characteristics dictate the usage of those data structures. In a nutshell, lists are used when the number of elements returned may vary. Tuples have a fixed size. Let's see two examples from the String module:
iex> String.split("hello world")
["hello", "world"]
iex> String.split("hello beautiful world")
["hello", "beautiful", "world"]
The String.split/1 function breaks a string into a list of strings on every whitespace character. Since the amount of elements returned depends on the input, we use a list.
On the other hand, String.split_at/2 splits a string in two parts at a given position. Since it always returns two entries, regardless of the input size, it returns tuples:
iex> String.split_at("hello world", 3)
{"hel", "lo world"}
iex> String.split_at("hello world", -4)
{"hello w", "orld"}
It is also very common to use tuples and atoms to create "tagged tuples", which is a handy return value when an operation may succeed or fail. For example, File.read/1 reads the contents of a file at a given path, which may or may not exist. It returns tagged tuples:
iex> File.read("path/to/existing/file")
{:ok, "... contents ..."}
iex> File.read("path/to/unknown/file")
{:error, :enoent}
If the path given to File.read/1 exists, it returns a tuple with the atom :ok as the first element and the file contents as the second. Otherwise, it returns a tuple with :error and the error description. As we will soon learn, Elixir allows us to pattern match on tagged tuples and effortlessly handle both success and failure cases.
Given Elixir consistently follows those rules, the choice between lists and tuples get clearer as you learn and use the language. Elixir often guides you to do the right thing. For example, there is an elem/2 function to access a tuple item:
iex> tuple = {:ok, "hello"}
{:ok, "hello"}
iex> elem(tuple, 1)
"hello"
However, given you often don't know the number of elements in a list, there is no built-in equivalent for accessing arbitrary entries in a lists, apart from its head.

  
    
  
  Size or length?


When counting the elements in a data structure, Elixir also abides by a simple rule: the function is named size if the operation is in constant time (the value is pre-calculated) or length if the operation is linear (calculating the length gets slower as the input grows). As a mnemonic, both "length" and "linear" start with "l".
For example, we have used 4 counting functions so far: byte_size/1 (for the number of bytes in a string), tuple_size/1 (for tuple size), length/1 (for list length) and String.length/1 (for the number of graphemes in a string). We use byte_size to get the number of bytes in a string, which is a cheap operation. Retrieving the number of Unicode graphemes, on the other hand, uses String.length/1, and may be expensive as it relies on a traversal of the entire string.
Now that we are familiar with the basic data-types in the language, let's learn important constructs for writing code, before we discuss more complex data structures.

Pattern matching
    

In this chapter, we will learn why the = operator in Elixir is called the match operator and how to use it to pattern match inside data structures. We will learn about the pin operator ^ used to access previously bound values.

  
    
  
  The match operator


We have used the = operator a couple times to assign variables in Elixir:
iex> x = 1
1
iex> x
1
In Elixir, the = operator is actually called the match operator. Let's see why:
iex> x = 1
1
iex> 1 = x
1
iex> 2 = x
** (MatchError) no match of right hand side value: 1
Notice that 1 = x is a valid expression, and it matched because both the left and right side are equal to 1. When the sides do not match, a MatchError is raised.
A variable can only be assigned on the left side of =:
iex> 1 = unknown
** (CompileError) iex:1: undefined variable "unknown"

  
    
  
  Pattern matching


The match operator is not only used to match against simple values, but it is also useful for destructuring more complex data types. For example, we can pattern match on tuples:
iex> {a, b, c} = {:hello, "world", 42}
{:hello, "world", 42}
iex> a
:hello
iex> b
"world"
A pattern match error will occur if the sides can't be matched, for example if the tuples have different sizes:
iex> {a, b, c} = {:hello, "world"}
** (MatchError) no match of right hand side value: {:hello, "world"}
And also when comparing different types, for example if matching a tuple on the left side with a list on the right side:
iex> {a, b, c} = [:hello, "world", 42]
** (MatchError) no match of right hand side value: [:hello, "world", 42]
More interestingly, we can match on specific values. The example below asserts that the left side will only match the right side when the right side is a tuple that starts with the atom :ok:
iex> {:ok, result} = {:ok, 13}
{:ok, 13}
iex> result
13

iex> {:ok, result} = {:error, :oops}
** (MatchError) no match of right hand side value: {:error, :oops}
We can pattern match on lists:
iex> [a, b, c] = [1, 2, 3]
[1, 2, 3]
iex> a
1
A list also supports matching on its own head and tail:
iex> [head | tail] = [1, 2, 3]
[1, 2, 3]
iex> head
1
iex> tail
[2, 3]
Similar to the hd/1 and tl/1 functions, we can't match an empty list with a head and tail pattern:
iex> [head | tail] = []
** (MatchError) no match of right hand side value: []
The [head | tail] format is not only used on pattern matching but also for prepending items to a list:
iex> list = [1, 2, 3]
[1, 2, 3]
iex> [0 | list]
[0, 1, 2, 3]
Pattern matching allows developers to easily destructure data types such as tuples and lists. As we will see in the following chapters, it is one of the foundations of recursion in Elixir and applies to other types as well, like maps and binaries.

  
    
  
  The pin operator


Variables in Elixir can be rebound:
iex> x = 1
1
iex> x = 2
2
However, there are times when we don't want variables to be rebound.
Use the pin operator ^ when you want to pattern match against a variable's existing value rather than rebinding the variable.
iex> x = 1
1
iex> ^x = 2
** (MatchError) no match of right hand side value: 2
Because we have pinned x when it was bound to the value of 1, it is equivalent to the following:
iex> 1 = 2
** (MatchError) no match of right hand side value: 2
Notice that we even see the exact same error message.
We can use the pin operator inside other pattern matches, such as tuples or lists:
iex> x = 1
1
iex> [^x, 2, 3] = [1, 2, 3]
[1, 2, 3]
iex> {y, ^x} = {2, 1}
{2, 1}
iex> y
2
iex> {y, ^x} = {2, 2}
** (MatchError) no match of right hand side value: {2, 2}
Because x was bound to the value of 1 when it was pinned, this last example could have been written as:
iex> {y, 1} = {2, 2}
** (MatchError) no match of right hand side value: {2, 2}
If a variable is mentioned more than once in a pattern, all references must bind to the same value:
iex> {x, x} = {1, 1}
{1, 1}
iex> {x, x} = {1, 2}
** (MatchError) no match of right hand side value: {1, 2}
In some cases, you don't care about a particular value in a pattern. It is a common practice to bind those values to the underscore, _. For example, if only the head of the list matters to us, we can assign the tail to underscore:
iex> [head | _] = [1, 2, 3]
[1, 2, 3]
iex> head
1
The variable _ is special in that it can never be read from. Trying to read from it gives a compile error:
iex> _
** (CompileError) iex:1: invalid use of _. "_" represents a value to be ignored in a pattern and cannot be used in expressions
Although pattern matching allows us to build powerful constructs, its usage is limited. For instance, you cannot make function calls on the left side of a match. The following example is invalid:
iex> length([1, [2], 3]) = 3
** (CompileError) iex:1: cannot invoke remote function :erlang.length/1 inside match
This finishes our introduction to pattern matching. As we will see in the next chapter, pattern matching is very common in many language constructs and they can be further augmented with guards.

case, cond, and if
    

In this chapter, we will learn about the case, cond, and if control flow structures.

  
    
  
  case


case allows us to compare a value against many patterns until we find a matching one:
iex> case {1, 2, 3} do
...>   {4, 5, 6} ->
...>     "This clause won't match"
...>   {1, x, 3} ->
...>     "This clause will match and bind x to 2 in this clause"
...>   _ ->
...>     "This clause would match any value"
...> end
"This clause will match and bind x to 2 in this clause"
If you want to pattern match against an existing variable, you need to use the ^ operator:
iex> x = 1
1
iex> case 10 do
...>   ^x -> "Won't match"
...>   _ -> "Will match"
...> end
"Will match"
Clauses also allow extra conditions to be specified via guards:
iex> case {1, 2, 3} do
...>   {1, x, 3} when x > 0 ->
...>     "Will match"
...>   _ ->
...>     "Would match, if guard condition were not satisfied"
...> end
"Will match"
The first clause above will only match when x is positive.
Keep in mind errors in guards do not leak but simply make the guard fail:
iex> hd(1)
** (ArgumentError) argument error
iex> case 1 do
...>   x when hd(x) -> "Won't match"
...>   x -> "Got #{x}"
...> end
"Got 1"
If none of the clauses match, an error is raised:
iex> case :ok do
...>   :error -> "Won't match"
...> end
** (CaseClauseError) no case clause matching: :ok
The documentation for the Kernel module lists all available guards in its sidebar. You can also consult the complete Patterns and Guards reference for in-depth documentation.

  
    
  
  if/unless


case builds on pattern matching and guards to destructure and match on certain conditions. However, patterns and guards are limited only to certain expressions which are optimized by the compiler. In many situations, you need to write conditions that go beyond what can be expressed with case. For those, if/2 (and unless/2) are useful alternatives:
iex> if true do
...>   "This works!"
...> end
"This works!"
iex> unless true do
...>   "This will never be seen"
...> end
nil
If the condition given to if/2 returns false or nil, the body given between do-end is not executed and instead it returns nil. The opposite happens with unless/2.
They also support else blocks (although using else with unless is generally discouraged):
iex> if nil do
...>   "This won't be seen"
...> else
...>   "This will"
...> end
"This will"
This is also a good opportunity to talk about variable scoping in Elixir. If any variable is declared or changed inside if, case, and similar constructs, the declaration and change will only be visible inside the construct. For example:
iex> x = 1
1
iex> if true do
...>   x = x + 1
...> end
2
iex> x
1
In said cases, if you want to change a value, you must return the value from the if:
iex> x = 1
1
iex> x = if true do
...>   x + 1
...> else
...>   x
...> end
2
if and unless are macros
An interesting note regarding if/2 and unless/2 is that they are implemented as macros in the language: they aren't special language constructs as they would be in many languages. You can check the documentation and their source for more information.

If you find yourself nesting several if/2 blocks, you may want to consider using cond/1 instead. Let's check it out.

  
    
  
  cond


If you need to check across several conditions and find the first one that does not evaluate to nil or false, cond/1 is a useful construct:
iex> cond do
...>   2 + 2 == 5 ->
...>     "This will not be true"
...>   2 * 2 == 3 ->
...>     "Nor this"
...>   1 + 1 == 2 ->
...>     "But this will"
...> end
"But this will"
This is equivalent to else if clauses in many imperative languages - although used less frequently in Elixir.
If all of the conditions return nil or false, an error (CondClauseError) is raised. For this reason, it may be necessary to add a final condition, equal to true, which will always match:
iex> cond do
...>   2 + 2 == 5 ->
...>     "This is never true"
...>   2 * 2 == 3 ->
...>     "Nor this"
...>   true ->
...>     "This is always true (equivalent to else)"
...> end
"This is always true (equivalent to else)"
Similar to if/2 and unless/2, cond considers any value besides nil and false to be true:
iex> cond do
...>   hd([1, 2, 3]) ->
...>     "1 is considered as true"
...> end
"1 is considered as true"

  
    
  
  Summing up


We have concluded the introduction to the most fundamental control-flow constructs in Elixir. Generally speaking, Elixir developers prefer pattern matching and guards, using case and function definitions (which we will explore in future chapters), as they are succinct and precise. When your logic cannot be outlined within patterns and guards, you may consider if/2, falling back to cond/1 when there are several conditions to check.

Anonymous functions
    

Anonymous functions allow us to store and pass executable code around as if it was an integer or a string. Let's learn more.

  
    
  
  Defining anonymous functions


Anonymous functions in Elixir are delimited by the keywords fn and end:
iex> add = fn a, b -> a + b end
#Function<12.71889879/2 in :erl_eval.expr/5>
iex> add.(1, 2)
3
iex> is_function(add)
true
In the example above, we defined an anonymous function that receives two arguments, a and b, and returns the result of a + b. The arguments are always on the left-hand side of -> and the code to be executed on the right-hand side. The anonymous function is stored in the variable add.
We can invoke anonymous functions by passing arguments to it. Note that a dot (.) between the variable and parentheses is required to invoke an anonymous function. The dot makes it clear when you are calling an anonymous function, stored in the variable add, opposed to a function named add/2. For example, if you have an anonymous function stored in the variable is_atom, there is no ambiguity between is_atom.(:foo) and is_atom(:foo). If both used the same is_atom(:foo) syntax, the only way to know the actual behavior of is_atom(:foo) would be by scanning all code thus far for a possible definition of the is_atom variable. This scanning hurts maintainability as it requires developers to track additional context in their head when reading and writing code.
Anonymous functions in Elixir are also identified by the number of arguments they receive. We can check if a function is of any given arity by using is_function/2:
# check if add is a function that expects exactly 2 arguments
iex> is_function(add, 2)
true
# check if add is a function that expects exactly 1 argument
iex> is_function(add, 1)
false

  
    
  
  Closures


Anonymous functions can also access variables that are in scope when the function is defined. This is typically referred to as closures, as they close over their scope. Let's define a new anonymous function that uses the add anonymous function we have previously defined:
iex> double = fn a -> add.(a, a) end
#Function<6.71889879/1 in :erl_eval.expr/5>
iex> double.(2)
4
A variable assigned inside a function does not affect its surrounding environment:
iex> x = 42
42
iex> (fn -> x = 0 end).()
0
iex> x
42

  
    
  
  Clauses and guards


Similar to case/2, we can pattern match on the arguments of anonymous functions as well as define multiple clauses and guards:
iex> f = fn
...>   x, y when x > 0 -> x + y
...>   x, y -> x * y
...> end
#Function<12.71889879/2 in :erl_eval.expr/5>
iex> f.(1, 3)
4
iex> f.(-1, 3)
-3
The number of arguments in each anonymous function clause needs to be the same, otherwise an error is raised.
iex> f2 = fn
...>   x, y when x > 0 -> x + y
...>   x, y, z -> x * y + z
...> end
** (CompileError) iex:1: cannot mix clauses with different arities in anonymous functions

  
    
  
  The capture operator


Throughout this guide, we have been using the notation name/arity to refer to functions. It happens that this notation can actually be used to capture an existing function into a data-type we can pass around, similar to how anonymous functions behave.
iex> fun = &is_atom/1
&:erlang.is_atom/1
iex> is_function(fun)
true
iex> fun.(:hello)
true
iex> fun.(123)
false
As you can see, once a function is captured, we can pass it as argument or invoke it using the anonymous function notation. The returned value above also hints we can capture functions defined in modules:
iex> fun = &String.length/1
&String.length/1
iex> fun.("hello")
5
You can also capture operators:
iex> add = &+/2
&:erlang.+/2
iex> add.(1, 2)
3
The capture syntax can also be used as a shortcut for creating functions. This is handy when you want to create functions that are mostly wrapping existing functions or operators:
iex> fun = &(&1 + 1)
#Function<6.71889879/1 in :erl_eval.expr/5>
iex> fun.(1)
2

iex> fun2 = &"Good #{&1}"
#Function<6.127694169/1 in :erl_eval.expr/5>
iex> fun2.("morning")
"Good morning"
The &1 represents the first argument passed into the function. &(&1 + 1) above is exactly the same as fn x -> x + 1 end. You can read more about the capture operator & in 

Binaries, strings, and charlists
    

In "Basic types", we learned a bit about strings and we used the is_binary/1 function for checks:
iex> string = "hello"
"hello"
iex> is_binary(string)
true
In this chapter, we will gain clarity on what exactly binaries are and how they relate to strings. We will also learn about charlists, ~c"like this", which are often used for interoperability with Erlang.
Although strings are one of the most common data types in computer languages, they are subtly complex and are often misunderstood. To understand strings in Elixir, let's first discuss Unicode and character encodings, specifically the UTF-8 encoding.

  
    
  
  Unicode and Code Points


In order to facilitate meaningful communication between computers across multiple languages, a standard is required so that the ones and zeros on one machine mean the same thing when they are transmitted to another. The Unicode Standard acts as an official registry of virtually all the characters we know: this includes characters from classical and historical texts, emoji, and formatting and control characters as well.
Unicode organizes all of the characters in its repertoire into code charts, and each character is given a unique numerical index. This numerical index is known as a Code Point.
In Elixir you can use a ? in front of a character literal to reveal its code point:
iex> ?a
97
iex> ?ł
322
Note that most Unicode code charts will refer to a code point by its hexadecimal (hex) representation, e.g. 97 translates to 0061 in hex, and we can represent any Unicode character in an Elixir string by using the \uXXXX notation and the hex representation of its code point number:
iex> "\u0061" == "a"
true
iex> 0x0061 = 97 = ?a
97
The hex representation will also help you look up information about a code point, e.g. https://codepoints.net/U+0061 has a data sheet all about the lower case a, a.k.a. code point 97.

  
    
  
  UTF-8 and Encodings


Now that we understand what the Unicode standard is and what code points are, we can finally talk about encodings. Whereas the code point is what we store, an encoding deals with how we store it: encoding is an implementation. In other words, we need a mechanism to convert the code point numbers into bytes so they can be stored in memory, written to disk, etc.
Elixir uses UTF-8 to encode its strings, which means that code points are encoded as a series of 8-bit bytes. UTF-8 is a variable width character encoding that uses one to four bytes to store each code point. It is capable of encoding all valid Unicode code points. Let's see an example:
iex> string = "héllo"
"héllo"
iex> String.length(string)
5
iex> byte_size(string)
6
Although the string above has 5 characters, it uses 6 bytes, as two bytes are used to represent the character é.
Note: if you are running on Windows, there is a chance your terminal does not use UTF-8 by default. You can change the encoding of your current session by running chcp 65001 before entering iex (iex.bat).

Besides defining characters, UTF-8 also provides a notion of graphemes. Graphemes may consist of multiple characters that are often perceived as one. For example, the woman firefighter emoji is represented as the combination of three characters: the woman emoji (👩), a hidden zero-width joiner, and the fire engine emoji (🚒):
iex> String.codepoints("👩‍🚒")
["👩", "‍", "🚒"]
iex> String.graphemes("👩‍🚒")
["👩‍🚒"]
However, Elixir is smart enough to know they are seen as a single character, and therefore the length is still one:
iex> String.length("👩‍🚒")
1
Note: if you can't see the emoji above in your terminal, you need to make sure your terminal supports emoji and that you are using a font that can render them.

Although these rules may sound complicated, UTF-8 encoded documents are everywhere. This page itself is encoded in UTF-8. The encoding information is given to your browser which then knows how to render all of the bytes, characters, and graphemes accordingly.
If you want to see the exact bytes that a string would be stored in a file, a common trick is to concatenate the null byte <<0>> to it:
iex> "hełło" <> <<0>>
<<104, 101, 197, 130, 197, 130, 111, 0>>
Alternatively, you can view a string's binary representation by using IO.inspect/2:
iex> IO.inspect("hełło", binaries: :as_binaries)
<<104, 101, 197, 130, 197, 130, 111>>
We are getting a little bit ahead of ourselves. Let's talk about bitstrings to learn about what exactly the <<>> constructor means.

  
    
  
  Bitstrings


Although we have covered code points and UTF-8 encoding, we still need to go a bit deeper into how exactly we store the encoded bytes, and this is where we introduce the bitstring. A bitstring is a fundamental data type in Elixir, denoted with the <<>> syntax. A bitstring is a contiguous sequence of bits in memory.
By default, 8 bits (i.e. 1 byte) is used to store each number in a bitstring, but you can manually specify the number of bits via a ::n modifier to denote the size in n bits, or you can use the more verbose declaration ::size(n):
iex> <<42>> == <<42::8>>
true
iex> <<3::4>>
<<3::size(4)>>
For example, the decimal number 3 when represented with 4 bits in base 2 would be 0011, which is equivalent to the values 0, 0, 1, 1, each stored using 1 bit:
iex> <<0::1, 0::1, 1::1, 1::1>> == <<3::4>>
true
Any value that exceeds what can be stored by the number of bits provisioned is truncated:
iex> <<1>> == <<257>>
true
Here, 257 in base 2 would be represented as 100000001, but since we have reserved only 8 bits for its representation (by default), the left-most bit is ignored and the value becomes truncated to 00000001, or simply 1 in decimal.
A complete reference for the bitstring constructor can be found in <<>>'s documentation.

  
    
  
  Binaries


A binary is a bitstring where the number of bits is divisible by 8. That means that every binary is a bitstring, but not every bitstring is a binary. We can use the is_bitstring/1 and is_binary/1 functions to demonstrate this.
iex> is_bitstring(<<3::4>>)
true
iex> is_binary(<<3::4>>)
false
iex> is_bitstring(<<0, 255, 42>>)
true
iex> is_binary(<<0, 255, 42>>)
true
iex> is_binary(<<42::16>>)
true
We can pattern match on binaries / bitstrings:
iex> <<0, 1, x>> = <<0, 1, 2>>
<<0, 1, 2>>
iex> x
2
iex> <<0, 1, x>> = <<0, 1, 2, 3>>
** (MatchError) no match of right hand side value: <<0, 1, 2, 3>>
Note that unless you explicitly use :: modifiers, each entry in the binary pattern is expected to match a single byte (exactly 8 bits). If we want to match on a binary of unknown size, we can use the binary modifier at the end of the pattern:
iex> <<0, 1, x::binary>> = <<0, 1, 2, 3>>
<<0, 1, 2, 3>>
iex> x
<<2, 3>>
There are a couple other modifiers that can be useful when doing pattern matches on binaries. The binary-size(n) modifier will match n bytes in a binary:
iex> <<head::binary-size(2), rest::binary>> = <<0, 1, 2, 3>>
<<0, 1, 2, 3>>
iex> head
<<0, 1>>
iex> rest
<<2, 3>>
A string is a UTF-8 encoded binary, where the code point for each character is encoded using 1 to 4 bytes. Thus every string is a binary, but due to the UTF-8 standard encoding rules, not every binary is a valid string.
iex> is_binary("hello")
true
iex> is_binary(<<239, 191, 19>>)
true
iex> String.valid?(<<239, 191, 19>>)
false
The string concatenation operator <> is actually a binary concatenation operator:
iex> "a" <> "ha"
"aha"
iex> <<0, 1>> <> <<2, 3>>
<<0, 1, 2, 3>>
Given that strings are binaries, we can also pattern match on strings:
iex> <<head, rest::binary>> = "banana"
"banana"
iex> head == ?b
true
iex> rest
"anana"
However, remember that binary pattern matching works on bytes, so matching on the string like "über" with multibyte characters won't match on the character, it will match on the first byte of that character:
iex> "ü" <> <<0>>
<<195, 188, 0>>
iex> <<x, rest::binary>> = "über"
"über"
iex> x == ?ü
false
iex> rest
<<188, 98, 101, 114>>
Above, x matched on only the first byte of the multibyte ü character.
Therefore, when pattern matching on strings, it is important to use the utf8 modifier:
iex> <<x::utf8, rest::binary>> = "über"
"über"
iex> x == ?ü
true
iex> rest
"ber"

  
    
  
  Charlists


Our tour of our bitstrings, binaries, and strings is nearly complete, but we have one more data type to explain: the charlist.
A charlist is a list of integers where all the integers are valid code points. In practice, you will not come across them often, only in specific scenarios such as interfacing with older Erlang libraries that do not accept binaries as arguments.
iex> ~c"hello"
~c"hello"
iex> [?h, ?e, ?l, ?l, ?o]
~c"hello"
The ~c sigil (we'll cover sigils later in the "Sigils" chapter) indicates the fact that we are dealing with a charlist and not a regular string.
Instead of containing bytes, a charlist contains integer code points. However, the list is only printed as a sigil if all code points are within the ASCII range:
iex> ~c"hełło"
[104, 101, 322, 322, 111]
iex> is_list(~c"hełło")
true
This is done to ease interoperability with Erlang, even though it may lead to some surprising behavior. For example, if you are storing a list of integers that happen to range between 0 and 127, by default IEx will interpret this as a charlist and it will display the corresponding ASCII characters.
iex> heartbeats_per_minute = [99, 97, 116]
~c"cat"
You can always force charlists to be printed in their list representation by calling the inspect/2 function:
iex> inspect(heartbeats_per_minute, charlists: :as_list)
"[99, 97, 116]"
Furthermore, you can convert a charlist to a string and back by using the to_string/1 and to_charlist/1:
iex> to_charlist("hełło")
[104, 101, 322, 322, 111]
iex> to_string(~c"hełło")
"hełło"
iex> to_string(:hello)
"hello"
iex> to_string(1)
"1"
The functions above are polymorphic, in other words, they accept many shapes: not only do they convert charlists to strings (and vice-versa), they can also convert integers, atoms, and so on.
String (binary) concatenation uses the <> operator but charlists, being lists, use the list concatenation operator ++:
iex> ~c"this " <> ~c"fails"
** (ArgumentError) expected binary argument in <> operator but got: ~c"this "
    (elixir) lib/kernel.ex:1821: Kernel.wrap_concatenation/3
    (elixir) lib/kernel.ex:1808: Kernel.extract_concatenations/2
    (elixir) expanding macro: Kernel.<>/2
    iex:1: (file)
iex> ~c"this " ++ ~c"works"
~c"this works"
iex> "he" ++ "llo"
** (ArgumentError) argument error
    :erlang.++("he", "llo")
iex> "he" <> "llo"
"hello"
With binaries, strings, and charlists out of the way, it is time to talk about key-value data structures.

Keyword lists and maps
    

Now let's talk about associative data structures. Associative data structures are able to associate a key to a certain value. Different languages call these different names like dictionaries, hashes, associative arrays, etc.
In Elixir, we have two main associative data structures: keyword lists and maps.

  
    
  
  Keyword lists


Keyword lists are a data-structure used to pass options to functions. Imagine you want to split a string of numbers. We can use String.split/2:
iex> String.split("1 2 3", " ")
["1", "2", "3"]
However, what happens if there is an additional space between the numbers:
iex> String.split("1  2  3", " ")
["1", "", "2", "", "3"]
As you can see, there are now empty strings in our results. Luckily, the String.split/3 function allows the trim option to be set to true:
iex> String.split("1  2  3", " ", [trim: true])
["1", "2", "3"]
[trim: true] is a keyword list. Furthermore, when a keyword list is the last argument of a function, we can skip the brackets and write:
iex> String.split("1  2  3", " ", trim: true)
["1", "2", "3"]
As shown in the example above, keyword lists are mostly used as optional arguments to functions.
As the name implies, keyword lists are simply lists. In particular, they are lists consisting of 2-item tuples where the first element (the key) is an atom and the second element can be any value. Both representations are the same:
iex> [{:trim, true}] == [trim: true]
true
Since keyword lists are lists, we can use all operations available to lists. For example, we can use ++ to add new values to a keyword list:
iex> list = [a: 1, b: 2]
[a: 1, b: 2]
iex> list ++ [c: 3]
[a: 1, b: 2, c: 3]
iex> [a: 0] ++ list
[a: 0, a: 1, b: 2]
You can read the value of a keyword list using the brackets syntax. This is also known as the access syntax, as it is defined by the Access module:
iex> list[:a]
1
iex> list[:b]
2
In case of duplicate keys, values added to the front are the ones fetched:
iex> new_list = [a: 0] ++ list
[a: 0, a: 1, b: 2]
iex> new_list[:a]
0
Keyword lists are important because they have three special characteristics:
	Keys must be atoms.
	Keys are ordered, as specified by the developer.
	Keys can be given more than once.

For example, the Ecto library makes use of these features to provide an elegant DSL for writing database queries:
query =
  from w in Weather,
    where: w.prcp > 0,
    where: w.temp < 20,
    select: w
Although we can pattern match on keyword lists, it is not done in practice since pattern matching on lists requires the number of items and their order to match:
iex> [a: a] = [a: 1]
[a: 1]
iex> a
1
iex> [a: a] = [a: 1, b: 2]
** (MatchError) no match of right hand side value: [a: 1, b: 2]
iex> [b: b, a: a] = [a: 1, b: 2]
** (MatchError) no match of right hand side value: [a: 1, b: 2]
Furthermore, given keyword lists are often used as optional arguments, they are used in situations where not all keys may be present, which would make it impossible to match on them. In a nutshell, do not pattern match on keyword lists.
In order to manipulate keyword lists, Elixir provides the Keyword module. Remember, though, keyword lists are simply lists, and as such they provide the same linear performance characteristics as them: the longer the list, the longer it will take to find a key, to count the number of items, and so on. If you need to store a large amount of keys in a key-value data structure, Elixir offers maps, which we will soon learn.

  
    
  
  do-blocks and keywords


As we have seen, keywords are mostly used in the language to pass optional values. In fact, we have used keywords in earlier chapters. For example, we have seen:
iex> if true do
...>   "This will be seen"
...> else
...>   "This won't"
...> end
"This will be seen"
It happens that do blocks are nothing more than a syntax convenience on top of keywords. We can rewrite the above to:
iex> if true, do: "This will be seen", else: "This won't"
"This will be seen"
Pay close attention to both syntaxes. In the keyword list format, we separate each key-value pair with commas, and each key is followed by :. In the do-blocks, we get rid of the colons, the commas, and separate each keyword by a newline. They are useful exactly because they remove the verbosity when writing blocks of code. Most of the time, you will use the block syntax, but it is good to know they are equivalent.
This plays an important role in the language as it allows Elixir syntax to stay small but still expressive. We only need few data structures to represent the language, a topic we will come back to when talking about optional syntax and go in-depth when discussing meta-programming.
With this out of the way, let's talk about maps.

  
    
  
  Maps as key-value pairs


Whenever you need to store key-value pairs, maps are the "go to" data structure in Elixir. A map is created using the %{} syntax:
iex> map = %{:a => 1, 2 => :b}
%{2 => :b, :a => 1}
iex> map[:a]
1
iex> map[2]
:b
iex> map[:c]
nil
Compared to keyword lists, we can already see two differences:
	Maps allow any value as a key.
	Maps' keys do not follow any ordering.

In contrast to keyword lists, maps are very useful with pattern matching. When a map is used in a pattern, it will always match on a subset of the given value:
iex> %{} = %{:a => 1, 2 => :b}
%{2 => :b, :a => 1}
iex> %{:a => a} = %{:a => 1, 2 => :b}
%{2 => :b, :a => 1}
iex> a
1
iex> %{:c => c} = %{:a => 1, 2 => :b}
** (MatchError) no match of right hand side value: %{2 => :b, :a => 1}
As shown above, a map matches as long as the keys in the pattern exist in the given map. Therefore, an empty map matches all maps.
The Map module provides a very similar API to the Keyword module with convenience functions to add, remove, and update maps keys:
iex> Map.get(%{:a => 1, 2 => :b}, :a)
1
iex> Map.put(%{:a => 1, 2 => :b}, :c, 3)
%{2 => :b, :a => 1, :c => 3}
iex> Map.to_list(%{:a => 1, 2 => :b})
[{2, :b}, {:a, 1}]

  
    
  
  Maps of predefined keys


In the previous section, we have used maps as a key-value data structure where keys can be added or removed at any time. However, it is also common to create maps with a pre-defined set of keys. Their values may be updated, but new keys are never added nor removed. This is useful when we know the shape of the data we are working with and, if we get a different key, it likely means a mistake was done elsewhere.
We define such maps using the same syntax as in the previous section, except that all keys must be atoms:
iex> map = %{:name => "John", :age => 23}
%{name: "John", age: 23}
As you can see from the printed result above, Elixir also allows you to write maps of atom keys using the same key: value syntax as keyword lists.
When the keys are atoms, in particular when working with maps of predefined keys, we can also access them using the map.key syntax:
iex> map = %{name: "John", age: 23}
%{name: "John", age: 23}

iex> map.name
"John"
iex> map.agee
** (KeyError) key :agee not found in: %{name: "John", age: 23}
There is also syntax for updating keys, which also raises if the key has not yet been defined:
iex> %{map | name: "Mary"}
%{name: "Mary", age: 23}
iex> %{map | agee: 27}
** (KeyError) key :agee not found in: %{name: "John", age: 23}
These operations have one large benefit in that they raise if the key does not exist in the map and the compiler may even detect and warn when possible. This makes them useful to get quick feedback and spot bugs and typos early on. This is also the syntax used to power another Elixir feature called "Structs", which we will learn later on.
Elixir developers typically prefer to use the map.key syntax and pattern matching instead of the functions in the Map module when working with maps because they lead to an assertive style of programming. This blog post by José Valim provides insight and examples on how you get more concise and faster software by writing assertive code in Elixir.

  
    
  
  Nested data structures


Often we will have maps inside maps, or even keywords lists inside maps, and so forth. Elixir provides conveniences for manipulating nested data structures via the get_in/1, put_in/2, update_in/2, and other macros giving the same conveniences you would find in imperative languages while keeping the immutable properties of the language.
Imagine you have the following structure:
iex> users = [
  john: %{name: "John", age: 27, languages: ["Erlang", "Ruby", "Elixir"]},
  mary: %{name: "Mary", age: 29, languages: ["Elixir", "F#", "Clojure"]}
]
[
  john: %{age: 27, languages: ["Erlang", "Ruby", "Elixir"], name: "John"},
  mary: %{age: 29, languages: ["Elixir", "F#", "Clojure"], name: "Mary"}
]
We have a keyword list of users where each value is a map containing the name, age and a list of programming languages each user likes. If we wanted to access the age for john, we could write:
iex> users[:john].age
27
It happens we can also use this same syntax for updating the value:
iex> users = put_in users[:john].age, 31
[
  john: %{age: 31, languages: ["Erlang", "Ruby", "Elixir"], name: "John"},
  mary: %{age: 29, languages: ["Elixir", "F#", "Clojure"], name: "Mary"}
]
The update_in/2 macro is similar but allows us to pass a function that controls how the value changes. For example, let's remove "Clojure" from Mary's list of languages:
iex> users = update_in users[:mary].languages, fn languages -> List.delete(languages, "Clojure") end
[
  john: %{age: 31, languages: ["Erlang", "Ruby", "Elixir"], name: "John"},
  mary: %{age: 29, languages: ["Elixir", "F#"], name: "Mary"}
]
There is more to learn about get_in/1, pop_in/1 and others, including the get_and_update_in/2 that allows us to extract a value and update the data structure at once. There are also get_in/3, put_in/3, update_in/3, get_and_update_in/3, pop_in/2 which allow dynamic access into the data structure.

  
    
  
  Summary


There are two different data structures for working with key-value stores in Elixir. Alongside the Access module and pattern matching, they provide a rich set of tools for manipulating complex, potentially nested, data structures.
As we conclude this chapter, remember that you should:
	Use keyword lists for passing optional values to functions

	Use maps for general key-value data structures

	Use maps when working with data that has a predefined set of keys


Now let's talk about modules and functions.

Modules and functions
    

In Elixir we group several functions into modules. We've already used many different modules in the previous chapters, such as the String module:
iex> String.length("hello")
5
In order to create our own modules in Elixir, we use the defmodule macro. The first letter of the module must be in uppercase. We use the def macro to define functions in that module. The first letter of every function must be in lowercase (or underscore):
iex> defmodule Math do
...>   def sum(a, b) do
...>     a + b
...>   end
...> end

iex> Math.sum(1, 2)
3
In this chapter we will define our own modules, with different levels of complexity. As our examples get longer in size, it can be tricky to type them all in the shell. It's about time for us to learn how to compile Elixir code and also how to run Elixir scripts.

  
    
  
  Compilation


Most of the time it is convenient to write modules into files so they can be compiled and reused. Let's assume we have a file named math.ex with the following contents:
defmodule Math do
  def sum(a, b) do
    a + b
  end
end
This file can be compiled using elixirc:
$ elixirc math.ex

This will generate a file named Elixir.Math.beam containing the bytecode for the defined module. If we start iex again, our module definition will be available (provided that iex is started in the same directory the bytecode file is in):
iex> Math.sum(1, 2)
3
Elixir projects are usually organized into three directories:
	_build - contains compilation artifacts
	lib - contains Elixir code (usually .ex files)
	test - contains tests (usually .exs files)

When working on actual projects, the build tool called mix will be responsible for compiling and setting up the proper paths for you. For learning and convenience purposes, Elixir also supports a scripting mode which is more flexible and does not generate any compiled artifacts.

  
    
  
  Scripting mode


In addition to the Elixir file extension .ex, Elixir also supports .exs files for scripting. Elixir treats both files exactly the same way, the only difference is in intention. .ex files are meant to be compiled while .exs files are used for scripting. This convention is followed by projects like mix.
For instance, we can create a file called math.exs:
defmodule Math do
  def sum(a, b) do
    a + b
  end
end

IO.puts Math.sum(1, 2)
And execute it as:
$ elixir math.exs

Because we used elixir instead of elixirc, the module was compiled and loaded into memory, but no .beam file was written to disk. In the following examples, we recommend you write your code into script files and execute them as shown above.

  
    
  
  Function definition


Inside a module, we can define functions with def/2 and private functions with defp/2. A function defined with def/2 can be invoked from other modules while a private function can only be invoked locally.
defmodule Math do
  def sum(a, b) do
    do_sum(a, b)
  end

  defp do_sum(a, b) do
    a + b
  end
end

IO.puts Math.sum(1, 2)    #=> 3
IO.puts Math.do_sum(1, 2) #=> ** (UndefinedFunctionError)
Function declarations also support guards and multiple clauses. If a function has several clauses, Elixir will try each clause until it finds one that matches. Here is an implementation of a function that checks if the given number is zero or not:
defmodule Math do
  def zero?(0) do
    true
  end

  def zero?(x) when is_integer(x) do
    false
  end
end

IO.puts Math.zero?(0)         #=> true
IO.puts Math.zero?(1)         #=> false
IO.puts Math.zero?([1, 2, 3]) #=> ** (FunctionClauseError)
IO.puts Math.zero?(0.0)       #=> ** (FunctionClauseError)
The trailing question mark in zero? means that this function returns a boolean. To learn more about the naming conventions for modules, function names, variables and more in Elixir, see Naming Conventions.
Giving an argument that does not match any of the clauses raises an error.
Similar to constructs like if, function definitions support both do: and do-block syntax, as we learned in the previous chapter. For example, we can edit math.exs to look like this:
defmodule Math do
  def zero?(0), do: true
  def zero?(x) when is_integer(x), do: false
end
And it will provide the same behavior. You may use do: for one-liners but always use do-blocks for functions spanning multiple lines. If you prefer to be consistent, you can use do-blocks throughout your codebase.

  
    
  
  Default arguments


Function definitions in Elixir also support default arguments:
defmodule Concat do
  def join(a, b, sep \\ " ") do
    a <> sep <> b
  end
end

IO.puts Concat.join("Hello", "world")      #=> Hello world
IO.puts Concat.join("Hello", "world", "_") #=> Hello_world
Any expression is allowed to serve as a default value, but it won't be evaluated during the function definition. Every time the function is invoked and any of its default values have to be used, the expression for that default value will be evaluated:
defmodule DefaultTest do
  def dowork(x \\ "hello") do
    x
  end
end
iex> DefaultTest.dowork()
"hello"
iex> DefaultTest.dowork(123)
123
iex> DefaultTest.dowork()
"hello"
If a function with default values has multiple clauses, it is required to create a function head (a function definition without a body) for declaring defaults:
defmodule Concat do
  # A function head declaring defaults
  def join(a, b \\ nil, sep \\ " ")

  def join(a, b, _sep) when is_nil(b) do
    a
  end

  def join(a, b, sep) do
    a <> sep <> b
  end
end

IO.puts Concat.join("Hello", "world")      #=> Hello world
IO.puts Concat.join("Hello", "world", "_") #=> Hello_world
IO.puts Concat.join("Hello")               #=> Hello
When a variable is not used by a function or a clause, we add a leading underscore (_) to its name to signal this intent. This rule is also covered in our Naming Conventions document.
When using default values, one must be careful to avoid overlapping function definitions. Consider the following example:
defmodule Concat do
  def join(a, b) do
    IO.puts "***First join"
    a <> b
  end

  def join(a, b, sep \\ " ") do
    IO.puts "***Second join"
    a <> sep <> b
  end
end
Elixir will emit the following warning:
warning: this clause cannot match because a previous clause at line 2 always matches
    concat.ex:7: Concat
The compiler is telling us that invoking the join function with two arguments will always choose the first definition of join whereas the second one will only be invoked when three arguments are passed:
$ iex concat.ex

iex> Concat.join "Hello", "world"
***First join
"Helloworld"
iex> Concat.join "Hello", "world", "_"
***Second join
"Hello_world"
Removing the default argument in this case will fix the warning.
This finishes our short introduction to modules. In the next chapters, we will learn how to use function definitions for recursion and later on explore more functionality related to modules.

Recursion
    

Elixir does not provide loop constructs. Instead we leverage recursion and high-level functions for working with collections. This chapter will explore the former.

  
    
  
  Loops through recursion


Due to immutability, loops in Elixir (as in any functional programming language) are written differently from imperative languages. For example, in an imperative language like C, one would write:
for(i = 0; i < sizeof(array); i++) {
  array[i] = array[i] * 2;
}
In the example above, we are mutating both the array and the variable i. However, data structures in Elixir are immutable. For this reason, functional languages rely on recursion: a function is called recursively until a condition is reached that stops the recursive action from continuing. No data is mutated in this process. Consider the example below that prints a string an arbitrary number of times:
defmodule Recursion do
  def print_multiple_times(msg, n) when n > 0 do
    IO.puts(msg)
    print_multiple_times(msg, n - 1)
  end

  def print_multiple_times(_msg, 0) do
    :ok
  end
end

Recursion.print_multiple_times("Hello!", 3)
# Hello!
# Hello!
# Hello!
:ok
Similar to case, a function may have many clauses. A particular clause is executed when the arguments passed to the function match the clause's argument patterns and its guards evaluate to true.
When print_multiple_times/2 is initially called in the example above, the argument n is equal to 3.
The first clause has a guard which says "use this definition if and only if n is more than 0". Since this is the case, it prints the msg and then calls itself passing n - 1 (2) as the second argument.
Now we execute the same function again, starting from the first clause. Given the second argument, n, is still more than 0, we print the message and call ourselves once more, now with the second argument set to 1. Then we print the message one last time and call print_multiple_times("Hello!", 0), starting from the top once again.
When the second argument is zero, the guard n > 0 evaluates to false, and the first function clause won't execute. Elixir then proceeds to try the next function clause, which explicitly matches on the case where n is 0. This clause, also known as the termination clause, ignores the message argument by assigning it to the _msg variable and returns the atom :ok.
Finally, if you pass an argument that does not match any clause, Elixir raises a FunctionClauseError:
iex> Recursion.print_multiple_times "Hello!", -1
** (FunctionClauseError) no function clause matching in Recursion.print_multiple_times/2

    The following arguments were given to Recursion.print_multiple_times/2:

        # 1
        "Hello!"

        # 2
        -1

    iex:1: Recursion.print_multiple_times/2

  
    
  
  Reduce and map algorithms


Let's now see how we can use the power of recursion to sum a list of numbers:
defmodule Math do
  def sum_list([head | tail], accumulator) do
    sum_list(tail, head + accumulator)
  end

  def sum_list([], accumulator) do
    accumulator
  end
end

IO.puts Math.sum_list([1, 2, 3], 0) #=> 6
We invoke sum_list with the list [1, 2, 3] and the initial value 0 as arguments. We will try each clause until we find one that matches according to the pattern matching rules. In this case, the list [1, 2, 3] matches against [head | tail] which binds head to 1 and tail to [2, 3]; accumulator is set to 0.
Then, we add the head of the list to the accumulator head + accumulator and call sum_list again, recursively, passing the tail of the list as its first argument. The tail will once again match [head | tail] until the list is empty, as seen below:
sum_list [1, 2, 3], 0
sum_list [2, 3], 1
sum_list [3], 3
sum_list [], 6
When the list is empty, it will match the final clause which returns the final result of 6.
The process of taking a list and reducing it down to one value is known as a reduce algorithm and is central to functional programming.
What if we instead want to double all of the values in our list?
defmodule Math do
  def double_each([head | tail]) do
    [head * 2 | double_each(tail)]
  end

  def double_each([]) do
    []
  end
end
$ iex math.exs

iex> Math.double_each([1, 2, 3]) #=> [2, 4, 6]
Here we have used recursion to traverse a list, doubling each element and returning a new list. The process of taking a list and mapping over it is known as a map algorithm.
Recursion and tail call optimization are an important part of Elixir and are commonly used to create loops. However, when programming in Elixir you will rarely use recursion as above to manipulate lists.
The Enum module, which we're going to see in the next chapter already provides many conveniences for working with lists. For instance, the examples above could be written as:
iex> Enum.reduce([1, 2, 3], 0, fn x, acc -> x + acc end)
6
iex> Enum.map([1, 2, 3], fn x -> x * 2 end)
[2, 4, 6]
Or, using the capture syntax:
iex> Enum.reduce([1, 2, 3], 0, &+/2)
6
iex> Enum.map([1, 2, 3], &(&1 * 2))
[2, 4, 6]
Let's take a deeper look at Enumerable and, while we're at it, its lazy counterpart, Stream.

Enumerables and Streams
    

While Elixir allows us to write recursive code, most operations we perform on collections is done with the help of the Enum and Stream modules. Let's learn how.

  
    
  
  Enumerables


Elixir provides the concept of enumerables and the Enum module to work with them. We have already learned two enumerables: lists and maps.
iex> Enum.map([1, 2, 3], fn x -> x * 2 end)
[2, 4, 6]
iex> Enum.map(%{1 => 2, 3 => 4}, fn {k, v} -> k * v end)
[2, 12]
The Enum module provides a huge range of functions to transform, sort, group, filter and retrieve items from enumerables. It is one of the modules developers use frequently in their Elixir code. For a general overview of all functions in the Enum module, see the Enum cheatsheet.
Elixir also provides ranges (see Range), which are also enumerable:
iex> Enum.map(1..3, fn x -> x * 2 end)
[2, 4, 6]
iex> Enum.reduce(1..3, 0, &+/2)
6
The functions in the Enum module are limited to, as the name says, enumerating values in data structures. For specific operations, like inserting and updating particular elements, you may need to reach for modules specific to the data type. For example, if you want to insert an element at a given position in a list, you should use the List.insert_at/3 function, as it would make little sense to insert a value into, for example, a range.
We say the functions in the Enum module are polymorphic because they can work with diverse data types. In particular, the functions in the Enum module can work with any data type that implements the Enumerable protocol. We are going to discuss Protocols in a later chapter, for now we are going to move on to a specific kind of enumerable called a stream.

  
    
  
  Eager vs Lazy


All the functions in the Enum module are eager. Many functions expect an enumerable and return a list back:
iex> odd? = fn x -> rem(x, 2) != 0 end
#Function<6.80484245/1 in :erl_eval.expr/5>
iex> Enum.filter(1..3, odd?)
[1, 3]
This means that when performing multiple operations with Enum, each operation is going to generate an intermediate list until we reach the result:
iex> 1..100_000 |> Enum.map(&(&1 * 3)) |> Enum.filter(odd?) |> Enum.sum()
7500000000
The example above has a pipeline of operations. We start with a range and then multiply each element in the range by 3. This first operation will now create and return a list with 100_000 items. Then we keep all odd elements from the list, generating a new list, now with 50_000 items, and then we sum all entries.

  
    
  
  The pipe operator


The |> symbol used in the snippet above is the pipe operator: it takes the output from the expression on its left side and passes it as the first argument to the function call on its right side. Its purpose is to highlight the data being transformed by a series of functions. To see how it can make the code cleaner, have a look at the example above rewritten without using the |> operator:
iex> Enum.sum(Enum.filter(Enum.map(1..100_000, &(&1 * 3)), odd?))
7500000000
Find more about the pipe operator by reading its documentation.

  
    
  
  Streams


As an alternative to Enum, Elixir provides the Stream module which supports lazy operations:
iex> 1..100_000 |> Stream.map(&(&1 * 3)) |> Stream.filter(odd?) |> Enum.sum()
7500000000
Streams are lazy, composable enumerables.
In the example above, 1..100_000 |> Stream.map(&(&1 * 3)) returns a data type, an actual stream, that represents the map computation over the range 1..100_000:
iex> 1..100_000 |> Stream.map(&(&1 * 3))
#Stream<[enum: 1..100000, funs: [#Function<34.16982430/1 in Stream.map/2>]]>
Furthermore, they are composable because we can pipe many stream operations:
iex> 1..100_000 |> Stream.map(&(&1 * 3)) |> Stream.filter(odd?)
#Stream<[enum: 1..100000, funs: [...]]>
Instead of generating intermediate lists, streams build a series of computations that are invoked only when we pass the underlying stream to the Enum module. Streams are useful when working with large, possibly infinite, collections.
Many functions in the Stream module accept any enumerable as an argument and return a stream as a result. It also provides functions for creating streams. For example, Stream.cycle/1 can be used to create a stream that cycles a given enumerable infinitely. Be careful to not call a function like Enum.map/2 on such streams, as they would cycle forever:
iex> stream = Stream.cycle([1, 2, 3])
#Function<15.16982430/2 in Stream.unfold/2>
iex> Enum.take(stream, 10)
[1, 2, 3, 1, 2, 3, 1, 2, 3, 1]
On the other hand, Stream.unfold/2 can be used to generate values from a given initial value:
iex> stream = Stream.unfold("hełło", &String.next_codepoint/1)
#Function<39.75994740/2 in Stream.unfold/2>
iex> Enum.take(stream, 3)
["h", "e", "ł"]
Another interesting function is Stream.resource/3 which can be used to wrap around resources, guaranteeing they are opened right before enumeration and closed afterwards, even in the case of failures. For example, File.stream!/1 builds on top of Stream.resource/3 to stream files:
iex> stream = File.stream!("path/to/file")
%File.Stream{
  line_or_bytes: :line,
  modes: [:raw, :read_ahead, :binary],
  path: "path/to/file",
  raw: true
}
iex> Enum.take(stream, 10)
The example above will fetch the first 10 lines of the file you have selected. This means streams can be very useful for handling large files or even slow resources like network resources.
The Enum and Stream modules provide a wide range of functions, but you don't have to know all of them by heart. Familiarize yourself with Enum.map/2, Enum.reduce/3 and other functions with either map or reduce in their names, and you will naturally build an intuition around the most important use cases. You may also focus on the Enum module first and only move to Stream for the particular scenarios where laziness is required, to either deal with slow resources or large, possibly infinite, collections.
Next, we'll look at a feature central to Elixir, Processes, which allows us to write concurrent, parallel and distributed programs in an easy and understandable way.

Processes
    

In Elixir, all code runs inside processes. Processes are isolated from each other, run concurrent to one another and communicate via message passing. Processes are not only the basis for concurrency in Elixir, but they also provide the means for building distributed and fault-tolerant programs.
Elixir's processes should not be confused with operating system processes. Processes in Elixir are extremely lightweight in terms of memory and CPU (even compared to threads as used in many other programming languages). Because of this, it is not uncommon to have tens or even hundreds of thousands of processes running simultaneously.
In this chapter, we will learn about the basic constructs for spawning new processes, as well as sending and receiving messages between processes.

  
    
  
  Spawning processes


The basic mechanism for spawning new processes is the auto-imported spawn/1 function:
iex> spawn(fn -> 1 + 2 end)
#PID<0.43.0>
spawn/1 takes a function which it will execute in another process.
Notice spawn/1 returns a PID (process identifier). At this point, the process you spawned is very likely dead. The spawned process will execute the given function and exit after the function is done:
iex> pid = spawn(fn -> 1 + 2 end)
#PID<0.44.0>
iex> Process.alive?(pid)
false
Note: you will likely get different process identifiers than the ones we are showing in our snippets.

We can retrieve the PID of the current process by calling self/0:
iex> self()
#PID<0.41.0>
iex> Process.alive?(self())
true
Processes get much more interesting when we are able to send and receive messages.

  
    
  
  Sending and receiving messages


We can send messages to a process with send/2 and receive them with receive/1:
iex> send(self(), {:hello, "world"})
{:hello, "world"}
iex> receive do
...>   {:hello, msg} -> msg
...>   {:world, _msg} -> "won't match"
...> end
"world"
When a message is sent to a process, the message is stored in the process mailbox. The receive/1 block goes through the current process mailbox searching for a message that matches any of the given patterns. receive/1 supports guards and many clauses, such as case/2.
The process that sends the message does not block on send/2, it puts the message in the recipient's mailbox and continues. In particular, a process can send messages to itself.
If there is no message in the mailbox matching any of the patterns, the current process will wait until a matching message arrives. A timeout can also be specified:
iex> receive do
...>   {:hello, msg}  -> msg
...> after
...>   1_000 -> "nothing after 1s"
...> end
"nothing after 1s"
A timeout of 0 can be given when you already expect the message to be in the mailbox.
Let's put it all together and send messages between processes:
iex> parent = self()
#PID<0.41.0>
iex> spawn(fn -> send(parent, {:hello, self()}) end)
#PID<0.48.0>
iex> receive do
...>   {:hello, pid} -> "Got hello from #{inspect pid}"
...> end
"Got hello from #PID<0.48.0>"
The inspect/1 function is used to convert a data structure's internal representation into a string, typically for printing. Notice that when the receive block gets executed the sender process we have spawned may already be dead, as its only instruction was to send a message.
While in the shell, you may find the helper flush/0 quite useful. It flushes and prints all the messages in the mailbox.
iex> send(self(), :hello)
:hello
iex> flush()
:hello
:ok

  
    
  
  Links


The majority of times we spawn processes in Elixir, we spawn them as linked processes. Before we show an example with spawn_link/1, let's see what happens when a process started with spawn/1 fails:
iex> spawn(fn -> raise "oops" end)
#PID<0.58.0>

[error] Process #PID<0.58.00> raised an exception
** (RuntimeError) oops
    (stdlib) erl_eval.erl:668: :erl_eval.do_apply/6
It merely logged an error but the parent process is still running. That's because processes are isolated. If we want the failure in one process to propagate to another one, we should link them. This can be done with spawn_link/1:
iex> self()
#PID<0.41.0>
iex> spawn_link(fn -> raise "oops" end)

** (EXIT from #PID<0.41.0>) evaluator process exited with reason: an exception was raised:
    ** (RuntimeError) oops
        (stdlib) erl_eval.erl:668: :erl_eval.do_apply/6

[error] Process #PID<0.289.0> raised an exception
** (RuntimeError) oops
    (stdlib) erl_eval.erl:668: :erl_eval.do_apply/6
Because processes are linked, we now see a message saying the parent process, which is the shell process, has received an EXIT signal from another process causing the shell to terminate. IEx detects this situation and starts a new shell session.
Linking can also be done manually by calling Process.link/1. We recommend that you take a look at the Process module for other functionality provided by processes.
Processes and links play an important role when building fault-tolerant systems. Elixir processes are isolated and don't share anything by default. Therefore, a failure in a process will never crash or corrupt the state of another process. Links, however, allow processes to establish a relationship in case of failure. We often link our processes to supervisors which will detect when a process dies and start a new process in its place.
While other languages would require us to catch/handle exceptions, in Elixir we are actually fine with letting processes fail because we expect supervisors to properly restart our systems. "Failing fast" (sometimes referred as "let it crash") is a common philosophy when writing Elixir software!
spawn/1 and spawn_link/1 are the basic primitives for creating processes in Elixir. Although we have used them exclusively so far, most of the time we are going to use abstractions that build on top of them. Let's see the most common one, called tasks.

  
    
  
  Tasks


Tasks build on top of the spawn functions to provide better error reports and introspection:
iex> Task.start(fn -> raise "oops" end)
{:ok, #PID<0.55.0>}

15:22:33.046 [error] Task #PID<0.55.0> started from #PID<0.53.0> terminating
** (RuntimeError) oops
    (stdlib) erl_eval.erl:668: :erl_eval.do_apply/6
    (elixir) lib/task/supervised.ex:85: Task.Supervised.do_apply/2
    (stdlib) proc_lib.erl:247: :proc_lib.init_p_do_apply/3
Function: #Function<20.99386804/0 in :erl_eval.expr/5>
    Args: []
Instead of spawn/1 and spawn_link/1, we use Task.start/1 and Task.start_link/1 which return {:ok, pid} rather than just the PID. This is what enables tasks to be used in supervision trees. Furthermore, Task provides convenience functions, like Task.async/1 and Task.await/1, and functionality to ease distribution.
We will explore tasks and other abstractions around processes in the "Mix and OTP guide".

  
    
  
  State


We haven't talked about state so far. If you are building an application that requires state, for example, to keep your application configuration, or you need to parse a file and keep it in memory, where would you store it?
Processes are the most common answer to this question. We can write processes that loop infinitely, maintain state, and send and receive messages. As an example, let's write a module that starts new processes that work as a key-value store in a file named kv.exs:
defmodule KV do
  def start_link do
    Task.start_link(fn -> loop(%{}) end)
  end

  defp loop(map) do
    receive do
      {:get, key, caller} ->
        send(caller, Map.get(map, key))
        loop(map)
      {:put, key, value} ->
        loop(Map.put(map, key, value))
    end
  end
end
Note that the start_link function starts a new process that runs the loop/1 function, starting with an empty map. The loop/1 (private) function then waits for messages and performs the appropriate action for each message. We made loop/1 private by using defp instead of def. In the case of a :get message, it sends a message back to the caller and calls loop/1 again, to wait for a new message. While the :put message actually invokes loop/1 with a new version of the map, with the given key and value stored.
Let's give it a try by running iex kv.exs:
iex> {:ok, pid} = KV.start_link()
{:ok, #PID<0.62.0>}
iex> send(pid, {:get, :hello, self()})
{:get, :hello, #PID<0.41.0>}
iex> flush()
nil
:ok
At first, the process map has no keys, so sending a :get message and then flushing the current process inbox returns nil. Let's send a :put message and try it again:
iex> send(pid, {:put, :hello, :world})
{:put, :hello, :world}
iex> send(pid, {:get, :hello, self()})
{:get, :hello, #PID<0.41.0>}
iex> flush()
:world
:ok
Notice how the process is keeping a state and we can get and update this state by sending the process messages. In fact, any process that knows the pid above will be able to send it messages and manipulate the state.
It is also possible to register the pid, giving it a name, and allowing everyone that knows the name to send it messages:
iex> Process.register(pid, :kv)
true
iex> send(:kv, {:get, :hello, self()})
{:get, :hello, #PID<0.41.0>}
iex> flush()
:world
:ok
Using processes to maintain state and name registration are very common patterns in Elixir applications. However, most of the time, we won't implement those patterns manually as above, but by using one of the many abstractions that ship with Elixir. For example, Elixir provides Agents, which are simple abstractions around state. Our code above could be directly written as:
iex> {:ok, pid} = Agent.start_link(fn -> %{} end)
{:ok, #PID<0.72.0>}
iex> Agent.update(pid, fn map -> Map.put(map, :hello, :world) end)
:ok
iex> Agent.get(pid, fn map -> Map.get(map, :hello) end)
:world
A :name option could also be given to Agent.start_link/2 and it would be automatically registered. Besides agents, Elixir provides an API for building generic servers (called GenServer), registries, and more, all powered by processes underneath. Those, along with supervision trees, will be explored with more detail in the "Mix and OTP guide", which will build a complete Elixir application from start to finish.
For now, let's move on and explore the world of I/O in Elixir.

IO and the file system
    

This chapter introduces the input/output mechanisms, file-system-related tasks, and related modules such as IO, File, and Path. The IO system provides a great opportunity to shed some light on some philosophies and curiosities of Elixir and the Erlang VM.

  
    
  
  The IO module


The IO module is the main mechanism in Elixir for reading and writing to standard input/output (:stdio), standard error (:stderr), files, and other IO devices. Usage of the module is pretty straightforward:
iex> IO.puts("hello world")
hello world
:ok
iex> IO.gets("yes or no? ")
yes or no? yes
"yes\n"
By default, functions in the IO module read from the standard input and write to the standard output. We can change that by passing, for example, :stderr as an argument (in order to write to the standard error device):
iex> IO.puts(:stderr, "hello world")
hello world
:ok

  
    
  
  The File module


The File module contains functions that allow us to open files as IO devices. By default, files are opened in binary mode, which requires developers to use the specific IO.binread/2 and IO.binwrite/2 functions from the IO module:
Potential data loss warning
The following code opens a file for writing. If an existing file is available at the given path, its contents will be deleted.

iex> {:ok, file} = File.open("path/to/file/hello", [:write])
{:ok, #PID<0.47.0>}
iex> IO.binwrite(file, "world")
:ok
iex> File.close(file)
:ok
iex> File.read("path/to/file/hello")
{:ok, "world"}
The file could be opened with the :append option, instead of :write, to preserve its contents. You may also pass the :utf8 option, which tells the File module to interpret the bytes read from the file as UTF-8-encoded bytes.
Besides functions for opening, reading and writing files, the File module has many functions to work with the file system. Those functions are named after their UNIX equivalents. For example, File.rm/1 can be used to remove files, File.mkdir/1 to create directories, File.mkdir_p/1 to create directories and all their parent chain. There are even File.cp_r/2 and File.rm_rf/1 to respectively copy and remove files and directories recursively (i.e., copying and removing the contents of the directories too).
You will also notice that functions in the File module have two variants: one "regular" variant and another variant with a trailing bang (!). For example, when we read the "hello" file in the example above, we use File.read/1. Alternatively, we can use File.read!/1:
iex> File.read("path/to/file/hello")
{:ok, "world"}
iex> File.read!("path/to/file/hello")
"world"
iex> File.read("path/to/file/unknown")
{:error, :enoent}
iex> File.read!("path/to/file/unknown")
** (File.Error) could not read file "path/to/file/unknown": no such file or directory
Notice that the version with ! returns the contents of the file instead of a tuple, and if anything goes wrong the function raises an error.
The version without ! is preferred when you want to handle different outcomes using pattern matching:
case File.read("path/to/file/hello") do
  {:ok, body} -> # do something with the `body`
  {:error, reason} -> # handle the error caused by `reason`
end
However, if you expect the file to be there, the bang variation is more useful as it raises a meaningful error message. Avoid writing:
{:ok, body} = File.read("path/to/file/unknown")
as, in case of an error, File.read/1 will return {:error, reason} and the pattern matching will fail. You will still get the desired result (a raised error), but the message will be about the pattern which doesn't match (thus being cryptic in respect to what the error actually is about).
Therefore, if you don't want to handle the error outcomes, prefer to use the functions ending with an exclamation mark, such as File.read!/1.

  
    
  
  The Path module


The majority of the functions in the File module expect paths as arguments. Most commonly, those paths will be regular binaries. The Path module provides facilities for working with such paths:
iex> Path.join("foo", "bar")
"foo/bar"
iex> Path.expand("~/hello")
"/Users/jose/hello"
Using functions from the Path module as opposed to directly manipulating strings is preferred since the Path module takes care of different operating systems transparently. Finally, keep in mind that Elixir will automatically convert slashes (/) into backslashes (\) on Windows when performing file operations.
With this, we have covered the main modules that Elixir provides for dealing with IO and interacting with the file system. In the next section, we will peek a bit under the covers and learn how the IO system is implemented in the VM.

  
    
  
  Processes


You may have noticed that File.open/2 returns a tuple like {:ok, pid}:
iex> {:ok, file} = File.open("hello")
{:ok, #PID<0.47.0>}
This happens because the IO module actually works with processes (see the previous chapter). Given a file is a process, when you write to a file that has been closed, you are actually sending a message to a process which has been terminated:
iex> File.close(file)
:ok
iex> IO.write(file, "is anybody out there")
** (ErlangError) Erlang error: :terminated:

  * 1st argument: the device has terminated

    (stdlib 5.0) io.erl:94: :io.put_chars(#PID<0.114.0>, "is anybody out there")
    iex:4: (file)
Let's see in more detail what happens when you request IO.write(pid, binary). The IO module sends a message to the process identified by pid with the desired operation. A small ad-hoc process can help us see it:
iex> pid = spawn(fn ->
...>  receive do: (msg -> IO.inspect(msg))
...> end)
#PID<0.57.0>
iex> IO.write(pid, "hello")
{:io_request, #PID<0.41.0>, #Reference<0.0.8.91>,
 {:put_chars, :unicode, "hello"}}
** (ErlangError) erlang error: :terminated
After IO.write/2, we can see the request sent by the IO module printed out (a four-elements tuple). Soon after that, we see that it fails since the IO module expected some kind of result, which we did not supply.
By modeling IO devices with processes, the Erlang VM allows us to even read and write to files across nodes. Neat!

  
    
  
  iodata and chardata


In all of the examples above, we used binaries when writing to files. However, most of the IO functions in Elixir also accept either "iodata" or "chardata".
One of the main reasons for using "iodata" and "chardata" is for performance. For example,
imagine you need to greet someone in your application:
name = "Mary"
IO.puts("Hello " <> name <> "!")
Given strings in Elixir are immutable, as most data structures, the example above will copy the string "Mary" into the new "Hello Mary!" string. While this is unlikely to matter for the short string as above, copying can be quite expensive for large strings! For this reason, the IO functions in Elixir allow you to pass instead a list of strings:
name = "Mary"
IO.puts(["Hello ", name, "!"])
In the example above, there is no copying. Instead we create a list that contains the original name. We call such lists either "iodata" or "chardata" and we will learn the precise difference between them soon.
Those lists are very useful because it can actually simplify the processing strings in several scenarios. For example, imagine you have a list of values, such as ["apple", "banana", "lemon"] that you want to write to disk separated by commas. How can you achieve this?
One option is to use Enum.join/2 and convert the values to a string:
iex> Enum.join(["apple", "banana", "lemon"], ",")
"apple,banana,lemon"
The above returns a new string by copying each value into the new string. However, with the knowledge in this section, we know that we can pass a list of strings to the IO/File functions. So instead we can do:
iex> Enum.intersperse(["apple", "banana", "lemon"], ",")
["apple", ",", "banana", ",", "lemon"]
"iodata" and "chardata" do not only contain strings, but they may contain arbitrary nested lists of strings too:
iex> IO.puts(["apple", [",", "banana", [",", "lemon"]]])
"iodata" and "chardata" may also contain integers. For example, we could print our comma separated list of values by using ?, as separator, which is the integer representing a comma (44):
iex> IO.puts(["apple", ?,, "banana", ?,, "lemon"])
The difference between "iodata" and "chardata" is precisely what said integer represents. For iodata, the integers represent bytes. For chardata, the integers represent Unicode codepoints. For ASCII characters, the byte representation is the same as the codepoint representation, so it fits both classifications. However, the default IO device works with chardata, which means we can do:
iex> IO.puts([?O, ?l, ?á, ?\s, "Mary", ?!])
Overall, integers in a list may represent either a bunch of bytes or a bunch of characters and which one to use depends on the encoding of the IO device. If the file is opened without encoding, the file is expected to be in raw mode, and the functions in the IO module starting with bin* must be used. Those functions expect an iodata as an argument, where integers in the list would represent bytes.
On the other hand, the default IO device (:stdio) and files opened with :utf8 encoding work with the remaining functions in the IO module. Those functions expect a chardata as an argument, where integers represent codepoints.
Although this is a subtle difference, you only need to worry about these details if you intend to pass lists containing integers to those functions. If you pass binaries, or list of binaries, then there is no ambiguity.
Finally, there is one last construct called charlist, which we discussed in earlier chapters. Charlists are a special case of chardata where all values are integers representing Unicode codepoints. They can be created with the ~c sigil:
iex> ~c"hello"
~c"hello"
Charlists mostly show up when interfacing with Erlang, as some Erlang APIs use charlist as their representation for strings. For this reason, any list containing printable ASCII codepoints will be printed as a charlist:
iex> [?a, ?b, ?c]
~c"abc"
We packed a lot into this small section, so let's break it down:
	iodata and chardata are lists of binaries and integers. Those binaries and integers can be arbitrarily nested inside lists. Their goal is to give flexibility and performance when working with IO devices and files;

	the choice between iodata and chardata depends on the encoding of the IO device. If the file is opened without encoding, the file expects iodata, and the functions in the IO module starting with bin* must be used. The default IO device (:stdio) and files opened with :utf8 encoding expect chardata and work with the remaining functions in the IO module;

	charlists are a special case of chardata, where it exclusively uses a list of integers Unicode codepoints. They can be created with the ~c sigil. Lists of integers are automatically printed using the ~c sigil if all integers in a list represent printable ASCII codepoints.


This finishes our tour of IO devices and IO related functionality. We have learned about three Elixir modules - IO, File, and Path - as well as how the VM uses processes for the underlying IO mechanisms and how to use chardata and iodata for IO operations.

alias, require, import, and use
    

In order to facilitate software reuse, Elixir provides three directives (alias, require, and import) plus a macro called use summarized below:
# Alias the module so it can be called as Bar instead of Foo.Bar
alias Foo.Bar, as: Bar

# Require the module in order to use its macros
require Foo

# Import functions from Foo so they can be called without the `Foo.` prefix
import Foo

# Invokes the custom code defined in Foo as an extension point
use Foo
We are going to explore them in detail now. Keep in mind the first three are called directives because they have lexical scope, while use is a common extension point that allows the used module to inject code.

  
    
  
  alias


alias allows you to set up aliases for any given module name.
Imagine a module uses a specialized list implemented in Math.List. The alias directive allows referring to Math.List just as List within the module definition:
defmodule Stats do
  alias Math.List, as: List
  # In the remaining module definition List expands to Math.List.
end
The original List can still be accessed within Stats by the fully-qualified name Elixir.List.
All modules defined in Elixir are defined inside the main Elixir namespace, such as Elixir.String. However, for convenience, you can omit "Elixir." when referencing them.

Aliases are frequently used to define shortcuts. In fact, calling alias without an :as option sets the alias automatically to the last part of the module name, for example:
alias Math.List
Is the same as:
alias Math.List, as: List
Note that alias is lexically scoped, which allows you to set aliases inside specific functions:
defmodule Math do
  def plus(a, b) do
    alias Math.List
    # ...
  end

  def minus(a, b) do
    # ...
  end
end
In the example above, since we are invoking alias inside the function plus/2, the alias will be valid only inside the function plus/2. minus/2 won't be affected at all.

  
    
  
  require


Elixir provides macros as a mechanism for meta-programming (writing code that generates code). Macros are expanded at compile time.
Public functions in modules are globally available, but in order to use macros, you need to opt-in by requiring the module they are defined in.
iex> Integer.is_odd(3)
** (UndefinedFunctionError) function Integer.is_odd/1 is undefined or private. However, there is a macro with the same name and arity. Be sure to require Integer if you intend to invoke this macro
    (elixir) Integer.is_odd(3)
iex> require Integer
Integer
iex> Integer.is_odd(3)
true
In Elixir, Integer.is_odd/1 is defined as a macro so that it can be used as a guard. This means that, in order to invoke Integer.is_odd/1, we need to first require the Integer module.
Note that like the alias directive, require is also lexically scoped. We will talk more about macros in a later chapter.

  
    
  
  import


We use import whenever we want to access functions or macros from other modules without using the fully-qualified name. Note we can only import public functions, as private functions are never accessible externally.
For example, if we want to use the duplicate/2 function from the List module several times, we can import it:
iex> import List, only: [duplicate: 2]
List
iex> duplicate(:ok, 3)
[:ok, :ok, :ok]
We imported only the function duplicate (with arity 2) from List. Although :only is optional, its usage is recommended in order to avoid importing all the functions of a given module inside the current scope. :except could also be given as an option in order to import everything in a module except a list of functions.
Note that import is lexically scoped too. This means that we can import specific macros or functions inside function definitions:
defmodule Math do
  def some_function do
    import List, only: [duplicate: 2]
    duplicate(:ok, 10)
  end
end
In the example above, the imported List.duplicate/2 is only visible within that specific function. duplicate/2 won't be available in any other function in that module (or any other module for that matter).
While imports can be a useful for frameworks and libraries to build abstractions, developers should generally prefer alias to import on their own codebases, as aliases make the origin of the function being invoked clearer.

  
    
  
  use


The use macro is frequently used as an extension point. This means that, when you use a module FooBar, you allow that module to inject any code in the current module, such as importing itself or other modules, defining new functions, setting a module state, etc.
For example, in order to write tests using the ExUnit framework, a developer should use the ExUnit.Case module:
defmodule AssertionTest do
  use ExUnit.Case, async: true

  test "always pass" do
    assert true
  end
end
Behind the scenes, use requires the given module and then calls the __using__/1 callback on it allowing the module to inject some code into the current context. Some modules (for example, the above ExUnit.Case, but also Supervisor and GenServer) use this mechanism to populate your module with some basic behaviour, which your module is intended to override or complete.
Generally speaking, the following module:
defmodule Example do
  use Feature, option: :value
end
is compiled into
defmodule Example do
  require Feature
  Feature.__using__(option: :value)
end
Since use allows any code to run, we can't really know the side-effects of using a module without reading its documentation. Therefore use this function with care and only if strictly required. Don't use use where an import or alias would do.

  
    
  
  Understanding Aliases


At this point, you may be wondering: what exactly is an Elixir alias and how is it represented?
An alias in Elixir is a capitalized identifier (like String, Keyword, etc) which is converted to an atom during compilation. For instance, the String alias translates by default to the atom :"Elixir.String":
iex> is_atom(String)
true
iex> to_string(String)
"Elixir.String"
iex> :"Elixir.String" == String
true
By using the alias/2 directive, we are changing the atom the alias expands to.
Aliases expand to atoms because in the Erlang Virtual Machine (and consequently Elixir) modules are always represented by atoms:
iex> List.flatten([1, [2], 3])
[1, 2, 3]
iex> :"Elixir.List".flatten([1, [2], 3])
[1, 2, 3]
That's the mechanism we use to call Erlang modules:
iex> :lists.flatten([1, [2], 3])
[1, 2, 3]

  
    
  
  Module nesting


Now that we have talked about aliases, we can talk about nesting and how it works in Elixir. Consider the following example:
defmodule Foo do
  defmodule Bar do
  end
end
The example above will define two modules: Foo and Foo.Bar. The second can be accessed as Bar inside Foo as long as they are in the same lexical scope.
If, later, the Bar module is moved outside the Foo module definition, it must be referenced by its full name (Foo.Bar) or an alias must be set using the alias directive discussed above.
Note: in Elixir, you don't have to define the Foo module before being able to define the Foo.Bar module, as they are effectively independent. The above could also be written as:
defmodule Foo.Bar do
end

defmodule Foo do
  alias Foo.Bar
  # Can still access it as `Bar`
end
Aliasing a nested module does not bring parent modules into scope. Consider the following example:
defmodule Foo do
  defmodule Bar do
    defmodule Baz do
    end
  end
end

alias Foo.Bar.Baz
# The module `Foo.Bar.Baz` is now available as `Baz`
# However, the module `Foo.Bar` is *not* available as `Bar`
As we will see in later chapters, aliases also play a crucial role in macros, to guarantee they are hygienic.

  
    
  
  Multi alias/import/require/use


It is possible to alias, import, require, or use multiple modules at once. This is particularly useful once we start nesting modules, which is very common when building Elixir applications. For example, imagine you have an application where all modules are nested under MyApp, you can alias the modules MyApp.Foo, MyApp.Bar and MyApp.Baz at once as follows:
alias MyApp.{Foo, Bar, Baz}
With this, we have finished our tour of Elixir modules. The next topic to cover is module attributes.

Module attributes
    

Module attributes in Elixir serve three purposes:
	as module and function annotations
	as temporary module storage to be used during compilation
	as compile-time constants

Let's check these examples.

  
    
  
  As annotations


Elixir brings the concept of module attributes from Erlang. For example:
defmodule MyServer do
  @moduledoc "My server code."
end
In the example above, we are defining the module documentation by using the module attribute syntax. Elixir has a handful of reserved attributes. Here are a few of them, the most commonly used ones:
	@moduledoc — provides documentation for the current module.
	@doc — provides documentation for the function or macro that follows the attribute.
	@spec — provides a typespec for the function that follows the attribute.
	@behaviour — (notice the British spelling) used for specifying an OTP or user-defined behaviour.

@moduledoc and @doc are by far the most used attributes, and we expect you to use them a lot. Elixir treats documentation as first-class and provides many functions to access documentation. We will cover them in their own chapter.
Let's go back to the Math module defined in the previous chapters, add some documentation and save it to the math.ex file:
defmodule Math do
  @moduledoc """
  Provides math-related functions.

  ## Examples

      iex> Math.sum(1, 2)
      3

  """

  @doc """
  Calculates the sum of two numbers.
  """
  def sum(a, b), do: a + b
end
Elixir promotes the use of Markdown with heredocs to write readable documentation. Heredocs are multi-line strings, they start and end with triple double-quotes, keeping the formatting of the inner text. We can access the documentation of any compiled module directly from IEx:
$ elixirc math.ex
$ iex

iex> h Math # Access the docs for the module Math
...
iex> h Math.sum # Access the docs for the sum function
...
We also provide a tool called ExDoc which is used to generate HTML pages from the documentation.
You can take a look at the docs for Module for a complete list of supported attributes. Elixir also uses attributes to annotate our code with typespecs.

  
    
  
  As temporary storage


So far, we have seen how to define attributes, but how can read them? Let's see an example:
defmodule MyServer do
  @service URI.parse("https://example.com")
  IO.inspect @service
end
Newlines
Do not add a newline between the attribute and its value, otherwise Elixir will assume you are reading the value, rather than setting it.

Trying to access an attribute that was not defined will print a warning:
defmodule MyServer do
  @unknown
end
warning: undefined module attribute @unknown, please remove access to @unknown or explicitly set it before access
Attributes can also be read inside functions:
defmodule MyApp.Status do
  @service URI.parse("https://example.com")
  def status(email) do
    SomeHttpClient.get(@service)
  end
end
The module attribute is defined at compilation time and its return value, not the function call itself, is what will be substituted in for the attribute. So the above will effectively compile to this:
defmodule MyApp.Status do
  def status(email) do
    SomeHttpClient.get(%URI{
      authority: "example.com",
      host: "example.com",
      port: 443,
      scheme: "https"
    })
  end
end
This can be useful for pre-computing values and then injecting its results into the module. This is what we mean by temporary storage: after the module is compiled, the module attribute is discarded, except for the functions that have read the attribute. Note you cannot invoke functions defined in the same module as part of the attribute itself, as those functions have not yet been defined.
Every time we read an attribute inside a function, Elixir takes a snapshot of its current value. Therefore if you read the same attribute multiple times inside multiple functions, you end-up increasing compilation times as Elixir now has to compile every snapshot. Generally speaking, you want to avoid reading the same attribute multiple times and instead move it to function. For example, instead of this:
def some_function, do: do_something_with(@example)
def another_function, do: do_something_else_with(@example)
Prefer this:
def some_function, do: do_something_with(example())
def another_function, do: do_something_else_with(example())
defp example, do: @example

  
    
  
  As compile-time constants


Module attributes may also be useful as compile-time constants. Generally speaking, functions themselves are sufficient for the role of constants in a codebase. For example, instead of defining:
@hours_in_a_day 24
You should prefer:
defp hours_in_a_day(), do: 24
You may even define a public function if it needs to be shared across modules. It is common in many projects to have a module called MyApp.Constants that defines all constants used throughout the codebase.
You can even have composite data structures as constants, as long as they are made exclusively of other data types (no function calls, no operators, and no other expressions). For example, you may specify a system configuration constant as follows:
defp system_config(), do: %{timezone: "Etc/UTC", locale: "pt-BR"}
Given data structures in Elixir are immutable, only a single instance of the data structure above is allocated and shared across all functions calls, as long as it doesn't have any executable expression.
The use case for module attributes arise when you need to do some work at compile-time and then inject its results inside a function. A common scenario is module attributes inside patterns and guards (as an alternative to defguard/1), since they only support a limited set of expressions:
# Inside pattern
@default_timezone "Etc/UTC"
def shift(@default_timezone), do: ...

# Inside guards
@time_periods [:am, :pm]
def shift(time, period) when period in @time_periods, do: ...
Module attributes as constants and as temporary storage are most often used together: the module attribute is used to compute and store an expensive value, and then exposed as constant from that module.

  
    
  
  Going further


Libraries and frameworks can leverage module attributes to provide custom annotations. To see an example, look no further than Elixir's unit test framework called ExUnit. ExUnit uses module attributes for multiple different purposes:
defmodule MyTest do
  use ExUnit.Case, async: true

  @tag :external
  @tag os: :unix
  test "contacts external service" do
    # ...
  end
end
In the example above, ExUnit stores the value of async: true in a module attribute to change how the module is compiled. Tags also work as annotations and they can be supplied multiple times, thanks to Elixir's ability to accumulate attribute. Then you can use tags to setup and filter tests, such as avoiding executing Unix specific tests while running your test suite on Windows.
To fully understand how ExUnit works, we'd need macros, so we will revisit this pattern in the Meta-programming guide and learn how to use module attributes as storage for custom annotations.
In the next chapters, we'll explore structs and protocols before moving to exception handling and other constructs like sigils and comprehensions.

Structs
    

We learned about maps in earlier chapters:
iex> map = %{a: 1, b: 2}
%{a: 1, b: 2}
iex> map[:a]
1
iex> %{map | a: 3}
%{a: 3, b: 2}
Structs are extensions built on top of maps that provide compile-time checks and default values.

  
    
  
  Defining structs


To define a struct, the defstruct/1 construct is used:
iex> defmodule User do
...>   defstruct name: "John", age: 27
...> end
The keyword list used with defstruct defines what fields the struct will have along with their default values. Structs take the name of the module they're defined in. In the example above, we defined a struct named User.
We can now create User structs by using a syntax similar to the one used to create maps:
iex> %User{}
%User{age: 27, name: "John"}
iex> %User{name: "Jane"}
%User{age: 27, name: "Jane"}
Structs provide compile-time guarantees that only the fields defined through defstruct will be allowed to exist in a struct:
iex> %User{oops: :field}
** (KeyError) key :oops not found expanding struct: User.__struct__/1

  
    
  
  Accessing and updating structs


Structs share the same syntax for accessing and updating fields as maps of fixed keys:
iex> john = %User{}
%User{age: 27, name: "John"}
iex> john.name
"John"
iex> jane = %{john | name: "Jane"}
%User{age: 27, name: "Jane"}
iex> %{jane | oops: :field}
** (KeyError) key :oops not found in: %User{age: 27, name: "Jane"}
When using the update syntax (|), Elixir is aware that no new keys will be added to the struct, allowing the maps underneath to share their structure in memory. In the example above, both john and jane share the same key structure in memory.
Structs can also be used in pattern matching, both for matching on the value of specific keys as well as for ensuring that the matching value is a struct of the same type as the matched value.
iex> %User{name: name} = john
%User{age: 27, name: "John"}
iex> name
"John"
iex> %User{} = %{}
** (MatchError) no match of right hand side value: %{}

  
    
  
  Structs are bare maps underneath


Structs are simply maps with a "special" field named __struct__ that holds the name of the struct:
iex> is_map(john)
true
iex> john.__struct__
User
However, structs do not inherit any of the protocols that maps do. For example, you can neither enumerate nor access a struct:
iex> john = %User{}
%User{age: 27, name: "John"}
iex> john[:name]
** (UndefinedFunctionError) function User.fetch/2 is undefined (User does not implement the Access behaviour)
             User.fetch(%User{age: 27, name: "John"}, :name)
iex> Enum.each(john, fn {field, value} -> IO.puts(value) end)
** (Protocol.UndefinedError) protocol Enumerable not implemented for %User{age: 27, name: "John"} of type User (a struct)
Structs alongside protocols provide one of the most important features for Elixir developers: data polymorphism. That's what we will explore in the next chapter.

  
    
  
  Default values and required keys


If you don't specify a default key value when defining a struct, nil will be assumed:
iex> defmodule Product do
...>   defstruct [:name]
...> end
iex> %Product{}
%Product{name: nil}
You can define a structure combining both fields with explicit default values, and implicit nil values. In this case you must first specify the fields which implicitly default to nil:
iex> defmodule User do
...>   defstruct [:email, name: "John", age: 27]
...> end
iex> %User{}
%User{age: 27, email: nil, name: "John"}
Doing it in reverse order will raise a syntax error:
iex> defmodule User do
...>   defstruct [name: "John", age: 27, :email]
...> end
** (SyntaxError) iex:107: unexpected expression after keyword list. Keyword lists must always come last in lists and maps.
You can also enforce that certain keys have to be specified when creating the struct via the @enforce_keys module attribute:
iex> defmodule Car do
...>   @enforce_keys [:make]
...>   defstruct [:model, :make]
...> end
iex> %Car{}
** (ArgumentError) the following keys must also be given when building struct Car: [:make]
    expanding struct: Car.__struct__/1
Enforcing keys provides a simple compile-time guarantee to aid developers when building structs. It is not enforced on updates and it does not provide any sort of value-validation.

Protocols
    

Protocols are a mechanism to achieve polymorphism in Elixir where you want the behavior to vary depending on the data type. We are already familiar with one way of solving this type of problem: via pattern matching and guard clauses. Consider a simple utility module that would tell us the type of input variable:
defmodule Utility do
  def type(value) when is_binary(value), do: "string"
  def type(value) when is_integer(value), do: "integer"
  # ... other implementations ...
end
If the use of this module were confined to your own project, you would be able to keep defining new type/1 functions for each new data type. However, this code could be problematic if it was shared as a dependency by multiple apps because there would be no easy way to extend its functionality.
This is where protocols can help us: protocols allow us to extend the original behavior for as many data types as we need. That's because dispatching on a protocol is available to any data type that has implemented the protocol and a protocol can be implemented by anyone, at any time.
Here's how we could write the same Utility.type/1 functionality as a protocol:
defprotocol Utility do
  @spec type(t) :: String.t()
  def type(value)
end

defimpl Utility, for: BitString do
  def type(_value), do: "string"
end

defimpl Utility, for: Integer do
  def type(_value), do: "integer"
end
We define the protocol using defprotocol/2 - its functions and specs may look similar to interfaces or abstract base classes in other languages. We can add as many implementations as we like using defimpl/2. The output is exactly the same as if we had a single module with multiple functions:
iex> Utility.type("foo")
"string"
iex> Utility.type(123)
"integer"
With protocols, however, we are no longer stuck having to continuously modify the same module to support more and more data types. For example, we could spread the defimpl calls above over multiple files and Elixir will dispatch the execution to the appropriate implementation based on the data type. Functions defined in a protocol may have more than one input, but the dispatching will always be based on the data type of the first input.
One of the most common protocols you may encounter is the String.Chars protocol: implementing its to_string/1 function for your custom structs will tell the Elixir kernel how to represent them as strings. We will explore all built-in protocols later. For now, let's implement our own.

  
    
  
  Example


Now that you have seen an example of the type of problem protocols help solve and how they solve them, let's look at a more in-depth example.
In Elixir, we have two idioms for checking how many items there are in a data structure: length and size. length means the information must be computed. For example, length(list) needs to traverse the whole list to calculate its length. On the other hand, tuple_size(tuple) and byte_size(binary) do not depend on the tuple and binary size as the size information is pre-computed in the data structure.
Even if we have type-specific functions for getting the size built into Elixir (such as tuple_size/1), we could implement a generic Size protocol that all data structures for which size is pre-computed would implement.
The protocol definition would look like this:
defprotocol Size do
  @doc "Calculates the size (and not the length!) of a data structure"
  def size(data)
end
The Size protocol expects a function called size that receives one argument (the data structure we want to know the size of) to be implemented. We can now implement this protocol for the data structures that would have a compliant implementation:
defimpl Size, for: BitString do
  def size(string), do: byte_size(string)
end

defimpl Size, for: Map do
  def size(map), do: map_size(map)
end

defimpl Size, for: Tuple do
  def size(tuple), do: tuple_size(tuple)
end
We didn't implement the Size protocol for lists as there is no "size" information pre-computed for lists, and the length of a list has to be computed (with length/1).
Now with the protocol defined and implementations in hand, we can start using it:
iex> Size.size("foo")
3
iex> Size.size({:ok, "hello"})
2
iex> Size.size(%{label: "some label"})
1
Passing a data type that doesn't implement the protocol raises an error:
iex> Size.size([1, 2, 3])
** (Protocol.UndefinedError) protocol Size not implemented for [1, 2, 3] of type List
It's possible to implement protocols for all Elixir data types:
	Atom
	BitString
	Float
	Function
	Integer
	List
	Map
	PID
	Port
	Reference
	Tuple


  
    
  
  Protocols and structs


The power of Elixir's extensibility comes when protocols and structs are used together.
In the previous chapter, we have learned that although structs are maps, they do not share protocol implementations with maps. For example, MapSets (sets based on maps) are implemented as structs. Let's try to use the Size protocol with a MapSet:
iex> Size.size(%{})
0
iex> set = %MapSet{} = MapSet.new
MapSet.new([])
iex> Size.size(set)
** (Protocol.UndefinedError) protocol Size not implemented for MapSet.new([]) of type MapSet (a struct)
Instead of sharing protocol implementation with maps, structs require their own protocol implementation. Since a MapSet has its size precomputed and accessible through MapSet.size/1, we can define a Size implementation for it:
defimpl Size, for: MapSet do
  def size(set), do: MapSet.size(set)
end
If desired, you could come up with your own semantics for the size of your struct. Not only that, you could use structs to build more robust data types, like queues, and implement all relevant protocols, such as Enumerable and possibly Size, for this data type.
defmodule User do
  defstruct [:name, :age]
end

defimpl Size, for: User do
  def size(_user), do: 2
end

  
    
  
  Implementing Any


Manually implementing protocols for all types can quickly become repetitive and tedious. In such cases, Elixir provides two options: we can explicitly derive the protocol implementation for our types or automatically implement the protocol for all types. In both cases, we need to implement the protocol for Any.

  
    
  
  Deriving


Elixir allows us to derive a protocol implementation based on the Any implementation. Let's first implement Any as follows:
defimpl Size, for: Any do
  def size(_), do: 0
end
The implementation above is arguably not a reasonable one. For example, it makes no sense to say that the size of a PID or an Integer is 0.
However, should we be fine with the implementation for Any, in order to use such implementation we would need to tell our struct to explicitly derive the Size protocol:
defmodule OtherUser do
  @derive [Size]
  defstruct [:name, :age]
end
When deriving, Elixir will implement the Size protocol for OtherUser based on the implementation provided for Any.

  
    
  
  Fallback to Any


Another alternative to @derive is to explicitly tell the protocol to fallback to Any when an implementation cannot be found. This can be achieved by setting @fallback_to_any to true in the protocol definition:
defprotocol Size do
  @fallback_to_any true
  def size(data)
end
As we said in the previous section, the implementation of Size for Any is not one that can apply to any data type. That's one of the reasons why @fallback_to_any is an opt-in behavior. For the majority of protocols, raising an error when a protocol is not implemented is the proper behavior. That said, assuming we have implemented Any as in the previous section:
defimpl Size, for: Any do
  def size(_), do: 0
end
Now all data types (including structs) that have not implemented the Size protocol will be considered to have a size of 0.
Which technique is best between deriving and falling back to Any depends on the use case but, given Elixir developers prefer explicit over implicit, you may see many libraries pushing towards the @derive approach.

  
    
  
  Built-in protocols


Elixir ships with some built-in protocols. In previous chapters, we have discussed the Enum module which provides many functions that work with any data structure that implements the Enumerable protocol:
iex> Enum.map([1, 2, 3], fn x -> x * 2 end)
[2, 4, 6]
iex> Enum.reduce(1..3, 0, fn x, acc -> x + acc end)
6
Another useful example is the String.Chars protocol, which specifies how to convert a data structure to its human representation as a string. It's exposed via the to_string function:
iex> to_string(:hello)
"hello"
Notice that string interpolation in Elixir calls the to_string function:
iex> "age: #{25}"
"age: 25"
The snippet above only works because numbers implement the String.Chars protocol. Passing a tuple, for example, will lead to an error:
iex> tuple = {1, 2, 3}
{1, 2, 3}
iex> "tuple: #{tuple}"
** (Protocol.UndefinedError) protocol String.Chars not implemented for {1, 2, 3} of type Tuple
When there is a need to "print" a more complex data structure, one can use the inspect function, based on the Inspect protocol:
iex> "tuple: #{inspect(tuple)}"
"tuple: {1, 2, 3}"
The Inspect protocol is the protocol used to transform any data structure into a readable textual representation. This is what tools like IEx use to print results:
iex> {1, 2, 3}
{1, 2, 3}
iex> %User{}
%User{name: "john", age: 27}
Keep in mind that, by convention, whenever the inspected value starts with #, it is representing a data structure in non-valid Elixir syntax. This means the inspect protocol is not reversible as information may be lost along the way:
iex> inspect &(&1+2)
"#Function<6.71889879/1 in :erl_eval.expr/5>"
There are other protocols in Elixir but this covers the most common ones. You can learn more about protocols and implementations in the Protocol module.

Comprehensions
    

In Elixir, it is common to loop over an Enumerable, often filtering out some results and mapping values into another list. Comprehensions are syntactic sugar for such constructs: they group those common tasks into the for special form.
For example, we can map a list of integers into their squared values:
iex> for n <- [1, 2, 3, 4], do: n * n
[1, 4, 9, 16]
A comprehension is made of three parts: generators, filters, and collectables.

  
    
  
  Generators and filters


In the expression above, n <- [1, 2, 3, 4] is the generator. It is literally generating values to be used in the comprehension. Any enumerable can be passed on the right-hand side of the generator expression:
iex> for n <- 1..4, do: n * n
[1, 4, 9, 16]
Generator expressions also support pattern matching on their left-hand side; all non-matching patterns are ignored. Imagine that, instead of a range, we have a keyword list where the key is the atom :good or :bad and we only want to compute the square of the :good values:
iex> values = [good: 1, good: 2, bad: 3, good: 4]
iex> for {:good, n} <- values, do: n * n
[1, 4, 16]
Alternatively to pattern matching, filters can be used to select some particular elements. For example, we can select the multiples of 3 and discard all others:
iex> for n <- 0..5, rem(n, 3) == 0, do: n * n
[0, 9]
Comprehensions discard all elements for which the filter expression returns false or nil; all other values are selected.
Comprehensions generally provide a much more concise representation than using the equivalent functions from the Enum and Stream modules. Furthermore, comprehensions also allow multiple generators and filters to be given. Here is an example that receives a list of directories and gets the size of each file in those directories:
dirs = ["/home/mikey", "/home/james"]

for dir <- dirs,
    file <- File.ls!(dir),
    path = Path.join(dir, file),
    File.regular?(path) do
  File.stat!(path).size
end
Multiple generators can also be used to calculate the Cartesian product of two lists:
iex> for i <- [:a, :b, :c], j <- [1, 2], do:  {i, j}
[a: 1, a: 2, b: 1, b: 2, c: 1, c: 2]
Finally, keep in mind that variable assignments inside the comprehension, be it in generators, filters or inside the block, are not reflected outside of the comprehension.

  
    
  
  Bitstring generators


Bitstring generators are also supported and are very useful when you need to comprehend over bitstring streams. The example below receives a list of pixels from a binary with their respective red, green and blue values and converts them into tuples of three elements each:
iex> pixels = <<213, 45, 132, 64, 76, 32, 76, 0, 0, 234, 32, 15>>
iex> for <<r::8, g::8, b::8 <- pixels>>, do: {r, g, b}
[{213, 45, 132}, {64, 76, 32}, {76, 0, 0}, {234, 32, 15}]
A bitstring generator can be mixed with "regular" enumerable generators, and supports filters as well.

  
    
  
  The :into option


In the examples above, all the comprehensions returned lists as their result. However, the result of a comprehension can be inserted into different data structures by passing the :into option to the comprehension.
For example, a bitstring generator can be used with the :into option in order to easily remove all spaces in a string:
iex> for <<c <- " hello world ">>, c != ?\s, into: "", do: <<c>>
"helloworld"
Sets, maps, and other dictionaries can also be given to the :into option. In general, :into accepts any structure that implements the Collectable protocol.
A common use case of :into can be transforming values in a map:
iex> for {key, val} <- %{"a" => 1, "b" => 2}, into: %{}, do: {key, val * val}
%{"a" => 1, "b" => 4}
Let's make another example using streams. Since the IO module provides streams (that are both Enumerables and Collectables), an echo terminal that echoes back the upcased version of whatever is typed can be implemented using comprehensions:
iex> stream = IO.stream(:stdio, :line)
iex> for line <- stream, into: stream do
...>   String.upcase(line) <> "\n"
...> end
Now type any string into the terminal and you will see that the same value will be printed in upper-case. Unfortunately, this example also got your IEx shell stuck in the comprehension, so you will need to hit Ctrl+C twice to get out of it. :)

  
    
  
  Other options


Comprehensions support other options, such as :reduce and :uniq. Here are additional resources to learn more about comprehensions:
	for official reference in Elixir documentation
	Mitchell Hanberg's comprehensive guide to Elixir's comprehensions

Sigils
    

Elixir provides double-quoted strings as well as a concept called charlists, which are defined using the ~c"hello world" sigil syntax. In this chapter, we will learn more about sigils and how to define our own.
One of Elixir's goals is extensibility: developers should be able to extend the language to fit any particular domain. Sigils provide the foundation for extending the language with custom textual representations. Sigils start with the tilde (~) character which is followed by either a single lower-case letter or one or more upper-case letters, and then a delimiter. Optional modifiers are added after the final delimiter.

  
    
  
  Regular expressions


The most common sigil in Elixir is ~r, which is used to create regular expressions:
# A regular expression that matches strings which contain "foo" or "bar":
iex> regex = ~r/foo|bar/
~r/foo|bar/
iex> "foo" =~ regex
true
iex> "bat" =~ regex
false
Elixir provides Perl-compatible regular expressions (regexes), as implemented by the PCRE library. Regexes also support modifiers. For example, the i modifier makes a regular expression case insensitive:
iex> "HELLO" =~ ~r/hello/
false
iex> "HELLO" =~ ~r/hello/i
true
Check out the Regex module for more information on other modifiers and the supported operations with regular expressions.
So far, all examples have used / to delimit a regular expression. However, sigils support 8 different delimiters:
~r/hello/
~r|hello|
~r"hello"
~r'hello'
~r(hello)
~r[hello]
~r{hello}
~r<hello>
The reason behind supporting different delimiters is to provide a way to write literals without escaped delimiters. For example, a regular expression with forward slashes like ~r(^https?://) reads arguably better than ~r/^https?:\/\//. Similarly, if the regular expression has forward slashes and capturing groups (that use ()), you may then choose double quotes instead of parentheses.

  
    
  
  Strings, charlists, and word lists sigils


Besides regular expressions, Elixir ships with three other sigils.

  
    
  
  Strings


The ~s sigil is used to generate strings, like double quotes are. The ~s sigil is useful when a string contains double quotes:
iex> ~s(this is a string with "double" quotes, not 'single' ones)
"this is a string with \"double\" quotes, not 'single' ones"

  
    
  
  Charlists


The ~c sigil is the regular way to represent charlists.
iex> [?c, ?a, ?t]
~c"cat"
iex> ~c(this is a char list containing "double quotes")
~c"this is a char list containing \"double quotes\""

  
    
  
  Word lists


The ~w sigil is used to generate lists of words (words are just regular strings). Inside the ~w sigil, words are separated by whitespace.
iex> ~w(foo bar bat)
["foo", "bar", "bat"]
The ~w sigil also accepts the c, s and a modifiers (for charlists, strings, and atoms, respectively), which specify the data type of the elements of the resulting list:
iex> ~w(foo bar bat)a
[:foo, :bar, :bat]

  
    
  
  Interpolation and escaping in string sigils


Elixir supports some sigil variants to deal with escaping characters and interpolation. In particular, uppercase letters sigils do not perform interpolation nor escaping. For example, although both ~s and ~S will return strings, the former allows escape codes and interpolation while the latter does not:
iex> ~s(String with escape codes \x26 #{"inter" <> "polation"})
"String with escape codes & interpolation"
iex> ~S(String without escape codes \x26 without #{interpolation})
"String without escape codes \\x26 without \#{interpolation}"
The following escape codes can be used in strings and charlists:
	\\ – single backslash
	\a – bell/alert
	\b – backspace
	\d - delete
	\e - escape
	\f - form feed
	\n – newline
	\r – carriage return
	\s – space
	\t – tab
	\v – vertical tab
	\0 - null byte
	\xDD - represents a single byte in hexadecimal (such as \x13)
	\uDDDD and \u{D...} - represents a Unicode codepoint in hexadecimal (such as \u{1F600})

In addition to those, a double quote inside a double-quoted string needs to be escaped as \", and, analogously, a single quote inside a single-quoted char list needs to be escaped as \'. Nevertheless, it is better style to change delimiters as seen above than to escape them.
Sigils also support heredocs, that is, three double-quotes or single-quotes as separators:
iex> ~s"""
...> this is
...> a heredoc string
...> """
The most common use case for heredoc sigils is when writing documentation. For example, writing escape characters in the documentation would soon become error prone because of the need to double-escape some characters:
@doc """
Converts double-quotes to single-quotes.

## Examples

    iex> convert("\\\"foo\\\"")
    "'foo'"

"""
def convert(...)
By using ~S, this problem can be avoided altogether:
@doc ~S"""
Converts double-quotes to single-quotes.

## Examples

    iex> convert("\"foo\"")
    "'foo'"

"""
def convert(...)

  
    
  
  Calendar sigils


Elixir offers several sigils to deal with various flavors of times and dates.

  
    
  
  Date


A %Date{} struct contains the fields year, month, day, and calendar. You can create one using the ~D sigil:
iex> d = ~D[2019-10-31]
~D[2019-10-31]
iex> d.day
31

  
    
  
  Time


The %Time{} struct contains the fields hour, minute, second, microsecond, and calendar. You can create one using the ~T sigil:
iex> t = ~T[23:00:07.0]
~T[23:00:07.0]
iex> t.second
7

  
    
  
  NaiveDateTime


The %NaiveDateTime{} struct contains fields from both Date and Time. You can create one using the ~N sigil:
iex> ndt = ~N[2019-10-31 23:00:07]
~N[2019-10-31 23:00:07]
Why is it called naive? Because it does not contain timezone information. Therefore, the given datetime may not exist at all or it may exist twice in certain timezones - for example, when we move the clock back and forward for daylight saving time.

  
    
  
  UTC DateTime


A %DateTime{} struct contains the same fields as a NaiveDateTime with the addition of fields to track timezones. The ~U sigil allows developers to create a DateTime in the UTC timezone:
iex> dt = ~U[2019-10-31 19:59:03Z]
~U[2019-10-31 19:59:03Z]
iex> %DateTime{minute: minute, time_zone: time_zone} = dt
~U[2019-10-31 19:59:03Z]
iex> minute
59
iex> time_zone
"Etc/UTC"

  
    
  
  Custom sigils


As hinted at the beginning of this chapter, sigils in Elixir are extensible. In fact, using the sigil ~r/foo/i is equivalent to calling sigil_r with a binary and a char list as the argument:
iex> sigil_r(<<"foo">>, [?i])
~r"foo"i
We can access the documentation for the ~r sigil via sigil_r:
iex> h sigil_r
...
We can also provide our own sigils by implementing functions that follow the sigil_{character} pattern. For example, let's implement the ~i sigil that returns an integer (with the optional n modifier to make it negative):
iex> defmodule MySigils do
...>   def sigil_i(string, []), do: String.to_integer(string)
...>   def sigil_i(string, [?n]), do: -String.to_integer(string)
...> end
iex> import MySigils
iex> ~i(13)
13
iex> ~i(42)n
-42
Custom sigils may be either a single lowercase character, or an uppercase character followed by more uppercase characters and digits.
Sigils can also be used to do compile-time work with the help of macros. For example, regular expressions in Elixir are compiled into an efficient representation during compilation of the source code, therefore skipping this step at runtime. If you're interested in the subject, you can learn more about macros and check out how sigils are implemented in the Kernel module (where the sigil_* functions are defined).

try, catch, and rescue
    

Elixir has three error mechanisms: errors, throws, and exits. In this chapter, we will explore each of them and include remarks about when each should be used.

  
    
  
  Errors


Errors (or exceptions) are used when exceptional things happen in the code. A sample error can be retrieved by trying to add a number to an atom:
iex> :foo + 1
** (ArithmeticError) bad argument in arithmetic expression
     :erlang.+(:foo, 1)
A runtime error can be raised any time by using raise/1:
iex> raise "oops"
** (RuntimeError) oops
Other errors can be raised with raise/2 passing the error name and a list of keyword arguments:
iex> raise ArgumentError, message: "invalid argument foo"
** (ArgumentError) invalid argument foo
You can also define your own errors by creating a module and using the defexception/1 construct inside it. This way, you'll create an error with the same name as the module it's defined in. The most common case is to define a custom exception with a message field:
iex> defmodule MyError do
iex>   defexception message: "default message"
iex> end
iex> raise MyError
** (MyError) default message
iex> raise MyError, message: "custom message"
** (MyError) custom message
Errors can be rescued using the try/rescue construct:
iex> try do
...>   raise "oops"
...> rescue
...>   e in RuntimeError -> e
...> end
%RuntimeError{message: "oops"}
The example above rescues the runtime error and returns the exception itself, which is then printed in the iex session.
If you don't have any use for the exception, you don't have to pass a variable to rescue:
iex> try do
...>   raise "oops"
...> rescue
...>   RuntimeError -> "Error!"
...> end
"Error!"
In practice, Elixir developers rarely use the try/rescue construct. For example, many languages would force you to rescue an error when a file cannot be opened successfully. Elixir instead provides a File.read/1 function which returns a tuple containing information about whether the file was opened successfully:
iex> File.read("hello")
{:error, :enoent}
iex> File.write("hello", "world")
:ok
iex> File.read("hello")
{:ok, "world"}
There is no try/rescue here. In case you want to handle multiple outcomes of opening a file, you can use pattern matching using the case construct:
iex> case File.read("hello") do
...>   {:ok, body} -> IO.puts("Success: #{body}")
...>   {:error, reason} -> IO.puts("Error: #{reason}")
...> end
For the cases where you do expect a file to exist (and the lack of that file is truly an error) you may use File.read!/1:
iex> File.read!("unknown")
** (File.Error) could not read file "unknown": no such file or directory
    (elixir) lib/file.ex:272: File.read!/1
At the end of the day, it's up to your application to decide if an error while opening a file is exceptional or not. That's why Elixir doesn't impose exceptions on File.read/1 and many other functions. Instead, it leaves it up to the developer to choose the best way to proceed.
Many functions in the standard library follow the pattern of having a counterpart that raises an exception instead of returning tuples to match against. The convention is to create a function (foo) which returns {:ok, result} or {:error, reason} tuples and another function (foo!, same name but with a trailing !) that takes the same arguments as foo but which raises an exception if there's an error. foo! should return the result (not wrapped in a tuple) if everything goes fine. The File module is a good example of this convention.

  
    
  
  Fail fast / Let it crash


One saying that is common in the Erlang community, as well as Elixir's, is "fail fast" / "let it crash". The idea behind let it crash is that, in case something unexpected happens, it is best to let the exception happen, without rescuing it.
It is important to emphasize the word unexpected. For example, imagine you are building a script to process files. Your script receives filenames as inputs. It is expected that users may make mistakes and provide unknown filenames. In this scenario, while you could use File.read!/1 to read files and let it crash in case of invalid filenames, it probably makes more sense to use File.read/1 and provide users of your script with a clear and precise feedback of what went wrong.
Other times, you may fully expect a certain file to exist, and in case it does not, it means something terribly wrong has happened elsewhere. In such cases, File.read!/1 is all you need.
The second approach also works because, as discussed in the Processes chapter, all Elixir code runs inside processes that are isolated and don't share anything by default. Therefore, an unhandled exception in a process will never crash or corrupt the state of another process. This allows us to define supervisor processes, which are meant to observe when a process terminates unexpectedly, and start a new one in its place.
At the end of the day, "fail fast" / "let it crash" is a way of saying that, when something unexpected happens, it is best to start from scratch within a new process, freshly started by a supervisor, rather than blindly trying to rescue all possible error cases without the full context of when and how they can happen.

  
    
  
  Reraise


While we generally avoid using try/rescue in Elixir, one situation where we may want to use such constructs is for observability/monitoring. Imagine you want to log that something went wrong, you could do:
try do
  ... some code ...
rescue
  e ->
    Logger.error(Exception.format(:error, e, __STACKTRACE__))
    reraise e, __STACKTRACE__
end
In the example above, we rescued the exception, logged it, and then re-raised it. We use the __STACKTRACE__ construct both when formatting the exception and when re-raising. This ensures we reraise the exception as is, without changing value or its origin.
Generally speaking, we take errors in Elixir literally: they are reserved for unexpected and/or exceptional situations, never for controlling the flow of our code. In case you actually need flow control constructs, throws should be used. That's what we are going to see next.

  
    
  
  Throws


In Elixir, a value can be thrown and later be caught. throw and catch are reserved for situations where it is not possible to retrieve a value unless by using throw and catch.
Those situations are quite uncommon in practice except when interfacing with libraries that do not provide a proper API. For example, let's imagine the Enum module did not provide any API for finding a value and that we needed to find the first multiple of 13 in a list of numbers:
iex> try do
...>   Enum.each(-50..50, fn x ->
...>     if rem(x, 13) == 0, do: throw(x)
...>   end)
...>   "Got nothing"
...> catch
...>   x -> "Got #{x}"
...> end
"Got -39"
Since Enum does provide a proper API, in practice Enum.find/2 is the way to go:
iex> Enum.find(-50..50, &(rem(&1, 13) == 0))
-39

  
    
  
  Exits


All Elixir code runs inside processes that communicate with each other. When a process dies of "natural causes" (e.g., unhandled exceptions), it sends an exit signal. A process can also die by explicitly sending an exit signal:
iex> spawn_link(fn -> exit(1) end)
** (EXIT from #PID<0.56.0>) shell process exited with reason: 1
In the example above, the linked process died by sending an exit signal with a value of 1. The Elixir shell automatically handles those messages and prints them to the terminal.
exit can also be "caught" using try/catch:
iex> try do
...>   exit("I am exiting")
...> catch
...>   :exit, _ -> "not really"
...> end
"not really"
catch can also be used within a function body without a matching try.
defmodule Example do
  def matched_catch do
    exit(:timeout)
  catch
    :exit, :timeout ->
      {:error, :timeout}
  end

  def mismatched_catch do
    exit(:timeout)
  catch
    # Since no clause matches, this catch will have no effect
    :exit, :explosion ->
      {:error, :explosion}
  end
end
However, using try/catch is already uncommon and using it to catch exits is even rarer.
exit signals are an important part of the fault tolerant system provided by the Erlang VM. Processes usually run under supervision trees which are themselves processes that listen to exit signals from the supervised processes. Once an exit signal is received, the supervision strategy kicks in and the supervised process is restarted.
It is exactly this supervision system that makes constructs like try/catch and try/rescue so uncommon in Elixir. Instead of rescuing an error, we'd rather "fail fast" since the supervision tree will guarantee our application will go back to a known initial state after the error.

  
    
  
  After


Sometimes it's necessary to ensure that a resource is cleaned up after some action that could potentially raise an error. The try/after construct allows you to do that. For example, we can open a file and use an after clause to close it -- even if something goes wrong:
iex> {:ok, file} = File.open("sample", [:utf8, :write])
iex> try do
...>   IO.write(file, "olá")
...>   raise "oops, something went wrong"
...> after
...>   File.close(file)
...> end
** (RuntimeError) oops, something went wrong
The after clause will be executed regardless of whether or not the tried block succeeds. Note, however, that if a linked process exits,
this process will exit and the after clause will not get run. Thus after provides only a soft guarantee. Luckily, files in Elixir are also linked to the current processes and therefore they will always get closed if the current process crashes, independent of the
after clause. You will find the same to be true for other resources like ETS tables, sockets, ports and more.
Sometimes you may want to wrap the entire body of a function in a try construct, often to guarantee some code will be executed afterwards. In such cases, Elixir allows you to omit the try line:
iex> defmodule RunAfter do
...>   def without_even_trying do
...>     raise "oops"
...>   after
...>     IO.puts "cleaning up!"
...>   end
...> end
iex> RunAfter.without_even_trying
cleaning up!
** (RuntimeError) oops
Elixir will automatically wrap the function body in a try whenever one of after, rescue or catch is specified.

  
    
  
  Else


If an else block is present, it will match on the results of the try block whenever the try block finishes without a throw or an error.
iex> x = 2
2
iex> try do
...>   1 / x
...> rescue
...>   ArithmeticError ->
...>     :infinity
...> else
...>   y when y < 1 and y > -1 ->
...>     :small
...>   _ ->
...>     :large
...> end
:small
Exceptions in the else block are not caught. If no pattern inside the else block matches, an exception will be raised; this exception is not caught by the current try/catch/rescue/after block.

  
    
  
  Variables scope


Similar to case, cond, if and other constructs in Elixir, variables defined inside try/catch/rescue/after blocks do not leak to the outer context. In other words, this code is invalid:
iex> try do
...>   raise "fail"
...>   what_happened = :did_not_raise
...> rescue
...>   _ -> what_happened = :rescued
...> end
iex> what_happened
** (CompileError) undefined variable "what_happened"
Instead, you should return the value of the try expression:
iex> what_happened =
...>   try do
...>     raise "fail"
...>     :did_not_raise
...>   rescue
...>     _ -> :rescued
...>   end
iex> what_happened
:rescued
Furthermore, variables defined in the do-block of try are not available inside rescue/after/else either. This is because the try block may fail at any moment and therefore the variables may have never been bound in the first place. So this also isn't valid:
iex> try do
...>   raise "fail"
...>   another_what_happened = :did_not_raise
...> rescue
...>   _ -> another_what_happened
...> end
** (CompileError) undefined variable "another_what_happened"
This finishes our introduction on try, catch, and rescue. You will find they are used less frequently in Elixir than in other languages. Next we will talk about a very important subject to Elixir developers: writing documentation.

Writing documentation
    

Elixir treats documentation as a first-class citizen. Documentation must be easy to write and easy to read. In this guide you will learn how to write documentation in Elixir, covering constructs like module attributes, style practices, and doctests.

  
    
  
  Markdown


Elixir documentation is written using Markdown. There are plenty of guides on Markdown online, we recommend the one from GitHub as a getting started point:
	Basic writing and formatting syntax


  
    
  
  Module Attributes


Documentation in Elixir is usually attached to module attributes. Let's see an example:
defmodule MyApp.Hello do
  @moduledoc """
  This is the Hello module.
  """
  @moduledoc since: "1.0.0"

  @doc """
  Says hello to the given `name`.

  Returns `:ok`.

  ## Examples

      iex> MyApp.Hello.world(:john)
      :ok

  """
  @doc since: "1.3.0"
  def world(name) do
    IO.puts("hello #{name}")
  end
end
The @moduledoc attribute is used to add documentation to the module. @doc is used before a function to provide documentation for it. Besides the attributes above, @typedoc can also be used to attach documentation to types defined as part of typespecs, which we will explore later on. Elixir also allows metadata to be attached to documentation, by passing a keyword list to @doc and friends.

  
    
  
  Function Arguments


When documenting a function, argument names are inferred by the compiler. For example:
def size(%{size: size}) do
  size
end
The compiler will infer this argument as map. Sometimes the inference will be suboptimal, especially if the function contains multiple clauses with the argument matching on different values each time. You can specify the proper names for documentation by declaring only the function head at any moment before the implementation:
def size(map_with_size)
def size(%{size: size}) do
  size
end

  
    
  
  Documentation metadata


Elixir allows developers to attach arbitrary metadata to the documentation. This is done by passing a keyword list to the relevant attribute (such as @moduledoc, @typedoc, and @doc). A commonly used metadata is :since, which annotates in which version that particular module, function, type, or callback was added, as shown in the example above.
Another common metadata is :deprecated, which emits a warning in the documentation, explaining that its usage is discouraged:
@doc deprecated: "Use Foo.bar/2 instead"
Note that the :deprecated key does not warn when a developer invokes the functions. If you want the code to also emit a warning, you can use the @deprecated attribute:
@deprecated "Use Foo.bar/2 instead"
Metadata can have any key. Documentation tools often use metadata to provide more data to readers and to enrich the user experience.

  
    
  
  Recommendations


When writing documentation:
	Keep the first paragraph of the documentation concise and simple, typically one-line. Tools like ExDoc use the first line to generate a summary.

	Reference modules by their full name.
Markdown uses backticks (`) to quote code. Elixir builds on top of that to automatically generate links when module or function names are referenced. For this reason, always use full module names. If you have a module called MyApp.Hello, always reference it as `MyApp.Hello` and never as `Hello`.

	Reference functions by name and arity if they are local, as in `world/1`, or by module, name and arity if pointing to an external module: `MyApp.Hello.world/1`.

	Reference a @callback by prepending c:, as in `c:world/1`.

	Reference a @type by prepending t:, as in `t:values/0`.

	Start new sections with second level Markdown headers ##. First level headers are reserved for module and function names.

	Place documentation before the first clause of multi-clause functions. Documentation is always per function and arity and not per clause.

	Use the :since key in the documentation metadata to annotate whenever new functions or modules are added to your API.



  
    
  
  Doctests


We recommend that developers include examples in their documentation, often under their own ## Examples heading. To ensure examples do not get out of date, Elixir's test framework (ExUnit) provides a feature called doctests that allows developers to test the examples in their documentation. Doctests work by parsing out code samples starting with iex> from the documentation. You can read more about them at ExUnit.DocTest.

  
    
  
  Documentation != Code comments


Elixir treats documentation and code comments as different concepts. Documentation is an explicit contract between you and users of your Application Programming Interface (API), be them third-party developers, co-workers, or your future self. Modules and functions must always be documented if they are part of your API.
Code comments are aimed at developers reading the code. They are useful for marking improvements, leaving notes (for example, why you had to resort to a workaround due to a bug in a library), and so forth. They are tied to the source code: you can completely rewrite a function and remove all existing code comments, and it will continue to behave the same, with no change to either its behavior or its documentation.
Because private functions cannot be accessed externally, Elixir will warn if a private function has a @doc attribute and will discard its content. However, you can add code comments to private functions, as with any other piece of code, and we recommend developers to do so whenever they believe it will add relevant information to the readers and maintainers of such code.
In summary, documentation is a contract with users of your API, who may not necessarily have access to the source code, whereas code comments are for those who interact directly with the source. You can learn and express different guarantees about your software by separating those two concepts.

  
    
  
  Hiding internal Modules and Functions


Besides the modules and functions libraries provide as part of their public interface, libraries may also implement important functionality that is not part of their API. While these modules and functions can be accessed, they are meant to be internal to the library and thus should not have documentation for end users.
Conveniently, Elixir allows developers to hide modules and functions from the documentation, by setting @doc false to hide a particular function, or @moduledoc false to hide the whole module. If a module is hidden, you may even document the functions in the module, but the module itself won't be listed in the documentation:
defmodule MyApp.Hidden do
  @moduledoc false

  @doc """
  This function won't be listed in docs.
  """
  def function_that_wont_be_listed_in_docs do
    # ...
  end
end
In case you don't want to hide a whole module, you can hide functions individually:
defmodule MyApp.Sample do
  @doc false
  def add(a, b), do: a + b
end
However, keep in mind @moduledoc false or @doc false do not make a function private. The function above can still be invoked as MyApp.Sample.add(1, 2). Not only that, if MyApp.Sample is imported, the add/2 function will also be imported into the caller. For those reasons, be cautious when adding @doc false to functions, instead use one of these two options:
	Move the undocumented function to a module with @moduledoc false, like MyApp.Hidden, ensuring the function won't be accidentally exposed or imported. Remember that you can use @moduledoc false to hide a whole module and still document each function with @doc. Tools will still ignore the module.

	Start the function name with one or two underscores, for example, __add__/2. Functions starting with underscore are automatically treated as hidden, although you can also be explicit and add @doc false. The compiler does not import functions with leading underscores and they hint to anyone reading the code of their intended private usage.



  
    
  
  Code.fetch_docs/1


Elixir stores documentation inside pre-defined chunks in the bytecode. Documentation is not loaded into memory when modules are loaded, instead, it can be read from the bytecode in disk using the Code.fetch_docs/1 function. The downside is that modules defined in-memory, like the ones defined in IEx, cannot have their documentation accessed as they do not write their bytecode to disk.

Optional syntax sheet
    

In the previous chapters, we learned that the Elixir syntax allows developers to omit delimiters in a few occasions to make code more readable. For example, we learned that parentheses are optional:
iex> length([1, 2, 3]) == length [1, 2, 3]
true
and that do-end blocks are equivalent to keyword lists:
# do-end blocks
iex> if true do
...>   :this
...> else
...>   :that
...> end
:this

# keyword lists
iex> if true, do: :this, else: :that
:this
Keyword lists use Elixir's regular notation for separating arguments, where we separate each key-value pair with commas, and each key is followed by :. In the do-blocks, we get rid of the colons, the commas, and separate each keyword by a newline. They are useful exactly because they remove the verbosity when writing blocks of code. Most of the time, we use the block syntax, but it is good to know they are equivalent.
Those conveniences, which we call here "optional syntax", allow the language syntax core to be small, without sacrificing the readability and expressiveness of your code.  In this brief chapter, we will review the four rules provided by the language, using a short snippet as playground.

  
    
  
  Walk-through


Take the following code:
if variable? do
  Call.this()
else
  Call.that()
end
Now let's remove the conveniences one by one:
	do-end blocks are equivalent to keywords:
if variable?, do: Call.this(), else: Call.that()

	Keyword lists as last argument do not require square brackets, but let's add them:
if variable?, [do: Call.this(), else: Call.that()]

	Keyword lists are the same as lists of two-element tuples:
if variable?, [{:do, Call.this()}, {:else, Call.that()}]

	Finally, parentheses are optional on function calls, but let's add them:
if(variable?, [{:do, Call.this()}, {:else, Call.that()}])


That's it! Those four rules outline the optional syntax available in Elixir.
To understand why these rules matter, we can briefly compare Elixir with many other programming languages. Most programming languages have several keywords for defining methods, functions, conditionals, loops, and so forth. Each of those keywords have their own syntax rules attached to them.
However, in Elixir, none of these language features require special "keywords", instead they all build from this small set of rules. The other benefit is that developers can also extend the language in a way that is consistent with the language itself, since the constructs for designing and extending the language are the same. We further explore this topic in the "Meta-programming" guide.
At the end of the day, those rules are what enables us to write:
defmodule Math do
  def add(a, b) do
    a + b
  end
end
instead of:
defmodule(Math, [
  {:do, def(add(a, b), [{:do, a + b}])}
])
Whenever you have any questions, this quick walk-through has you covered.
Finally, if you are concerned about when to apply these rules, it's worth noting that the Elixir formatter handles those concerns for you. Most Elixir developers use the mix format task to format their codebases according to a well-defined set of rules defined by the Elixir team and the community. For instance, mix format will always add parentheses to function calls unless explicitly configured not to do so. This helps to maintain consistency across all codebases within organizations and the wider community.


  

  
    
    Erlang libraries - Elixir v1.17.3
    
    

    


  
  

    
Erlang libraries
    

Elixir provides excellent interoperability with Erlang libraries. In fact, Elixir discourages simply wrapping Erlang libraries in favor of directly interfacing with Erlang code. In this section, we will present some of the most common and useful Erlang functionality that is not found in Elixir.
Erlang modules have a different naming convention than in Elixir and start in lowercase. In both cases, module names are atoms and we invoke functions by dispatching to the module name:
iex> is_atom(String)
true
iex> String.first("hello")
"h"
iex> is_atom(:binary)
true
iex> :binary.first("hello")
104
As you grow more proficient in Elixir, you may want to explore the Erlang STDLIB Reference Manual in more detail.

  
    
  
  The binary module


The built-in Elixir String module handles binaries that are UTF-8 encoded. The :binary module is useful when you are dealing with binary data that is not necessarily UTF-8 encoded.
iex> String.to_charlist("Ø")
[216]
iex> :binary.bin_to_list("Ø")
[195, 152]
The above example shows the difference; the String module returns Unicode codepoints, while :binary deals with raw data bytes.

  
    
  
  Formatted text output


Elixir does not contain a function similar to printf found in C and other languages. Luckily, the Erlang standard library functions :io.format/2 and :io_lib.format/2 may be used. The first formats to terminal output, while the second formats to an iolist. The format specifiers differ from printf, refer to the Erlang documentation for details.
iex> :io.format("Pi is approximately given by:~10.3f~n", [:math.pi])
Pi is approximately given by:     3.142
:ok
iex> to_string(:io_lib.format("Pi is approximately given by:~10.3f~n", [:math.pi]))
"Pi is approximately given by:     3.142\n"

  
    
  
  The crypto module


The :crypto module contains hashing functions, digital signatures, encryption and more:
iex> Base.encode16(:crypto.hash(:sha256, "Elixir"))
"3315715A7A3AD57428298676C5AE465DADA38D951BDFAC9348A8A31E9C7401CB"
The :crypto module is part of the :crypto application that ships with Erlang. This means you must list the :crypto application as an additional application in your project configuration. To do this, edit your mix.exs file to include:
def application do
  [extra_applications: [:crypto]]
end
Any module that is not part of the :kernel or :stdlib Erlang applications must have their application explicitly listed in your mix.exs. You can find the application name of any Erlang module in the Erlang documentation, immediately below the Erlang logo in the sidebar.

  
    
  
  The digraph module


The :digraph and :digraph_utils modules contain functions for dealing with directed graphs built of vertices and edges. After constructing the graph, the algorithms in there will help find, for instance, the shortest path between two vertices, or loops in the graph.
Given three vertices, find the shortest path from the first to the last.
iex> digraph = :digraph.new()
iex> coords = [{0.0, 0.0}, {1.0, 0.0}, {1.0, 1.0}]
iex> [v0, v1, v2] = (for c <- coords, do: :digraph.add_vertex(digraph, c))
iex> :digraph.add_edge(digraph, v0, v1)
iex> :digraph.add_edge(digraph, v1, v2)
iex> :digraph.get_short_path(digraph, v0, v2)
[{0.0, 0.0}, {1.0, 0.0}, {1.0, 1.0}]
Note that the functions in :digraph alter the graph structure in-place, this
is possible because they are implemented as ETS tables, explained next.

  
    
  
  Erlang Term Storage


The modules :ets and :dets handle storage of large data structures in memory or on disk respectively.
ETS lets you create a table containing tuples. By default, ETS tables are protected, which means only the owner process may write to the table but any other process can read. ETS has some functionality to allow a table to be used as a simple database, a key-value store or as a cache mechanism.
The functions in the ets module will modify the state of the table as a side-effect.
iex> table = :ets.new(:ets_test, [])
# Store as tuples with {name, population}
iex> :ets.insert(table, {"China", 1_374_000_000})
iex> :ets.insert(table, {"India", 1_284_000_000})
iex> :ets.insert(table, {"USA", 322_000_000})
iex> :ets.i(table)
<1   > {<<"India">>,1284000000}
<2   > {<<"USA">>,322000000}
<3   > {<<"China">>,1374000000}

  
    
  
  The math module


The :math module contains common mathematical operations covering trigonometry, exponential, and logarithmic functions.
iex> angle_45_deg = :math.pi() * 45.0 / 180.0
iex> :math.sin(angle_45_deg)
0.7071067811865475
iex> :math.exp(55.0)
7.694785265142018e23
iex> :math.log(7.694785265142018e23)
55.0

  
    
  
  The queue module


The :queue module provides a data structure that implements (double-ended) FIFO (first-in first-out) queues efficiently:
iex> q = :queue.new
iex> q = :queue.in("A", q)
iex> q = :queue.in("B", q)
iex> {value, q} = :queue.out(q)
iex> value
{:value, "A"}
iex> {value, q} = :queue.out(q)
iex> value
{:value, "B"}
iex> {value, q} = :queue.out(q)
iex> value
:empty

  
    
  
  The rand module


The :rand has functions for returning random values and setting the random seed.
iex> :rand.uniform()
0.8175669086010815
iex> _ = :rand.seed(:exs1024, {123, 123534, 345345})
iex> :rand.uniform()
0.5820506340260994
iex> :rand.uniform(6)
6

  
    
  
  The zip and zlib modules


The :zip module lets you read and write ZIP files to and from disk or memory, as well as extracting file information.
This code counts the number of files in a ZIP file:
iex> :zip.foldl(fn _, _, _, acc -> acc + 1 end, 0, :binary.bin_to_list("file.zip"))
{:ok, 633}
The :zlib module deals with data compression in zlib format, as found in the gzip command line utility found in Unix systems.
iex> song = "
...> Mary had a little lamb,
...> His fleece was white as snow,
...> And everywhere that Mary went,
...> The lamb was sure to go."
iex> compressed = :zlib.compress(song)
iex> byte_size(song)
110
iex> byte_size(compressed)
99
iex> :zlib.uncompress(compressed)
"\nMary had a little lamb,\nHis fleece was white as snow,\nAnd everywhere that Mary went,\nThe lamb was sure to go."

  
    
  
  Learning Erlang


If you want to get deeper into Erlang, here's a list of online resources that cover Erlang's fundamentals and its more advanced features:
	This Erlang Syntax: A Crash Course provides a concise intro to Erlang's syntax. Each code snippet is accompanied by equivalent code in Elixir. This is an opportunity for you to not only get some exposure to Erlang's syntax but also review what you learned about Elixir.

	Erlang's official website has a short tutorial. There is a chapter with pictures briefly describing Erlang's primitives for concurrent programming.

	Learn You Some Erlang for Great Good! is an excellent introduction to Erlang, its design principles, standard library, best practices, and much more. Once you have read through the crash course mentioned above, you'll be able to safely skip the first couple of chapters in the book that mostly deal with the syntax. When you reach The Hitchhiker's Guide to Concurrency chapter, that's where the real fun starts.


Our last step is to take a look at existing Elixir (and Erlang) libraries you might use while debugging.

Debugging
    

There are a number of ways to debug code in Elixir. In this chapter we will cover some of the more common ways of doing so.

  
    
  
  IO.inspect/2


What makes IO.inspect(item, opts \\ []) really useful in debugging is that it returns the item argument passed to it without affecting the behavior of the original code. Let's see an example.
(1..10)
|> IO.inspect()
|> Enum.map(fn x -> x * 2 end)
|> IO.inspect()
|> Enum.sum()
|> IO.inspect()
Prints:
1..10
[2, 4, 6, 8, 10, 12, 14, 16, 18, 20]
110
As you can see IO.inspect/2 makes it possible to "spy" on values almost anywhere in your code without altering the result, making it very helpful inside of a pipeline like in the above case.
IO.inspect/2 also provides the ability to decorate the output with a label option. The label will be printed before the inspected item:
[1, 2, 3]
|> IO.inspect(label: "before")
|> Enum.map(&(&1 * 2))
|> IO.inspect(label: "after")
|> Enum.sum
Prints:
before: [1, 2, 3]
after: [2, 4, 6]
It is also very common to use IO.inspect/2 with binding/0, which returns all variable names and their values:
def some_fun(a, b, c) do
  IO.inspect binding()
  ...
end
When some_fun/3 is invoked with :foo, "bar", :baz it prints:
[a: :foo, b: "bar", c: :baz]
See IO.inspect/2 and Inspect.Opts respectively to learn more about the function and read about all supported options.

  
    
  
  dbg/2


Elixir v1.14 introduced dbg/2. dbg is similar to IO.inspect/2 but specifically tailored for debugging. It prints the value passed to it and returns it (just like IO.inspect/2), but it also prints the code and location.
# In my_file.exs
feature = %{name: :dbg, inspiration: "Rust"}
dbg(feature)
dbg(Map.put(feature, :in_version, "1.14.0"))
The code above prints this:
[my_file.exs:2: (file)]
feature #=> %{inspiration: "Rust", name: :dbg}
[my_file.exs:3: (file)]
Map.put(feature, :in_version, "1.14.0") #=> %{in_version: "1.14.0", inspiration: "Rust", name: :dbg}

When talking about IO.inspect/2, we mentioned its usefulness when placed between steps of |> pipelines. dbg does it better: it understands Elixir code, so it will print values at every step of the pipeline.
# In dbg_pipes.exs
__ENV__.file
|> String.split("/", trim: true)
|> List.last()
|> File.exists?()
|> dbg()
This code prints:
[dbg_pipes.exs:5: (file)]
__ENV__.file #=> "/home/myuser/dbg_pipes.exs"
|> String.split("/", trim: true) #=> ["home", "myuser", "dbg_pipes.exs"]
|> List.last() #=> "dbg_pipes.exs"
|> File.exists?() #=> true

While dbg provides conveniences around Elixir constructs, you will need IEx if you want to execute code and set breakpoints while debugging.

  
    
  
  Pry


When using IEx, you may pass --dbg pry as an option to "stop" the code execution where the dbg call is:
$ iex --dbg pry

Or to debug inside a of a project:
$ iex --dbg pry -S mix

Now any call to dbg will ask if you want to pry the existing code. If you accept, you'll be able to access all variables, as well as imports and aliases from the code, directly from IEx. This is called "prying". While the pry session is running, the code execution stops, until continue (or c) or next (or n) are called. Remember you can always run iex in the context of a project with iex -S mix TASK.


  
    
  
  Breakpoints


dbg calls require us to change the code we intend to debug and has limited stepping functionality. Luckily IEx also provides a IEx.break!/2 function which allows you to set and manage breakpoints on any Elixir code without modifying its source:

Similar to dbg, once a breakpoint is reached, code execution stops until continue (or c) or next (or n) are invoked. Breakpoints can navigate line-by-line by default, however, they do not have access to aliases and imports when breakpoints are set on compiled modules.
The mix test task direct integration with breakpoints via the -b/--breakpoints flag. When the flag is used, a breakpoint is set at the beginning of every test that will run:

Here are some commands you can use in practice:
# Debug all failed tests
$ iex -S mix test --breakpoints --failed
# Debug the test at the given file:line
$ iex -S mix test -b path/to/file:line


  
    
  
  Observer


For debugging complex systems, jumping at the code is not enough. It is necessary to have an understanding of the whole virtual machine, processes, applications, as well as set up tracing mechanisms. Luckily this can be achieved in Erlang with :observer. In your application:
$ iex
iex> :observer.start()
Missing dependencies
When running iex inside a project with iex -S mix, observer won't be available as a dependency. To do so, you will need to call the following functions before:
iex> Mix.ensure_application!(:wx)             # Not necessary on Erlang/OTP 27+
iex> Mix.ensure_application!(:runtime_tools)  # Not necessary on Erlang/OTP 27+
iex> Mix.ensure_application!(:observer)
iex> :observer.start()
If any of the calls above fail, here is what may have happened: some package managers default to installing a minimized Erlang without WX bindings for GUI support. In some package managers, you may be able to replace the headless Erlang with a more complete package (look for packages named erlang vs erlang-nox on Debian/Ubuntu/Arch). In others managers, you may need to install a separate erlang-wx (or similarly named) package.

The above will open another Graphical User Interface that provides many panes to fully understand and navigate the runtime and your project.
We explore the Observer in the context of an actual project in the Dynamic Supervisor chapter of the Mix & OTP guide. This is one of the debugging techniques the Phoenix framework used to achieve 2 million connections on a single machine.
If you are using the Phoenix web framework, it ships with the Phoenix LiveDashboard, a web dashboard for production nodes which provides similar features to Observer.
Finally, remember you can also get a mini-overview of the runtime info by calling runtime_info/0 directly in IEx.

  
    
  
  Other tools and community


We have just scratched the surface of what the Erlang VM has to offer, for example:
	Alongside the observer application, Erlang also includes a :crashdump_viewer to view crash dumps

	Integration with OS level tracers, such as Linux Trace Toolkit, DTRACE, and SystemTap

	Microstate accounting measures how much time the runtime spends in several low-level tasks in a short time interval

	Mix ships with many tasks under the profile namespace, such as mix profile.cprof and mix profile.fprof

	For more advanced use cases, we recommend the excellent Erlang in Anger, which is available as a free ebook


Happy debugging!

Enum cheatsheet
    

A quick reference into the Enum module, a module for working with collections (known as enumerables). Most of the examples below use the following data structure:
cart = [
  %{fruit: "apple", count: 3},
  %{fruit: "banana", count: 1},
  %{fruit: "orange", count: 6}
]
Some examples use the string =~ part operator, which checks the string on the left contains the part on the right.

  
    
  
  Predicates



  
    
  
  any?(enum, fun)


iex> Enum.any?(cart, & &1.fruit == "orange")
true
iex> Enum.any?(cart, & &1.fruit == "pear")
false
any? with an empty collection is always false:
iex> Enum.any?([], & &1.fruit == "orange")
false

  
    
  
  all?(enum, fun)


iex> Enum.all?(cart, & &1.count > 0)
true
iex> Enum.all?(cart, & &1.count > 1)
false
all? with an empty collection is always true:
iex> Enum.all?([], & &1.count > 0)
true

  
    
  
  member?(enum, value)


iex> Enum.member?(cart, %{fruit: "apple", count: 3})
true
iex> Enum.member?(cart, :something_else)
false
item in enum is equivalent to Enum.member?(enum, item):
iex> %{fruit: "apple", count: 3} in cart
true
iex> :something_else in cart
false

  
    
  
  empty?(enum)


iex> Enum.empty?(cart)
false
iex> Enum.empty?([])
true

  
    
  
  Filtering



  
    
  
  filter(enum, fun)


iex> Enum.filter(cart, &(&1.fruit =~ "o"))
[%{fruit: "orange", count: 6}]
iex> Enum.filter(cart, &(&1.fruit =~ "e"))
[
  %{fruit: "apple", count: 3},
  %{fruit: "orange", count: 6}
]

  
    
  
  reject(enum, fun)


iex> Enum.reject(cart, &(&1.fruit =~ "o"))
[
  %{fruit: "apple", count: 3},
  %{fruit: "banana", count: 1}
]

  
    
  
  Comprehension


Filtering can also be done with comprehensions:
iex> for item <- cart, item.fruit =~ "e" do
...>   item
...> end
[
  %{fruit: "apple", count: 3},
  %{fruit: "orange", count: 6}
]
Pattern-matching in comprehensions acts as a filter as well:
iex> for %{count: 1, fruit: fruit} <- cart do
...>   fruit
...> end
["banana"]

  
    
  
  Mapping



  
    
  
  map(enum, fun)


iex> Enum.map(cart, & &1.fruit)
["apple", "banana", "orange"]
iex> Enum.map(cart, fn item ->
...>   %{item | count: item.count + 10}
...> end)
[
  %{fruit: "apple", count: 13},
  %{fruit: "banana", count: 11},
  %{fruit: "orange", count: 16}
]

  
    
  
  map_every(enum, nth, fun)


iex> Enum.map_every(cart, 2, fn item ->
...>   %{item | count: item.count + 10}
...> end)
[
  %{fruit: "apple", count: 13},
  %{fruit: "banana", count: 1},
  %{fruit: "orange", count: 16}
]

  
    
  
  Comprehension


Mapping can also be done with comprehensions:
iex> for item <- cart do
...>   item.fruit
...> end
["apple", "banana", "orange"]
You can also filter and map at once:
iex> for item <- cart, item.fruit =~ "e" do
...>   item.fruit
...> end
["apple", "orange"]

  
    
  
  Side-effects



  
    
  
  each(enum, fun)


iex> Enum.each(cart, &IO.puts(&1.fruit))
apple
banana
orange
:ok
Enum.each/2 is used exclusively for side-effects.

  
    
  
  Accumulating



  
    
  
  reduce(enum, acc, fun)


iex> Enum.reduce(cart, 0, fn item, acc ->
...>   item.count + acc
...> end)
10

  
    
  
  map_reduce(enum, acc, fun)


iex> Enum.map_reduce(cart, 0, fn item, acc ->
...>   {item.fruit, item.count + acc}
...> end)
{["apple", "banana", "orange"], 10}

  
    
  
  scan(enum, acc, fun)


iex> Enum.scan(cart, 0, fn item, acc ->
...>   item.count + acc
...> end)
[3, 4, 10]

  
    
  
  reduce_while(enum, acc, fun)


iex> Enum.reduce_while(cart, 0, fn item, acc ->
...>   if item.fruit == "orange" do
...>     {:halt, acc}
...>   else
...>     {:cont, item.count + acc}
...>   end
...> end)
4

  
    
  
  Comprehension


Reducing can also be done with comprehensions:
iex> for item <- cart, reduce: 0 do
...>   acc -> item.count + acc
...> end
10
You can also filter and reduce at once:
iex> for item <- cart, item.fruit =~ "e", reduce: 0 do
...>   acc -> item.count + acc
...> end
9

  
    
  
  Aggregations



  
    
  
  count(enum)


iex> Enum.count(cart)
3
See Enum.count_until/2 to count until a limit.

  
    
  
  frequencies(enum)


iex> Enum.frequencies(["apple", "banana", "orange", "apple"])
%{"apple" => 2, "banana" => 1, "orange" => 1}

  
    
  
  frequencies_by(enum, key_fun)


Frequencies of the last letter of the fruit:
iex> Enum.frequencies_by(cart, &String.last(&1.fruit))
%{"a" => 1, "e" => 2}

  
    
  
  count(enum, fun)


iex> Enum.count(cart, &(&1.fruit =~ "e"))
2
iex> Enum.count(cart, &(&1.fruit =~ "y"))
0
See Enum.count_until/3 to count until a limit with a function.

  
    
  
  sum(enum)


iex> cart |> Enum.map(& &1.count) |> Enum.sum()
10

  
    
  
  product(enum)


iex> cart |> Enum.map(& &1.count) |> Enum.product()
18

  
    
  
  Sorting



  
    
  
  sort(enum, sorter \\ :asc)


iex> cart |> Enum.map(& &1.fruit) |> Enum.sort()
["apple", "banana", "orange"]
iex> cart |> Enum.map(& &1.fruit) |> Enum.sort(:desc)
["orange", "banana", "apple"]
When sorting structs, use Enum.sort/2 with a module as sorter.

  
    
  
  sort_by(enum, mapper, sorter \\ :asc)


iex> Enum.sort_by(cart, & &1.count)
[
  %{fruit: "banana", count: 1},
  %{fruit: "apple", count: 3},
  %{fruit: "orange", count: 6}
]
iex> Enum.sort_by(cart, & &1.count, :desc)
[
  %{fruit: "orange", count: 6},
  %{fruit: "apple", count: 3},
  %{fruit: "banana", count: 1}
]
When the sorted by value is a struct, use Enum.sort_by/3 with a module as sorter.

  
    
  
  min(enum)


iex> cart |> Enum.map(& &1.count) |> Enum.min()
1
When comparing structs, use Enum.min/2 with a module as sorter.

  
    
  
  min_by(enum, mapper)


iex> Enum.min_by(cart, & &1.count)
%{fruit: "banana", count: 1}
When comparing structs, use Enum.min_by/3 with a module as sorter.

  
    
  
  max(enum)


iex> cart |> Enum.map(& &1.count) |> Enum.max()
6
When comparing structs, use Enum.max/2 with a module as sorter.

  
    
  
  max_by(enum, mapper)


iex> Enum.max_by(cart, & &1.count)
%{fruit: "orange", count: 6}
When comparing structs, use Enum.max_by/3 with a module as sorter.

  
    
  
  Concatenating & flattening



  
    
  
  concat(enums)


iex> Enum.concat([[1, 2, 3], [4, 5, 6], [7, 8, 9]])
[1, 2, 3, 4, 5, 6, 7, 8, 9]

  
    
  
  concat(left, right)


iex> Enum.concat([1, 2, 3], [4, 5, 6])
[1, 2, 3, 4, 5, 6]

  
    
  
  flat_map(enum, fun)


iex> Enum.flat_map(cart, fn item ->
...>   List.duplicate(item.fruit, item.count)
...> end)
["apple", "apple", "apple", "banana", "orange",
 "orange", "orange", "orange", "orange", "orange"]

  
    
  
  flat_map_reduce(enum, acc, fun)


iex> Enum.flat_map_reduce(cart, 0, fn item, acc ->
...>   list = List.duplicate(item.fruit, item.count)
...>   acc = acc + item.count
...>   {list, acc}
...> end)
{["apple", "apple", "apple", "banana", "orange",
  "orange", "orange", "orange", "orange", "orange"], 10}

  
    
  
  Comprehension


Flattening can also be done with comprehensions:
iex> for item <- cart,
...>     fruit <- List.duplicate(item.fruit, item.count) do
...>   fruit
...> end
["apple", "apple", "apple", "banana", "orange",
 "orange", "orange", "orange", "orange", "orange"]

  
    
  
  Conversion



  
    
  
  into(enum, collectable)


iex> pairs = [{"apple", 3}, {"banana", 1}, {"orange", 6}]
iex> Enum.into(pairs, %{})
%{"apple" => 3, "banana" => 1, "orange" => 6}

  
    
  
  into(enum, collectable, transform)


iex> Enum.into(cart, %{}, fn item ->
...>   {item.fruit, item.count}
...> end)
%{"apple" => 3, "banana" => 1, "orange" => 6}

  
    
  
  to_list(enum)


iex> Enum.to_list(1..5)
[1, 2, 3, 4, 5]

  
    
  
  Comprehension


Conversion can also be done with comprehensions:
iex> for item <- cart, into: %{} do
...>   {item.fruit, item.count}
...> end
%{"apple" => 3, "banana" => 1, "orange" => 6}

  
    
  
  Duplicates & uniques



  
    
  
  dedup(enum)


dedup only removes contiguous duplicates:
iex> Enum.dedup([1, 2, 2, 3, 3, 3, 1, 2, 3])
[1, 2, 3, 1, 2, 3]

  
    
  
  dedup_by(enum, fun)


Remove contiguous entries given a property:
iex> Enum.dedup_by(cart, & &1.fruit =~ "a")
[%{fruit: "apple", count: 3}]
iex> Enum.dedup_by(cart, & &1.count < 5)
[
  %{fruit: "apple", count: 3},
  %{fruit: "orange", count: 6}
]

  
    
  
  uniq(enum)


uniq applies to the whole collection:
iex> Enum.uniq([1, 2, 2, 3, 3, 3, 1, 2, 3])
[1, 2, 3]
Comprehensions also support the uniq: true option.

  
    
  
  uniq_by(enum, fun)


Get entries which are unique by the last letter of the fruit:
iex> Enum.uniq_by(cart, &String.last(&1.fruit))
[
  %{fruit: "apple", count: 3},
  %{fruit: "banana", count: 1}
]

  
    
  
  Indexing



  
    
  
  at(enum, index, default \\ nil)


iex> Enum.at(cart, 0)
%{fruit: "apple", count: 3}
iex> Enum.at(cart, 10)
nil
iex> Enum.at(cart, 10, :none)
:none
Accessing a list by index in a loop is discouraged.

  
    
  
  fetch(enum, index)


iex> Enum.fetch(cart, 0)
{:ok, %{fruit: "apple", count: 3}}
iex> Enum.fetch(cart, 10)
:error

  
    
  
  fetch!(enum, index)


iex> Enum.fetch!(cart, 0)
%{fruit: "apple", count: 3}
iex> Enum.fetch!(cart, 10)
** (Enum.OutOfBoundsError) out of bounds error

  
    
  
  with_index(enum)


iex> Enum.with_index(cart)
[
  {%{fruit: "apple", count: 3}, 0},
  {%{fruit: "banana", count: 1}, 1},
  {%{fruit: "orange", count: 6}, 2}
]

  
    
  
  with_index(enum, fun)


iex> Enum.with_index(cart, fn item, index ->
...>   {item.fruit, index}
...> end)
[
  {"apple", 0},
  {"banana", 1},
  {"orange", 2}
]

  
    
  
  Finding



  
    
  
  find(enum, default \\ nil, fun)


iex> Enum.find(cart, &(&1.fruit =~ "o"))
%{fruit: "orange", count: 6}
iex> Enum.find(cart, &(&1.fruit =~ "y"))
nil
iex> Enum.find(cart, :none, &(&1.fruit =~ "y"))
:none

  
    
  
  find_index(enum, fun)


iex> Enum.find_index(cart, &(&1.fruit =~ "o"))
2
iex> Enum.find_index(cart, &(&1.fruit =~ "y"))
nil

  
    
  
  find_value(enum, default \\ nil, fun)


iex> Enum.find_value(cart, fn item ->
...>   if item.count == 1, do: item.fruit, else: nil
...> end)
"banana"
iex> Enum.find_value(cart, :none, fn item ->
...>   if item.count == 100, do: item.fruit, else: nil
...> end)
:none

  
    
  
  Grouping



  
    
  
  group_by(enum, key_fun)


Group by the last letter of the fruit:
iex> Enum.group_by(cart, &String.last(&1.fruit))
%{
  "a" => [%{fruit: "banana", count: 1}],
  "e" => [
    %{fruit: "apple", count: 3},
    %{fruit: "orange", count: 6}
  ]
}

  
    
  
  group_by(enum, key_fun, value_fun)


Group by the last letter of the fruit with custom value:
iex> Enum.group_by(cart, &String.last(&1.fruit), & &1.fruit)
%{
  "a" => ["banana"],
  "e" => ["apple", "orange"]
}

  
    
  
  Joining & interspersing



  
    
  
  join(enum, joiner \\ "")


iex> Enum.join(["apple", "banana", "orange"], ", ")
"apple, banana, orange"

  
    
  
  map_join(enum, joiner \\ "", mapper)


iex> Enum.map_join(cart, ", ", & &1.fruit)
"apple, banana, orange"

  
    
  
  intersperse(enum, separator \\ "")


iex> Enum.intersperse(["apple", "banana", "orange"], ", ")
["apple", ", ", "banana", ", ", "orange"]

  
    
  
  map_intersperse(enum, separator \\ "", mapper)


iex> Enum.map_intersperse(cart, ", ", & &1.fruit)
["apple", ", ", "banana", ", ", "orange"]

  
    
  
  Slicing



  
    
  
  slice(enum, index_range)


iex> Enum.slice(cart, 0..1)
[
  %{fruit: "apple", count: 3},
  %{fruit: "banana", count: 1}
]
Negative ranges count from the back:
iex> Enum.slice(cart, -2..-1)
[
  %{fruit: "banana", count: 1},
  %{fruit: "orange", count: 6}
]

  
    
  
  slice(enum, start_index, amount)


iex> Enum.slice(cart, 1, 2)
[
  %{fruit: "banana", count: 1},
  %{fruit: "orange", count: 6}
]

  
    
  
  slide(enum, range_or_single_index, insertion_index)


fruits = ["apple", "banana", "grape", "orange", "pear"]
iex> Enum.slide(fruits, 2, 0)
["grape", "apple", "banana", "orange", "pear"]
iex> Enum.slide(fruits, 2, 4)
["apple", "banana", "orange", "pear", "grape"]
iex> Enum.slide(fruits, 1..3, 0)
["banana", "grape", "orange", "apple", "pear"]
iex> Enum.slide(fruits, 1..3, 4)
["apple", "pear", "banana", "grape", "orange"]

  
    
  
  Reversing



  
    
  
  reverse(enum)


iex> Enum.reverse(cart)
[
  %{fruit: "orange", count: 6},
  %{fruit: "banana", count: 1},
  %{fruit: "apple", count: 3}
]

  
    
  
  reverse(enum, tail)


iex> Enum.reverse(cart, [:this_will_be, :the_tail])
[
  %{fruit: "orange", count: 6},
  %{fruit: "banana", count: 1},
  %{fruit: "apple", count: 3},
  :this_will_be,
  :the_tail
]

  
    
  
  reverse_slice(enum, start_index, count)


iex> Enum.reverse_slice(cart, 1, 2)
[
  %{fruit: "apple", count: 3},
  %{fruit: "orange", count: 6},
  %{fruit: "banana", count: 1}
]

  
    
  
  Splitting



  
    
  
  split(enum, amount)


iex> Enum.split(cart, 1)
{[%{fruit: "apple", count: 3}],
 [
   %{fruit: "banana", count: 1},
   %{fruit: "orange", count: 6}
 ]}
Negative indexes count from the back:
iex> Enum.split(cart, -1)
{[
  %{fruit: "apple", count: 3},
  %{fruit: "banana", count: 1}
 ],
 [%{fruit: "orange", count: 6}]}

  
    
  
  split_while(enum, fun)


Stops splitting as soon as it is false:
iex> Enum.split_while(cart, &(&1.fruit =~ "e"))
{[%{fruit: "apple", count: 3}],
 [
   %{fruit: "banana", count: 1},
   %{fruit: "orange", count: 6}
 ]}

  
    
  
  split_with(enum, fun)


Splits the whole collection:
iex> Enum.split_with(cart, &(&1.fruit =~ "e"))
{[
  %{fruit: "apple", count: 3},
  %{fruit: "orange", count: 6}
 ],
 [%{fruit: "banana", count: 1}]}

  
    
  
  Splitting (drop and take)



  
    
  
  drop(enum, amount)


iex> Enum.drop(cart, 1)
[
  %{fruit: "banana", count: 1},
  %{fruit: "orange", count: 6}
]
Negative indexes count from the back:
iex> Enum.drop(cart, -1)
[
  %{fruit: "apple", count: 3},
  %{fruit: "banana", count: 1}
]

  
    
  
  drop_every(enum, nth)


iex> Enum.drop_every(cart, 2)
[%{fruit: "banana", count: 1}]

  
    
  
  drop_while(enum, fun)


iex> Enum.drop_while(cart, &(&1.fruit =~ "e"))
[
  %{fruit: "banana", count: 1},
  %{fruit: "orange", count: 6}
]

  
    
  
  take(enum, amount)


iex> Enum.take(cart, 1)
[%{fruit: "apple", count: 3}]
Negative indexes count from the back:
iex> Enum.take(cart, -1)
[%{fruit: "orange", count: 6}]

  
    
  
  take_every(enum, nth)


iex> Enum.take_every(cart, 2)
[
  %{fruit: "apple", count: 3},
  %{fruit: "orange", count: 6}
]

  
    
  
  take_while(enum, fun)


iex> Enum.take_while(cart, &(&1.fruit =~ "e"))
[%{fruit: "apple", count: 3}]

  
    
  
  Random



  
    
  
  random(enum)


Results will vary on every call:
iex> Enum.random(cart)
%{fruit: "orange", count: 6}

  
    
  
  take_random(enum, count)


Results will vary on every call:
iex> Enum.take_random(cart, 2)
[
  %{fruit: "orange", count: 6},
  %{fruit: "apple", count: 3}
]

  
    
  
  shuffle(enum)


Results will vary on every call:
iex> Enum.shuffle(cart)
[
  %{fruit: "orange", count: 6},
  %{fruit: "apple", count: 3},
  %{fruit: "banana", count: 1}
]

  
    
  
  Chunking



  
    
  
  chunk_by(enum, fun)


iex> Enum.chunk_by(cart, &String.length(&1.fruit))
[
  [%{fruit: "apple", count: 3}],
  [
    %{fruit: "banana", count: 1},
    %{fruit: "orange", count: 6}
  ]
]

  
    
  
  chunk_every(enum, count)


iex> Enum.chunk_every(cart, 2)
[
  [
    %{fruit: "apple", count: 3},
    %{fruit: "banana", count: 1}
  ],
  [%{fruit: "orange", count: 6}]
]

  
    
  
  chunk_every(enum, count, step, leftover \\ [])


iex> Enum.chunk_every(cart, 2, 2, [:elements, :to_complete])
[
  [
    %{fruit: "apple", count: 3},
    %{fruit: "banana", count: 1}
  ],
  [
    %{fruit: "orange", count: 6},
    :elements
  ]
]
iex> Enum.chunk_every(cart, 2, 1, :discard)
[
  [
    %{fruit: "apple", count: 3},
    %{fruit: "banana", count: 1}
  ],
  [
    %{fruit: "banana", count: 1},
    %{fruit: "orange", count: 6}
  ]
]
See Enum.chunk_while/4 for custom chunking.

  
    
  
  Zipping



  
    
  
  zip(enum1, enum2)


iex> fruits = ["apple", "banana", "orange"]
iex> counts = [3, 1, 6]
iex> Enum.zip(fruits, counts)
[{"apple", 3}, {"banana", 1}, {"orange", 6}]
See Enum.zip/1 for zipping many collections at once.

  
    
  
  zip_with(enum1, enum2, fun)


iex> fruits = ["apple", "banana", "orange"]
iex> counts = [3, 1, 6]
iex> Enum.zip_with(fruits, counts, fn fruit, count ->
...>   %{fruit: fruit, count: count}
...> end)
[
  %{fruit: "apple", count: 3},
  %{fruit: "banana", count: 1},
  %{fruit: "orange", count: 6}
]
See Enum.zip_with/2 for zipping many collections at once.

  
    
  
  zip_reduce(left, right, acc, fun)


iex> fruits = ["apple", "banana", "orange"]
iex> counts = [3, 1, 6]
iex> Enum.zip_reduce(fruits, counts, 0, fn fruit, count, acc ->
...>   price = if fruit =~ "e", do: count * 2, else: count
...>   acc + price
...> end)
19
See Enum.zip_reduce/3 for zipping many collections at once.

  
    
  
  unzip(list)


iex> cart |> Enum.map(&{&1.fruit, &1.count}) |> Enum.unzip()
{["apple", "banana", "orange"], [3, 1, 6]}

What are anti-patterns?
    

Anti-patterns describe common mistakes or indicators of problems in code.
They are also known as "code smells".
The goal of these guides is to document potential anti-patterns found in Elixir software
and teach developers how to identify them and their pitfalls. If an existing piece
of code matches an anti-pattern, it does not mean your code must be rewritten.
Sometimes, even if a snippet matches a potential anti-pattern and its limitations,
it may be the best approach to the problem at hand. No codebase is free of anti-patterns
and one should not aim to remove all of them.
The anti-patterns in these guides are broken into 4 main categories:
	Code-related anti-patterns: related to your code and particular
language idioms and features;

	Design-related anti-patterns: related to your modules, functions,
and the role they play within a codebase;

	Process-related anti-patterns: related to processes and process-based
abstractions;

	Meta-programming anti-patterns: related to meta-programming.


Each anti-pattern is documented using the following structure:
	Name: Unique identifier of the anti-pattern. This name is important to facilitate
communication between developers;

	Problem: How the anti-pattern can harm code quality and what impacts this can have
for developers;

	Example: Code and textual descriptions to illustrate the occurrence of the anti-pattern;

	Refactoring: Ways to change your code to improve its qualities. Examples of refactored
code are presented to illustrate these changes.


An additional section with "Additional Remarks" may be provided. Those may include known scenarios where the anti-pattern does not apply.
The initial catalog of anti-patterns was proposed by Lucas Vegi and Marco Tulio Valente, from ASERG/DCC/UFMG. For more info, see Understanding Code Smells in Elixir Functional Language and the associated code repository.
Additionally, the Security Working Group of the Erlang Ecosystem Foundation publishes documents with security resources and best-practices of both Erlang and Elixir, including detailed guides for web applications.

Code-related anti-patterns
    

This document outlines potential anti-patterns related to your code and particular Elixir idioms and features.

  
    
  
  Comments overuse


Problem
When you overuse comments or comment self-explanatory code, it can have the effect of making code less readable.
Example
# Returns the Unix timestamp of 5 minutes from the current time
defp unix_five_min_from_now do
  # Get the current time
  now = DateTime.utc_now()

  # Convert it to a Unix timestamp
  unix_now = DateTime.to_unix(now, :second)

  # Add five minutes in seconds
  unix_now + (60 * 5)
end
Refactoring
Prefer clear and self-explanatory function names, module names, and variable names when possible. In the example above, the function name explains well what the function does, so you likely won't need the comment before it. The code also explains the operations well through variable names and clear function calls.
You could refactor the code above like this:
@five_min_in_seconds 60 * 5

defp unix_five_min_from_now do
  now = DateTime.utc_now()
  unix_now = DateTime.to_unix(now, :second)
  unix_now + @five_min_in_seconds
end
We removed the unnecessary comments. We also added a @five_min_in_seconds module attribute, which serves the additional purpose of giving a name to the "magic" number 60 * 5, making the code clearer and more expressive.
Additional remarks
Elixir makes a clear distinction between documentation and code comments. The language has built-in first-class support for documentation through @doc, @moduledoc, and more. See the "Writing documentation" guide for more information.

  
    
  
  Complex else clauses in with


Problem
This anti-pattern refers to with statements that flatten all its error clauses into a single complex else block. This situation is harmful to the code readability and maintainability because it's difficult to know from which clause the error value came.
Example
An example of this anti-pattern, as shown below, is a function open_decoded_file/1 that reads a Base64-encoded string content from a file and returns a decoded binary string. This function uses a with statement that needs to handle two possible errors, all of which are concentrated in a single complex else block.
def open_decoded_file(path) do
  with {:ok, encoded} <- File.read(path),
       {:ok, decoded} <- Base.decode64(encoded) do
    {:ok, String.trim(decoded)}
  else
    {:error, _} -> {:error, :badfile}
    :error -> {:error, :badencoding}
  end
end
In the code above, it is unclear how each pattern on the left side of <- relates to their error at the end. The more patterns in a with, the less clear the code gets, and the more likely it is that unrelated failures will overlap each other.
Refactoring
In this situation, instead of concentrating all error handling within a single complex else block, it is better to normalize the return types in specific private functions. In this way, with can focus on the success case and the errors are normalized closer to where they happen, leading to better organized and maintainable code.
def open_decoded_file(path) do
  with {:ok, encoded} <- file_read(path),
       {:ok, decoded} <- base_decode64(encoded) do
    {:ok, String.trim(decoded)}
  end
end

defp file_read(path) do
  case File.read(path) do
    {:ok, contents} -> {:ok, contents}
    {:error, _} -> {:error, :badfile}
  end
end

defp base_decode64(contents) do
  case Base.decode64(contents) do
    {:ok, decoded} -> {:ok, decoded}
    :error -> {:error, :badencoding}
  end
end

  
    
  
  Complex extractions in clauses


Problem
When we use multi-clause functions, it is possible to extract values in the clauses for further usage and for pattern matching/guard checking. This extraction itself does not represent an anti-pattern, but when you have extractions made across several clauses and several arguments of the same function, it becomes hard to know which extracted parts are used for pattern/guards and what is used only inside the function body. This anti-pattern is related to Unrelated multi-clause function, but with implications of its own. It impairs the code readability in a different way.
Example
The multi-clause function drive/1 is extracting fields of an %User{} struct for usage in the clause expression (age) and for usage in the function body (name):
def drive(%User{name: name, age: age}) when age >= 18 do
  "#{name} can drive"
end

def drive(%User{name: name, age: age}) when age < 18 do
  "#{name} cannot drive"
end
While the example above is small and does not constitute an anti-pattern, it is an example of mixed extraction and pattern matching. A situation where drive/1 was more complex, having many more clauses, arguments, and extractions, would make it hard to know at a glance which variables are used for pattern/guards and which ones are not.
Refactoring
As shown below, a possible solution to this anti-pattern is to extract only pattern/guard related variables in the signature once you have many arguments or multiple clauses:
def drive(%User{age: age} = user) when age >= 18 do
  %User{name: name} = user
  "#{name} can drive"
end

def drive(%User{age: age} = user) when age < 18 do
  %User{name: name} = user
  "#{name} cannot drive"
end

  
    
  
  Dynamic atom creation


Problem
An Atom is an Elixir basic type whose value is its own name. Atoms are often useful to identify resources or express the state, or result, of an operation. Creating atoms dynamically is not an anti-pattern by itself; however, atoms are not garbage collected by the Erlang Virtual Machine, so values of this type live in memory during a software's entire execution lifetime. The Erlang VM limits the number of atoms that can exist in an application by default to 1_048_576, which is more than enough to cover all atoms defined in a program, but attempts to serve as an early limit for applications which are "leaking atoms" through dynamic creation.
For these reason, creating atoms dynamically can be considered an anti-pattern when the developer has no control over how many atoms will be created during the software execution. This unpredictable scenario can expose the software to unexpected behavior caused by excessive memory usage, or even by reaching the maximum number of atoms possible.
Example
Picture yourself implementing code that converts string values into atoms. These strings could have been received from an external system, either as part of a request into our application, or as part of a response to your application. This dynamic and unpredictable scenario poses a security risk, as these uncontrolled conversions can potentially trigger out-of-memory errors.
defmodule MyRequestHandler do
  def parse(%{"status" => status, "message" => message} = _payload) do
    %{status: String.to_atom(status), message: message}
  end
end
iex> MyRequestHandler.parse(%{"status" => "ok", "message" => "all good"})
%{status: :ok, message: "all good"}
When we use the String.to_atom/1 function to dynamically create an atom, it essentially gains potential access to create arbitrary atoms in our system, causing us to lose control over adhering to the limits established by the BEAM. This issue could be exploited by someone to create enough atoms to shut down a system.
Refactoring
To eliminate this anti-pattern, developers must either perform explicit conversions by mapping strings to atoms or replace the use of String.to_atom/1 with String.to_existing_atom/1. An explicit conversion could be done as follows:
defmodule MyRequestHandler do
  def parse(%{"status" => status, "message" => message} = _payload) do
    %{status: convert_status(status), message: message}
  end

  defp convert_status("ok"), do: :ok
  defp convert_status("error"), do: :error
  defp convert_status("redirect"), do: :redirect
end
iex> MyRequestHandler.parse(%{"status" => "status_not_seen_anywhere", "message" => "all good"})
** (FunctionClauseError) no function clause matching in MyRequestHandler.convert_status/1
By explicitly listing all supported statuses, you guarantee only a limited number of conversions may happen. Passing an invalid status will lead to a function clause error.
An alternative is to use String.to_existing_atom/1, which will only convert a string to atom if the atom already exists in the system:
defmodule MyRequestHandler do
  def parse(%{"status" => status, "message" => message} = _payload) do
    %{status: String.to_existing_atom(status), message: message}
  end
end
iex> MyRequestHandler.parse(%{"status" => "status_not_seen_anywhere", "message" => "all good"})
** (ArgumentError) errors were found at the given arguments:

  * 1st argument: not an already existing atom
In such cases, passing an unknown status will raise as long as the status was not defined anywhere as an atom in the system. However, assuming status can be either :ok, :error, or :redirect, how can you guarantee those atoms exist? You must ensure those atoms exist somewhere in the same module where String.to_existing_atom/1 is called. For example, if you had this code:
defmodule MyRequestHandler do
  def parse(%{"status" => status, "message" => message} = _payload) do
    %{status: String.to_existing_atom(status), message: message}
  end

  def handle(%{status: status}) do
    case status do
      :ok -> ...
      :error -> ...
      :redirect -> ...
    end
  end
end
All valid statuses are defined as atoms within the same module, and that's enough. If you want to be explicit, you could also have a function that lists them:
def valid_statuses do
  [:ok, :error, :redirect]
end
However, keep in mind using a module attribute or defining the atoms in the module body, outside of a function, are not sufficient, as the module body is only executed during compilation and it is not necessarily part of the compiled module loaded at runtime.

  
    
  
  Long parameter list


Problem
In a functional language like Elixir, functions tend to explicitly receive all inputs and return all relevant outputs, instead of relying on mutations or side-effects. As functions grow in complexity, the amount of arguments (parameters) they need to work with may grow, to a point where the function's interface becomes confusing and prone to errors during use.
Example
In the following example, the loan/6 functions takes too many arguments, causing its interface to be confusing and potentially leading developers to introduce errors during calls to this function.
defmodule Library do
  # Too many parameters that can be grouped!
  def loan(user_name, email, password, user_alias, book_title, book_ed) do
    ...
  end
end
Refactoring
To address this anti-pattern, related arguments can be grouped using key-value data structures, such as maps, structs, or even keyword lists in the case of optional arguments. This effectively reduces the number of arguments and the key-value data structures adds clarity to the caller.
For this particular example, the arguments to loan/6 can be grouped into two different maps, thereby reducing its arity to loan/2:
defmodule Library do
  def loan(%{name: name, email: email, password: password, alias: alias} = user, %{title: title, ed: ed} = book) do
    ...
  end
end
In some cases, the function with too many arguments may be a private function, which gives us more flexibility over how to separate the function arguments. One possible suggestion for such scenarios is to split the arguments in two maps (or tuples): one map keeps the data that may change, and the other keeps the data that won't change (read-only). This gives us a mechanical option to refactor the code.
Other times, a function may legitimately take half a dozen or more completely unrelated arguments. This may suggest the function is trying to do too much and would be better broken into multiple functions, each responsible for a smaller piece of the overall responsibility.

  
    
  
  Namespace trespassing


Problem
This anti-pattern manifests when a package author or a library defines modules outside of its "namespace". A library should use its name as a "prefix" for all of its modules. For example, a package named :my_lib should define all of its modules within the MyLib namespace, such as MyLib.User, MyLib.SubModule, MyLib.Application, and MyLib itself.
This is important because the Erlang VM can only load one instance of a module at a time. So if there are multiple libraries that define the same module, then they are incompatible with each other due to this limitation. By always using the library name as a prefix, it avoids module name clashes due to the unique prefix.
Example
This problem commonly manifests when writing an extension of another library. For example, imagine you are writing a package that adds authentication to Plug called :plug_auth. You must avoid defining modules within the Plug namespace:
defmodule Plug.Auth do
  # ...
end
Even if Plug does not currently define a Plug.Auth module, it may add such a module in the future, which would ultimately conflict with plug_auth's definition.
Refactoring
Given the package is named :plug_auth, it must define modules inside the PlugAuth namespace:
defmodule PlugAuth do
  # ...
end
Additional remarks
There are few known exceptions to this anti-pattern:
	Protocol implementations are, by design, defined under the protocol namespace

	In some scenarios, the namespace owner may allow exceptions to this rule. For example, in Elixir itself, you defined custom Mix tasks by placing them under the Mix.Tasks namespace, such as Mix.Tasks.PlugAuth

	If you are the maintainer for both plug and plug_auth, then you may allow plug_auth to define modules with the Plug namespace, such as Plug.Auth. However, you are responsible for avoiding or managing any conflicts that may arise in the future



  
    
  
  Non-assertive map access


Problem
In Elixir, it is possible to access values from Maps, which are key-value data structures, either statically or dynamically.
When a key is expected to exist in a map, it must be accessed using the map.key notation, making it clear to developers (and the compiler) that the key must exist. If the key does not exist, an exception is raised (and in some cases also compiler warnings). This is also known as the static notation, as the key is known at the time of writing the code.
When a key is optional, the map[:key] notation must be used instead. This way, if the informed key does not exist, nil is returned. This is the dynamic notation, as it also supports dynamic key access, such as map[some_var].
When you use map[:key] to access a key that always exists in the map, you are making the code less clear for developers and for the compiler, as they now need to work with the assumption the key may not be there. This mismatch may also make it harder to track certain bugs. If the key is unexpectedly missing, you will have a nil value propagate through the system, instead of raising on map access.
Example
The function plot/1 tries to draw a graphic to represent the position of a point in a Cartesian plane. This function receives a parameter of Map type with the point attributes, which can be a point of a 2D or 3D Cartesian coordinate system. This function uses dynamic access to retrieve values for the map keys:
defmodule Graphics do
  def plot(point) do
    # Some other code...
    {point[:x], point[:y], point[:z]}
  end
end
iex> point_2d = %{x: 2, y: 3}
%{x: 2, y: 3}
iex> point_3d = %{x: 5, y: 6, z: 7}
%{x: 5, y: 6, z: 7}
iex> Graphics.plot(point_2d)
{2, 3, nil}
iex> Graphics.plot(point_3d)
{5, 6, 7}
Given we want to plot both 2D and 3D points, the behavior above is expected. But what happens if we forget to pass a point with either :x or :y?
iex> bad_point = %{y: 3, z: 4}
%{y: 3, z: 4}
iex> Graphics.plot(bad_point)
{nil, 3, 4}
The behavior above is unexpected because our function should not work with points without a :x key. This leads to subtle bugs, as we may now pass nil to another function, instead of raising early on.
Refactoring
To remove this anti-pattern, we must use the dynamic map[:key] syntax and the static map.key notation according to our requirements. We expect :x and :y to always exist, but not :z. The next code illustrates the refactoring of plot/1, removing this anti-pattern:
defmodule Graphics do
  def plot(point) do
    # Some other code...
    {point.x, point.y, point[:z]}
  end
end
iex> Graphics.plot(point_2d)
{2, 3, nil}
iex> Graphics.plot(bad_point)
** (KeyError) key :x not found in: %{y: 3, z: 4}
  graphic.ex:4: Graphics.plot/1
Overall, the usage of map.key and map[:key] encode important information about your data structure, allowing developers to be clear about their intent. See both Map and Access module documentation for more information and examples.
An alternative to refactor this anti-pattern is to use pattern matching, defining explicit clauses for 2d vs 3d points:
defmodule Graphics do
  # 3d
  def plot(%{x: x, y: y, z: z}) do
    # Some other code...
    {x, y, z}
  end

  # 2d
  def plot(%{x: x, y: y}) do
    # Some other code...
    {x, y}
  end
end
Pattern-matching is specially useful when matching over multiple keys as well as on the values themselves at once.
Another option is to use structs. By default, structs only support static access to its fields. In such scenarios, you may consider defining structs for both 2D and 3D points:
defmodule Point2D do
  @enforce_keys [:x, :y]
  defstruct [x: nil, y: nil]
end
Generally speaking, structs are useful when sharing data structures across modules, at the cost of adding a compile time dependency between these modules. If module A uses a struct defined in module B, A must be recompiled if the fields in the struct B change.
Additional remarks
This anti-pattern was formerly known as Accessing non-existent map/struct fields.

  
    
  
  Non-assertive pattern matching


Problem
Overall, Elixir systems are composed of many supervised processes, so the effects of an error are localized to a single process, and don't propagate to the entire application. A supervisor detects the failing process, reports it, and possibly restarts it. This anti-pattern arises when developers write defensive or imprecise code, capable of returning incorrect values which were not planned for, instead of programming in an assertive style through pattern matching and guards.
Example
The function get_value/2 tries to extract a value from a specific key of a URL query string. As it is not implemented using pattern matching, get_value/2 always returns a value, regardless of the format of the URL query string passed as a parameter in the call. Sometimes the returned value will be valid. However, if a URL query string with an unexpected format is used in the call, get_value/2 will extract incorrect values from it:
defmodule Extract do
  def get_value(string, desired_key) do
    parts = String.split(string, "&")

    Enum.find_value(parts, fn pair ->
      key_value = String.split(pair, "=")
      Enum.at(key_value, 0) == desired_key && Enum.at(key_value, 1)
    end)
  end
end
# URL query string with the planned format - OK!
iex> Extract.get_value("name=Lucas&university=UFMG&lab=ASERG", "lab")
"ASERG"
iex> Extract.get_value("name=Lucas&university=UFMG&lab=ASERG", "university")
"UFMG"
# Unplanned URL query string format - Unplanned value extraction!
iex> Extract.get_value("name=Lucas&university=institution=UFMG&lab=ASERG", "university")
"institution"   # <= why not "institution=UFMG"? or only "UFMG"?
Refactoring
To remove this anti-pattern, get_value/2 can be refactored through the use of pattern matching. So, if an unexpected URL query string format is used, the function will crash instead of returning an invalid value. This behavior, shown below, allows clients to decide how to handle these errors and doesn't give a false impression that the code is working correctly when unexpected values are extracted:
defmodule Extract do
  def get_value(string, desired_key) do
    parts = String.split(string, "&")

    Enum.find_value(parts, fn pair ->
      [key, value] = String.split(pair, "=") # <= pattern matching
      key == desired_key && value
    end)
  end
end
# URL query string with the planned format - OK!
iex> Extract.get_value("name=Lucas&university=UFMG&lab=ASERG", "name")
"Lucas"
# Unplanned URL query string format - Crash explaining the problem to the client!
iex> Extract.get_value("name=Lucas&university=institution=UFMG&lab=ASERG", "university")
** (MatchError) no match of right hand side value: ["university", "institution", "UFMG"]
  extract.ex:7: anonymous fn/2 in Extract.get_value/2 # <= left hand: [key, value] pair
iex> Extract.get_value("name=Lucas&university&lab=ASERG", "university")
** (MatchError) no match of right hand side value: ["university"]
  extract.ex:7: anonymous fn/2 in Extract.get_value/2 # <= left hand: [key, value] pair
Elixir and pattern matching promote an assertive style of programming where you handle the known cases. Once an unexpected scenario arises, you can decide to address it accordingly based on practical examples, or conclude the scenario is indeed invalid and the exception is the desired choice.
case/2 is another important construct in Elixir that help us write assertive code, by matching on specific patterns. For example, if a function returns {:ok, ...} or {:error, ...}, prefer to explicitly match on both patterns:
case some_function(arg) do
  {:ok, value} -> # ...
  {:error, _} -> # ...
end
In particular, avoid matching solely on _, as shown below:
case some_function(arg) do
  {:ok, value} -> # ...
  _ -> # ...
end
 Matching on _ is less clear in intent and it may hide bugs if some_function/1 adds new return values in the future.
Additional remarks
This anti-pattern was formerly known as Speculative assumptions.

  
    
  
  Non-assertive truthiness


Problem
Elixir provides the concept of truthiness: nil and false are considered "falsy" and all other values are "truthy". Many constructs in the language, such as 

This document outlines potential anti-patterns related to your modules, functions, and the role they play within a codebase.

  
    
  
  Alternative return types


Problem
This anti-pattern refers to functions that receive options (typically as a keyword list parameter) that drastically change their return type. Because options are optional and sometimes set dynamically, if they also change the return type, it may be hard to understand what the function actually returns.
Example
An example of this anti-pattern, as shown below, is when a function has many alternative return types, depending on the options received as a parameter.
defmodule AlternativeInteger do
  @spec parse(String.t(), keyword()) :: integer() | {integer(), String.t()} | :error
  def parse(string, options \\ []) when is_list(options) do
    if Keyword.get(options, :discard_rest, false) do
      case Integer.parse(string) do
        {int, _rest} -> int
        :error -> :error
      end
    else
      Integer.parse(string)
    end
  end
end
iex> AlternativeInteger.parse("13")
{13, ""}
iex> AlternativeInteger.parse("13", discard_rest: false)
{13, ""}
iex> AlternativeInteger.parse("13", discard_rest: true)
13
Refactoring
To refactor this anti-pattern, as shown next, add a specific function for each return type (for example, parse_discard_rest/1), no longer delegating this to options passed as arguments.
defmodule AlternativeInteger do
  @spec parse(String.t()) :: {integer(), String.t()} | :error
  def parse(string) do
    Integer.parse(string)
  end

  @spec parse_discard_rest(String.t()) :: integer() | :error
  def parse_discard_rest(string) do
    case Integer.parse(string) do
      {int, _rest} -> int
      :error -> :error
    end
  end
end
iex> AlternativeInteger.parse("13")
{13, ""}
iex> AlternativeInteger.parse_discard_rest("13")
13

  
    
  
  Boolean obsession


Problem
This anti-pattern happens when booleans are used instead of atoms to encode information. The usage of booleans themselves is not an anti-pattern, but whenever multiple booleans are used with overlapping states, replacing the booleans by atoms (or composite data types such as tuples) may lead to clearer code.
This is a special case of Primitive obsession, specific to boolean values.
Example
An example of this anti-pattern is a function that receives two or more options, such as editor: true and admin: true, to configure its behavior in overlapping ways. In the code below, the :editor option has no effect if :admin is set, meaning that the :admin option has higher priority than :editor, and they are ultimately related.
defmodule MyApp do
  def process(invoice, options \\ []) do
    cond do
      options[:admin] ->  # Is an admin
      options[:editor] -> # Is an editor
      true ->          # Is none
    end
  end
end
Refactoring
Instead of using multiple options, the code above could be refactored to receive a single option, called :role, that can be either :admin, :editor, or :default:
defmodule MyApp do
  def process(invoice, options \\ []) do
    case Keyword.get(options, :role, :default) do
      :admin ->   # Is an admin
      :editor ->  # Is an editor
      :default -> # Is none
    end
  end
end
This anti-pattern may also happen in our own data structures. For example, we may define a User struct with two boolean fields, :editor and :admin, while a single field named :role may be preferred.
Finally, it is worth noting that using atoms may be preferred even when we have a single boolean argument/option. For example, consider an invoice which may be set as approved/unapproved. One option is to provide a function that expects a boolean:
MyApp.update(invoice, approved: true)
However, using atoms may read better and make it simpler to add further states (such as pending) in the future:
MyApp.update(invoice, status: :approved)
Remember booleans are internally represented as atoms. Therefore there is no performance penalty in one approach over the other.

  
    
  
  Exceptions for control-flow


Problem
This anti-pattern refers to code that uses Exceptions for control flow. Exception handling itself does not represent an anti-pattern, but developers must prefer to use case and pattern matching to change the flow of their code, instead of try/rescue. In turn, library authors should provide developers with APIs to handle errors without relying on exception handling. When developers have no freedom to decide if an error is exceptional or not, this is considered an anti-pattern.
Example
An example of this anti-pattern, as shown below, is using try/rescue to deal with file operations:
defmodule MyModule do
  def print_file(file) do
    try do
      IO.puts(File.read!(file))
    rescue
      e -> IO.puts(:stderr, Exception.message(e))
    end
  end
end
iex> MyModule.print_file("valid_file")
This is a valid file!
:ok
iex> MyModule.print_file("invalid_file")
could not read file "invalid_file": no such file or directory
:ok
Refactoring
To refactor this anti-pattern, as shown next, use File.read/1, which returns tuples instead of raising when a file cannot be read:
defmodule MyModule do
  def print_file(file) do
    case File.read(file) do
      {:ok, binary} -> IO.puts(binary)
      {:error, reason} -> IO.puts(:stderr, "could not read file #{file}: #{reason}")
    end
  end
end
This is only possible because the File module provides APIs for reading files with tuples as results (File.read/1), as well as a version that raises an exception (File.read!/1). The bang (exclamation point) is effectively part of Elixir's naming conventions.
Library authors are encouraged to follow the same practices. In practice, the bang variant is implemented on top of the non-raising version of the code. For example, File.read!/1 is implemented as:
def read!(path) do
  case read(path) do
    {:ok, binary} ->
      binary

    {:error, reason} ->
      raise File.Error, reason: reason, action: "read file", path: IO.chardata_to_string(path)
  end
end
A common practice followed by the community is to make the non-raising version return {:ok, result} or {:error, Exception.t}. For example, an HTTP client may return {:ok, %HTTP.Response{}} on success cases and {:error, %HTTP.Error{}} for failures, where HTTP.Error is implemented as an exception. This makes it convenient for anyone to raise an exception by simply calling Kernel.raise/1.
Additional remarks
This anti-pattern is of special importance to library authors and whenever writing functions that will be invoked by other developers and third-party code. Nevertheless, there are still scenarios where developers can afford to raise exceptions directly, for example:
	invalid arguments: it is expected that functions will raise for invalid arguments, as those are structural error and not semantic errors. For example, File.read(123) will always raise, because 123 is never a valid filename

	during tests, scripts, etc: those are common scenarios where you want your code to fail as soon as possible in case of errors. Using ! functions, such as File.read!/1, allows you to do so quickly and with clear error messages

	some frameworks, such as Phoenix, allow developers to raise exceptions in their code and uses a protocol to convert these exceptions into semantic HTTP responses


This anti-pattern was formerly known as Using exceptions for control-flow.

  
    
  
  Primitive obsession


Problem
This anti-pattern happens when Elixir basic types (for example, integer, float, and string) are excessively used to carry structured information, rather than creating specific composite data types (for example, tuples, maps, and structs) that can better represent a domain.
Example
An example of this anti-pattern is the use of a single string to represent an Address. An Address is a more complex structure than a simple basic (aka, primitive) value.
defmodule MyApp do
  def extract_postal_code(address) when is_binary(address) do
    # Extract postal code with address...
  end

  def fill_in_country(address) when is_binary(address) do
    # Fill in missing country...
  end
end
While you may receive the address as a string from a database, web request, or a third-party, if you find yourself frequently manipulating or extracting information from the string, it is a good indicator you should convert the address into structured data:
Another example of this anti-pattern is using floating numbers to model money and currency, when richer data structures should be preferred.
Refactoring
Possible solutions to this anti-pattern is to use maps or structs to model our address. The example below creates an Address struct, better representing this domain through a composite type. Additionally, we introduce a parse/1 function, that converts the string into an Address, which will simplify the logic of remaining functions. With this modification, we can extract each field of this composite type individually when needed.
defmodule Address do
  defstruct [:street, :city, :state, :postal_code, :country]
end
defmodule MyApp do
  def parse(address) when is_binary(address) do
    # Returns %Address{}
  end

  def extract_postal_code(%Address{} = address) do
    # Extract postal code with address...
  end

  def fill_in_country(%Address{} = address) do
    # Fill in missing country...
  end
end

  
    
  
  Unrelated multi-clause function


Problem
Using multi-clause functions is a powerful Elixir feature. However, some developers may abuse this feature to group unrelated functionality, which is an anti-pattern.
Example
A frequent example of this usage of multi-clause functions occurs when developers mix unrelated business logic into the same function definition, in a way that the behavior of each clause becomes completely distinct from the others. Such functions often have too broad specifications, making it difficult for other developers to understand and maintain them.
Some developers may use documentation mechanisms such as @doc annotations to compensate for poor code readability, however the documentation itself may end-up full of conditionals to describe how the function behaves for each different argument combination. This is a good indicator that the clauses are ultimately unrelated.
@doc """
Updates a struct.

If given a product, it will...

If given an animal, it will...
"""
def update(%Product{count: count, material: material})  do
  # ...
end

def update(%Animal{count: count, skin: skin})  do
  # ...
end
If updating an animal is completely different from updating a product and requires a different set of rules, it may be worth splitting those over different functions or even different modules.
Refactoring
As shown below, a possible solution to this anti-pattern is to break the business rules that are mixed up in a single unrelated multi-clause function in simple functions. Each function can have a specific name and @doc, describing its behavior and parameters received. While this refactoring sounds simple, it can impact the function's callers, so be careful!
@doc """
Updates a product.

It will...
"""
def update_product(%Product{count: count, material: material}) do
  # ...
end

@doc """
Updates an animal.

It will...
"""
def update_animal(%Animal{count: count, skin: skin}) do
  # ...
end
These functions may still be implemented with multiple clauses, as long as the clauses group related functionality. For example, update_product could be in practice implemented as follows:
def update_product(%Product{count: 0}) do
  # ...
end

def update_product(%Product{material: material})
    when material in ["metal", "glass"] do
  # ...
end

def update_product(%Product{material: material})
    when material not in ["metal", "glass"] do
  # ...
end
You can see this pattern in practice within Elixir itself. The +/2 operator can add Integers and Floats together, but not Strings, which instead use the <>/2 operator. In this sense, it is reasonable to handle integers and floats in the same operation, but strings are unrelated enough to deserve their own function.
You will also find examples in Elixir of functions that work with any struct, which would seemingly be an occurrence of this anti-pattern, such as struct/2:
iex> struct(URI.parse("/foo/bar"), path: "/bar/baz")
%URI{
  scheme: nil,
  userinfo: nil,
  host: nil,
  port: nil,
  path: "/bar/baz",
  query: nil,
  fragment: nil
}
The difference here is that the struct/2 function behaves precisely the same for any struct given, therefore there is no question of how the function handles different inputs. If the behavior is clear and consistent for all inputs, then the anti-pattern does not take place.

  
    
  
  Using application configuration for libraries


Problem
The application environment can be used to parameterize global values that can be used in an Elixir system. This mechanism can be very useful and therefore is not considered an anti-pattern by itself. However, library authors should avoid using the application environment to configure their library. The reason is exactly that the application environment is a global state, so there can only be a single value for each key in the environment for an application. This makes it impossible for multiple applications depending on the same library to configure the same aspect of the library in different ways.
Example
The DashSplitter module represents a library that configures the behavior of its functions through the global application environment. These configurations are concentrated in the config/config.exs file, shown below:
import Config

config :app_config,
  parts: 3

import_config "#{config_env()}.exs"
One of the functions implemented by the DashSplitter library is split/1. This function aims to separate a string received via a parameter into a certain number of parts. The character used as a separator in split/1 is always "-" and the number of parts the string is split into is defined globally by the application environment. This value is retrieved by the split/1 function by calling Application.fetch_env!/2, as shown next:
defmodule DashSplitter do
  def split(string) when is_binary(string) do
    parts = Application.fetch_env!(:app_config, :parts) # <= retrieve parameterized value
    String.split(string, "-", parts: parts)             # <= parts: 3
  end
end
Due to this parameterized value used by the DashSplitter library, all applications dependent on it can only use the split/1 function with identical behavior about the number of parts generated by string separation. Currently, this value is equal to 3, as we can see in the use examples shown below:
iex> DashSplitter.split("Lucas-Francisco-Vegi")
["Lucas", "Francisco", "Vegi"]
iex> DashSplitter.split("Lucas-Francisco-da-Matta-Vegi")
["Lucas", "Francisco", "da-Matta-Vegi"]
Refactoring
To remove this anti-pattern, this type of configuration should be performed using a parameter passed to the function. The code shown below performs the refactoring of the split/1 function by accepting keyword lists as a new optional parameter. With this new parameter, it is possible to modify the default behavior of the function at the time of its call, allowing multiple different ways of using split/2 within the same application:
defmodule DashSplitter do
  def split(string, opts \\ []) when is_binary(string) and is_list(opts) do
    parts = Keyword.get(opts, :parts, 2) # <= default config of parts == 2
    String.split(string, "-", parts: parts)
  end
end
iex> DashSplitter.split("Lucas-Francisco-da-Matta-Vegi", [parts: 5])
["Lucas", "Francisco", "da", "Matta", "Vegi"]
iex> DashSplitter.split("Lucas-Francisco-da-Matta-Vegi") #<= default config is used!
["Lucas", "Francisco-da-Matta-Vegi"]
Of course, not all uses of the application environment by libraries are incorrect. One example is using configuration to replace a component (or dependency) of a library by another that must behave the exact same. Consider a library that needs to parse CSV files. The library author may pick one package to use as default parser but allow its users to swap to different implementations via the application environment. At the end of the day, choosing a different CSV parser should not change the outcome, and library authors can even enforce this by defining behaviours with the exact semantics they expect.
Additional remarks: Supervision trees
In practice, libraries may require additional configuration beyond keyword lists. For example, if a library needs to start a supervision tree, how can the user of said library customize its supervision tree? Given the supervision tree itself is global (as it belongs to the library), library authors may be tempted to use the application configuration once more.
One solution is for the library to provide its own child specification, instead of starting the supervision tree itself. This allows the user to start all necessary processes under its own supervision tree, potentially passing custom configuration options during initialization.
You can see this pattern in practice in projects like Nx and DNS Cluster. These libraries require that you list processes under your own supervision tree:
children = [
  {DNSCluster, query: "my.subdomain"}
]
In such cases, if the users of DNSCluster need to configure DNSCluster per environment, they can be the ones reading from the application environment, without the library forcing them to:
children = [
  {DNSCluster, query: Application.get_env(:my_app, :dns_cluster_query) || :ignore}
]
Some libraries, such as Ecto, allow you to pass your application name as an option (called :otp_app or similar) and then automatically read the environment from your application. While this addresses the issue with the application environment being global, as they read from each individual application, it comes at the cost of some indirection, compared to the example above where users explicitly read their application environment from their own code, whenever desired.
Additional remarks: Compile-time configuration
A similar discussion entails compile-time configuration. What if a library author requires some configuration to be provided at compilation time?
Once again, instead of forcing users of your library to provide compile-time configuration, you may want to allow users of your library to generate the code themselves. That's the approach taken by libraries such as Ecto:
defmodule MyApp.Repo do
  use Ecto.Repo, adapter: Ecto.Adapters.Postgres
end
Instead of forcing developers to share a single repository, Ecto allows its users to define as many repositories as they want. Given the :adapter configuration is required at compile-time, it is a required value on use Ecto.Repo. If developers want to configure the adapter per environment, then it is their choice:
defmodule MyApp.Repo do
  use Ecto.Repo, adapter: Application.compile_env(:my_app, :repo_adapter)
end
On the other hand, code generation comes with its own anti-patterns, and must be considered carefully. That's to say: while using the application environment for libraries is discouraged, especially compile-time configuration, in some cases they may be the best option. For example, consider a library needs to parse CSV or JSON files to generate code based on data files. In such cases, it is best to provide reasonable defaults and make them customizable via the application environment, instead of asking each user of your library to generate the exact same code.
Additional remarks: Mix tasks
For Mix tasks and related tools, it may be necessary to provide per-project configuration. For example, imagine you have a :linter project, which supports setting the output file and the verbosity level. You may choose to configure it through application environment:
config :linter,
  output_file: "/path/to/output.json",
  verbosity: 3
However, Mix allows tasks to read per-project configuration via Mix.Project.config/0. In this case, you can configure the :linter directly in the mix.exs file:
def project do
  [
    app: :my_app,
    version: "1.0.0",
    linter: [
      output_file: "/path/to/output.json",
      verbosity: 3
    ],
    ...
  ]
end
Additionally, if a Mix task is available, you can also accept these options as command line arguments (see OptionParser):
mix linter --output-file /path/to/output.json --verbosity 3

Process-related anti-patterns
    

This document outlines potential anti-patterns related to processes and process-based abstractions.

  
    
  
  Code organization by process


Problem
This anti-pattern refers to code that is unnecessarily organized by processes. A process itself does not represent an anti-pattern, but it should only be used to model runtime properties (such as concurrency, access to shared resources, error isolation, etc). When you use a process for code organization, it can create bottlenecks in the system.
Example
An example of this anti-pattern, as shown below, is a module that implements arithmetic operations (like add and subtract) by means of a GenServer process. If the number of calls to this single process grows, this code organization can compromise the system performance, therefore becoming a bottleneck.
defmodule Calculator do
  @moduledoc """
  Calculator that performs basic arithmetic operations.

  This code is unnecessarily organized in a GenServer process.
  """

  use GenServer

  def add(a, b, pid) do
    GenServer.call(pid, {:add, a, b})
  end

  def subtract(a, b, pid) do
    GenServer.call(pid, {:subtract, a, b})
  end

  @impl GenServer
  def init(init_arg) do
    {:ok, init_arg}
  end

  @impl GenServer
  def handle_call({:add, a, b}, _from, state) do
    {:reply, a + b, state}
  end

  def handle_call({:subtract, a, b}, _from, state) do
    {:reply, a - b, state}
  end
end
iex> {:ok, pid} = GenServer.start_link(Calculator, :init)
{:ok, #PID<0.132.0>}
iex> Calculator.add(1, 5, pid)
6
iex> Calculator.subtract(2, 3, pid)
-1
Refactoring
In Elixir, as shown next, code organization must be done only through modules and functions. Whenever possible, a library should not impose specific behavior (such as parallelization) on its users. It is better to delegate this behavioral decision to the developers of clients, thus increasing the potential for code reuse of a library.
defmodule Calculator do
  def add(a, b) do
    a + b
  end

  def subtract(a, b) do
    a - b
  end
end
iex> Calculator.add(1, 5)
6
iex> Calculator.subtract(2, 3)
-1

  
    
  
  Scattered process interfaces


Problem
In Elixir, the use of an Agent, a GenServer, or any other process abstraction is not an anti-pattern in itself. However, when the responsibility for direct interaction with a process is spread throughout the entire system, it can become problematic. This bad practice can increase the difficulty of code maintenance and make the code more prone to bugs.
Example
The following code seeks to illustrate this anti-pattern. The responsibility for interacting directly with the Agent is spread across four different modules (A, B, C, and D).
defmodule A do
  def update(process) do
    # Some other code...
    Agent.update(process, fn _list -> 123 end)
  end
end
defmodule B do
  def update(process) do
    # Some other code...
    Agent.update(process, fn content -> %{a: content} end)
  end
end
defmodule C do
  def update(process) do
    # Some other code...
    Agent.update(process, fn content -> [:atom_value | content] end)
  end
end
defmodule D do
  def get(process) do
    # Some other code...
    Agent.get(process, fn content -> content end)
  end
end
This spreading of responsibility can generate duplicated code and make code maintenance more difficult. Also, due to the lack of control over the format of the shared data, complex composed data can be shared. This freedom to use any format of data is dangerous and can induce developers to introduce bugs.
# start an agent with initial state of an empty list
iex> {:ok, agent} = Agent.start_link(fn -> [] end)
{:ok, #PID<0.135.0>}

# many data formats (for example, List, Map, Integer, Atom) are
# combined through direct access spread across the entire system
iex> A.update(agent)
iex> B.update(agent)
iex> C.update(agent)

# state of shared information
iex> D.get(agent)
[:atom_value, %{a: 123}]
For a GenServer and other behaviours, this anti-pattern will manifest when scattering calls to GenServer.call/3 and GenServer.cast/2 throughout multiple modules, instead of encapsulating all the interaction with the GenServer in a single place.
Refactoring
Instead of spreading direct access to a process abstraction, such as Agent, over many places in the code, it is better to refactor this code by centralizing the responsibility for interacting with a process in a single module. This refactoring improves maintainability by removing duplicated code; it also allows you to limit the accepted format for shared data, reducing bug-proneness. As shown below, the module Foo.Bucket is centralizing the responsibility for interacting with the Agent. Any other place in the code that needs to access shared data must now delegate this action to Foo.Bucket. Also, Foo.Bucket now only allows data to be shared in Map format.
defmodule Foo.Bucket do
  use Agent

  def start_link(_opts) do
    Agent.start_link(fn -> %{} end)
  end

  def get(bucket, key) do
    Agent.get(bucket, &Map.get(&1, key))
  end

  def put(bucket, key, value) do
    Agent.update(bucket, &Map.put(&1, key, value))
  end
end
The following are examples of how to delegate access to shared data (provided by an Agent) to Foo.Bucket.
# start an agent through `Foo.Bucket`
iex> {:ok, bucket} = Foo.Bucket.start_link(%{})
{:ok, #PID<0.114.0>}

# add shared values to the keys `milk` and `beer`
iex> Foo.Bucket.put(bucket, "milk", 3)
iex> Foo.Bucket.put(bucket, "beer", 7)

# access shared data of specific keys
iex> Foo.Bucket.get(bucket, "beer")
7
iex> Foo.Bucket.get(bucket, "milk")
3
Additional remarks
This anti-pattern was formerly known as Agent obsession.

  
    
  
  Sending unnecessary data


Problem
Sending a message to a process can be an expensive operation if the message is big enough. That's because that message will be fully copied to the receiving process, which may be CPU and memory intensive. This is due to Erlang's "share nothing" architecture, where each process has its own memory, which simplifies and speeds up garbage collection.
This is more obvious when using send/2, GenServer.call/3, or the initial data in GenServer.start_link/3. Notably this also happens when using spawn/1, Task.async/1, Task.async_stream/3, and so on. It is more subtle here as the anonymous function passed to these functions captures the variables it references, and all captured variables will be copied over. By doing this, you can accidentally send way more data to a process than you actually need.
Example
Imagine you were to implement some simple reporting of IP addresses that made requests against your application. You want to do this asynchronously and not block processing, so you decide to use spawn/1. It may seem like a good idea to hand over the whole connection because we might need more data later. However passing the connection results in copying a lot of unnecessary data like the request body, params, etc.
# log_request_ip send the ip to some external service
spawn(fn -> log_request_ip(conn) end)
This problem also occurs when accessing only the relevant parts:
spawn(fn -> log_request_ip(conn.remote_ip) end)
This will still copy over all of conn, because the conn variable is being captured inside the spawned function. The function then extracts the remote_ip field, but only after the whole conn has been copied over.
send/2 and the GenServer APIs also rely on message passing. In the example below, the conn is once again copied to the underlying GenServer:
GenServer.cast(pid, {:report_ip_address, conn})
Refactoring
This anti-pattern has many potential remedies:
	Limit the data you send to the absolute necessary minimum instead of sending an entire struct. For example, don't send an entire conn struct if all you need is a couple of fields.

	If the only process that needs data is the one you are sending to, consider making the process fetch that data instead of passing it.

	Some abstractions, such as :persistent_term, allows you to share data between processes, as long as such data changes infrequently.


In our case, limiting the input data is a reasonable strategy. If all we need right now is the IP address, then let's only work with that and make sure we're only passing the IP address into the closure, like so:
ip_address = conn.remote_ip
spawn(fn -> log_request_ip(ip_address) end)
Or in the GenServer case:
GenServer.cast(pid, {:report_ip_address, conn.remote_ip})

  
    
  
  Unsupervised processes


Problem
In Elixir, creating a process outside a supervision tree is not an anti-pattern in itself. However, when you spawn many long-running processes outside of supervision trees, this can make visibility and monitoring of these processes difficult, preventing developers from fully controlling their applications.
Example
The following code example seeks to illustrate a library responsible for maintaining a numerical Counter through a GenServer process outside a supervision tree. Multiple counters can be created simultaneously by a client (one process for each counter), making these unsupervised processes difficult to manage. This can cause problems with the initialization, restart, and shutdown of a system.
defmodule Counter do
  @moduledoc """
  Global counter implemented through a GenServer process.
  """

  use GenServer

  @doc "Starts a counter process."
  def start_link(opts \\ []) do
    initial_value = Keyword.get(opts, :initial_value, 0)
    name = Keyword.get(opts, :name, __MODULE__)
    GenServer.start(__MODULE__, initial_value, name: name)
  end

  @doc "Gets the current value of the given counter."
  def get(pid_name \\ __MODULE__) do
    GenServer.call(pid_name, :get)
  end

  @doc "Bumps the value of the given counter."
  def bump(pid_name \\ __MODULE__, value) do
    GenServer.call(pid_name, {:bump, value})
  end

  @impl true
  def init(counter) do
    {:ok, counter}
  end

  @impl true
  def handle_call(:get, _from, counter) do
    {:reply, counter, counter}
  end

  def handle_call({:bump, value}, _from, counter) do
    {:reply, counter, counter + value}
  end
end
iex> Counter.start_link()
{:ok, #PID<0.115.0>}
iex> Counter.get()
0
iex> Counter.start_link(initial_value: 15, name: :other_counter)
{:ok, #PID<0.120.0>}
iex> Counter.get(:other_counter)
15
iex> Counter.bump(:other_counter, -3)
12
iex> Counter.bump(Counter, 7)
7
Refactoring
To ensure that clients of a library have full control over their systems, regardless of the number of processes used and the lifetime of each one, all processes must be started inside a supervision tree. As shown below, this code uses a Supervisor as a supervision tree. When this Elixir application is started, two different counters (Counter and :other_counter) are also started as child processes of the Supervisor named App.Supervisor. One is initialized with 0, the other with 15. By means of this supervision tree, it is possible to manage the life cycle of all child processes (stopping or restarting each one), improving the visibility of the entire app.
defmodule SupervisedProcess.Application do
  use Application

  @impl true
  def start(_type, _args) do
    children = [
      # With the default values for counter and name
      Counter,
      # With custom values for counter, name, and a custom ID
      Supervisor.child_spec(
        {Counter, name: :other_counter, initial_value: 15},
        id: :other_counter
      )
    ]

    Supervisor.start_link(children, strategy: :one_for_one, name: App.Supervisor)
  end
end
iex> Supervisor.count_children(App.Supervisor)
%{active: 2, specs: 2, supervisors: 0, workers: 2}
iex> Counter.get(Counter)
0
iex> Counter.get(:other_counter)
15
iex> Counter.bump(Counter, 7)
7
iex> Supervisor.terminate_child(App.Supervisor, Counter)
iex> Supervisor.count_children(App.Supervisor) # Only one active child
%{active: 1, specs: 2, supervisors: 0, workers: 2}
iex> Counter.get(Counter) # The process was terminated
** (EXIT) no process: the process is not alive...
iex> Supervisor.restart_child(App.Supervisor, Counter)
iex> Counter.get(Counter) # After the restart, this process can be used again
0

Meta-programming anti-patterns
    

This document outlines potential anti-patterns related to meta-programming.

  
    
  
  Large code generation


Problem
This anti-pattern is related to macros that generate too much code. When a macro generates a large amount of code, it impacts how the compiler and/or the runtime work. The reason for this is that Elixir may have to expand, compile, and execute the code multiple times, which will make compilation slower and the resulting compiled artifacts larger.
Example
Imagine you are defining a router for a web application, where you could have macros like get/2. On every invocation of the macro (which could be hundreds), the code inside get/2 will be expanded and compiled, which can generate a large volume of code overall.
defmodule Routes do
  defmacro get(route, handler) do
    quote do
      route = unquote(route)
      handler = unquote(handler)

      if not is_binary(route) do
        raise ArgumentError, "route must be a binary"
      end

      if not is_atom(handler) do
        raise ArgumentError, "handler must be a module"
      end

      @store_route_for_compilation {route, handler}
    end
  end
end
Refactoring
To remove this anti-pattern, the developer should simplify the macro, delegating part of its work to other functions. As shown below, by encapsulating the code inside quote/1 inside the function __define__/3 instead, we reduce the code that is expanded and compiled on every invocation of the macro, and instead we dispatch to a function to do the bulk of the work.
defmodule Routes do
  defmacro get(route, handler) do
    quote do
      Routes.__define__(__MODULE__, unquote(route), unquote(handler))
    end
  end

  def __define__(module, route, handler) do
    if not is_binary(route) do
      raise ArgumentError, "route must be a binary"
    end

    if not is_atom(handler) do
      raise ArgumentError, "handler must be a module"
    end

    Module.put_attribute(module, :store_route_for_compilation, {route, handler})
  end
end

  
    
  
  Unnecessary macros


Problem
Macros are powerful meta-programming mechanisms that can be used in Elixir to extend the language. While using macros is not an anti-pattern in itself, this meta-programming mechanism should only be used when absolutely necessary. Whenever a macro is used, but it would have been possible to solve the same problem using functions or other existing Elixir structures, the code becomes unnecessarily more complex and less readable. Because macros are more difficult to implement and reason about, their indiscriminate use can compromise the evolution of a system, reducing its maintainability.
Example
The MyMath module implements the sum/2 macro to perform the sum of two numbers received as parameters. While this code has no syntax errors and can be executed correctly to get the desired result, it is unnecessarily more complex. By implementing this functionality as a macro rather than a conventional function, the code became less clear:
defmodule MyMath do
  defmacro sum(v1, v2) do
    quote do
      unquote(v1) + unquote(v2)
    end
  end
end
iex> require MyMath
MyMath
iex> MyMath.sum(3, 5)
8
iex> MyMath.sum(3 + 1, 5 + 6)
15
Refactoring
To remove this anti-pattern, the developer must replace the unnecessary macro with structures that are simpler to write and understand, such as named functions. The code shown below is the result of the refactoring of the previous example. Basically, the sum/2 macro has been transformed into a conventional named function. Note that the require/2 call is no longer needed:
defmodule MyMath do
  def sum(v1, v2) do # <= The macro became a named function
    v1 + v2
  end
end
iex> MyMath.sum(3, 5)
8
iex> MyMath.sum(3+1, 5+6)
15

  
    
  
  use instead of import


Problem
Elixir has mechanisms such as import/1, alias/1, and use/1 to establish dependencies between modules. Code implemented with these mechanisms does not characterize a smell by itself. However, while the import/1 and alias/1 directives have lexical scope and only facilitate a module calling functions of another, the use/1 directive has a broader scope, which can be problematic.
The use/1 directive allows a module to inject any type of code into another, including propagating dependencies. In this way, using the use/1 directive makes code harder to read, because to understand exactly what will happen when it references a module, it is necessary to have knowledge of the internal details of the referenced module.
Example
The code shown below is an example of this anti-pattern. It defines three modules -- ModuleA, Library, and ClientApp. ClientApp is reusing code from the Library via the use/1 directive, but is unaware of its internal details. This makes it harder for the author of ClientApp to visualize which modules and functionality are now available within its module. To make matters worse, Library also imports ModuleA, which defines a foo/0 function that conflicts with a local function defined in ClientApp:
defmodule ModuleA do
  def foo do
    "From Module A"
  end
end
defmodule Library do
  defmacro __using__(_opts) do
    quote do
      import Library
      import ModuleA  # <= propagating dependencies!
    end
  end

  def from_lib do
    "From Library"
  end
end
defmodule ClientApp do
  use Library

  def foo do
    "Local function from client app"
  end

  def from_client_app do
    from_lib() <> " - " <> foo()
  end
end
When we try to compile ClientApp, Elixir detects the conflict and throws the following error:
error: imported ModuleA.foo/0 conflicts with local function
  └ client_app.ex:4:
Refactoring
To remove this anti-pattern, we recommend library authors avoid providing __using__/1 callbacks whenever it can be replaced by alias/1 or import/1 directives. In the following code, we assume use Library is no longer available and ClientApp was refactored in this way, and with that, the code is clearer and the conflict as previously shown no longer exists:
defmodule ClientApp do
  import Library

  def foo do
    "Local function from client app"
  end

  def from_client_app do
    from_lib() <> " - " <> foo()
  end
end
iex> ClientApp.from_client_app()
"From Library - Local function from client app"
Additional remarks
In situations where you need to do more than importing and aliasing modules, providing use MyModule may be necessary, as it provides a common extension point within the Elixir ecosystem.
Therefore, to provide guidance and clarity, we recommend library authors to include an admonition block in their @moduledoc that explains how use MyModule impacts the developer's code. As an example, the GenServer documentation outlines:
use GenServer
When you use GenServer, the GenServer module will
set @behaviour GenServer and define a child_spec/1
function, so your module can be used as a child
in a supervision tree.

Think of this summary as a "Nutrition facts label" for code generation. Make sure to only list changes made to the public API of the module. For example, if use Library sets an internal attribute called @_some_module_info and this attribute is never meant to be public, avoid documenting it in the nutrition facts.
For convenience, the markup notation to generate the admonition block above is this:
> #### `use GenServer` {: .info}
>
> When you `use GenServer`, the `GenServer` module will
> set `@behaviour GenServer` and define a `child_spec/1`
> function, so your module can be used as a child
> in a supervision tree.

Quote and unquote
    

This guide aims to introduce the meta-programming techniques available in Elixir. The ability to represent an Elixir program by its own data structures is at the heart of meta-programming. This chapter starts by exploring those structures and the associated quote/2 and unquote/1 constructs, so we can take a look at macros in the next guide, and finally build our own domain specific language.

  
    
  
  Quoting


The building block of an Elixir program is a tuple with three elements. For example, the function call sum(1, 2, 3) is represented internally as:
{:sum, [], [1, 2, 3]}
You can get the representation of any expression by using the quote/2 macro:
iex> quote do: sum(1, 2, 3)
{:sum, [], [1, 2, 3]}
The first element is the function name, the second is a keyword list containing metadata, and the third is the arguments list.
Operators are also represented as such tuples:
iex> quote do: 1 + 2
{:+, [context: Elixir, import: Kernel], [1, 2]}
Even a map is represented as a call to %{}:
iex> quote do: %{1 => 2}
{:%{}, [], [{1, 2}]}
Variables are represented using such triplets, with the difference that the last element is an atom, instead of a list:
iex> quote do: x
{:x, [], Elixir}
When quoting more complex expressions, we can see that the code is represented in such tuples, which are often nested inside each other in a structure resembling a tree. Many languages would call such representations an Abstract Syntax Tree (AST). Elixir calls them quoted expressions:
iex> quote do: sum(1, 2 + 3, 4)
{:sum, [], [1, {:+, [context: Elixir, import: Kernel], [2, 3]}, 4]}
Sometimes, when working with quoted expressions, it may be useful to get the textual code representation back. This can be done with Macro.to_string/1:
iex> Macro.to_string(quote do: sum(1, 2 + 3, 4))
"sum(1, 2 + 3, 4)"
In general, the tuples above are structured according to the following format:
{atom | tuple, list, list | atom}
	The first element is an atom or another tuple in the same representation;
	The second element is a keyword list containing metadata, like numbers and contexts;
	The third element is either a list of arguments for the function call or an atom. When this element is an atom, it means the tuple represents a variable.

Besides the tuple defined above, there are five Elixir literals that, when quoted, return themselves (and not a tuple). They are:
:sum         #=> Atoms
1.0          #=> Numbers
[1, 2]       #=> Lists
"strings"    #=> Strings
{key, value} #=> Tuples with two elements
Most Elixir code has a straight-forward translation to its underlying quoted expression. We recommend you try out different code samples and see what the results are. For example, what does String.upcase("foo") expand to? We have also learned that if(true, do: :this, else: :that) is the same as if true do :this else :that end. How does this affirmation hold with quoted expressions?

  
    
  
  Unquoting


Quoting is about retrieving the inner representation of some particular chunk of code. However, sometimes it may be necessary to inject some other particular chunk of code inside the representation we want to retrieve.
For example, imagine you have a variable called number which contains the number you want to inject inside a quoted expression.
iex> number = 13
iex> Macro.to_string(quote do: 11 + number)
"11 + number"
That's not what we wanted, since the value of the number variable has not been injected and number has been quoted in the expression. In order to inject the value of the number variable, unquote/1 has to be used inside the quoted representation:
iex> number = 13
iex> Macro.to_string(quote do: 11 + unquote(number))
"11 + 13"
unquote/1 can even be used to inject function names:
iex> fun = :hello
iex> Macro.to_string(quote do: unquote(fun)(:world))
"hello(:world)"
In some cases, it may be necessary to inject many values inside a list. For example, imagine you have a list containing [1, 2, 6], and we want to inject [3, 4, 5] into it. Using unquote/1 won't yield the desired result:
iex> inner = [3, 4, 5]
iex> Macro.to_string(quote do: [1, 2, unquote(inner), 6])
"[1, 2, [3, 4, 5], 6]"
That's when unquote_splicing/1 comes in handy:
iex> inner = [3, 4, 5]
iex> Macro.to_string(quote do: [1, 2, unquote_splicing(inner), 6])
"[1, 2, 3, 4, 5, 6]"
Unquoting is very useful when working with macros. When writing macros, developers are able to receive code chunks and inject them inside other code chunks, which can be used to transform code or write code that generates code during compilation.

  
    
  
  Escaping


As we saw at the beginning of this chapter, only some values are valid quoted expressions in Elixir. For example, a map is not a valid quoted expression. Neither is a tuple with four elements. However, such values can be expressed as a quoted expression:
iex> quote do: %{1 => 2}
{:%{}, [], [{1, 2}]}
In some cases, you may need to inject such values into quoted expressions. To do that, we need to first escape those values into quoted expressions with the help of Macro.escape/1:
iex> map = %{hello: :world}
iex> Macro.escape(map)
{:%{}, [], [hello: :world]}
Macros receive quoted expressions and must return quoted expressions. However, sometimes during the execution of a macro, you may need to work with values and making a distinction between values and quoted expressions will be required.
In other words, it is important to make a distinction between a regular Elixir value (like a list, a map, a process, a reference, and so on) and a quoted expression. Some values, such as integers, atoms, and strings, have a quoted expression equal to the value itself. Other values, like maps, need to be explicitly converted. Finally, values like functions and references cannot be converted to a quoted expression at all.
When working with macros and code that generates code, check out the documentation for the Macro module, which contains many functions to work with Elixir's AST.
In this introduction, we have laid the groundwork to finally write our first macro. You can check that out in the next guide.

Macros
    

Even though Elixir attempts its best to provide a safe environment for macros, most of the responsibility of writing clean code with macros falls on developers. Macros are harder to write than ordinary Elixir functions, and it's considered to be bad style to use them when they're not necessary. Write macros responsibly.
Elixir already provides mechanisms to write your everyday code in a simple and readable fashion by using its data structures and functions. Macros should only be used as a last resort. Remember that explicit is better than implicit. Clear code is better than concise code.

  
    
  
  Our first macro


Macros in Elixir are defined via defmacro/2.
For this guide, we will be using files instead of running code samples in IEx. That's because the code samples will span multiple lines of code and typing them all in IEx can be counter-productive. You should be able to run the code samples by saving them into a macros.exs file and running it with elixir macros.exs or iex macros.exs.

In order to better understand how macros work, let's create a new module where we are going to implement unless (which does the opposite of if/2), as a macro and as a function:
defmodule Unless do
  def fun_unless(clause, do: expression) do
    if(!clause, do: expression)
  end

  defmacro macro_unless(clause, do: expression) do
    quote do
      if(!unquote(clause), do: unquote(expression))
    end
  end
end
The function receives the arguments and passes them to if/2. However, as we learned in the previous guide, the macro will receive quoted expressions, inject them into the quote, and finally return another quoted expression.
Let's start iex with the module above:
$ iex macros.exs

and play with those definitions:
iex> require Unless
iex> Unless.macro_unless(true, do: IO.puts "this should never be printed")
nil
iex> Unless.fun_unless(true, do: IO.puts "this should never be printed")
"this should never be printed"
nil
In our macro implementation, the sentence was not printed, although it was printed in our function implementation. That's because the arguments to a function call are evaluated before calling the function. However, macros do not evaluate their arguments. Instead, they receive the arguments as quoted expressions which are then transformed into other quoted expressions. In this case, we have rewritten our unless macro to become an if/2 behind the scenes.
In other words, when invoked as:
Unless.macro_unless(true, do: IO.puts "this should never be printed")
Our macro_unless macro received the following:
macro_unless(true, [do: {{:., [], [{:__aliases__, [alias: false], [:IO]}, :puts]}, [], ["this should never be printed"]}])
and it then returned a quoted expression as follows:
{:if, [],
 [{:!, [], [true]},
  [do: {{:., [],
     [{:__aliases__,
       [], [:IO]},
      :puts]}, [], ["this should never be printed"]}]]}
We can actually verify that this is the case by using Macro.expand_once/2:
iex> expr = quote do: Unless.macro_unless(true, do: IO.puts("this should never be printed"))
iex> res  = Macro.expand_once(expr, __ENV__)
iex> IO.puts(Macro.to_string(res))
if(!true) do
  IO.puts("this should never be printed")
end
:ok
Macro.expand_once/2 receives a quoted expression and expands it according to the current environment. In this case, it expanded/invoked the Unless.macro_unless/2 macro and returned its result. We then proceeded to convert the returned quoted expression to a string and print it (we will talk about __ENV__ later in this chapter).
That's what macros are all about. They are about receiving quoted expressions and transforming them into something else. In fact, unless/2 in Elixir is implemented as a macro:
defmacro unless(clause, do: expression) do
  quote do
    if(!unquote(clause), do: unquote(expression))
  end
end
Constructs such as unless/2, defmacro/2, def/2, defprotocol/2, and many others used throughout the Elixir standard library are written in pure Elixir, often as a macro. This means that the constructs being used to build the language can be used by developers to extend the language to the domains they are working on.
We can define any function and macro we want, including ones that override the built-in definitions provided by Elixir. The only exceptions are Elixir special forms which are not implemented in Elixir and therefore cannot be overridden. The full list of special forms is available in Kernel.SpecialForms.

  
    
  
  Macro hygiene


Elixir macros have "late resolution". This guarantees that a variable defined inside a quote won't conflict with a variable defined in the context where that macro is expanded. For example:
defmodule Hygiene do
  defmacro no_interference do
    quote do: a = 1
  end
end

defmodule HygieneTest do
  def go do
    require Hygiene
    a = 13
    Hygiene.no_interference()
    a
  end
end

HygieneTest.go()
# => 13
In the example above, even though the macro injects a = 1, it does not affect the variable a defined by the go/0 function. If a macro wants to explicitly affect the context, it can use var!/1:
defmodule Hygiene do
  defmacro interference do
    quote do: var!(a) = 1
  end
end

defmodule HygieneTest do
  def go do
    require Hygiene
    a = 13
    Hygiene.interference()
    a
  end
end

HygieneTest.go()
# => 1
The code above will work but issue a warning: variable "a" is unused. The macro is overriding the original value and the original value is never used.
Variable hygiene only works because Elixir annotates variables with their context. For example, a variable x defined on line 3 of a module would be represented as:
{:x, [line: 3], nil}
However, a quoted variable would be represented as:
defmodule Sample do
  def quoted do
    quote do: x
  end
end

Sample.quoted() #=> {:x, [line: 3], Sample}
Notice that the third element in the quoted variable is the atom Sample, instead of nil, which marks the variable as coming from the Sample module. Therefore, Elixir considers these two variables as coming from different contexts and handles them accordingly.
Elixir provides similar mechanisms for imports and aliases too. This guarantees that a macro will behave as specified by its source module rather than conflicting with the target module where the macro is expanded. Hygiene can be bypassed under specific situations by using macros like var!/2 and alias!/1, although one must be careful when using those as they directly change the user environment.
Sometimes variable names might be dynamically created. In such cases, Macro.var/2 can be used to define new variables:
defmodule Sample do
  defmacro initialize_to_char_count(variables) do
    Enum.map(variables, fn name ->
      var = Macro.var(name, nil)
      length = name |> Atom.to_string() |> String.length()

      quote do
        unquote(var) = unquote(length)
      end
    end)
  end

  def run do
    initialize_to_char_count([:red, :green, :yellow])
    [red, green, yellow]
  end
end

> Sample.run() #=> [3, 5, 6]
Take note of the second argument to Macro.var/2. This is the context being used and will determine hygiene as described in the next section. Check out also Macro.unique_var/2, for cases when you need to generate variables with unique names.

  
    
  
  The environment


When calling Macro.expand_once/2 earlier in this chapter, we used the special form __ENV__/0.
__ENV__/0 returns a Macro.Env struct which contains useful information about the compilation environment, including the current module, file, and line, all variables defined in the current scope, as well as imports, requires, and more:
iex> __ENV__.module
nil
iex> __ENV__.file
"iex"
iex> __ENV__.requires
[IEx.Helpers, Kernel, Kernel.Typespec]
iex> require Integer
nil
iex> __ENV__.requires
[IEx.Helpers, Integer, Kernel, Kernel.Typespec]
Many of the functions in the Macro module expect a Macro.Env environment. You can read more about these functions in Macro and learn more about the compilation environment in the Macro.Env.

  
    
  
  Private macros


Elixir also supports private macros via defmacrop. Like private functions, these macros are only available inside the module that defines them, and only at compilation time.
It is important that a macro is defined before its usage. Failing to define a macro before its invocation will raise an error at runtime, since the macro won't be expanded and will be translated to a function call:
iex> defmodule Sample do
...>  def four, do: two() + two()
...>  defmacrop two, do: 2
...> end
** (CompileError) iex:2: function two/0 undefined

  
    
  
  Write macros responsibly


Macros are a powerful construct and Elixir provides many mechanisms to ensure they are used responsibly.
	Macros are hygienic: by default, variables defined inside a macro are not going to affect the user code. Furthermore, function calls and aliases available in the macro context are not going to leak into the user context.

	Macros are lexical: it is impossible to inject code or macros globally. In order to use a macro, you need to explicitly require or import the module that defines the macro.

	Macros are explicit: it is impossible to run a macro without explicitly invoking it. For example, some languages allow developers to completely rewrite functions behind the scenes, often via parse transforms or via some reflection mechanisms. In Elixir, a macro must be explicitly invoked in the caller during compilation time.

	Macros' language is clear: many languages provide syntax shortcuts for quote and unquote. In Elixir, we preferred to have them explicitly spelled out, in order to clearly delimit the boundaries of a macro definition and its quoted expressions.


Even with such guarantees, the developer plays a big role when writing macros responsibly. If you are confident you need to resort to macros, remember that macros are not your API. Keep your macro definitions short, including their quoted contents. For example, instead of writing a macro like this:
defmodule MyModule do
  defmacro my_macro(a, b, c) do
    quote do
      do_this(unquote(a))
      # ...
      do_that(unquote(b))
      # ...
      and_that(unquote(c))
    end
  end
end
write:
defmodule MyModule do
  defmacro my_macro(a, b, c) do
    quote do
      # Keep what you need to do here to a minimum
      # and move everything else to a function
      MyModule.do_this_that_and_that(unquote(a), unquote(b), unquote(c))
    end
  end

  def do_this_that_and_that(a, b, c) do
    do_this(a)
    ...
    do_that(b)
    ...
    and_that(c)
  end
end
This makes your code clearer and easier to test and maintain, as you can invoke and test do_this_that_and_that/3 directly. It also helps you design an actual API for developers that do not want to rely on macros.
With this guide, we finish our introduction to macros. The next guide is a brief discussion on DSLs that shows how we can mix macros and module attributes to annotate and extend modules and functions.

Domain-Specific Languages (DSLs)
    

Domain-specific Languages (DSLs) are languages tailored to a specific application domain. You don't need macros in order to have a DSL: every data structure and every function you define in your module is part of your domain-specific language.
For example, imagine we want to implement a Validator module which provides a data validation domain-specific language. We could implement it using data structures, functions, or macros. Let's see what those different DSLs would look like:
# 1. Data structures
import Validator
validate user, name: [length: 1..100], email: [matches: ~r/@/]

# 2. Functions
import Validator
user
|> validate_length(:name, 1..100)
|> validate_matches(:email, ~r/@/)

# 3. Macros + modules
defmodule MyValidator do
  use Validator
  validate_length :name, 1..100
  validate_matches :email, ~r/@/
end

MyValidator.validate(user)
Of all the approaches above, the first is definitely the most flexible. If our domain rules can be encoded with data structures, they are by far the easiest to compose and implement, as Elixir's standard library is filled with functions for manipulating different data types.
The second approach uses function calls which better suits more complex APIs (for example, if you need to pass many options) and reads nicely in Elixir thanks to the pipe operator.
The third approach uses macros, and is by far the most complex. It will take more lines of code to implement, it is hard and expensive to test (compared to testing simple functions), and it limits how the user may use the library since all validations need to be defined inside a module.
To drive the point home, imagine you want to validate a certain attribute only if a given condition is met. We could easily achieve it with the first solution, by manipulating the data structure accordingly, or with the second solution by using conditionals (if/else) before invoking the function. However, it is impossible to do so with the macros approach unless its DSL is augmented.
In other words:
data > functions > macros
That said, there are still cases where using macros and modules to build domain-specific languages is useful. Since we have explored data structures and function definitions in the Getting Started guide, this chapter will explore how to use macros and module attributes to tackle more complex DSLs.

  
    
  
  Building our own test case


The goal in this chapter is to build a module named TestCase that allows us to write the following:
defmodule MyTest do
  use TestCase

  test "arithmetic operations" do
    4 = 2 + 2
  end

  test "list operations" do
    [1, 2, 3] = [1, 2] ++ [3]
  end
end

MyTest.run()
In the example above, by using TestCase, we can write tests using the test macro, which defines a function named run to automatically run all tests for us. Our prototype will rely on the match operator (=) as a mechanism to do assertions.

  
    
  
  The test macro


Let's start by creating a module that defines and imports the test macro when used:
defmodule TestCase do
  # Callback invoked by `use`.
  #
  # For now it returns a quoted expression that
  # imports the module itself into the user code.
  @doc false
  defmacro __using__(_opts) do
    quote do
      import TestCase
    end
  end

  @doc """
  Defines a test case with the given description.

  ## Examples

      test "arithmetic operations" do
        4 = 2 + 2
      end

  """
  defmacro test(description, do: block) do
    function_name = String.to_atom("test " <> description)
    quote do
      def unquote(function_name)(), do: unquote(block)
    end
  end
end
Assuming we defined TestCase in a file named tests.exs, we can open it up by running iex tests.exs and define our first tests:
iex> defmodule MyTest do
...>   use TestCase
...>
...>   test "hello" do
...>     "hello" = "world"
...>   end
...> end
For now, we don't have a mechanism to run tests, but we know that a function named test hello was defined behind the scenes. When we invoke it, it should fail:
iex> MyTest."test hello"()
** (MatchError) no match of right hand side value: "world"

  
    
  
  Storing information with attributes


In order to finish our TestCase implementation, we need to be able to access all defined test cases. One way of doing this is by retrieving the tests at runtime via __MODULE__.__info__(:functions), which returns a list of all functions in a given module. However, considering that we may want to store more information about each test besides the test name, a more flexible approach is required.
When discussing module attributes in earlier chapters, we mentioned how they can be used as temporary storage. That's exactly the property we will apply in this section.
In the __using__/1 implementation, we will initialize a module attribute named @tests to an empty list, then store the name of each defined test in this attribute so the tests can be invoked from the run function.
Here is the updated code for the TestCase module:
defmodule TestCase do
  @doc false
  defmacro __using__(_opts) do
    quote do
      import TestCase

      # Initialize @tests to an empty list
      @tests []

      # Invoke TestCase.__before_compile__/1 before the module is compiled
      @before_compile TestCase
    end
  end

  @doc """
  Defines a test case with the given description.

  ## Examples

      test "arithmetic operations" do
        4 = 2 + 2
      end

  """
  defmacro test(description, do: block) do
    function_name = String.to_atom("test " <> description)
    quote do
      # Prepend the newly defined test to the list of tests
      @tests [unquote(function_name) | @tests]
      def unquote(function_name)(), do: unquote(block)
    end
  end

  # This will be invoked right before the target module is compiled
  # giving us the perfect opportunity to inject the `run/0` function
  @doc false
  defmacro __before_compile__(_env) do
    quote do
      def run do
        Enum.each(@tests, fn name ->
          IO.puts("Running #{name}")
          apply(__MODULE__, name, [])
        end)
      end
    end
  end
end
By starting a new IEx session, we can now define our tests and run them:
iex> defmodule MyTest do
...>   use TestCase
...>
...>   test "hello" do
...>     "hello" = "world"
...>   end
...> end
iex> MyTest.run()
Running test hello
** (MatchError) no match of right hand side value: "world"
Although we have overlooked some details, this is the main idea behind creating domain-specific languages in Elixir via modules and macros. Macros enable us to return quoted expressions that are executed in the caller, which we can then use to transform code and store relevant information in the target module via module attributes. Finally, callbacks such as @before_compile allow us to inject code into the module when its definition is complete.
Besides @before_compile, there are other useful module attributes like @on_definition and @after_compile, which you can read more about in the docs for Module. You can also find useful information about macros and the compilation environment in the documentation for the Macro and Macro.Env.

Introduction to Mix
    

In this guide, we will build a complete Elixir application, with its own supervision tree, configuration, tests, and more.
The requirements for this guide are (see elixir -v):
	Elixir 1.15.0 onwards
	Erlang/OTP 24 onwards

The application works as a distributed key-value store. We are going to organize key-value pairs into buckets and distribute those buckets across multiple nodes. We will also build a simple client that allows us to connect to any of those nodes and send requests such as:
CREATE shopping
OK

PUT shopping milk 1
OK

PUT shopping eggs 3
OK

GET shopping milk
1
OK

DELETE shopping eggs
OK
In order to build our key-value application, we are going to use three main tools:
	OTP (Open Telecom Platform) is a set of libraries that ships with Erlang. Erlang developers use OTP to build robust, fault-tolerant applications. In this chapter we will explore how many aspects from OTP integrate with Elixir, including supervision trees, event managers and more;

	Mix is a build tool that ships with Elixir that provides tasks for creating, compiling, testing your application, managing its dependencies and much more;

	ExUnit is a test-unit based framework that ships with Elixir.


In this chapter, we will create our first project using Mix and explore different features in OTP, Mix, and ExUnit as we go.
Source code
The final code for the application built in this guide is in this repository and can be used as a reference.

Is this guide required reading?
This guide is not required reading in your Elixir journey. We'll explain.
As an Elixir developer, you will most likely use one of the many existing frameworks when writing your Elixir code. Phoenix covers web applications, Ecto communicates with databases, you can craft embedded software with Nerves, Nx powers machine learning and AI projects, Membrane assembles audio/video processing pipelines, Broadway handles data ingestion and processing, and many more. These frameworks handle the lower level details of concurrency, distribution, and fault-tolerance, so you, as a user, can focus on your own needs and demands.
On the other hand, if you want to learn the foundations these frameworks are built upon, and the abstractions that power the Elixir ecosystem, this guide will give you a tour through several important concepts.


  
    
  
  Our first project


When you install Elixir, besides getting the elixir, elixirc, and iex executables, you also get an executable Elixir script named mix.
Let's create our first project by invoking mix new from the command line. We'll pass the project path as the argument (kv, in this case). By default, the application name and module name will be retrieved from the path. So we tell Mix that our main module should be the all-uppercase KV, instead of the default, which would have been Kv:
$ mix new kv --module KV

Mix will create a directory named kv with a few files in it:
* creating README.md
* creating .formatter.exs
* creating .gitignore
* creating mix.exs
* creating lib
* creating lib/kv.ex
* creating test
* creating test/test_helper.exs
* creating test/kv_test.exs
Let's take a brief look at those generated files.
Executables in the PATH
Mix is an Elixir executable. This means that in order to run mix, you need to have both mix and elixir executables in your PATH. That's what happens when you install Elixir.


  
    
  
  Project compilation


A file named mix.exs was generated inside our new project folder (kv) and its main responsibility is to configure our project. Let's take a look at it:
defmodule KV.MixProject do
  use Mix.Project

  def project do
    [
      app: :kv,
      version: "0.1.0",
      elixir: "~> 1.11",
      start_permanent: Mix.env() == :prod,
      deps: deps()
    ]
  end

  # Run "mix help compile.app" to learn about applications
  def application do
    [
      extra_applications: [:logger]
    ]
  end

  # Run "mix help deps" to learn about dependencies
  defp deps do
    [
      # {:dep_from_hexpm, "~> 0.3.0"},
      # {:dep_from_git, git: "https://github.com/elixir-lang/my_dep.git", tag: "0.1.0"},
    ]
  end
end
Our mix.exs defines two public functions: project, which returns project configuration like the project name and version, and application, which is used to generate an application file.
There is also a private function named deps, which is invoked from the project function, that defines our project dependencies. Defining deps as a separate function is not required, but it helps keep the project configuration tidy.
Mix also generates a file at lib/kv.ex with a module containing exactly one function, called hello:
defmodule KV do
  @moduledoc """
  Documentation for KV.
  """

  @doc """
  Hello world.

  ## Examples

      iex> KV.hello()
      :world

  """
  def hello do
    :world
  end
end

This structure is enough to compile our project:
$ cd kv
$ mix compile

Will output:
Compiling 1 file (.ex)
Generated kv app
The lib/kv.ex file was compiled and an application manifest named kv.app was generated. All compilation artifacts are placed inside the _build directory using the options defined in the mix.exs file.
Once the project is compiled, you can start a iex session inside the project by running the command below. The -S mix is necessary to load the project in the interactive shell:
$ iex -S mix

We are going to work on this kv project, making modifications and trying out the latest changes from a iex session. While you may start a new session whenever there are changes to the project source code, you can also recompile the project from within iex with the recompile helper, like this:
iex> recompile()
Compiling 1 file (.ex)
:ok
iex> recompile()
:noop
If anything had to be compiled, you see some informative text, and get the :ok atom back, otherwise the function is silent, and returns :noop.

  
    
  
  Running tests


Mix also generated the appropriate structure for running our project tests. Mix projects usually follow the convention of having a <filename>_test.exs file in the test directory for each file in the lib directory. For this reason, we can already find a test/kv_test.exs corresponding to our lib/kv.ex file. It doesn't do much at this point:
defmodule KVTest do
  use ExUnit.Case
  doctest KV

  test "greets the world" do
    assert KV.hello() == :world
  end
end
It is important to note a couple of things:
	the test file is an Elixir script file (.exs). This is convenient because we don't need to compile test files before running them;

	we define a test module named KVTest, in which we use ExUnit.Case to inject the testing API;

	we use one of the imported macros, ExUnit.DocTest.doctest/1, to indicate that the KV module contains doctests (we will discuss those in a later chapter);

	we use the ExUnit.Case.test/2 macro to define a simple test;


Mix also generated a file named test/test_helper.exs which is responsible for setting up the test framework:
ExUnit.start()
This file will be required by Mix every time before we run our tests. We can run tests with:
$ mix test
Compiled lib/kv.ex
Generated kv app
Running ExUnit with seed: 540224, max_cases: 16
..

Finished in 0.04 seconds
1 doctest, 1 test, 0 failures

Notice that by running mix test, Mix has compiled the source files and generated the application manifest once again. This happens because Mix supports multiple environments, which we will discuss later in this chapter.
Furthermore, you can see that ExUnit prints a dot for each successful test and automatically randomizes tests too. Let's make the test fail on purpose and see what happens.
Change the assertion in test/kv_test.exs to the following:
assert KV.hello() == :oops
Now run mix test again (notice this time there will be no compilation):
  1) test greets the world (KVTest)
     test/kv_test.exs:5
     Assertion with == failed
     code:  assert KV.hello() == :oops
     left:  :world
     right: :oops
     stacktrace:
       test/kv_test.exs:6: (test)

.

Finished in 0.05 seconds
1 doctest, 1 test, 1 failure
For each failure, ExUnit prints a detailed report, containing the test name with the test case, the code that failed and the values for the left side and right side (RHS) of the == operator.
In the second line of the failure, right below the test name, there is the location where the test was defined. If you copy the test location in full, including the file and line number, and append it to mix test, Mix will load and run just that particular test:
$ mix test test/kv_test.exs:5

This shortcut will be extremely useful as we build our project, allowing us to quickly iterate by running a single test.
Finally, the stacktrace relates to the failure itself, giving information about the test and often the place the failure was generated from within the source files.

  
    
  
  Automatic code formatting


One of the files generated by mix new is the .formatter.exs. Elixir ships with a code formatter that is capable of automatically formatting our codebase according to a consistent style. The formatter is triggered with the mix format task. The generated .formatter.exs file configures which files should be formatted when mix format runs.
To give the formatter a try, change a file in the lib or test directories to include extra spaces or extra newlines, such as def hello do, and then run mix format.
Most editors provide built-in integration with the formatter, allowing a file to be formatted on save or via a chosen keybinding. If you are learning Elixir, editor integration gives you useful and quick feedback when learning the Elixir syntax.
For companies and teams, we recommend developers to run mix format --check-formatted on their continuous integration servers, ensuring all current and future code follows the standard.
You can learn more about the code formatter by checking the format task documentation or by reading the release announcement for Elixir v1.6, the first version to include the formatter.

  
    
  
  Environments


Mix provides the concept of "environments". They allow a developer to customize compilation and other options for specific scenarios. By default, Mix understands three environments:
	:dev — the one in which Mix tasks (like compile) run by default
	:test — used by mix test
	:prod — the one you will use to run your project in production

The environment applies only to the current project. As we will see in future chapters, any dependency you add to your project will by default run in the :prod environment.
Customization per environment can be done by accessing the Mix.env/0 in your mix.exs file, which returns the current environment as an atom. That's what we have used in the :start_permanent options:
def project do
  [
    ...,
    start_permanent: Mix.env() == :prod,
    ...
  ]
end
When true, the :start_permanent option starts your application in permanent mode, which means the Erlang VM will crash if your application's supervision tree shuts down. Notice we don't want this behavior in dev and test because it is useful to keep the VM instance running in those environments for troubleshooting purposes.
Mix will default to the :dev environment, except for the test task that will default to the :test environment. The environment can be changed via the MIX_ENV environment variable:
$ MIX_ENV=prod mix compile

Or on Windows:
> set "MIX_ENV=prod" && mix compile
Mix in production
Mix is a build tool and, as such, it is not expected to be available in production. Therefore, it is recommended to access Mix.env/0 only in configuration files and inside mix.exs, never in your application code (lib).


  
    
  
  Exploring


There is much more to Mix, and we will continue to explore it as we build our project. A general overview is available on the Mix documentation and you can always invoke the help task to list all available tasks:
$ mix help
$ mix help compile

Now let's move forward and add the first modules and functions to our application.

Simple state management with agents
    

In this chapter, we will learn how to keep and share state between multiple entities. If you have previous programming experience, you may think of globally shared variables, but the model we will learn here is quite different. The next chapters will generalize the concepts introduced here.
If you have skipped the Getting Started guide or read it long ago, be sure to re-read the Processes chapter. We will use it as a starting point.

  
    
  
  The trouble with (mutable) state


Elixir is an immutable language where nothing is shared by default. If we want to share information, which can be read and modified from multiple places, we have two main options in Elixir:
	Using processes and message passing
	ETS (Erlang Term Storage)

We covered processes in the Getting Started guide. ETS (Erlang Term Storage) is a new topic that we will explore in later chapters. When it comes to processes though, we rarely hand-roll our own, instead we use the abstractions available in Elixir and OTP:
	Agent — Simple wrappers around state.
	GenServer — "Generic servers" (processes) that encapsulate state, provide sync and async calls, support code reloading, and more.
	Task — Asynchronous units of computation that allow spawning a process and potentially retrieving its result at a later time.

We will explore these abstractions as we move forward. Keep in mind that they are all implemented on top of processes using the basic features provided by the VM, like send/2, receive/1, spawn/1 and Process.link/1.
Here, we will use agents, and create a module named KV.Bucket, responsible for storing our key-value entries in a way that allows them to be read and modified by other processes.

  
    
  
  Agents 101


Agents are simple wrappers around state. If all you want from a process is to keep state, agents are a great fit. Let's start a iex session inside the project with:
$ iex -S mix

And play a bit with agents:
iex> {:ok, agent} = Agent.start_link(fn -> [] end)
{:ok, #PID<0.57.0>}
iex> Agent.update(agent, fn list -> ["eggs" | list] end)
:ok
iex> Agent.get(agent, fn list -> list end)
["eggs"]
iex> Agent.stop(agent)
:ok
We started an agent with an initial state of an empty list. We updated the agent's state, adding our new item to the head of the list. The second argument of Agent.update/3 is a function that takes the agent's current state as input and returns its desired new state. Finally, we retrieved the whole list. The second argument of Agent.get/3 is a function that takes the state as input and returns the value that Agent.get/3 itself will return. Once we are done with the agent, we can call Agent.stop/3 to terminate the agent process.
The Agent.update/3 function accepts as a second argument any function that receives one argument and returns a value:
iex> {:ok, agent} = Agent.start_link(fn -> [] end)
{:ok, #PID<0.338.0>}
iex> Agent.update(agent, fn _list -> 123 end)
:ok
iex> Agent.update(agent, fn content -> %{a: content} end)
:ok
iex> Agent.update(agent, fn content -> [12 | [content]] end)
:ok
iex> Agent.update(agent, fn list -> [:nop | list] end)
:ok
iex> Agent.get(agent, fn content -> content end)
[:nop, 12, %{a: 123}]
As you can see, we can modify the agent state in any way we want. Therefore, we most likely don't want to access the Agent API throughout many different places in our code. Instead, we want to encapsulate all Agent-related functionality in a single module, which we will call KV.Bucket. Before we implement it, let's write some tests which will outline the API exposed by our module.
Create a file at test/kv/bucket_test.exs (remember the .exs extension) with the following:
defmodule KV.BucketTest do
  use ExUnit.Case, async: true

  test "stores values by key" do
    {:ok, bucket} = KV.Bucket.start_link([])
    assert KV.Bucket.get(bucket, "milk") == nil

    KV.Bucket.put(bucket, "milk", 3)
    assert KV.Bucket.get(bucket, "milk") == 3
  end
end
use ExUnit.Case is responsible for setting up our module for testing and imports many test-related functionality, such as the test/2 macro.
Our first test starts a new KV.Bucket by calling the start_link/1 and passing an empty list of options. Then we perform some get/2 and put/3 operations on it, asserting the result.
Also note the async: true option passed to ExUnit.Case. This option makes the test case run in parallel with other :async test cases by using multiple cores in our machine. This is extremely useful to speed up our test suite. However, :async must only be set if the test case does not rely on or change any global values. For example, if the test requires writing to the file system or access a database, keep it synchronous (omit the :async option) to avoid race conditions between tests.
Async or not, our new test should obviously fail, as none of the functionality is implemented in the module being tested:
** (UndefinedFunctionError) function KV.Bucket.start_link/1 is undefined (module KV.Bucket is not available)
In order to fix the failing test, let's create a file at lib/kv/bucket.ex with the contents below. Feel free to give a try at implementing the KV.Bucket module yourself using agents before peeking at the implementation below.
defmodule KV.Bucket do
  use Agent

  @doc """
  Starts a new bucket.
  """
  def start_link(_opts) do
    Agent.start_link(fn -> %{} end)
  end

  @doc """
  Gets a value from the `bucket` by `key`.
  """
  def get(bucket, key) do
    Agent.get(bucket, &Map.get(&1, key))
  end

  @doc """
  Puts the `value` for the given `key` in the `bucket`.
  """
  def put(bucket, key, value) do
    Agent.update(bucket, &Map.put(&1, key, value))
  end
end
The first step in our implementation is to call use Agent. Most of the functionality we will learn, such as GenServer and Supervisor, follow this pattern. For all of them, calling use generates a child_spec/1 function with default configuration, which will be handy when we start supervising processes in chapter 4.
Then we define a start_link/1 function, which will effectively start the agent. It is a convention to define a start_link/1 function that always accepts a list of options. We don't plan on using any options right now, but we might later on. We then proceed to call Agent.start_link/1, which receives an anonymous function that returns the Agent's initial state.
We are keeping a map inside the agent to store our keys and values. Getting and putting values on the map is done with the Agent API and the capture operator &, introduced in the Getting Started guide. The agent passes its state to the anonymous function via the &1 argument when Agent.get/2 and Agent.update/2 are called.
Now that the KV.Bucket module has been defined, our test should pass! You can try it yourself by running: mix test.

  
    
  
  Test setup with ExUnit callbacks


Before moving on and adding more features to KV.Bucket, let's talk about ExUnit callbacks. As you may expect, all KV.Bucket tests will require a bucket agent to be up and running. Luckily, ExUnit supports callbacks that allow us to skip such repetitive tasks.
Let's rewrite the test case to use callbacks:
defmodule KV.BucketTest do
  use ExUnit.Case, async: true

  setup do
    {:ok, bucket} = KV.Bucket.start_link([])
    %{bucket: bucket}
  end

  test "stores values by key", %{bucket: bucket} do
    assert KV.Bucket.get(bucket, "milk") == nil

    KV.Bucket.put(bucket, "milk", 3)
    assert KV.Bucket.get(bucket, "milk") == 3
  end
end
We have first defined a setup callback with the help of the setup/1 macro. The setup/1 macro defines a callback that is run before every test, in the same process as the test itself.
Note that we need a mechanism to pass the bucket PID from the callback to the test. We do so by using the test context. When we return %{bucket: bucket} from the callback, ExUnit will merge this map into the test context. Since the test context is a map itself, we can pattern match the bucket out of it, providing access to the bucket inside the test:
test "stores values by key", %{bucket: bucket} do
  # `bucket` is now the bucket from the setup block
end
You can read more about ExUnit cases in the ExUnit.Case module documentation and more about callbacks in ExUnit.Callbacks.

  
    
  
  Other agent actions


Besides getting a value and updating the agent state, agents allow us to get a value and update the agent state in one function call via Agent.get_and_update/2. Let's implement a KV.Bucket.delete/2 function that deletes a key from the bucket, returning its current value:
@doc """
Deletes `key` from `bucket`.

Returns the current value of `key`, if `key` exists.
"""
def delete(bucket, key) do
  Agent.get_and_update(bucket, &Map.pop(&1, key))
end
Now it is your turn to write a test for the functionality above! Also, be sure to explore the documentation for the Agent module to learn more about them.

  
    
  
  Client/server in agents


Before we move on to the next chapter, let's discuss the client/server dichotomy in agents. Let's expand the delete/2 function we have just implemented:
def delete(bucket, key) do
  Agent.get_and_update(bucket, fn dict ->
    Map.pop(dict, key)
  end)
end
Everything that is inside the function we passed to the agent happens in the agent process. In this case, since the agent process is the one receiving and responding to our messages, we say the agent process is the server. Everything outside the function is happening in the client.
This distinction is important. If there are expensive actions to be done, you must consider if it will be better to perform these actions on the client or on the server. For example:
def delete(bucket, key) do
  Process.sleep(1000) # puts client to sleep
  Agent.get_and_update(bucket, fn dict ->
    Process.sleep(1000) # puts server to sleep
    Map.pop(dict, key)
  end)
end
When a long action is performed on the server, all other requests to that particular server will wait until the action is done, which may cause some clients to timeout.
In the next chapter, we will explore GenServers, where the segregation between clients and servers is made more apparent.

Client-server communication with GenServer
    

In the previous chapter, we used agents to represent our buckets. In the introduction to mix, we specified we would like to name each bucket so we can do the following:
CREATE shopping
OK

PUT shopping milk 1
OK

GET shopping milk
1
OK
In the session above we interacted with the "shopping" bucket.
Since agents are processes, each bucket has a process identifier (PID), but buckets do not have a name. Back in the Process chapter, we have learned that we can register processes in Elixir by giving them atom names:
iex> Agent.start_link(fn -> %{} end, name: :shopping)
{:ok, #PID<0.43.0>}
iex> KV.Bucket.put(:shopping, "milk", 1)
:ok
iex> KV.Bucket.get(:shopping, "milk")
1
However, naming dynamic processes with atoms is a terrible idea! If we use atoms, we would need to convert the bucket name (often received from an external client) to atoms, and we should never convert user input to atoms. This is because atoms are not garbage collected. Once an atom is created, it is never reclaimed. Generating atoms from user input would mean the user can inject enough different names to exhaust our system memory!
In practice, it is more likely you will reach the Erlang VM limit for the maximum number of atoms before you run out of memory, which will bring your system down regardless.
Instead of abusing the built-in name facility, we will create our own process registry that associates the bucket name to the bucket process.
The registry needs to guarantee that it is always up to date. For example, if one of the bucket processes crashes due to a bug, the registry must notice this change and avoid serving stale entries. In Elixir, we say the registry needs to monitor each bucket. Because our registry needs to be able to receive and handle ad-hoc messages from the system, the Agent API is not enough.
We will use a GenServer to create a registry process that can monitor the bucket processes. GenServer provides industrial strength functionality for building servers in both Elixir and OTP.
Please read the GenServer module documentation for an overview if you haven't yet. Once you do so, we are ready to proceed.

  
    
  
  GenServer callbacks


A GenServer is a process that invokes a limited set of functions under specific conditions. When we used a Agent, we would keep both the client code and the server code side by side, like this:
def put(bucket, key, value) do
  Agent.update(bucket, &Map.put(&1, key, value))
end
Let's break that code apart a bit:
def put(bucket, key, value) do
  # Here is the client code
  Agent.update(bucket, fn state ->
    # Here is the server code
    Map.put(state, key, value)
  end)
  # Back to the client code
end
In the code above, we have a process, which we call "the client" sending a request to an agent, "the server". The request contains an anonymous function, which must be executed by the server.
In a GenServer, the code above would be two separate functions, roughly like this:
def put(bucket, key, value) do
  # Send the server a :put "instruction"
  GenServer.call(bucket, {:put, key, value})
end

# Server callback

def handle_call({:put, key, value}, _from, state) do
  {:reply, :ok, Map.put(state, key, value)}
end
There is quite a bit more ceremony in the GenServer code but, as we will see, it brings some benefits too.
For now, we will write only the server callbacks for our bucket registering logic, without providing a proper API, which we will do later.
Create a new file at lib/kv/registry.ex with the following contents:
defmodule KV.Registry do
  use GenServer

  ## Missing Client API - will add this later

  ## Defining GenServer Callbacks

  @impl true
  def init(:ok) do
    {:ok, %{}}
  end

  @impl true
  def handle_call({:lookup, name}, _from, names) do
    {:reply, Map.fetch(names, name), names}
  end

  @impl true
  def handle_cast({:create, name}, names) do
    if Map.has_key?(names, name) do
      {:noreply, names}
    else
      {:ok, bucket} = KV.Bucket.start_link([])
      {:noreply, Map.put(names, name, bucket)}
    end
  end
end
There are two types of requests you can send to a GenServer: calls and casts. Calls are synchronous and the server must send a response back to such requests. While the server computes the response, the client is waiting. Casts are asynchronous: the server won't send a response back and therefore the client won't wait for one. Both requests are messages sent to the server, and will be handled in sequence. In the above implementation, we pattern-match on the :create messages, to be handled as cast, and on the :lookup messages, to be handled as call.
In order to invoke the callbacks above, we need to go through the corresponding GenServer functions. Let's start a registry, create a named bucket, and then look it up:
iex> {:ok, registry} = GenServer.start_link(KV.Registry, :ok)
{:ok, #PID<0.136.0>}
iex> GenServer.cast(registry, {:create, "shopping"})
:ok
iex> {:ok, bucket} = GenServer.call(registry, {:lookup, "shopping"})
{:ok, #PID<0.174.0>}
Our KV.Registry process received a cast with {:create, "shopping"} and a call with {:lookup, "shopping"}, in this sequence. GenServer.cast will immediately return, as soon as the message is sent to the registry. The GenServer.call on the other hand, is where we would be waiting for an answer, provided by the above KV.Registry.handle_call callback.
You may also have noticed that we have added @impl true before each callback. The @impl true informs the compiler that our intention for the subsequent function definition is to define a callback. If by any chance we make a mistake in the function name or in the number of arguments, like we define a handle_call/2, the compiler would warn us there isn't any handle_call/2 to define, and would give us the complete list of known callbacks for the GenServer module.
This is all good and well, but we still want to offer our users an API that allows us to hide our implementation details.

  
    
  
  The Client API


A GenServer is implemented in two parts: the client API and the server callbacks. You can either combine both parts into a single module or you can separate them into a client module and a server module. The client is any process that invokes the client function. The server is always the process identifier or process name that we will explicitly pass as argument to the client API. Here we'll use a single module for both the server callbacks and the client API.
Edit the file at lib/kv/registry.ex, filling in the blanks for the client API:
  ## Client API

  @doc """
  Starts the registry.
  """
  def start_link(opts) do
    GenServer.start_link(__MODULE__, :ok, opts)
  end

  @doc """
  Looks up the bucket pid for `name` stored in `server`.

  Returns `{:ok, pid}` if the bucket exists, `:error` otherwise.
  """
  def lookup(server, name) do
    GenServer.call(server, {:lookup, name})
  end

  @doc """
  Ensures there is a bucket associated with the given `name` in `server`.
  """
  def create(server, name) do
    GenServer.cast(server, {:create, name})
  end
The first function is start_link/1, which starts a new GenServer passing a list of options. start_link/1 calls out to GenServer.start_link/3, which takes three arguments:
	The module where the server callbacks are implemented, in this case __MODULE__ (meaning the current module)

	The initialization arguments, in this case the atom :ok

	A list of options which can be used to specify things like the name of the server. For now, we forward the list of options that we receive on start_link/1 to GenServer.start_link/3


The next two functions, lookup/2 and create/2, are responsible for sending these requests to the server.  In this case, we have used {:lookup, name} and {:create, name} respectively.  Requests are often specified as tuples, like this, in order to provide more than one "argument" in that first argument slot. It's common to specify the action being requested as the first element of a tuple, and arguments for that action in the remaining elements. Note that the requests must match the first argument to handle_call/3 or handle_cast/2.
That's it for the client API. On the server side, we can implement a variety of callbacks to guarantee the server initialization, termination, and handling of requests. Those callbacks are optional and for now, we have only implemented the ones we care about. Let's recap.
The first is the init/1 callback, that receives the second argument given to GenServer.start_link/3 and returns {:ok, state}, where state is a new map. We can already notice how the GenServer API makes the client/server segregation more apparent. start_link/3 happens in the client, while init/1 is the respective callback that runs on the server.
For call/2 requests, we implement a handle_call/3 callback that receives the request, the process from which we received the request (_from), and the current server state (names). The handle_call/3 callback returns a tuple in the format {:reply, reply, new_state}. The first element of the tuple, :reply, indicates that the server should send a reply back to the client. The second element, reply, is what will be sent to the client while the third, new_state is the new server state.
For cast/2 requests, we implement a handle_cast/2 callback that receives the request and the current server state (names). The handle_cast/2 callback returns a tuple in the format {:noreply, new_state}. Note that in a real application we would have probably implemented the callback for :create with a synchronous call instead of an asynchronous cast. We are doing it this way to illustrate how to implement a cast callback.
There are other tuple formats both handle_call/3 and handle_cast/2 callbacks may return. There are other callbacks like terminate/2 and code_change/3 that we could implement. You are welcome to explore the full GenServer documentation to learn more about those.
For now, let's write some tests to guarantee our GenServer works as expected.

  
    
  
  Testing a GenServer


Testing a GenServer is not much different from testing an agent. We will spawn the server on a setup callback and use it throughout our tests. Create a file at test/kv/registry_test.exs with the following:
defmodule KV.RegistryTest do
  use ExUnit.Case, async: true

  setup do
    registry = start_supervised!(KV.Registry)
    %{registry: registry}
  end

  test "spawns buckets", %{registry: registry} do
    assert KV.Registry.lookup(registry, "shopping") == :error

    KV.Registry.create(registry, "shopping")
    assert {:ok, bucket} = KV.Registry.lookup(registry, "shopping")

    KV.Bucket.put(bucket, "milk", 1)
    assert KV.Bucket.get(bucket, "milk") == 1
  end
end
Our test case first asserts there are no buckets in our registry, creates a named bucket, looks it up, and asserts it behaves as a bucket.
There is one important difference between the setup block we wrote for KV.Registry and the one we wrote for KV.Bucket. Instead of starting the registry by hand by calling KV.Registry.start_link/1, we instead called the ExUnit.Callbacks.start_supervised!/2 function, passing the KV.Registry module.
The start_supervised! function was injected into our test module by use ExUnit.Case. It does the job of starting the KV.Registry process, by calling its start_link/1 function. The advantage of using start_supervised! is that ExUnit will guarantee that the registry process will be shutdown before the next test starts. In other words, it helps guarantee that the state of one test is not going to interfere with the next one in case they depend on shared resources.
When starting processes during your tests, we should always prefer to use start_supervised!. We recommend you to change the setup block in bucket_test.exs to use start_supervised! too.
Run the tests and they should all pass!

  
    
  
  The need for monitoring


Everything we have done so far could have been implemented with a Agent. In this section, we will see one of many things that we can achieve with a GenServer that is not possible with an Agent.
Let's start with a test that describes how we want the registry to behave if a bucket stops or crashes:
test "removes buckets on exit", %{registry: registry} do
  KV.Registry.create(registry, "shopping")
  {:ok, bucket} = KV.Registry.lookup(registry, "shopping")
  Agent.stop(bucket)
  assert KV.Registry.lookup(registry, "shopping") == :error
end
The test above will fail on the last assertion as the bucket name remains in the registry even after we stop the bucket process.
In order to fix this bug, we need the registry to monitor every bucket it spawns. Once we set up a monitor, the registry will receive a notification every time a bucket process exits, allowing us to clean the registry up.
Let's first play with monitors by starting a new console with iex -S mix:
iex> {:ok, pid} = KV.Bucket.start_link([])
{:ok, #PID<0.66.0>}
iex> Process.monitor(pid)
#Reference<0.0.0.551>
iex> Agent.stop(pid)
:ok
iex> flush()
{:DOWN, #Reference<0.0.0.551>, :process, #PID<0.66.0>, :normal}
Note Process.monitor(pid) returns a unique reference that allows us to match upcoming messages to that monitoring reference. After we stop the agent, we can flush/0 all messages and notice a :DOWN message arrived, with the exact reference returned by monitor, notifying that the bucket process exited with reason :normal.
Let's reimplement the server callbacks to fix the bug and make the test pass. First, we will modify the GenServer state to two dictionaries: one that contains name -> pid and another that holds ref -> name. Then we need to monitor the buckets on handle_cast/2 as well as implement a handle_info/2 callback to handle the monitoring messages. The full server callbacks implementation is shown below:
## Server callbacks

@impl true
def init(:ok) do
  names = %{}
  refs = %{}
  {:ok, {names, refs}}
end

@impl true
def handle_call({:lookup, name}, _from, state) do
  {names, _} = state
  {:reply, Map.fetch(names, name), state}
end

@impl true
def handle_cast({:create, name}, {names, refs}) do
  if Map.has_key?(names, name) do
    {:noreply, {names, refs}}
  else
    {:ok, bucket} = KV.Bucket.start_link([])
    ref = Process.monitor(bucket)
    refs = Map.put(refs, ref, name)
    names = Map.put(names, name, bucket)
    {:noreply, {names, refs}}
  end
end

@impl true
def handle_info({:DOWN, ref, :process, _pid, _reason}, {names, refs}) do
  {name, refs} = Map.pop(refs, ref)
  names = Map.delete(names, name)
  {:noreply, {names, refs}}
end

@impl true
def handle_info(msg, state) do
  require Logger
  Logger.debug("Unexpected message in KV.Registry: #{inspect(msg)}")
  {:noreply, state}
end
Observe that we were able to considerably change the server implementation without changing any of the client API. That's one of the benefits of explicitly segregating the server and the client.
Finally, different from the other callbacks, we have defined a "catch-all" clause for handle_info/2 that discards and logs any unknown message. To understand why, let's move on to the next section.

  
    
  
  call, cast or info?


So far we have used three callbacks: handle_call/3, handle_cast/2 and handle_info/2. Here is what we should consider when deciding when to use each:
	handle_call/3 must be used for synchronous requests. This should be the default choice as waiting for the server reply is a useful back-pressure mechanism.

	handle_cast/2 must be used for asynchronous requests, when you don't care about a reply. A cast does not guarantee the server has received the message and, for this reason, should be used sparingly. For example, the create/2 function we have defined in this chapter should have used call/2. We have used cast/2 for didactic purposes.

	handle_info/2 must be used for all other messages a server may receive that are not sent via GenServer.call/2 or GenServer.cast/2, including regular messages sent with send/2. The monitoring :DOWN messages are an example of this.


Since any message, including the ones sent via send/2, go to handle_info/2, there is a chance that unexpected messages will arrive to the server. Therefore, if we don't define the catch-all clause, those messages could cause our registry to crash, because no clause would match. We don't need to worry about such cases for handle_call/3 and handle_cast/2 though. Calls and casts are only done via the GenServer API, so an unknown message is quite likely a developer mistake.
To help developers remember the differences between call, cast and info, the supported return values and more, we have a tiny GenServer cheat sheet.

  
    
  
  Monitors or links?


We have previously learned about links in the Process chapter. Now, with the registry complete, you may be wondering: when should we use monitors and when should we use links?
Links are bi-directional. If you link two processes and one of them crashes, the other side will crash too (unless it is trapping exits). A monitor is uni-directional: only the monitoring process will receive notifications about the monitored one. In other words: use links when you want linked crashes, and monitors when you just want to be informed of crashes, exits, and so on.
Returning to our handle_cast/2 implementation, you can see the registry is both linking and monitoring the buckets:
{:ok, bucket} = KV.Bucket.start_link([])
ref = Process.monitor(bucket)
This is a bad idea, as we don't want the registry to crash when a bucket crashes. The proper fix is to actually not link the bucket to the registry. Instead, we will link each bucket to a special type of process called Supervisors, which are explicitly designed to handle failures and crashes. We will learn more about them in the next chapter.

In the previous chapter about GenServer, we implemented KV.Registry to manage buckets. At some point, we started monitoring buckets so we were able to take action whenever a KV.Bucket crashed. Although the change was relatively small, it introduced a question which is frequently asked by Elixir developers: what happens when something fails?
Before we added monitoring, if a bucket crashed, the registry would forever point to a bucket that no longer exists. If a user tried to read or write to the crashed bucket, it would fail. Any attempt at creating a new bucket with the same name would just return the PID of the crashed bucket. In other words, that registry entry for that bucket would forever be in a bad state. Once we added monitoring, the registry automatically removes the entry for the crashed bucket. Trying to lookup the crashed bucket now (correctly) says the bucket does not exist and a user of the system can successfully create a new one if desired.
In practice, we are not expecting the processes working as buckets to fail. But, if it does happen, for whatever reason, we can rest assured that our system will continue to work as intended.
If you have prior programming experience, you may be wondering: "could we just guarantee the bucket does not crash in the first place?". As we will see, Elixir developers tend to refer to those practices as "defensive programming". That's because a live production system has dozens of different reasons why something can go wrong. The disk can fail, memory can be corrupted, bugs, the network may stop working for a second, etc. If we were to write software that attempted to protect or circumvent all of those errors, we would spend more time handling failures than writing our own software!
Therefore, an Elixir developer prefers to "let it crash" or "fail fast". And one of the most common ways we can recover from a failure is by restarting whatever part of the system crashed.
For example, imagine your computer, router, printer, or whatever device is not working properly. How often do you fix it by restarting it? Once we restart the device, we reset the device back to its initial state, which is well-tested and guaranteed to work. In Elixir, we apply this same approach to software: whenever a process crashes, we start a new process to perform the same job as the crashed process.
In Elixir, this is done by a Supervisor. A Supervisor is a process that supervises other processes and restarts them whenever they crash. To do so, Supervisors manage the whole life cycle of any supervised processes, including startup and shutdown.
In this chapter, we will learn how to put those concepts into practice by supervising the KV.Registry process. After all, if something goes wrong with the registry, the whole registry is lost and no bucket could ever be found! To address this, we will define a KV.Supervisor module that guarantees that our KV.Registry is up and running at any given moment.
At the end of the chapter, we will also talk about Applications. As we will see, Mix has been packaging all of our code into an application, and we will learn how to customize our application to guarantee that our Supervisor and the Registry are up and running whenever our system starts.

  
    
  
  Our first supervisor


A supervisor is a process which supervises other processes, which we refer to as child processes. The act of supervising a process includes three distinct responsibilities. The first one is to start child processes. Once a child process is running, the supervisor may restart a child process, either because it terminated abnormally or because a certain condition was reached. For example, a supervisor may restart all children if any child dies. Finally, a supervisor is also responsible for shutting down the child processes when the system is shutting down. Please see the Supervisor module for a more in-depth discussion.
Creating a supervisor is not much different from creating a GenServer. We are going to define a module named KV.Supervisor, which will use the Supervisor behaviour, inside the lib/kv/supervisor.ex file:
defmodule KV.Supervisor do
  use Supervisor

  def start_link(opts) do
    Supervisor.start_link(__MODULE__, :ok, opts)
  end

  @impl true
  def init(:ok) do
    children = [
      KV.Registry
    ]

    Supervisor.init(children, strategy: :one_for_one)
  end
end
Our supervisor has a single child so far: KV.Registry. After we define a list of children, we call Supervisor.init/2, passing the children and the supervision strategy.
The supervision strategy dictates what happens when one of the children crashes. :one_for_one means that if a child dies, it will be the only one restarted. Since we have only one child now, that's all we need. The Supervisor behaviour supports several strategies, which we will discuss in this chapter.
Once the supervisor starts, it will traverse the list of children and it will invoke the child_spec/1 function on each module.
The child_spec/1 function returns the child specification which describes how to start the process, if the process is a worker or a supervisor, if the process is temporary, transient or permanent and so on. The child_spec/1 function is automatically defined when we use Agent, use GenServer, use Supervisor, etc. Let's give it a try in the terminal with iex -S mix:
iex> KV.Registry.child_spec([])
%{id: KV.Registry, start: {KV.Registry, :start_link, [[]]}}
We will learn those details as we move forward on this guide. If you would rather peek ahead, check the Supervisor docs.
After the supervisor retrieves all child specifications, it proceeds to start its children one by one, in the order they were defined, using the information in the :start key in the child specification. For our current specification, it will call KV.Registry.start_link([]).
Let's take the supervisor for a spin:
iex> {:ok, sup} = KV.Supervisor.start_link([])
{:ok, #PID<0.148.0>}
iex> Supervisor.which_children(sup)
[{KV.Registry, #PID<0.150.0>, :worker, [KV.Registry]}]
So far we have started the supervisor and listed its children. Once the supervisor started, it also started all of its children.
What happens if we intentionally crash the registry started by the supervisor? Let's do so by sending it a bad input on call:
iex> [{_, registry, _, _}] = Supervisor.which_children(sup)
[{KV.Registry, #PID<0.150.0>, :worker, [KV.Registry]}]
iex> GenServer.call(registry, :bad_input)
08:52:57.311 [error] GenServer #PID<0.150.0> terminating
** (FunctionClauseError) no function clause matching in KV.Registry.handle_call/3
iex> Supervisor.which_children(sup)
[{KV.Registry, #PID<0.157.0>, :worker, [KV.Registry]}]
Notice how the supervisor automatically started a new registry, with a new PID, in place of the first one once we caused it to crash due to a bad input.
In the previous chapters, we have always started processes directly. For example, we would call KV.Registry.start_link([]), which would return {:ok, pid}, and that would allow us to interact with the registry via its pid. Now that processes are started by the supervisor, we have to directly ask the supervisor who its children are, and fetch the PID from the returned list of children. In practice, doing so every time would be very expensive. To address this, we often give names to processes, allowing them to be uniquely identified in a single machine from anywhere in our code.
Let's learn how to do that.

  
    
  
  Naming processes


While our application will have many buckets, it will only have a single registry. Therefore, whenever we start the registry, we want to give it a unique name so we can reach out to it from anywhere. We do so by passing a :name option to KV.Registry.start_link/1.
Let's slightly change our children definition (in KV.Supervisor.init/1) to be a list of tuples instead of a list of atoms:
  def init(:ok) do
    children = [
      {KV.Registry, name: KV.Registry}
    ]
With this in place, the supervisor will now start KV.Registry by calling KV.Registry.start_link(name: KV.Registry).
If you revisit the KV.Registry.start_link/1 implementation, you will remember it simply passes the options to GenServer:
  def start_link(opts) do
    GenServer.start_link(__MODULE__, :ok, opts)
  end
which in turn will register the process with the given name. The :name option expects an atom for locally named processes (locally named means it is available to this machine — there are other options, which we won't discuss here). Since module identifiers are atoms (try i(KV.Registry) in IEx), we can name a process after the module that implements it, provided there is only one process for that name. This helps when debugging and introspecting the system.
Let's give the updated supervisor a try inside iex -S mix:
iex> KV.Supervisor.start_link([])
{:ok, #PID<0.66.0>}
iex> KV.Registry.create(KV.Registry, "shopping")
:ok
iex> KV.Registry.lookup(KV.Registry, "shopping")
{:ok, #PID<0.70.0>}
This time the supervisor started a named registry, allowing us to create buckets without having to explicitly fetch the PID from the supervisor. You should also know how to make the registry crash again, without looking up its PID: give it a try.
At this point, you may be wondering: should you also locally name bucket processes? Remember buckets are started dynamically based on user input. Since local names MUST be atoms, we would have to dynamically create atoms, which is a bad idea since once an atom is defined, it is never erased nor garbage collected. This means that, if we create atoms dynamically based on user input, we will eventually run out of memory (or to be more precise, the VM will crash because it imposes a hard limit on the number of atoms). This limitation is precisely why we created our own registry (or why one would use Elixir's built-in Registry module).

We are getting closer and closer to a fully working system. The supervisor automatically starts the registry. But how can we automatically start the supervisor whenever our system starts? To answer this question, let's talk about applications.

  
    
  
  Understanding applications


We have been working inside an application this entire time. Every time we changed a file and ran mix compile, we could see a Generated kv app message in the compilation output.
We can find the generated .app file at _build/dev/lib/kv/ebin/kv.app. Let's have a look at its contents:
{application,kv,
             [{applications,[kernel,stdlib,elixir,logger]},
              {description,"kv"},
              {modules,['Elixir.KV','Elixir.KV.Bucket','Elixir.KV.Registry',
                        'Elixir.KV.Supervisor']},
              {registered,[]},
              {vsn,"0.1.0"}]}.
This file contains Erlang terms (written using Erlang syntax). Even though we are not familiar with Erlang, it is easy to guess this file holds our application definition. It contains our application version, all the modules defined by it, as well as a list of applications we depend on, like Erlang's kernel, elixir itself, and logger.
The logger application ships as part of Elixir. We stated that our application needs it by specifying it in the :extra_applications list in mix.exs. See the official documentation for more information.

In a nutshell, an application consists of all the modules defined in the .app file, including the .app file itself. An application has generally only two directories: ebin, for Elixir artifacts, such as .beam and .app files, and priv, with any other artifact or asset you may need in your application.
Although Mix generates and maintains the .app file for us, we can customize its contents by adding new entries to the application/0 function inside the mix.exs project file. We are going to do our first customization soon.

  
    
  
  Starting applications


Each application in our system can be started and stopped. The rules for starting and stopping an application are also defined in the .app file. When we invoke iex -S mix, Mix compiles our application and then starts it.
Let's see this in practice. Start a console with iex -S mix and try:
iex> Application.start(:kv)
{:error, {:already_started, :kv}}
Oops, it's already started. Mix starts the current application and all of its dependencies automatically. This is also true for mix test and many other Mix commands.
We can, however, stop our :kv application, as well as the :logger application:
iex> Application.stop(:kv)
:ok
iex> Application.stop(:logger)
:ok
And let's try to start our application again:
iex> Application.start(:kv)
{:error, {:not_started, :logger}}
Now we get an error because an application that :kv depends on (:logger in this case) isn't started. We need to either start each application manually in the correct order or call Application.ensure_all_started/1 as follows:
iex> Application.ensure_all_started(:kv)
{:ok, [:logger, :kv]}
In practice, our tools always start our applications for us, but there is an API available if you need fine-grained control.

  
    
  
  The application callback


Whenever we invoke iex -S mix, Mix automatically starts our application by calling Application.start(:kv). But can we customize what happens when our application starts? As a matter of fact, we can! To do so, we define an application callback.
The first step is to tell our application definition (for example, our .app file) which module is going to implement the application callback. Let's do so by opening mix.exs and changing def application to the following:
  def application do
    [
      extra_applications: [:logger],
      mod: {KV, []}
    ]
  end
The :mod option specifies the "application callback module", followed by the arguments to be passed on application start. The application callback module can be any module that implements the Application behaviour.
To implement the Application behaviour, we have to use Application and define a start/2 function. The goal of start/2 is to start a supervisor, which will then start any child services or execute any other code our application may need. Let's use this opportunity to start the KV.Supervisor we have implemented earlier in this chapter.
Since we have specified KV as the module callback, let's change the KV module defined in lib/kv.ex to implement a start/2 function:
defmodule KV do
  use Application

  @impl true
  def start(_type, _args) do
    # Although we don't use the supervisor name below directly,
    # it can be useful when debugging or introspecting the system.
    KV.Supervisor.start_link(name: KV.Supervisor)
  end
end
Please note that by doing this, we are breaking the boilerplate test case which tested the hello function in KV. You can simply remove that test case.

When we use Application, we may define a couple of functions, similar to when we used Supervisor or GenServer. This time we only had to define a start/2 function. The Application behaviour also has a stop/1 callback, but it is rarely used in practice. You can check the documentation for more information.
Now that you have defined an application callback which starts our supervisor, we expect the KV.Registry process to be up and running as soon as we start iex -S mix. Let's give it another try:
iex> KV.Registry.create(KV.Registry, "shopping")
:ok
iex> KV.Registry.lookup(KV.Registry, "shopping")
{:ok, #PID<0.88.0>}
Let's recap what is happening. Whenever we invoke iex -S mix, it automatically starts our application by calling Application.start(:kv), which then invokes the application callback. The application callback's job is to start a supervision tree. Right now, our supervisor has a single child named KV.Registry, started with name KV.Registry. Our supervisor could have other children, and some of these children could be their own supervisors with their own children, leading to the so-called supervision trees.

  
    
  
  Projects or applications?


Mix makes a distinction between projects and applications. Based on the contents of our mix.exs file, we would say we have a Mix project that defines the :kv application. As we will see in later chapters, there are projects that don't define any application.
When we say "project" you should think about Mix. Mix is the tool that manages your project. It knows how to compile your project, test your project and more. It also knows how to compile and start the application relevant to your project.
When we talk about applications, we talk about OTP. Applications are the entities that are started and stopped as a whole by the runtime. You can learn more about applications and how they relate to booting and shutting down of your system as a whole in the documentation for the Application module.

  
    
  
  Next steps


Although this chapter was the first time we implemented a supervisor, it was not the first time we used one! In the previous chapter, when we used start_supervised! to start the registry during our tests, ExUnit started the registry under a supervisor managed by the ExUnit framework itself. By defining our own supervisor, we provide more structure on how we initialize, shutdown and supervise processes in our applications, aligning our production code and tests with best practices.
But we are not done yet. So far we are supervising the registry but our application is also starting buckets. Since buckets are started dynamically, we can use a special type of supervisor called DynamicSupervisor, which is optimized to handle such scenarios. Let's explore it next.

We have now successfully defined our supervisor which is automatically started (and stopped) as part of our application life cycle.
Remember, however, that our KV.Registry is both linking (via start_link) and monitoring (via monitor) bucket processes in the handle_cast/2 callback:
{:ok, bucket} = KV.Bucket.start_link([])
ref = Process.monitor(bucket)
Links are bidirectional, which implies that a crash in a bucket will crash the registry. Although we now have the supervisor, which guarantees the registry will be back up and running, crashing the registry still means we lose all data associating bucket names to their respective processes.
In other words, we want the registry to keep on running even if a bucket crashes. Let's write a new registry test:
test "removes bucket on crash", %{registry: registry} do
  KV.Registry.create(registry, "shopping")
  {:ok, bucket} = KV.Registry.lookup(registry, "shopping")

  # Stop the bucket with non-normal reason
  Agent.stop(bucket, :shutdown)
  assert KV.Registry.lookup(registry, "shopping") == :error
end
The test is similar to "removes bucket on exit" except that we are being a bit more harsh by sending :shutdown as the exit reason instead of :normal. If a process terminates with a reason other than :normal, all linked processes receive an EXIT signal, causing the linked process to also terminate unless it is trapping exits.
Since the bucket terminated, the registry also stopped, and our test fails when trying to GenServer.call/3 it:
  1) test removes bucket on crash (KV.RegistryTest)
     test/kv/registry_test.exs:26
     ** (exit) exited in: GenServer.call(#PID<0.148.0>, {:lookup, "shopping"}, 5000)
         ** (EXIT) no process: the process is not alive or there's no process currently associated with the given name, possibly because its application isn't started
     code: assert KV.Registry.lookup(registry, "shopping") == :error
     stacktrace:
       (elixir) lib/gen_server.ex:770: GenServer.call/3
       test/kv/registry_test.exs:33: (test)
We are going to solve this issue by defining a new supervisor that will spawn and supervise all buckets. Opposite to the previous Supervisor we defined, the children are not known upfront, but they are rather started dynamically. For those situations, we use a supervisor optimized to such use cases called DynamicSupervisor. The DynamicSupervisor does not expect a list of children during initialization; instead each child is started manually via DynamicSupervisor.start_child/2.

  
    
  
  The bucket supervisor


Since a DynamicSupervisor does not define any children during initialization, the DynamicSupervisor also allows us to skip the work of defining a whole separate module with the usual start_link function and the init callback. Instead, we can define a DynamicSupervisor directly in the supervision tree, by giving it a name and a strategy.
Open up lib/kv/supervisor.ex and add the dynamic supervisor as a child as follows:
  def init(:ok) do
    children = [
      {KV.Registry, name: KV.Registry},
      {DynamicSupervisor, name: KV.BucketSupervisor, strategy: :one_for_one}
    ]

    Supervisor.init(children, strategy: :one_for_one)
  end
Remember that the name of a process can be any atom. So far, we have named processes with the same name as the modules that define their implementation. For example, the process defined by KV.Registry was given a process name of KV.Registry. This is simply a convention: If later there is an error in your system that says, "process named KV.Registry crashed with reason", we know exactly where to investigate.
In this case, there is no module, so we picked the name KV.BucketSupervisor. It could have been any other name. We also chose the :one_for_one strategy, which is currently the only available strategy for dynamic supervisors.
Run iex -S mix so we can give our dynamic supervisor a try:
iex> {:ok, bucket} = DynamicSupervisor.start_child(KV.BucketSupervisor, KV.Bucket)
{:ok, #PID<0.72.0>}
iex> KV.Bucket.put(bucket, "eggs", 3)
:ok
iex> KV.Bucket.get(bucket, "eggs")
3
DynamicSupervisor.start_child/2 expects the name of the supervisor and the child specification of the child to be started.
The last step is to change the registry to use the dynamic supervisor:
  def handle_cast({:create, name}, {names, refs}) do
    if Map.has_key?(names, name) do
      {:noreply, {names, refs}}
    else
      {:ok, pid} = DynamicSupervisor.start_child(KV.BucketSupervisor, KV.Bucket)
      ref = Process.monitor(pid)
      refs = Map.put(refs, ref, name)
      names = Map.put(names, name, pid)
      {:noreply, {names, refs}}
    end
  end
That's enough for our tests to pass but there is a resource leakage in our application. When a bucket terminates, the supervisor will start a new bucket in its place. After all, that's the role of the supervisor!
However, when the supervisor restarts the new bucket, the registry does not know about it. So we will have an empty bucket in the supervisor that nobody can access! To solve this, we want to say that buckets are actually temporary. If they crash, regardless of the reason, they should not be restarted.
We can do this by passing the restart: :temporary option to use Agent in KV.Bucket:
defmodule KV.Bucket do
  use Agent, restart: :temporary
Let's also add a test to test/kv/bucket_test.exs that guarantees the bucket is temporary:
  test "are temporary workers" do
    assert Supervisor.child_spec(KV.Bucket, []).restart == :temporary
  end
Our test uses the Supervisor.child_spec/2 function to retrieve the child specification out of a module and then assert its restart value is :temporary. At this point, you may be wondering why use a supervisor if it never restarts its children. It happens that supervisors provide more than restarts, they are also responsible for guaranteeing proper startup and shutdown, especially in case of crashes in a supervision tree.

  
    
  
  Supervision trees


When we added KV.BucketSupervisor as a child of KV.Supervisor, we began to have supervisors that supervise other supervisors, forming so-called "supervision trees".
Every time you add a new child to a supervisor, it is important to evaluate if the supervisor strategy is correct as well as the order of child processes. In this case, we are using :one_for_one and the KV.Registry is started before KV.BucketSupervisor.
One flaw that shows up right away is the ordering issue. Since KV.Registry invokes KV.BucketSupervisor, then the KV.BucketSupervisor must be started before KV.Registry. Otherwise, it may happen that the registry attempts to reach the bucket supervisor before it has started.
The second flaw is related to the supervision strategy. If KV.Registry dies, all information linking KV.Bucket names to bucket processes is lost. Therefore the KV.BucketSupervisor and all children must terminate too - otherwise we will have orphan processes.
In light of this observation, we should consider moving to another supervision strategy. The two other candidates are :one_for_all and :rest_for_one. A supervisor using the :rest_for_one strategy will kill and restart child processes which were started after the crashed child. In this case, we would want KV.BucketSupervisor to terminate if KV.Registry terminates. This would require the bucket supervisor to be placed after the registry which violates the ordering constraints we have established two paragraphs above.
So our last option is to go all in and pick the :one_for_all strategy: the supervisor will kill and restart all of its children processes whenever any one of them dies. This is a completely reasonable approach for our application, since the registry can't work without the bucket supervisor, and the bucket supervisor should terminate without the registry. Let's reimplement init/1 in KV.Supervisor to encode those properties:
  def init(:ok) do
    children = [
      {DynamicSupervisor, name: KV.BucketSupervisor, strategy: :one_for_one},
      {KV.Registry, name: KV.Registry}
    ]

    Supervisor.init(children, strategy: :one_for_all)
  end
There are two topics left before we move on to the next chapter.

  
    
  
  Shared state in tests


So far we have been starting one registry per test to ensure they are isolated:
setup do
  registry = start_supervised!(KV.Registry)
  %{registry: registry}
end
Since we have changed our registry to use KV.BucketSupervisor, our tests are now relying on this shared supervisor even though each test has its own registry. The question is: should we?
It depends. It is ok to rely on shared state as long as we depend only on a non-shared partition of this state. Although multiple registries may start buckets on the shared bucket supervisor, those buckets and registries are isolated from each other. We would only run into concurrency issues if we used a function like DynamicSupervisor.count_children(KV.BucketSupervisor) which would count all buckets from all registries, potentially giving different results when tests run concurrently.
Since we have relied only on a non-shared partition of the bucket supervisor so far, we don't need to worry about concurrency issues in our test suite. In case it ever becomes a problem, we can start a supervisor per test and pass it as an argument to the registry start_link function.

  
    
  
  Observer


Now that we have defined our supervision tree, it is a great opportunity to introduce the Observer tool that ships with Erlang. Start your application with iex -S mix and key this in:
iex> :observer.start()
Missing dependencies
When running iex inside a project with iex -S mix, observer won't be available as a dependency. To do so, you will need to call the following functions before:
iex> Mix.ensure_application!(:wx)             # Not necessary on Erlang/OTP 27+
iex> Mix.ensure_application!(:runtime_tools)  # Not necessary on Erlang/OTP 27+
iex> Mix.ensure_application!(:observer)
iex> :observer.start()
If any of the calls above fail, here is what may have happened: some package managers default to installing a minimized Erlang without WX bindings for GUI support. In some package managers, you may be able to replace the headless Erlang with a more complete package (look for packages named erlang vs erlang-nox on Debian/Ubuntu/Arch). In others managers, you may need to install a separate erlang-wx (or similarly named) package.
There are conversations to improve this experience in future releases.

A GUI should pop up containing all sorts of information about our system, from general statistics to load charts as well as a list of all running processes and applications.
In the Applications tab, you will see all applications currently running in your system alongside their supervision tree. You can select the kv application to explore it further:
[image: Observer GUI screenshot]Not only that, as you create new buckets on the terminal, you should see new processes spawned in the supervision tree shown in Observer:
iex> KV.Registry.create(KV.Registry, "shopping")
:ok
We will leave it up to you to further explore what Observer provides. Note you can double-click any process in the supervision tree to retrieve more information about it, as well as right-click a process to send "a kill signal", a perfect way to emulate failures and see if your supervisor reacts as expected.
At the end of the day, tools like Observer are one of the reasons you want to always start processes inside supervision trees, even if they are temporary, to ensure they are always reachable and introspectable.
Now that our buckets are properly linked and supervised, let's see how we can speed things up.

Speeding up with ETS
    

Every time we need to look up a bucket, we need to send a message to the registry. In case our registry is being accessed concurrently by multiple processes, the registry may become a bottleneck!
In this chapter, we will learn about ETS (Erlang Term Storage) and how to use it as a cache mechanism.
Warning! Don't use ETS as a cache prematurely! Log and analyze your application performance and identify which parts are bottlenecks, so you know whether you should cache, and what you should cache. This chapter is merely an example of how ETS can be used, once you've determined the need.


  
    
  
  ETS as a cache


ETS allows us to store any Elixir term in an in-memory table. Working with ETS tables is done via Erlang's :ets module:
iex> table = :ets.new(:buckets_registry, [:set, :protected])
#Reference<0.1885502827.460455937.234656>
iex> :ets.insert(table, {"foo", self()})
true
iex> :ets.lookup(table, "foo")
[{"foo", #PID<0.41.0>}]
When creating an ETS table, two arguments are required: the table name and a set of options. From the available options, we passed the table type and its access rules. We have chosen the :set type, which means that keys cannot be duplicated. We've also set the table's access to :protected, meaning only the process that created the table can write to it, but all processes can read from it. The possible access controls:
  :public — Read/Write available to all processes.
  :protected — Read available to all processes. Only writable by owner process. This is the default.
  :private — Read/Write limited to owner process.
Be aware that if your Read/Write call violates the access control, the operation will raise ArgumentError. Finally, since :set and :protected are the default values, we will skip them from now on.
ETS tables can also be named, allowing us to access them by a given name:
iex> :ets.new(:buckets_registry, [:named_table])
:buckets_registry
iex> :ets.insert(:buckets_registry, {"foo", self()})
true
iex> :ets.lookup(:buckets_registry, "foo")
[{"foo", #PID<0.41.0>}]
Let's change the KV.Registry to use ETS tables. The first change is to modify our registry to require a name argument, we will use it to name the ETS table and the registry process itself. ETS names and process names are stored in different locations, so there is no chance of conflicts.
Open up lib/kv/registry.ex, and let's change its implementation. We've added comments to the source code to highlight the changes we've made:
defmodule KV.Registry do
  use GenServer

  ## Client API

  @doc """
  Starts the registry with the given options.

  `:name` is always required.
  """
  def start_link(opts) do
    # 1. Pass the name to GenServer's init
    server = Keyword.fetch!(opts, :name)
    GenServer.start_link(__MODULE__, server, opts)
  end

  @doc """
  Looks up the bucket pid for `name` stored in `server`.

  Returns `{:ok, pid}` if the bucket exists, `:error` otherwise.
  """
  def lookup(server, name) do
    # 2. Lookup is now done directly in ETS, without accessing the server
    case :ets.lookup(server, name) do
      [{^name, pid}] -> {:ok, pid}
      [] -> :error
    end
  end

  @doc """
  Ensures there is a bucket associated with the given `name` in `server`.
  """
  def create(server, name) do
    GenServer.cast(server, {:create, name})
  end

  ## Server callbacks

  @impl true
  def init(table) do
    # 3. We have replaced the names map by the ETS table
    names = :ets.new(table, [:named_table, read_concurrency: true])
    refs  = %{}
    {:ok, {names, refs}}
  end

  # 4. The previous handle_call callback for lookup was removed

  @impl true
  def handle_cast({:create, name}, {names, refs}) do
    # 5. Read and write to the ETS table instead of the map
    case lookup(names, name) do
      {:ok, _pid} ->
        {:noreply, {names, refs}}

      :error ->
        {:ok, pid} = DynamicSupervisor.start_child(KV.BucketSupervisor, KV.Bucket)
        ref = Process.monitor(pid)
        refs = Map.put(refs, ref, name)
        :ets.insert(names, {name, pid})
        {:noreply, {names, refs}}
    end
  end

  @impl true
  def handle_info({:DOWN, ref, :process, _pid, _reason}, {names, refs}) do
    # 6. Delete from the ETS table instead of the map
    {name, refs} = Map.pop(refs, ref)
    :ets.delete(names, name)
    {:noreply, {names, refs}}
  end

  @impl true
  def handle_info(_msg, state) do
    {:noreply, state}
  end
end
Notice that before our changes KV.Registry.lookup/2 sent requests to the server, but now it reads directly from the ETS table, which is shared across all processes. That's the main idea behind the cache mechanism we are implementing.
In order for the cache mechanism to work, the created ETS table needs to have access :protected (the default), so all clients can read from it, while only the KV.Registry process writes to it. We have also set read_concurrency: true when starting the table, optimizing the table for the common scenario of concurrent read operations.
The changes we have performed above have broken our tests because the registry requires the :name option when starting up. Furthermore, some registry operations such as lookup/2 require the name to be given as an argument, instead of a PID, so we can do the ETS table lookup. Let's change the setup function in test/kv/registry_test.exs to fix both issues:
  setup context do
    _ = start_supervised!({KV.Registry, name: context.test})
    %{registry: context.test}
  end
Since each test has a unique name, we use the test name to name our registries. This way, we no longer need to pass the registry PID around, instead we identify it by the test name. Also note we assigned the result of start_supervised! to underscore (_). This idiom is often used to signal that we are not interested in the result of start_supervised!.
Once we change setup, some tests will continue to fail. You may even notice tests pass and fail inconsistently between runs. For example, the "spawns buckets" test:
test "spawns buckets", %{registry: registry} do
  assert KV.Registry.lookup(registry, "shopping") == :error

  KV.Registry.create(registry, "shopping")
  assert {:ok, bucket} = KV.Registry.lookup(registry, "shopping")

  KV.Bucket.put(bucket, "milk", 1)
  assert KV.Bucket.get(bucket, "milk") == 1
end
may be failing on this line:
{:ok, bucket} = KV.Registry.lookup(registry, "shopping")
How can this line fail if we just created the bucket in the previous line?
The reason those failures are happening is because, for didactic purposes, we have made two mistakes:
	We are prematurely optimizing (by adding this cache layer)
	We are using cast/2 (while we should be using call/2)


  
    
  
  Race conditions?


Developing in Elixir does not make your code free of race conditions. However, Elixir's abstractions where nothing is shared by default make it easier to spot a race condition's root cause.
What is happening in our tests is that there is a delay in between an operation and the time we can observe this change in the ETS table. Here is what we were expecting to happen:
	We invoke KV.Registry.create(registry, "shopping")
	The registry creates the bucket and updates the cache table
	We access the information from the table with KV.Registry.lookup(registry, "shopping")
	The command above returns {:ok, bucket}

However, since KV.Registry.create/2 is a cast operation, the command will return before we actually write to the table! In other words, this is happening:
	We invoke KV.Registry.create(registry, "shopping")
	We access the information from the table with KV.Registry.lookup(registry, "shopping")
	The command above returns :error
	The registry creates the bucket and updates the cache table

To fix the failure we need to make KV.Registry.create/2 synchronous by using call/2 rather than cast/2. This will guarantee that the client will only continue after changes have been made to the table. Let's back to lib/kv/registry.ex and change the function and its callback as follows:
def create(server, name) do
  GenServer.call(server, {:create, name})
end
@impl true
def handle_call({:create, name}, _from, {names, refs}) do
  case lookup(names, name) do
    {:ok, pid} ->
      {:reply, pid, {names, refs}}

    :error ->
      {:ok, pid} = DynamicSupervisor.start_child(KV.BucketSupervisor, KV.Bucket)
      ref = Process.monitor(pid)
      refs = Map.put(refs, ref, name)
      :ets.insert(names, {name, pid})
      {:reply, pid, {names, refs}}
  end
end
We changed the callback from handle_cast/2 to handle_call/3 and changed it to reply with the PID of the created bucket. Generally speaking, Elixir developers prefer to use call/2 instead of cast/2 as it also provides back-pressure — you block until you get a reply. Using cast/2 when not necessary can also be considered a premature optimization.
Let's run the tests once again. This time though, we will pass the --trace option:
$ mix test --trace

The --trace option is useful when your tests are deadlocking or there are race conditions, as it runs all tests synchronously (async: true has no effect) and shows detailed information about each test. If you run the tests multiple times you may see this intermittent failure:
  1) test removes buckets on exit (KV.RegistryTest)
     test/kv/registry_test.exs:19
     Assertion with == failed
     code:  assert KV.Registry.lookup(registry, "shopping") == :error
     left:  {:ok, #PID<0.109.0>}
     right: :error
     stacktrace:
       test/kv/registry_test.exs:23
According to the failure message, we are expecting that the bucket no longer exists on the table, but it still does! This problem is the opposite of the one we have just solved: while previously there was a delay between the command to create a bucket and updating the table, now there is a delay between the bucket process dying and its entry being removed from the table. Since this is a race condition, you may not be able to reproduce it on your machine, but it is there.
Last time we fixed the race condition by replacing the asynchronous operation, a cast, by a call, which is synchronous. Unfortunately, the handle_info/2 callback we are using to receive the :DOWN message and delete the entry from the ETS table does not have a synchronous equivalent. This time, we need to find a way to guarantee the registry has processed the :DOWN notification sent when the bucket process terminated.
An easy way to do so is by sending a synchronous request to the registry before we do the bucket lookup. The Agent.stop/2 operation is synchronous and only returns after the bucket process terminates. Therefore, once Agent.stop/2 returns, the registry has received the :DOWN message but it may not have processed it yet. In order to guarantee the processing of the :DOWN message, we can do a synchronous request. Since messages are processed in order, once the registry replies to the synchronous request, then the :DOWN message will definitely have been processed.
Let's do so by creating a "bogus" bucket, which is a synchronous request, after Agent.stop/2 in both "remove" tests at test/kv/registry_test.exs:
  test "removes buckets on exit", %{registry: registry} do
    KV.Registry.create(registry, "shopping")
    {:ok, bucket} = KV.Registry.lookup(registry, "shopping")
    Agent.stop(bucket)

    # Do a call to ensure the registry processed the DOWN message
    _ = KV.Registry.create(registry, "bogus")
    assert KV.Registry.lookup(registry, "shopping") == :error
  end

  test "removes bucket on crash", %{registry: registry} do
    KV.Registry.create(registry, "shopping")
    {:ok, bucket} = KV.Registry.lookup(registry, "shopping")

    # Stop the bucket with non-normal reason
    Agent.stop(bucket, :shutdown)

    # Do a call to ensure the registry processed the DOWN message
    _ = KV.Registry.create(registry, "bogus")
    assert KV.Registry.lookup(registry, "shopping") == :error
  end
Our tests should now (always) pass!
This concludes our optimization chapter. We have used ETS as a cache mechanism where reads can happen from any processes but writes are still serialized through a single process. More importantly, we have also learned that once data can be read asynchronously, we need to be aware of the race conditions it might introduce.
In practice, if you find yourself in a position where you need a registry for dynamic processes, you should use the Registry module provided as part of Elixir. It provides functionality similar to the one we have built using a GenServer + :ets while also being able to perform both writes and reads concurrently. It has been benchmarked to scale across all cores even on machines with 40 cores.
Next, let's discuss external and internal dependencies and how Mix helps us manage large codebases.

In this chapter, we will discuss how to manage dependencies in Mix.
Our kv application is complete, so it's time to implement the server that will handle the requests we defined in the first chapter:
CREATE shopping
OK

PUT shopping milk 1
OK

PUT shopping eggs 3
OK

GET shopping milk
1
OK

DELETE shopping eggs
OK
However, instead of adding more code to the kv application, we are going to build the TCP server as another application that is a client of the kv application. Since the whole runtime and Elixir ecosystem are geared towards applications, it makes sense to break our projects into smaller applications that work together rather than building a big, monolithic app.
Before creating our new application, we must discuss how Mix handles dependencies. In practice, there are two kinds of dependencies we usually work with: internal and external dependencies. Mix supports mechanisms to work with both.

  
    
  
  External dependencies


External dependencies are the ones not tied to your business domain. For example, if you need an HTTP API for your distributed KV application, you can use the Plug project as an external dependency.
Installing external dependencies is simple. Most commonly, we use the Hex Package Manager, by listing the dependency inside the deps function in our mix.exs file:
def deps do
  [{:plug, "~> 1.0"}]
end
This dependency refers to the latest version of Plug in the 1.x.x version series that has been pushed to Hex. This is indicated by the ~> preceding the version number. For more information on specifying version requirements, see the documentation for the Version module.
Typically, stable releases are pushed to Hex. If you want to depend on an external dependency still in development, Mix is able to manage Git dependencies too:
def deps do
  [{:plug, git: "https://github.com/elixir-lang/plug.git"}]
end
You will notice that when you add a dependency to your project, Mix generates a mix.lock file that guarantees repeatable builds. The lock file must be checked in to your version control system, to guarantee that everyone who uses the project will use the same dependency versions as you.
Mix provides many tasks for working with dependencies, which can be seen in mix help:
$ mix help
mix deps              # Lists dependencies and their status
mix deps.clean        # Deletes the given dependencies' files
mix deps.compile      # Compiles dependencies
mix deps.get          # Gets all out of date dependencies
mix deps.tree         # Prints the dependency tree
mix deps.unlock       # Unlocks the given dependencies
mix deps.update       # Updates the given dependencies

The most common tasks are mix deps.get and mix deps.update. Once fetched, dependencies are automatically compiled for you. You can read more about deps by typing mix help deps, and in the documentation for the Mix.Tasks.Deps module.

  
    
  
  Internal dependencies


Internal dependencies are the ones that are specific to your project. They usually don't make sense outside the scope of your project/company/organization. Most of the time, you want to keep them private, whether due to technical, economic or business reasons.
If you have an internal dependency, Mix supports two methods to work with them: Git repositories or umbrella projects.
For example, if you push the kv project to a Git repository, you'll need to list it in your deps code in order to use it:
def deps do
  [{:kv, git: "https://github.com/YOUR_ACCOUNT/kv.git"}]
end
If the repository is private though, you may need to specify the private URL git@github.com:YOUR_ACCOUNT/kv.git. In any case, Mix will be able to fetch it for you as long as you have the proper credentials.
Using Git repositories for internal dependencies is somewhat discouraged in Elixir. Remember that the runtime and the Elixir ecosystem already provide the concept of applications. As such, we expect you to frequently break your code into applications that can be organized logically, even within a single project.
However, if you push every application as a separate project to a Git repository, your projects may become very hard to maintain as you will spend a lot of time managing those Git repositories rather than writing your code.
For this reason, Mix supports "umbrella projects". Umbrella projects are used to build applications that run together in a single repository. That is exactly the style we are going to explore in the next sections.
Let's create a new Mix project. We are going to creatively name it kv_umbrella, and this new project will have both the existing kv application and the new kv_server application inside. The directory structure will look like this:
+ kv_umbrella
  + apps
    + kv
    + kv_server
The interesting thing about this approach is that Mix has many conveniences for working with such projects, such as the ability to compile and test all applications inside apps with a single command. However, even though they are all listed together inside apps, they are still decoupled from each other, so you can build, test and deploy each application in isolation if you want to.
So let's get started!

  
    
  
  Umbrella projects


Let's start a new project using mix new. This new project will be named kv_umbrella and we need to pass the --umbrella option when creating it. Do not create this new project inside the existing kv project!
$ mix new kv_umbrella --umbrella
* creating README.md
* creating .formatter.exs
* creating .gitignore
* creating mix.exs
* creating apps
* creating config
* creating config/config.exs

From the printed information, we can see far fewer files are generated. The generated mix.exs file is different too. Let's take a look (comments have been removed):
defmodule KvUmbrella.MixProject do
  use Mix.Project

  def project do
    [
      apps_path: "apps",
      start_permanent: Mix.env() == :prod,
      deps: deps()
    ]
  end

  defp deps do
    []
  end
end
What makes this project different from the previous one is the apps_path: "apps" entry in the project definition. This means this project will act as an umbrella. Such projects do not have source files nor tests, although they can have their own dependencies. Each child application must be defined inside the apps directory.
Let's move inside the apps directory and start building kv_server. This time, we are going to pass the --sup flag, which will tell Mix to generate a supervision tree automatically for us, instead of building one manually as we did in previous chapters:
$ cd kv_umbrella/apps
$ mix new kv_server --module KVServer --sup

The generated files are similar to the ones we first generated for kv, with a few differences. Let's open up mix.exs:
defmodule KVServer.MixProject do
  use Mix.Project

  def project do
    [
      app: :kv_server,
      version: "0.1.0",
      build_path: "../../_build",
      config_path: "../../config/config.exs",
      deps_path: "../../deps",
      lockfile: "../../mix.lock",
      elixir: "~> 1.14",
      start_permanent: Mix.env() == :prod,
      deps: deps()
    ]
  end

  # Run "mix help compile.app" to learn about applications
  def application do
    [
      extra_applications: [:logger],
      mod: {KVServer.Application, []}
    ]
  end

  # Run "mix help deps" to learn about dependencies
  defp deps do
    [
      # {:dep_from_hexpm, "~> 0.3.0"},
      # {:dep_from_git, git: "https://github.com/elixir-lang/my_dep.git", tag: "0.1.0"},
      # {:sibling_app_in_umbrella, in_umbrella: true},
    ]
  end
end
First of all, since we generated this project inside kv_umbrella/apps, Mix automatically detected the umbrella structure and added four lines to the project definition:
build_path: "../../_build",
config_path: "../../config/config.exs",
deps_path: "../../deps",
lockfile: "../../mix.lock",
Those options mean all dependencies will be checked out to kv_umbrella/deps, and they will share the same build, config, and lock files. We haven't talked about configuration yet, but from here we can build the intuition that all configuration and dependencies are shared across all projects in an umbrella, and it is not per application.
The second change is in the application function inside mix.exs:
def application do
  [
    extra_applications: [:logger],
    mod: {KVServer.Application, []}
  ]
end
Because we passed the --sup flag, Mix automatically added mod: {KVServer.Application, []}, specifying that KVServer.Application is our application callback module. KVServer.Application will start our application supervision tree.
In fact, let's open up lib/kv_server/application.ex:
defmodule KVServer.Application do
  # See https://hexdocs.pm/elixir/Application.html
  # for more information on OTP Applications
  @moduledoc false

  use Application

  @impl true
  def start(_type, _args) do
    # List all child processes to be supervised
    children = [
      # Starts a worker by calling: KVServer.Worker.start_link(arg)
      # {KVServer.Worker, arg},
    ]

    # See https://hexdocs.pm/elixir/Supervisor.html
    # for other strategies and supported options
    opts = [strategy: :one_for_one, name: KVServer.Supervisor]
    Supervisor.start_link(children, opts)
  end
end
Notice that it defines the application callback function, start/2, and instead of defining a supervisor named KVServer.Supervisor that uses the Supervisor module, it conveniently defined the supervisor inline! You can read more about such supervisors by reading the Supervisor module documentation.
We can already try out our first umbrella child. We could run tests inside the apps/kv_server directory, but that wouldn't be much fun. Instead, go to the root of the umbrella project and run mix test:
$ mix test

And it works!
Since we want kv_server to eventually use the functionality we defined in kv, we need to add kv as a dependency to our application.

  
    
  
  Dependencies within an umbrella project


Dependencies between applications in an umbrella project must still be explicitly defined and Mix makes it easy to do so. Open up apps/kv_server/mix.exs and change the deps/0 function to the following:
defp deps do
  [{:kv, in_umbrella: true}]
end
The line above makes :kv available as a dependency inside :kv_server and automatically starts the :kv application before the server starts.
Finally, copy the kv application we have built so far to the apps directory in our new umbrella project. The final directory structure should match the structure we mentioned earlier:
+ kv_umbrella
  + apps
    + kv
    + kv_server
We now need to modify apps/kv/mix.exs to contain the umbrella entries we have seen in apps/kv_server/mix.exs. Open up apps/kv/mix.exs and add to the project/0 function:
build_path: "../../_build",
config_path: "../../config/config.exs",
deps_path: "../../deps",
lockfile: "../../mix.lock",
Now you can run tests for both projects from the umbrella root with mix test. Sweet!

  
    
  
  Don't drink the kool aid


Umbrella projects are a convenience to help you organize and manage multiple applications. While it provides a degree of separation between applications, those applications are not fully decoupled, as they share the same configuration and the same dependencies.
The pattern of keeping multiple applications in the same repository is known as "mono-repo". Umbrella projects maximize this pattern by providing conveniences to compile, test and run multiple applications at once.
If you find yourself in a position where you want to use different configurations in each application for the same dependency or use different dependency versions, then it is likely your codebase has grown beyond what umbrellas can provide.
The good news is that breaking an umbrella apart is quite straightforward, as you simply need to move applications outside of the umbrella project's apps/ directory and update the project's mix.exs file to no longer set the build_path, config_path, deps_path, and lockfile configuration. You can depend on private projects outside of the umbrella in multiple ways:
	Move it to a separate folder within the same repository and point to it using a path dependency (the mono-repo pattern)
	Move the repository to a separate Git repository and depend on it
	Publish the project to a private Hex.pm organization


  
    
  
  Summing up


In this chapter, we have learned more about Mix dependencies and umbrella projects. While we may run kv without a server, our kv_server depends directly on kv. By breaking them into separate applications, we gain more control in how they are developed and tested.
When using umbrella applications, it is important to have a clear boundary between them. Our upcoming kv_server must only access public APIs defined in kv. Think of your umbrella apps as any other dependency or even Elixir itself: you can only access what is public and documented. Reaching into private functionality in your dependencies is a poor practice that will eventually cause your code to break when a new version is up.
Umbrella applications can also be used as a stepping stone for eventually extracting an application from your codebase. For example, imagine a web application that has to send "push notifications" to its users. The whole "push notifications system" can be developed as a separate application in the umbrella, with its own supervision tree and APIs. If you ever run into a situation where another project needs the push notifications system, the system can be moved to a private repository or a Hex package.
Finally, keep in mind that applications in an umbrella project all share the same configurations and dependencies. If two applications in your umbrella need to configure the same dependency in drastically different ways or even use different versions, you have probably outgrown the benefits brought by umbrellas. Remember you can break the umbrella and still leverage the benefits behind "mono-repos".
With our umbrella project up and running, it is time to start writing our server.

Task and gen_tcp
    

In this chapter, we are going to learn how to use Erlang's :gen_tcp module to serve requests. This provides a great opportunity to explore Elixir's Task module. In future chapters, we will expand our server so that it can actually serve the commands.

  
    
  
  Echo server


We will start our TCP server by first implementing an echo server. It will send a response with the text it received in the request. We will slowly improve our server until it is supervised and ready to handle multiple connections.
A TCP server, in broad strokes, performs the following steps:
	Listens to a port until the port is available and it gets hold of the socket
	Waits for a client connection on that port and accepts it
	Reads the client request and writes a response back

Let's implement those steps. Move to the apps/kv_server application, open up lib/kv_server.ex, and add the following functions:
defmodule KVServer do
  require Logger

  def accept(port) do
    # The options below mean:
    #
    # 1. `:binary` - receives data as binaries (instead of lists)
    # 2. `packet: :line` - receives data line by line
    # 3. `active: false` - blocks on `:gen_tcp.recv/2` until data is available
    # 4. `reuseaddr: true` - allows us to reuse the address if the listener crashes
    #
    {:ok, socket} =
      :gen_tcp.listen(port, [:binary, packet: :line, active: false, reuseaddr: true])
    Logger.info("Accepting connections on port #{port}")
    loop_acceptor(socket)
  end

  defp loop_acceptor(socket) do
    {:ok, client} = :gen_tcp.accept(socket)
    serve(client)
    loop_acceptor(socket)
  end

  defp serve(socket) do
    socket
    |> read_line()
    |> write_line(socket)

    serve(socket)
  end

  defp read_line(socket) do
    {:ok, data} = :gen_tcp.recv(socket, 0)
    data
  end

  defp write_line(line, socket) do
    :gen_tcp.send(socket, line)
  end
end
We are going to start our server by calling KVServer.accept(4040), where 4040 is the port. The first step in accept/1 is to listen to the port until the socket becomes available and then call loop_acceptor/1. loop_acceptor/1 is a loop accepting client connections. For each accepted connection, we call serve/1.
serve/1 is another loop that reads a line from the socket and writes those lines back to the socket. Note that the serve/1 function uses the pipe operator |>/2 to express this flow of operations. The pipe operator evaluates the left side and passes its result as the first argument to the function on the right side. The example above:
socket |> read_line() |> write_line(socket)
is equivalent to:
write_line(read_line(socket), socket)
The read_line/1 implementation receives data from the socket using :gen_tcp.recv/2 and write_line/2 writes to the socket using :gen_tcp.send/2.
Note that serve/1 is an infinite loop called sequentially inside loop_acceptor/1, so the tail call to loop_acceptor/1 is never reached and could be avoided. However, as we shall see, we will need to execute serve/1 in a separate process, so we will need that tail call soon.
This is pretty much all we need to implement our echo server. Let's give it a try!
Start an IEx session inside the kv_server application with iex -S mix. Inside IEx, run:
iex> KVServer.accept(4040)
The server is now running, and you will even notice the console is blocked. Let's use a telnet client to access our server. There are clients available on most operating systems, and their command lines are generally similar:
$ telnet 127.0.0.1 4040
Trying 127.0.0.1...
Connected to localhost.
Escape character is '^]'.
hello
hello
is it me
is it me
you are looking for?
you are looking for?

Type "hello", press enter, and you will get "hello" back. Excellent!
My particular telnet client can be exited by typing ctrl + ], typing quit, and pressing <Enter>, but your client may require different steps.
Once you exit the telnet client, you will likely see an error in the IEx session:
** (MatchError) no match of right hand side value: {:error, :closed}
    (kv_server) lib/kv_server.ex:45: KVServer.read_line/1
    (kv_server) lib/kv_server.ex:37: KVServer.serve/1
    (kv_server) lib/kv_server.ex:30: KVServer.loop_acceptor/1
That's because we were expecting data from :gen_tcp.recv/2 but the client closed the connection. We need to handle such cases better in future revisions of our server.
For now, there is a more important bug we need to fix: what happens if our TCP acceptor crashes? Since there is no supervision, the server dies and we won't be able to serve more requests, because it won't be restarted. That's why we must move our server to a supervision tree.

  
    
  
  Tasks


We have learned about agents, generic servers, and supervisors. They are all meant to work with multiple messages or manage state. But what do we use when we only need to execute some task and that is it?
The Task module provides this functionality exactly. For example, it has a Task.start_link/1 function that receives an anonymous function and executes it inside a new process that will be part of a supervision tree.
Let's give it a try. Open up lib/kv_server/application.ex, and let's change the supervisor in the start/2 function to the following:
  def start(_type, _args) do
    children = [
      {Task, fn -> KVServer.accept(4040) end}
    ]

    opts = [strategy: :one_for_one, name: KVServer.Supervisor]
    Supervisor.start_link(children, opts)
  end
As usual, we've passed a two-element tuple as a child specification, which in turn will invoke Task.start_link/1.
With this change, we are saying that we want to run KVServer.accept(4040) as a task. We are hardcoding the port for now but this could be changed in a few ways, for example, by reading the port out of the system environment when starting the application:
port = String.to_integer(System.get_env("PORT") || "4040")
# ...
{Task, fn -> KVServer.accept(port) end}
Insert these changes in your code and now you may start your application using the following command PORT=4321 mix run --no-halt, notice how we are passing the port as a variable, but still defaults to 4040 if none is given.
Now that the server is part of the supervision tree, it should start automatically when we run the application. Start your server, now passing the port, and once again use the telnet client to make sure that everything still works:
$ telnet 127.0.0.1 4321
Trying 127.0.0.1...
Connected to localhost.
Escape character is '^]'.
say you
say you
say me
say me

Yes, it works! However, does it scale?
Try to connect two telnet clients at the same time. When you do so, you will notice that the second client doesn't echo:
$ telnet 127.0.0.1 4321
Trying 127.0.0.1...
Connected to localhost.
Escape character is '^]'.
hello
hello?
HELLOOOOOO?

It doesn't seem to work at all. That's because we are serving requests in the same process that are accepting connections. When one client is connected, we can't accept another client.

  
    
  
  Task supervisor


In order to make our server handle simultaneous connections, we need to have one process working as an acceptor that spawns other processes to serve requests. One solution would be to change:
defp loop_acceptor(socket) do
  {:ok, client} = :gen_tcp.accept(socket)
  serve(client)
  loop_acceptor(socket)
end
to also use Task.start_link/1:
defp loop_acceptor(socket) do
  {:ok, client} = :gen_tcp.accept(socket)
  Task.start_link(fn -> serve(client) end)
  loop_acceptor(socket)
end
We are starting a linked Task directly from the acceptor process. But we've already made this mistake once. Do you remember?
This is similar to the mistake we made when we called KV.Bucket.start_link/1 straight from the registry. That meant a failure in any bucket would bring the whole registry down.
The code above would have the same flaw: if we link the serve(client) task to the acceptor, a crash when serving a request would bring the acceptor, and consequently all other connections, down.
We fixed the issue for the registry by using a simple one for one supervisor. We are going to use the same tactic here, except that this pattern is so common with tasks that Task already comes with a solution: a simple one for one supervisor that starts temporary tasks as part of our supervision tree.
Let's change start/2 once again, to add a supervisor to our tree:
  def start(_type, _args) do
    port = String.to_integer(System.get_env("PORT") || "4040")

    children = [
      {Task.Supervisor, name: KVServer.TaskSupervisor},
      {Task, fn -> KVServer.accept(port) end}
    ]

    opts = [strategy: :one_for_one, name: KVServer.Supervisor]
    Supervisor.start_link(children, opts)
  end
We'll now start a Task.Supervisor process with name KVServer.TaskSupervisor. Remember, since the acceptor task depends on this supervisor, the supervisor must be started first.
Now we need to change loop_acceptor/1 to use Task.Supervisor to serve each request:
defp loop_acceptor(socket) do
  {:ok, client} = :gen_tcp.accept(socket)
  {:ok, pid} = Task.Supervisor.start_child(KVServer.TaskSupervisor, fn -> serve(client) end)
  :ok = :gen_tcp.controlling_process(client, pid)
  loop_acceptor(socket)
end
You might notice that we added a line, :ok = :gen_tcp.controlling_process(client, pid). This makes the child process the "controlling process" of the client socket. If we didn't do this, the acceptor would bring down all the clients if it crashed because sockets would be tied to the process that accepted them (which is the default behavior).
Start a new server with PORT=4040 mix run --no-halt and we can now open up many concurrent telnet clients. You will also notice that quitting a client does not bring the acceptor down. Excellent!
Here is the full echo server implementation:
defmodule KVServer do
  require Logger

  @doc """
  Starts accepting connections on the given `port`.
  """
  def accept(port) do
    {:ok, socket} = :gen_tcp.listen(port,
                      [:binary, packet: :line, active: false, reuseaddr: true])
    Logger.info "Accepting connections on port #{port}"
    loop_acceptor(socket)
  end

  defp loop_acceptor(socket) do
    {:ok, client} = :gen_tcp.accept(socket)
    {:ok, pid} = Task.Supervisor.start_child(KVServer.TaskSupervisor, fn -> serve(client) end)
    :ok = :gen_tcp.controlling_process(client, pid)
    loop_acceptor(socket)
  end

  defp serve(socket) do
    socket
    |> read_line()
    |> write_line(socket)

    serve(socket)
  end

  defp read_line(socket) do
    {:ok, data} = :gen_tcp.recv(socket, 0)
    data
  end

  defp write_line(line, socket) do
    :gen_tcp.send(socket, line)
  end
end
Since we have changed the supervisor specification, we need to ask: is our supervision strategy still correct?
In this case, the answer is yes: if the acceptor crashes, there is no need to crash the existing connections. On the other hand, if the task supervisor crashes, there is no need to crash the acceptor too.
However, there is still one concern left, which are the restart strategies. Tasks, by default, have the :restart value set to :temporary, which means they are not restarted. This is an excellent default for the connections started via the Task.Supervisor, as it makes no sense to restart a failed connection, but it is a bad choice for the acceptor. If the acceptor crashes, we want to bring the acceptor up and running again.
Let's fix this. We know that for a child of shape {Task, fun}, Elixir will invoke Task.child_spec(fun) to retrieve the underlying child specification. Therefore, one might imagine that to change the {Task, fun} specification to have a :restart of :permanent, we would need to change the Task module. However, that's impossible to do, as the Task module is defined as part of Elixir's standard library (and even if it was possible, it is unlikely it would be a good idea).
Luckily, this can be done by using Supervisor.child_spec/2, which allows us to configure a child specification with new values. Let's rewrite start/2 in KVServer.Application once more:
  def start(_type, _args) do
    port = String.to_integer(System.get_env("PORT") || "4040")

    children = [
      {Task.Supervisor, name: KVServer.TaskSupervisor},
      Supervisor.child_spec({Task, fn -> KVServer.accept(port) end}, restart: :permanent)
    ]

    opts = [strategy: :one_for_one, name: KVServer.Supervisor]
    Supervisor.start_link(children, opts)
  end
Now we have an always running acceptor that starts temporary task processes under an always running task supervisor.
In the next chapter, we will start parsing the client requests and sending responses, finishing our server.

Doctests, patterns, and with
    

In this chapter, we will implement the code that parses the commands we described in the first chapter:
CREATE shopping
OK

PUT shopping milk 1
OK

PUT shopping eggs 3
OK

GET shopping milk
1
OK

DELETE shopping eggs
OK
After the parsing is done, we will update our server to dispatch the parsed commands to the :kv application we built previously.

  
    
  
  Doctests


On the language homepage, we mention that Elixir makes documentation a first-class citizen in the language. We have explored this concept many times throughout this guide, be it via mix help or by typing h Enum or another module in an IEx console.
In this section, we will implement the parsing functionality, document it and make sure our documentation is up to date with doctests. This helps us provide documentation with accurate code samples.
Let's create our command parser at lib/kv_server/command.ex and start with the doctest:
defmodule KVServer.Command do
  @doc ~S"""
  Parses the given `line` into a command.

  ## Examples

      iex> KVServer.Command.parse("CREATE shopping\r\n")
      {:ok, {:create, "shopping"}}

  """
  def parse(_line) do
    :not_implemented
  end
end
Doctests are specified by an indentation of four spaces followed by the iex> prompt in a documentation string. If a command spans multiple lines, you can use ...>, as in IEx. The expected result should start at the next line after iex> or ...> line(s) and is terminated either by a newline or a new iex> prefix.
Also, note that we started the documentation string using @doc ~S""". The ~S prevents the \r\n characters from being converted to a carriage return and line feed until they are evaluated in the test.
To run our doctests, we'll create a file at test/kv_server/command_test.exs and call doctest KVServer.Command in the test case:
defmodule KVServer.CommandTest do
  use ExUnit.Case, async: true
  doctest KVServer.Command
end
Run the test suite and the doctest should fail:
  1) doctest KVServer.Command.parse/1 (1) (KVServer.CommandTest)
     test/kv_server/command_test.exs:3
     Doctest failed
     doctest:
       iex> KVServer.Command.parse("CREATE shopping\r\n")
       {:ok, {:create, "shopping"}}
     code: KVServer.Command.parse "CREATE shopping\r\n" === {:ok, {:create, "shopping"}}
     left:  :not_implemented
     right: {:ok, {:create, "shopping"}}
     stacktrace:
       lib/kv_server/command.ex:7: KVServer.Command (module)
Excellent!
Now let's make the doctest pass. Let's implement the parse/1 function:
def parse(line) do
  case String.split(line) do
    ["CREATE", bucket] -> {:ok, {:create, bucket}}
  end
end
Our implementation splits the line on whitespace and then matches the command against a list. Using String.split/1 means our commands will be whitespace-insensitive. Leading and trailing whitespace won't matter, nor will consecutive spaces between words. Let's add some new doctests to test this behavior along with the other commands:
@doc ~S"""
Parses the given `line` into a command.

## Examples

    iex> KVServer.Command.parse "CREATE shopping\r\n"
    {:ok, {:create, "shopping"}}

    iex> KVServer.Command.parse "CREATE  shopping  \r\n"
    {:ok, {:create, "shopping"}}

    iex> KVServer.Command.parse "PUT shopping milk 1\r\n"
    {:ok, {:put, "shopping", "milk", "1"}}

    iex> KVServer.Command.parse "GET shopping milk\r\n"
    {:ok, {:get, "shopping", "milk"}}

    iex> KVServer.Command.parse "DELETE shopping eggs\r\n"
    {:ok, {:delete, "shopping", "eggs"}}

Unknown commands or commands with the wrong number of
arguments return an error:

    iex> KVServer.Command.parse "UNKNOWN shopping eggs\r\n"
    {:error, :unknown_command}

    iex> KVServer.Command.parse "GET shopping\r\n"
    {:error, :unknown_command}

"""
With doctests at hand, it is your turn to make tests pass! Once you're ready, you can compare your work with our solution below:
def parse(line) do
  case String.split(line) do
    ["CREATE", bucket] -> {:ok, {:create, bucket}}
    ["GET", bucket, key] -> {:ok, {:get, bucket, key}}
    ["PUT", bucket, key, value] -> {:ok, {:put, bucket, key, value}}
    ["DELETE", bucket, key] -> {:ok, {:delete, bucket, key}}
    _ -> {:error, :unknown_command}
  end
end
Notice how we were able to elegantly parse the commands without adding a bunch of if/else clauses that check the command name and number of arguments!
Finally, you may have observed that each doctest corresponds to a different test in our suite, which now reports a total of 7 doctests. That is because ExUnit considers the following to define two different doctests:
iex> KVServer.Command.parse("UNKNOWN shopping eggs\r\n")
{:error, :unknown_command}

iex> KVServer.Command.parse("GET shopping\r\n")
{:error, :unknown_command}
Without new lines, as seen below, ExUnit compiles it into a single doctest:
iex> KVServer.Command.parse("UNKNOWN shopping eggs\r\n")
{:error, :unknown_command}
iex> KVServer.Command.parse("GET shopping\r\n")
{:error, :unknown_command}
As the name says, doctest is documentation first and a test later. Their goal is not to replace tests but to provide up-to-date documentation. You can read more about doctests in the ExUnit.DocTest documentation.

  
    
  
  with


As we are now able to parse commands, we can finally start implementing the logic that runs the commands. Let's add a stub definition for this function for now:
defmodule KVServer.Command do
  @doc """
  Runs the given command.
  """
  def run(command) do
    {:ok, "OK\r\n"}
  end
end
Before we implement this function, let's change our server to start using our new parse/1 and run/1 functions. Remember, our read_line/1 function was also crashing when the client closed the socket, so let's take the opportunity to fix it, too. Open up lib/kv_server.ex and replace the existing server definition:
defp serve(socket) do
  socket
  |> read_line()
  |> write_line(socket)

  serve(socket)
end

defp read_line(socket) do
  {:ok, data} = :gen_tcp.recv(socket, 0)
  data
end

defp write_line(line, socket) do
  :gen_tcp.send(socket, line)
end
by the following:
defp serve(socket) do
  msg =
    case read_line(socket) do
      {:ok, data} ->
        case KVServer.Command.parse(data) do
          {:ok, command} ->
            KVServer.Command.run(command)
          {:error, _} = err ->
            err
        end
      {:error, _} = err ->
        err
    end

  write_line(socket, msg)
  serve(socket)
end

defp read_line(socket) do
  :gen_tcp.recv(socket, 0)
end

defp write_line(socket, {:ok, text}) do
  :gen_tcp.send(socket, text)
end

defp write_line(socket, {:error, :unknown_command}) do
  # Known error; write to the client
  :gen_tcp.send(socket, "UNKNOWN COMMAND\r\n")
end

defp write_line(_socket, {:error, :closed}) do
  # The connection was closed, exit politely
  exit(:shutdown)
end

defp write_line(socket, {:error, error}) do
  # Unknown error; write to the client and exit
  :gen_tcp.send(socket, "ERROR\r\n")
  exit(error)
end
If we start our server, we can now send commands to it. For now, we will get two different responses: "OK" when the command is known and "UNKNOWN COMMAND" otherwise:
$ telnet 127.0.0.1 4040
Trying 127.0.0.1...
Connected to localhost.
Escape character is '^]'.
CREATE shopping
OK
HELLO
UNKNOWN COMMAND

This means our implementation is going in the correct direction, but it doesn't look very elegant, does it?
The previous implementation used pipelines which made the logic straightforward to follow. However, now that we need to handle different error codes along the way, our server logic is nested inside many case calls.
Thankfully, Elixir v1.2 introduced the with construct, which allows you to simplify code like the above, replacing nested case calls with a chain of matching clauses. Let's rewrite the serve/1 function to use with:
defp serve(socket) do
  msg =
    with {:ok, data} <- read_line(socket),
         {:ok, command} <- KVServer.Command.parse(data),
         do: KVServer.Command.run(command)

  write_line(socket, msg)
  serve(socket)
end
Much better! with will retrieve the value returned by the right-side of <- and match it against the pattern on the left side. If the value matches the pattern, with moves on to the next expression. In case there is no match, the non-matching value is returned.
In other words, we converted each expression given to case/2 as a step in with. As soon as any of the steps return something that does not match {:ok, x}, with aborts, and returns the non-matching value.
You can read more about with/1 in our documentation.

  
    
  
  Running commands


The last step is to implement KVServer.Command.run/1, to run the parsed commands against the :kv application. Its implementation is shown below:
@doc """
Runs the given command.
"""
def run(command)

def run({:create, bucket}) do
  KV.Registry.create(KV.Registry, bucket)
  {:ok, "OK\r\n"}
end

def run({:get, bucket, key}) do
  lookup(bucket, fn pid ->
    value = KV.Bucket.get(pid, key)
    {:ok, "#{value}\r\nOK\r\n"}
  end)
end

def run({:put, bucket, key, value}) do
  lookup(bucket, fn pid ->
    KV.Bucket.put(pid, key, value)
    {:ok, "OK\r\n"}
  end)
end

def run({:delete, bucket, key}) do
  lookup(bucket, fn pid ->
    KV.Bucket.delete(pid, key)
    {:ok, "OK\r\n"}
  end)
end

defp lookup(bucket, callback) do
  case KV.Registry.lookup(KV.Registry, bucket) do
    {:ok, pid} -> callback.(pid)
    :error -> {:error, :not_found}
  end
end
Every function clause dispatches the appropriate command to the KV.Registry server that we registered during the :kv application startup. Since our :kv_server depends on the :kv application, it is completely fine to depend on the services it provides.
You might have noticed we have a function head, def run(command), without a body. In the Modules and Functions chapter, we learned that a bodiless function can be used to declare default arguments for a multi-clause function. Here is another use case where we use a function without a body to document what the arguments are.
Note that we have also defined a private function named lookup/2 to help with the common functionality of looking up a bucket and returning its pid if it exists, {:error, :not_found} otherwise.
By the way, since we are now returning {:error, :not_found}, we should amend the write_line/2 function in KVServer to print such error as well:
defp write_line(socket, {:error, :not_found}) do
  :gen_tcp.send(socket, "NOT FOUND\r\n")
end
Our server functionality is almost complete. Only tests are missing. This time, we have left tests for last because there are some important considerations to be made.
KVServer.Command.run/1's implementation is sending commands directly to the server named KV.Registry, which is registered by the :kv application. This means this server is global and if we have two tests sending messages to it at the same time, our tests will conflict with each other (and likely fail). We need to decide between having unit tests that are isolated and can run asynchronously, or writing integration tests that work on top of the global state, but exercise our application's full stack as it is meant to be exercised in production.
So far we have only written unit tests, typically testing a single module directly. However, in order to make KVServer.Command.run/1 testable as a unit we would need to change its implementation to not send commands directly to the KV.Registry process but instead pass a server as an argument. For example, we would need to change run's signature to def run(command, pid) and then change all clauses accordingly:
def run({:create, bucket}, pid) do
  KV.Registry.create(pid, bucket)
  {:ok, "OK\r\n"}
end

# ... other run clauses ...
Feel free to go ahead and do the changes above and write some unit tests. The idea is that your tests will start an instance of the KV.Registry and pass it as an argument to run/2 instead of relying on the global KV.Registry. This has the advantage of keeping our tests asynchronous as there is no shared state.
But let's also try something different. Let's write integration tests that rely on the global server names to exercise the whole stack from the TCP server to the bucket. Our integration tests will rely on global state and must be synchronous. With integration tests, we get coverage on how the components in our application work together at the cost of test performance. They are typically used to test the main flows in your application. For example, we should avoid using integration tests to test an edge case in our command parsing implementation.
Our integration test will use a TCP client that sends commands to our server and assert we are getting the desired responses.
Let's implement the integration test in test/kv_server_test.exs as shown below:
defmodule KVServerTest do
  use ExUnit.Case

  setup do
    Application.stop(:kv)
    :ok = Application.start(:kv)
  end

  setup do
    opts = [:binary, packet: :line, active: false]
    {:ok, socket} = :gen_tcp.connect('localhost', 4040, opts)
    %{socket: socket}
  end

  test "server interaction", %{socket: socket} do
    assert send_and_recv(socket, "UNKNOWN shopping\r\n") ==
           "UNKNOWN COMMAND\r\n"

    assert send_and_recv(socket, "GET shopping eggs\r\n") ==
           "NOT FOUND\r\n"

    assert send_and_recv(socket, "CREATE shopping\r\n") ==
           "OK\r\n"

    assert send_and_recv(socket, "PUT shopping eggs 3\r\n") ==
           "OK\r\n"

    # GET returns two lines
    assert send_and_recv(socket, "GET shopping eggs\r\n") == "3\r\n"
    assert send_and_recv(socket, "") == "OK\r\n"

    assert send_and_recv(socket, "DELETE shopping eggs\r\n") ==
           "OK\r\n"

    # GET returns two lines
    assert send_and_recv(socket, "GET shopping eggs\r\n") == "\r\n"
    assert send_and_recv(socket, "") == "OK\r\n"
  end

  defp send_and_recv(socket, command) do
    :ok = :gen_tcp.send(socket, command)
    {:ok, data} = :gen_tcp.recv(socket, 0, 1000)
    data
  end
end
Our integration test checks all server interaction, including unknown commands and not found errors. It is worth noting that, as with ETS tables and linked processes, there is no need to close the socket. Once the test process exits, the socket is automatically closed.
This time, since our test relies on global data, we have not given async: true to use ExUnit.Case. Furthermore, in order to guarantee our test is always in a clean state, we stop and start the :kv application before each test. In fact, stopping the :kv application even prints a warning on the terminal:
18:12:10.698 [info] Application kv exited: :stopped
To avoid printing log messages during tests, ExUnit provides a neat feature called :capture_log. By setting @tag :capture_log before each test or @moduletag :capture_log for the whole test module, ExUnit will automatically capture anything that is logged while the test runs. In case our test fails, the captured logs will be printed alongside the ExUnit report.
Between use ExUnit.Case and setup, add the following call:
@moduletag :capture_log
In case the test crashes, you will see a report as follows:
  1) test server interaction (KVServerTest)
     test/kv_server_test.exs:17
     ** (RuntimeError) oops
     stacktrace:
       test/kv_server_test.exs:29

     The following output was logged:

     13:44:10.035 [notice] Application kv exited: :stopped
With this simple integration test, we start to see why integration tests may be slow. Not only can this test not run asynchronously, but it also requires the expensive setup of stopping and starting the :kv application.
At the end of the day, it is up to you and your team to figure out the best testing strategy for your applications. You need to balance code quality, confidence, and test suite runtime. For example, we may start with testing the server only with integration tests, but if the server continues to grow in future releases, or it becomes a part of the application with frequent bugs, it is important to consider breaking it apart and writing more intensive unit tests that don't have the weight of an integration test.
Let's move to the next chapter. We will finally make our system distributed by adding a bucket routing mechanism. We will use this opportunity to also improve our testing chops.


  

  
    
    Distributed tasks and tags - Elixir v1.17.3
    
    

    


  
  

    
Distributed tasks and tags
    

In this chapter, we will go back to the :kv application and add a routing layer that will allow us to distribute requests between nodes based on the bucket name.
The routing layer will receive a routing table of the following format:
[
  {?a..?m, :"foo@computer-name"},
  {?n..?z, :"bar@computer-name"}
]
The router will check the first byte of the bucket name against the table and dispatch to the appropriate node based on that. For example, a bucket starting with the letter "a" (?a represents the Unicode codepoint of the letter "a") will be dispatched to node foo@computer-name.
If the matching entry points to the node evaluating the request, then we've finished routing, and this node will perform the requested operation. If the matching entry points to a different node, we'll pass the request to said node, which will look at its own routing table (which may be different from the one in the first node) and act accordingly. If no entry matches, an error will be raised.
Note: we will be using two nodes in the same machine throughout this chapter. You are free to use two (or more) different machines on the same network but you need to do some prep work. First of all, you need to ensure all machines have a ~/.erlang.cookie file with exactly the same value. Then you need to guarantee epmd is running on a port that is not blocked (you can run epmd -d for debug info).


  
    
  
  Our first distributed code


Elixir ships with facilities to connect nodes and exchange information between them. In fact, we use the same concepts of processes, message passing and receiving messages when working in a distributed environment because Elixir processes are location transparent. This means that when sending a message, it doesn't matter if the recipient process is on the same node or on another node, the VM will be able to deliver the message in both cases.
In order to run distributed code, we need to start the VM with a name. The name can be short (when in the same network) or long (requires the full computer address). Let's start a new IEx session:
$ iex --sname foo

You can see now the prompt is slightly different and shows the node name followed by the computer name:
Interactive Elixir - press Ctrl+C to exit (type h() ENTER for help)
iex(foo@jv)1>
My computer is named jv, so I see foo@jv in the example above, but you will get a different result. We will use foo@computer-name in the following examples and you should update them accordingly when trying out the code.
Let's define a module named Hello in this shell:
iex> defmodule Hello do
...>   def world, do: IO.puts "hello world"
...> end
If you have another computer on the same network with both Erlang and Elixir installed, you can start another shell on it. If you don't, you can start another IEx session in another terminal. In either case, give it the short name of bar:
$ iex --sname bar

Note that inside this new IEx session, we cannot access Hello.world/0:
iex> Hello.world
** (UndefinedFunctionError) function Hello.world/0 is undefined (module Hello is not available)
    Hello.world()
However, we can spawn a new process on foo@computer-name from bar@computer-name! Let's give it a try (where @computer-name is the one you see locally):
iex> Node.spawn_link(:"foo@computer-name", fn -> Hello.world() end)
#PID<9014.59.0>
hello world
Elixir spawned a process on another node and returned its PID. The code then executed on the other node where the Hello.world/0 function exists and invoked that function. Note that the result of "hello world" was printed on the current node bar and not on foo. In other words, the message to be printed was sent back from foo to bar. This happens because the process spawned on the other node (foo) knows all the output should be sent back to the original node!
We can send and receive messages from the PID returned by Node.spawn_link/2 as usual. Let's try a quick ping-pong example:
iex> pid = Node.spawn_link(:"foo@computer-name", fn ->
...>   receive do
...>     {:ping, client} -> send(client, :pong)
...>   end
...> end)
#PID<9014.59.0>
iex> send(pid, {:ping, self()})
{:ping, #PID<0.73.0>}
iex> flush()
:pong
:ok
From our quick exploration, we could conclude that we should use Node.spawn_link/2 to spawn processes on a remote node every time we need to do a distributed computation. However, we have learned throughout this guide that spawning processes outside of supervision trees should be avoided if possible, so we need to look for other options.
There are three better alternatives to Node.spawn_link/2 that we could use in our implementation:
	We could use Erlang's :erpc module to execute functions on a remote node. Inside the bar@computer-name shell above, you can call :erpc.call(:"foo@computer-name", Hello, :world, []) and it will print "hello world"

	We could have a server running on the other node and send requests to that node via the GenServer API. For example, you can call a server on a remote node by using GenServer.call({name, node}, arg) or passing the remote process PID as the first argument

	We could use tasks, which we have learned about in a previous chapter, as they can be spawned on both local and remote nodes


The options above have different properties. The GenServer would serialize your requests on a single server, while tasks are effectively running asynchronously on the remote node, with the only serialization point being the spawning done by the supervisor.
For our routing layer, we are going to use tasks, but feel free to explore the other alternatives too.

  
    
  
  async/await


So far we have explored tasks that are started and run in isolation, without regard to their return value. However, sometimes it is useful to run a task to compute a value and read its result later on. For this, tasks also provide the async/await pattern:
task = Task.async(fn -> compute_something_expensive() end)
res  = compute_something_else()
res + Task.await(task)
async/await provides a very simple mechanism to compute values concurrently. Not only that, async/await can also be used with the same Task.Supervisor we have used in previous chapters. We just need to call Task.Supervisor.async/2 instead of Task.Supervisor.start_child/2 and use Task.await/2 to read the result later on.

  
    
  
  Distributed tasks


Distributed tasks are exactly the same as supervised tasks. The only difference is that we pass the node name when spawning the task on the supervisor. Open up lib/kv/supervisor.ex from the :kv application. Let's add a task supervisor as the last child of the tree:
{Task.Supervisor, name: KV.RouterTasks},
Now, let's start two named nodes again, but inside the :kv application:
$ iex --sname foo -S mix
$ iex --sname bar -S mix

From inside bar@computer-name, we can now spawn a task directly on the other node via the supervisor:
iex> task = Task.Supervisor.async({KV.RouterTasks, :"foo@computer-name"}, fn ->
...>   {:ok, node()}
...> end)
%Task{
  mfa: {:erlang, :apply, 2},
  owner: #PID<0.122.0>,
  pid: #PID<12467.88.0>,
  ref: #Reference<0.0.0.400>
}
iex> Task.await(task)
{:ok, :"foo@computer-name"}
Our first distributed task retrieves the name of the node the task is running on. Notice we have given an anonymous function to Task.Supervisor.async/2 but, in distributed cases, it is preferable to give the module, function, and arguments explicitly:
iex> task = Task.Supervisor.async({KV.RouterTasks, :"foo@computer-name"}, Kernel, :node, [])
%Task{
  mfa: {Kernel, :node, 0},
  owner: #PID<0.122.0>,
  pid: #PID<12467.89.0>,
  ref: #Reference<0.0.0.404>
}
iex> Task.await(task)
:"foo@computer-name"
The difference is that anonymous functions require the target node to have exactly the same code version as the caller. Using module, function, and arguments is more robust because you only need to find a function with matching arity in the given module.
With this knowledge in hand, let's finally write the routing code.

  
    
  
  Routing layer


Create a file at lib/kv/router.ex with the following contents:
defmodule KV.Router do
  @doc """
  Dispatch the given `mod`, `fun`, `args` request
  to the appropriate node based on the `bucket`.
  """
  def route(bucket, mod, fun, args) do
    # Get the first byte of the binary
    first = :binary.first(bucket)

    # Try to find an entry in the table() or raise
    entry =
      Enum.find(table(), fn {enum, _node} ->
        first in enum
      end) || no_entry_error(bucket)

    # If the entry node is the current node
    if elem(entry, 1) == node() do
      apply(mod, fun, args)
    else
      {KV.RouterTasks, elem(entry, 1)}
      |> Task.Supervisor.async(KV.Router, :route, [bucket, mod, fun, args])
      |> Task.await()
    end
  end

  defp no_entry_error(bucket) do
    raise "could not find entry for #{inspect bucket} in table #{inspect table()}"
  end

  @doc """
  The routing table.
  """
  def table do
    # Replace computer-name with your local machine name
    [{?a..?m, :"foo@computer-name"}, {?n..?z, :"bar@computer-name"}]
  end
end
Let's write a test to verify our router works. Create a file named test/kv/router_test.exs containing:
defmodule KV.RouterTest do
  use ExUnit.Case, async: true

  test "route requests across nodes" do
    assert KV.Router.route("hello", Kernel, :node, []) ==
             :"foo@computer-name"
    assert KV.Router.route("world", Kernel, :node, []) ==
             :"bar@computer-name"
  end

  test "raises on unknown entries" do
    assert_raise RuntimeError, ~r/could not find entry/, fn ->
      KV.Router.route(<<0>>, Kernel, :node, [])
    end
  end
end
The first test invokes Kernel.node/0, which returns the name of the current node, based on the bucket names "hello" and "world". According to our routing table so far, we should get foo@computer-name and bar@computer-name as responses, respectively.
The second test checks that the code raises for unknown entries.
In order to run the first test, we need to have two nodes running. Move into apps/kv and let's restart the node named bar which is going to be used by tests.
$ iex --sname bar -S mix

And now run tests with:
$ elixir --sname foo -S mix test

The test should pass.

  
    
  
  Test filters and tags


Although our tests pass, our testing structure is getting more complex. In particular, running tests with only mix test causes failures in our suite, since our test requires a connection to another node.
Luckily, ExUnit ships with a facility to tag tests, allowing us to run specific callbacks or even filter tests altogether based on those tags. We have already used the :capture_log tag in the previous chapter, which has its semantics specified by ExUnit itself.
This time let's add a :distributed tag to test/kv/router_test.exs:
@tag :distributed
test "route requests across nodes" do
Writing @tag :distributed is equivalent to writing @tag distributed: true.
With the test properly tagged, we can now check if the node is alive on the network and, if not, we can exclude all distributed tests. Open up test/test_helper.exs inside the :kv application and add the following:
exclude =
  if Node.alive?(), do: [], else: [distributed: true]

ExUnit.start(exclude: exclude)
Now run tests with mix test:
$ mix test
Excluding tags: [distributed: true]

.......

Finished in 0.05 seconds
9 tests, 0 failures, 1 excluded

This time all tests passed and ExUnit warned us that distributed tests were being excluded. If you run tests with $ elixir --sname foo -S mix test, one extra test should run and successfully pass as long as the bar@computer-name node is available.
The mix test command also allows us to dynamically include and exclude tags. For example, we can run $ mix test --include distributed to run distributed tests regardless of the value set in test/test_helper.exs. We could also pass --exclude to exclude a particular tag from the command line. Finally, --only can be used to run only tests with a particular tag:
$ elixir --sname foo -S mix test --only distributed

You can read more about filters, tags, and the default tags in the ExUnit.Case module documentation.

  
    
  
  Wiring it all up


Now with our routing system in place, let's change KVServer to use the router. Replace the lookup/2 function in KVServer.Command from this:
defp lookup(bucket, callback) do
  case KV.Registry.lookup(KV.Registry, bucket) do
    {:ok, pid} -> callback.(pid)
    :error -> {:error, :not_found}
  end
end
by this:
defp lookup(bucket, callback) do
  case KV.Router.route(bucket, KV.Registry, :lookup, [KV.Registry, bucket]) do
    {:ok, pid} -> callback.(pid)
    :error -> {:error, :not_found}
  end
end
Instead of directly looking up the registry, we are using the router instead to match a specific node. Then we get a pid that can be from any process in our cluster. From now on, GET, PUT and DELETE requests are all routed to the appropriate node.
Let's also make sure that when a new bucket is created it ends up on the correct node. Replace the run/1 function in KVServer.Command, the one that matches the :create command, with the following:
def run({:create, bucket}) do
  case KV.Router.route(bucket, KV.Registry, :create, [KV.Registry, bucket]) do
    pid when is_pid(pid) -> {:ok, "OK\r\n"}
    _ -> {:error, "FAILED TO CREATE BUCKET"}
  end
end
Now if you run the tests, you will see that an existing test that checks the server interaction will fail, as it will attempt to use the routing table. To address this failure, change the test_helper.exs for :kv_server application as we did for :kv and add @tag :distributed to this test too:
@tag :distributed
test "server interaction", %{socket: socket} do
However, keep in mind that by making the test distributed, we will likely run it less frequently, since we may not do the distributed setup on every test run. We will learn how to address this in the next chapter, by effectively learning how to make the routing table configurable.

  
    
  
  Summing up


We have only scratched the surface of what is possible when it comes to distribution.
In all of our examples, we relied on Erlang's ability to automatically connect nodes whenever there is a request. For example, when we invoked Node.spawn_link(:"foo@computer-name", fn -> Hello.world() end), Erlang automatically connected to said node and started a new process. However, you may also want to take a more explicit approach to connections, by using Node.connect/1 and Node.disconnect/1.
By default, Erlang establishes a fully meshed network, which means all nodes are connected to each other. Under this topology, the Erlang distribution is known to scale to several dozens of nodes in the same cluster. Erlang also has the concept of hidden nodes, which can allow developers to assemble custom topologies as seen in projects such as Partisan.
In production, you may have nodes connecting and disconnecting at any time. In such scenarios, you need to provide node discoverability. Libraries such as libcluster and dns_cluster provide several strategies for node discoverability using DNS, Kubernetes, etc.
Distributed key-value stores, used in real-life, need to consider the fact nodes may go up and down at any time and also migrate the bucket across nodes. Even further, buckets often need to be duplicated between nodes, so a failure in a node does not lead to the whole bucket being lost. This process is called replication. Our implementation won't attempt to tackle such problems. Instead, we assume there is a fixed number of nodes and therefore use a fixed routing table.
These topics can be daunting at first but remember that most Elixir frameworks abstract those concerns for you. For example, when using the Phoenix web framework, its plug-and-play abstractions take care of sending messages and tracking how users join and leave a cluster. However, if you are interested in distributed systems after all, there is much to explore. Here are some additional references:
	The excellent Distribunomicon chapter from Learn You Some Erlang
	Erlang's global module, which can provide global names and global locks, allowing unique names and unique locks in a whole cluster of machines
	Erlang's pg module, which allows process to join different groups shared across the whole cluster
	Phoenix PubSub project, which provides a distributed messaging system and a distributed presence system for tracking users and processes in a cluster

You will also find many libraries for building distributed systems within the overall Erlang ecosystem. For now, it is time to go back to our simple distributed key-value store and learn how to configure and package it for production.

Configuration and releases
    

In this last guide, we will make the routing table for our distributed key-value store configurable, and then finally package the software for production.
Let's do this.

  
    
  
  Application environment


So far we have hard-coded the routing table into the KV.Router module. However, we would like to make the table dynamic. This allows us not only to configure development/test/production, but also to allow different nodes to run with different entries in the routing table. There is a feature of OTP that does exactly that: the application environment.
Each application has an environment that stores the application's specific configuration by key. For example, we could store the routing table in the :kv application environment, giving it a default value and allowing other applications to change the table as needed.
Open up apps/kv/mix.exs and change the application/0 function to return the following:
def application do
  [
    extra_applications: [:logger],
    env: [routing_table: []],
    mod: {KV, []}
  ]
end
We have added a new :env key to the application. It returns the application default environment, which has an entry of key :routing_table and value of an empty list. It makes sense for the application environment to ship with an empty table, as the specific routing table depends on the testing/deployment structure.
In order to use the application environment in our code, we need to replace KV.Router.table/0 with the definition below:
@doc """
The routing table.
"""
def table do
  Application.fetch_env!(:kv, :routing_table)
end
We use Application.fetch_env!/2 to read the entry for :routing_table in :kv's environment. You can find more information and other functions to manipulate the app environment in the Application module.
Since our routing table is now empty, our distributed tests should fail. Restart the apps and re-run tests to see the failure:
$ iex --sname bar -S mix
$ elixir --sname foo -S mix test --only distributed

We need a way to configure the application environment. That's when we use configuration files.

  
    
  
  Configuration


Configuration files provide a mechanism for us to configure the environment of any application. Elixir provides two configuration entry points:
	config/config.exs — this file is read at build time, before we compile our application and before we even load our dependencies. This means we can't access the code in our application nor in our dependencies. However, it means we can control how they are compiled

	config/runtime.exs — this file is read after our application and dependencies are compiled and therefore it can configure how our application works at runtime. If you want to read system environment variables (via System.get_env/1) or any sort of external configuration, this is the appropriate place to do so


For example, we can configure IEx default prompt to another value. Let's create the config/runtime.exs file with the following content:
import Config
config :iex, default_prompt: ">>>"
Start IEx with iex -S mix and you can see that the IEx prompt has changed.
This means we can also configure our :routing_table directly in the config/runtime.exs file. However, which configuration value should we use?
Currently we have two tests tagged with @tag :distributed. The "server interaction" test in KVServerTest, and the "route requests across nodes" in KV.RouterTest. Both tests are failing since they require a routing table, which is currently empty.
For simplicity, we will define a routing table that always points to the current node. That's the table we will use for development and most of our tests. Back in config/runtime.exs, add this line:
config :kv, :routing_table, [{?a..?z, node()}]
With such a simple table available, we can now remove @tag :distributed from the test in test/kv_server_test.exs. If you run the complete suite, the test should now pass.
However, for the tests in KV.RouterTest, we effectively need two nodes in our routing table. To do so, we will write a setup block that runs before all tests in that file. The setup block will change the application environment and revert it back once we are done, like this:
defmodule KV.RouterTest do
  use ExUnit.Case

  setup_all do
    current = Application.get_env(:kv, :routing_table)

    Application.put_env(:kv, :routing_table, [
      {?a..?m, :"foo@computer-name"},
      {?n..?z, :"bar@computer-name"}
    ])

    on_exit fn -> Application.put_env(:kv, :routing_table, current) end
  end

  @tag :distributed
  test "route requests across nodes" do
Note we removed async: true from use ExUnit.Case. Since the application environment is a global storage, tests that modify it cannot run concurrently. With all changes in place, all tests should pass, including the distributed one.

  
    
  
  Releases


Now that our application runs distributed, you may be wondering how we can package our application to run in production. After all, all of our code so far depends on Erlang and Elixir versions that are installed in your current system. To achieve this goal, Elixir provides releases.
A release is a self-contained directory that consists of your application code, all of its dependencies, plus the whole Erlang Virtual Machine (VM) and runtime. Once a release is assembled, it can be packaged and deployed to a target as long as the target runs on the same operating system (OS) distribution and version as the machine that assembled the release.
In a regular project, we can assemble a release by simply running mix release. However, we have an umbrella project, and in such cases Elixir requires some extra input from us. Let's see what is necessary:
$ MIX_ENV=prod mix release
** (Mix) Umbrella projects require releases to be explicitly defined with a non-empty applications key that chooses which umbrella children should be part of the releases:

releases: [
  foo: [
    applications: [child_app_foo: :permanent]
  ],
  bar: [
    applications: [child_app_bar: :permanent]
  ]
]

Alternatively you can perform the release from the children applications

That's because an umbrella project gives us plenty of options when deploying the software. We can:
	deploy all applications in the umbrella to a node that will work as both TCP server and key-value storage

	deploy the :kv_server application to work only as a TCP server as long as the routing table points only to other nodes

	deploy only the :kv application when we want a node to work only as storage (no TCP access)


As a starting point, let's define a release that includes both :kv_server and :kv applications. We will also add a version to it. Open up the mix.exs in the umbrella root and add inside def project:
releases: [
  foo: [
    version: "0.0.1",
    applications: [kv_server: :permanent, kv: :permanent]
  ]
]
That defines a release named foo with both kv_server and kv applications. Their mode is set to :permanent, which means that, if those applications crash, the whole node terminates. That's reasonable since those applications are essential to our system.
Before we assemble the release, let's also define our routing table for production. Given we expect to have two nodes, we need to update config/runtime.exs to look like this:
import Config

config :kv, :routing_table, [{?a..?z, node()}]

if config_env() == :prod do
  config :kv, :routing_table, [
    {?a..?m, :"foo@computer-name"},
    {?n..?z, :"bar@computer-name"}
  ]
end
We have hard-coded the table and node names, which is good enough for our example, but you would likely move it to an external configuration system in an actual production setup. We have also wrapped it in a config_env() == :prod check, so this configuration does not apply to other environments.
With the configuration in place, let's give assembling the release another try:
$ MIX_ENV=prod mix release foo
* assembling foo-0.0.1 on MIX_ENV=prod
* skipping runtime configuration (config/runtime.exs not found)

Release created at _build/prod/rel/foo!

    # To start your system
    _build/prod/rel/foo/bin/foo start

Once the release is running:

    # To connect to it remotely
    _build/prod/rel/foo/bin/foo remote

    # To stop it gracefully (you may also send SIGINT/SIGTERM)
    _build/prod/rel/foo/bin/foo stop

To list all commands:

    _build/prod/rel/foo/bin/foo

Excellent! A release was assembled in _build/prod/rel/foo. Inside the release, there will be a bin/foo file which is the entry point to your system. It supports multiple commands, such as:
	bin/foo start, bin/foo start_iex, bin/foo restart, and bin/foo stop — for general management of the release

	bin/foo rpc COMMAND and bin/foo remote — for running commands on the running system or to connect to the running system

	bin/foo eval COMMAND — to start a fresh system that runs a single command and then shuts down

	bin/foo daemon and bin/foo daemon_iex — to start the system as a daemon on Unix-like systems

	bin/foo install — to install the system as a service on Windows machines


If you run bin/foo start, it will start the system using a short name (--sname) equal to the release name, which in this case is foo. The next step is to start a system named bar, so we can connect foo and bar together, like we did in the previous chapter. But before we achieve this, let's talk a bit about the benefits of releases.

  
    
  
  Why releases?


Releases allow developers to precompile and package all of their code and the runtime into a single unit. The benefits of releases are:
	Code preloading. The VM has two mechanisms for loading code: interactive and embedded. By default, it runs in the interactive mode which dynamically loads modules when they are used for the first time. The first time your application calls Enum.map/2, the VM will find the Enum module and load it. There's a downside. When you start a new server in production, it may need to load many other modules, causing the first requests to have an unusual spike in response time. Releases run in embedded mode, which loads all available modules upfront, guaranteeing your system is ready to handle requests after booting.

	Configuration and customization. Releases give developers fine grained control over system configuration and the VM flags used to start the system.

	Self-contained. A release does not require the source code to be included in your production artifacts. All of the code is precompiled and packaged. Releases do not even require Erlang or Elixir on your servers, as they include the Erlang VM and its runtime by default. Furthermore, both Erlang and Elixir standard libraries are stripped to bring only the parts you are actually using.

	Multiple releases. You can assemble different releases with different configuration per application or even with different applications altogether.


We have written extensive documentation on releases, so please check the official documentation for more information. For now, we will continue exploring some of the features outlined above.

  
    
  
  Assembling multiple releases


So far, we have assembled a release named foo, but our routing table contains information for both foo and bar. Let's start foo:
$ _build/prod/rel/foo/bin/foo start
16:58:58.508 [info]  Accepting connections on port 4040

And let's connect to it and issue a request in another terminal:
$ telnet 127.0.0.1 4040
Trying 127.0.0.1...
Connected to localhost.
Escape character is '^]'.
CREATE bitsandpieces
OK
PUT bitsandpieces sword 1
OK
GET bitsandpieces sword
1
OK
GET shopping foo
Connection closed by foreign host.

Our application works already when we operate on the bucket named "bitsandpieces". But since the "shopping" bucket would be stored on bar, the request fails as bar is not available. If you go back to the terminal running foo, you will see:
17:16:19.555 [error] Task #PID<0.622.0> started from #PID<0.620.0> terminating
** (stop) exited in: GenServer.call({KV.RouterTasks, :"bar@computer-name"}, {:start_task, [{:"foo@josemac-2", #PID<0.622.0>, #PID<0.622.0>}, [#PID<0.622.0>, #PID<0.620.0>, #PID<0.618.0>], :monitor, {KV.Router, :route, ["shopping", KV.Registry, :lookup, [KV.Registry, "shopping"]]}], :temporary, nil}, :infinity)
    ** (EXIT) no connection to bar@computer-name
    (elixir) lib/gen_server.ex:1010: GenServer.call/3
    (elixir) lib/task/supervisor.ex:454: Task.Supervisor.async/6
    (kv) lib/kv/router.ex:21: KV.Router.route/4
    (kv_server) lib/kv_server/command.ex:74: KVServer.Command.lookup/2
    (kv_server) lib/kv_server.ex:29: KVServer.serve/1
    (elixir) lib/task/supervised.ex:90: Task.Supervised.invoke_mfa/2
    (stdlib) proc_lib.erl:249: :proc_lib.init_p_do_apply/3
Function: #Function<0.128611034/0 in KVServer.loop_acceptor/1>
    Args: []
Let's now define a release for :bar. One first step could be to define a release exactly like foo inside mix.exs. Additionally we will set the cookie option on both releases to weknoweachother in order for them to allow connections from each other. See the Distributed Erlang Documentation for further information on this topic:
releases: [
  foo: [
    version: "0.0.1",
    applications: [kv_server: :permanent, kv: :permanent],
    cookie: "weknoweachother"
  ],
  bar: [
    version: "0.0.1",
    applications: [kv_server: :permanent, kv: :permanent],
    cookie: "weknoweachother"
  ]
]
And now let's assemble both releases:
$ MIX_ENV=prod mix release foo
$ MIX_ENV=prod mix release bar

Stop foo if it's still running and re-start it to load the cookie:
$ _build/prod/rel/foo/bin/foo start

And start bar in another terminal:
$ _build/prod/rel/bar/bin/bar start

You should see an error like the error below happen 5 times, before the application finally shuts down:
    17:21:57.567 [error] Task #PID<0.620.0> started from KVServer.Supervisor terminating
    ** (MatchError) no match of right hand side value: {:error, :eaddrinuse}
        (kv_server) lib/kv_server.ex:12: KVServer.accept/1
        (elixir) lib/task/supervised.ex:90: Task.Supervised.invoke_mfa/2
        (stdlib) proc_lib.erl:249: :proc_lib.init_p_do_apply/3
    Function: #Function<0.98032413/0 in KVServer.Application.start/2>
        Args: []
That's happening because the release foo is already listening on port 4040 and bar is trying to do the same! One option could be to move the :port configuration to the application environment, like we did for the routing table, and setup different ports per node.
But let's try something else. Let's make it so the bar release contains only the :kv application. So it works as a storage but it won't have a front-end. Change the :bar information to this:
releases: [
  foo: [
    version: "0.0.1",
    applications: [kv_server: :permanent, kv: :permanent],
    cookie: "weknoweachother"
  ],
  bar: [
    version: "0.0.1",
    applications: [kv: :permanent],
    cookie: "weknoweachother"
  ]
]
And now let's assemble bar once more:
$ MIX_ENV=prod mix release bar

And finally successfully boot it:
$ _build/prod/rel/bar/bin/bar start

If you connect to localhost once again and perform another request, now everything should work, as long as the routing table contains the correct node names. Outstanding!
With releases, we were able to "cut different slices" of our project and prepared them to run in production, all packaged into a single directory.

  
    
  
  Configuring releases


Releases also provide built-in hooks for configuring almost every need of the production system:
	config/config.exs — provides build-time application configuration, which is executed before our application compiles. This file often imports configuration files based on the environment, such as config/dev.exs and config/prod.exs.

	config/runtime.exs — provides runtime application configuration. It is executed every time the release boots and is further extensible via config providers.

	rel/env.sh.eex and rel/env.bat.eex — template files that are copied into every release and executed on every command to set up environment variables, including ones specific to the VM, and the general environment.

	rel/vm.args.eex — a template file that is copied into every release and provides static configuration of the Erlang Virtual Machine and other runtime flags.


As we have seen, config/config.exs and config/runtime.exs are loaded during releases and regular Mix commands. On the other hand, rel/env.sh.eex and rel/vm.args.eex are specific to releases. Let's take a look.

  
    
  
  Operating System environment configuration


Every release contains an environment file, named env.sh on Unix-like systems and env.bat on Windows machines, that executes before the Elixir system starts. In this file, you can execute any OS-level code, such as invoke other applications, set environment variables and so on. Some of those environment variables can even configure how the release itself runs.
For instance, releases run using short-names (--sname). However, if you want to actually run a distributed key-value store in production, you will need multiple nodes and start the release with the --name option. We can achieve this by setting the RELEASE_DISTRIBUTION environment variable inside the env.sh and env.bat files. Mix already has a template for said files which we can customize, so let's ask Mix to copy them to our application:
$ mix release.init
* creating rel/vm.args.eex
* creating rel/remote.vm.args.eex
* creating rel/env.sh.eex
* creating rel/env.bat.eex

If you open up rel/env.sh.eex, you will see:
#!/bin/sh

# # Sets and enables heart (recommended only in daemon mode)
# case $RELEASE_COMMAND in
#   daemon*)
#     HEART_COMMAND="$RELEASE_ROOT/bin/$RELEASE_NAME $RELEASE_COMMAND"
#     export HEART_COMMAND
#     export ELIXIR_ERL_OPTIONS="-heart"
#     ;;
#   *)
#     ;;
# esac

# # Set the release to load code on demand (interactive) instead of preloading (embedded).
# export RELEASE_MODE=interactive

# # Set the release to work across nodes.
# # RELEASE_DISTRIBUTION must be "sname" (local), "name" (distributed) or "none".
# export RELEASE_DISTRIBUTION=name
# export RELEASE_NODE=<%= @release.name %>

The steps necessary to work across nodes is already commented out as an example. You can enable full distribution by uncommenting the last two lines by removing the leading #.
If you are on Windows, you will have to open up rel/env.bat.eex, where you will find this:
@echo off
rem Set the release to load code on demand (interactive) instead of preloading (embedded).
rem set RELEASE_MODE=interactive

rem Set the release to work across nodes.
rem RELEASE_DISTRIBUTION must be "sname" (local), "name" (distributed) or "none".
rem set RELEASE_DISTRIBUTION=name
rem set RELEASE_NODE=<%= @release.name %>
Once again, uncomment the last two lines by removing the leading rem to enable full distribution. And that's all!

  
    
  
  VM arguments


The rel/vm.args.eex allows you to specify low-level flags that control how the Erlang VM and its runtime operate. You specify entries as if you were specifying arguments in the command line with code comments also supported. Here is the default generated file:
## Customize flags given to the VM: https://www.erlang.org/doc/man/erl.html
## -mode/-name/-sname/-setcookie are configured via env vars, do not set them here

## Increase number of concurrent ports/sockets
##+Q 65536

## Tweak GC to run more often
##-env ERL_FULLSWEEP_AFTER 10
You can see a complete list of VM arguments and flags in the Erlang documentation.

  
    
  
  Summing up


Throughout the guide, we have built a very simple distributed key-value store as an opportunity to explore many constructs like generic servers, supervisors, tasks, agents, applications and more. Not only that, we have written tests for the whole application, got familiar with ExUnit, and learned how to use the Mix build tool to accomplish a wide range of tasks.
If you are looking for a distributed key-value store to use in production, you should definitely look into Riak, which also runs in the Erlang VM. In Riak, the buckets are replicated, to avoid data loss, and instead of a router, they use consistent hashing to map a bucket to a node. A consistent hashing algorithm helps reduce the amount of data that needs to be migrated when new storage nodes are added to your live system.
Of course, Elixir can be used for much more than distributed key-value stores. Embedded systems, data-processing and data-ingestion, web applications, audio/video streaming systems, and others are many of the different domains Elixir excels at. We hope this guide has prepared you to explore any of those domains or any future domain you may desire to bring Elixir into.
Happy coding!

Compatibility and deprecations
    

Elixir is versioned according to a vMAJOR.MINOR.PATCH schema.
Elixir is currently at major version v1. A new backwards compatible minor release happens every 6 months. Patch releases are not scheduled and are made whenever there are bug fixes or security patches.
Elixir applies bug fixes only to the latest minor branch. Security patches are available for the last 5 minor branches:
	Elixir version	Support
	1.17	Bug fixes and security patches
	1.16	Security patches only
	1.15	Security patches only
	1.14	Security patches only
	1.13	Security patches only

New releases are announced in the read-only announcements mailing list. All security releases will be tagged with [security].
There are currently no plans for a major v2 release.

  
    
  
  Between non-major Elixir versions


Elixir minor and patch releases are backwards compatible: well-defined behaviors and documented APIs in a given version will continue working on future versions.
Although we expect the vast majority of programs to remain compatible over time, it is impossible to guarantee that no future change will break any program. Under some unlikely circumstances, we may introduce changes that break existing code:
	Security: a security issue in the implementation may arise whose resolution requires backwards incompatible changes. We reserve the right to address such security issues.

	Bugs: if an API has undesired behavior, a program that depends on the buggy behavior may break if the bug is fixed. We reserve the right to fix such bugs.

	Compiler front-end: improvements may be done to the compiler, introducing new warnings for ambiguous modes and providing more detailed error messages. Those can lead to compilation errors (when running with --warning-as-errors) or tooling failures when asserting on specific error messages (although one should avoid such). We reserve the right to do such improvements.

	Imports: new functions may be added to the Kernel module, which is auto-imported. They may collide with local functions defined in your modules. Collisions can be resolved in a backwards compatible fashion using import Kernel, except: [...] with a list of all functions you don't want to be imported from Kernel. We reserve the right to do such additions.


In order to continue evolving the language without introducing breaking changes, Elixir will rely on deprecations to demote certain practices and promote new ones. Our deprecation policy is outlined in the "Deprecations" section.
The only exception to the compatibility guarantees above are experimental features, which will be explicitly marked as such, and do not provide any compatibility guarantee until they are stabilized.

  
    
  
  Between Elixir and Erlang/OTP


Erlang/OTP versioning is independent from the versioning of Elixir. Erlang releases a new major version yearly. Our goal is to support the last three Erlang major versions by the time Elixir is released. The compatibility table is shown below.
	Elixir version	Supported Erlang/OTP versions
	1.17	25 - 27
	1.16	24 - 26
	1.15	24 - 26
	1.14	23 - 25 (and Erlang/OTP 26 from v1.14.5)
	1.13	22 - 24 (and Erlang/OTP 25 from v1.13.4)
	1.12	22 - 24
	1.11	21 - 23 (and Erlang/OTP 24 from v1.11.4)
	1.10	21 - 22 (and Erlang/OTP 23 from v1.10.3)
	1.9	20 - 22
	1.8	20 - 22
	1.7	19 - 22
	1.6	19 - 20 (and Erlang/OTP 21 from v1.6.6)
	1.5	18 - 20
	1.4	18 - 19 (and Erlang/OTP 20 from v1.4.5)
	1.3	18 - 19
	1.2	18 - 18 (and Erlang/OTP 19 from v1.2.6)
	1.1	17 - 18
	1.0	17 - 17 (and Erlang/OTP 18 from v1.0.5)

Elixir may add compatibility to new Erlang/OTP versions on patch releases, such as support for Erlang/OTP 20 in v1.4.5. Those releases are made for convenience and typically contain the minimum changes for Elixir to run without errors, if any changes are necessary. Only the next minor release, in this case v1.5.0, effectively leverages the new features provided by the latest Erlang/OTP release.

  
    
  
  Deprecations



  
    
  
  Policy


Elixir deprecations happen in 3 steps:
	The feature is soft-deprecated. It means both CHANGELOG and documentation must list the feature as deprecated but no warning is effectively emitted by running the code. There is no requirement to soft-deprecate a feature.

	The feature is effectively deprecated by emitting warnings on usage. This is also known as hard-deprecation. In order to deprecate a feature, the proposed alternative MUST exist for AT LEAST THREE minor versions. For example, Enum.uniq/2 was soft-deprecated in favor of Enum.uniq_by/2 in Elixir v1.1. This means a deprecation warning may only be emitted by Elixir v1.4 or later.

	The feature is removed. This can only happen on major releases. This means deprecated features in Elixir v1.x shall only be removed by Elixir v2.x.



  
    
  
  Table of deprecations


The first column is the version the feature was hard deprecated. The second column shortly describes the deprecated feature and the third column explains the replacement and from which the version the replacement is available from.
	Version	Deprecated feature	Replaced by (available since)
	v1.17	Single-quoted charlists ('foo')	~c"foo" (v1.0)
	v1.17	left..right in patterns and guards	left..right//step (v1.11)
	v1.17	ExUnit.Case.register_test/4	register_test/6 (v1.10)
	v1.17	:all in IO.read/2 and IO.binread/2	:eof (v1.13)
	v1.16	~R/.../	~r/.../ (v1.0)
	v1.16	Ranges with negative steps in Enum.slice/2	Explicit steps in ranges (v1.11)
	v1.16	Ranges with negative steps in String.slice/2	Explicit steps in ranges (v1.11)
	v1.15	Calendar.ISO.day_of_week/3	Calendar.ISO.day_of_week/4 (v1.11)
	v1.15	Exception.exception?/1	Kernel.is_exception/1 (v1.11)
	v1.15	Regex.regex?/1	Kernel.is_struct/2 (Kernel.is_struct(term, Regex)) (v1.11)
	v1.15	Logger.warn/2	Logger.warning/2 (v1.11)
	v1.14	use Bitwise	import Bitwise (v1.0)
	v1.14	~~~/1	bnot/2 (v1.0)
	v1.14	Application.get_env/3 and similar in module body	Application.compile_env/3 (v1.10)
	v1.14	Compiled patterns in String.starts_with?/2	Pass a list of strings instead (v1.0)
	v1.14	Mix.Tasks.Xref.calls/1	Compilation tracers (outlined in Code) (v1.10)
	v1.14	$levelpad in Logger	None
	v1.14	<|> as a custom operator	Another custom operator (v1.0)
	v1.13	! and != in Version requirements	~> or >= (v1.0)
	v1.13	Mix.Config	Config (v1.9)
	v1.13	:strip_beam config to mix escript.build	:strip_beams (v1.9)
	v1.13	Macro.to_string/2	Macro.to_string/1 (v1.0)
	v1.13	System.get_pid/0	System.pid/0 (v1.9)
	v1.12	^^^/2	bxor/2 (v1.0)
	v1.12	@foo() to read module attributes	Remove the parenthesis (v1.0)
	v1.12	use EEx.Engine	Explicitly delegate to EEx.Engine instead (v1.0)
	v1.12	:xref compiler in Mix	Nothing (it always runs as part of the compiler now)
	v1.11	Mix.Project.compile/2	Mix.Task.run("compile", args) (v1.0)
	v1.11	Supervisor.Spec.worker/3 and Supervisor.Spec.supervisor/3	The new child specs outlined in Supervisor (v1.5)
	v1.11	Supervisor.start_child/2 and Supervisor.terminate_child/2	DynamicSupervisor (v1.6)
	v1.11	System.stacktrace/1	__STACKTRACE__ in try/catch/rescue (v1.7)
	v1.10	Code.ensure_compiled?/1	Code.ensure_compiled/1 (v1.0)
	v1.10	Code.load_file/2	Code.require_file/2 (v1.0) or Code.compile_file/2 (v1.7)
	v1.10	Code.loaded_files/0	Code.required_files/0 (v1.7)
	v1.10	Code.unload_file/1	Code.unrequire_files/1 (v1.7)
	v1.10	Passing non-chardata to Logger.log/2	Explicitly convert to string with to_string/1 (v1.0)
	v1.10	:compile_time_purge_level in Logger app environment	:compile_time_purge_matching in Logger app environment (v1.7)
	v1.10	Supervisor.Spec.supervise/2	The new child specs outlined in Supervisor (v1.5)
	v1.10	:simple_one_for_one strategy in Supervisor	DynamicSupervisor (v1.6)
	v1.10	:restart and :shutdown in Task.Supervisor.start_link/1	:restart and :shutdown in Task.Supervisor.start_child/3 (v1.6)
	v1.9	Enumerable keys in Map.drop/2, Map.split/2, and Map.take/2	Call Enum.to_list/1 on the second argument before hand (v1.0)
	v1.9	Mix.Project.load_paths/1	Mix.Project.compile_path/1 (v1.0)
	v1.9	Passing :insert_replaced to String.replace/4	Use :binary.replace/4 (v1.0)
	v1.8	Passing a non-empty list to Collectable.into/1	++/2 or Keyword.merge/2 (v1.0)
	v1.8	Passing a non-empty list to :into in for/1	++/2 or Keyword.merge/2 (v1.0)
	v1.8	Passing a non-empty list to Enum.into/2	++/2 or Keyword.merge/2 (v1.0)
	v1.8	Time units in its plural form, such as: :seconds, :milliseconds, and the like	Use the singular form, such as: :second, :millisecond, and so on (v1.4)
	v1.8	Inspect.Algebra.surround/3	Inspect.Algebra.concat/2 and Inspect.Algebra.nest/2 (v1.0)
	v1.8	Inspect.Algebra.surround_many/6	Inspect.Algebra.container_doc/6 (v1.6)
	v1.9	--detached in Kernel.CLI	--erl "-detached" (v1.0)
	v1.8	Kernel.ParallelCompiler.files/2	Kernel.ParallelCompiler.compile/2 (v1.6)
	v1.8	Kernel.ParallelCompiler.files_to_path/2	Kernel.ParallelCompiler.compile_to_path/2 (v1.6)
	v1.8	Kernel.ParallelRequire.files/2	Kernel.ParallelCompiler.require/2 (v1.6)
	v1.8	Returning {:ok, contents} or :error from Mix.Compilers.Erlang.compile/6's callback	Return {:ok, contents, warnings} or {:error, errors, warnings} (v1.6)
	v1.8	System.cwd/0 and System.cwd!/0	File.cwd/0 and File.cwd!/0 (v1.0)
	v1.7	Code.get_docs/2	Code.fetch_docs/1 (v1.7)
	v1.7	Enum.chunk/2,3,4	Enum.chunk_every/2 and Enum.chunk_every/3,4 (v1.5)
	v1.7	Calling super/1 inGenServer callbacks	Implementing the behaviour explicitly without calling super/1 (v1.0)
	v1.7	not left in right	left not in right (v1.5)
	v1.7	Registry.start_link/3	Registry.start_link/1 (v1.5)
	v1.7	Stream.chunk/2,3,4	Stream.chunk_every/2 and Stream.chunk_every/3,4 (v1.5)
	v1.6	Enum.partition/2	Enum.split_with/2 (v1.4)
	v1.6	Macro.unescape_tokens/1,2	Use Enum.map/2 to traverse over the arguments (v1.0)
	v1.6	Module.add_doc/6	@doc module attribute (v1.0)
	v1.6	Range.range?/1	Pattern match on _.._ (v1.0)
	v1.5	() to mean nil	nil (v1.0)
	v1.5	char_list/0 type	charlist/0 type (v1.3)
	v1.5	Atom.to_char_list/1	Atom.to_charlist/1 (v1.3)
	v1.5	Enum.filter_map/3	Enum.filter/2 + Enum.map/2 or for/1 comprehensions (v1.0)
	v1.5	Float.to_char_list/1	Float.to_charlist/1 (v1.3)
	v1.5	GenEvent module	Supervisor and GenServer (v1.0);<br/>GenStage (v1.3);<br/>:gen_event (Erlang/OTP 17)
	v1.5	<%= in middle and end expressions in EEx	Use <% (<%= is allowed only in start expressions) (v1.0)
	v1.5	:as_char_lists value in Inspect.Opts.t/0 type	:as_charlists value (v1.3)
	v1.5	:char_lists key in Inspect.Opts.t/0 type	:charlists key (v1.3)
	v1.5	Integer.to_char_list/1,2	Integer.to_charlist/1 and Integer.to_charlist/2 (v1.3)
	v1.5	to_char_list/1	to_charlist/1 (v1.3)
	v1.5	List.Chars.to_char_list/1	List.Chars.to_charlist/1 (v1.3)
	v1.5	@compile {:parse_transform, _} in Module	None
	v1.5	Stream.filter_map/3	Stream.filter/2 + Stream.map/2 (v1.0)
	v1.5	String.ljust/3 and String.rjust/3	Use String.pad_leading/3 and String.pad_trailing/3 with a binary padding (v1.3)
	v1.5	String.lstrip/1 and String.rstrip/1	String.trim_leading/1 and String.trim_trailing/1 (v1.3)
	v1.5	String.lstrip/2 and String.rstrip/2	Use String.trim_leading/2 and String.trim_trailing/2 with a binary as second argument (v1.3)
	v1.5	String.strip/1 and String.strip/2	String.trim/1 and String.trim/2 (v1.3)
	v1.5	String.to_char_list/1	String.to_charlist/1 (v1.3)
	v1.4	Anonymous functions with no expression after ->	Use an expression or explicitly return nil (v1.0)
	v1.4	Support for making private functions overridable	Use public functions (v1.0)
	v1.4	Variable used as function call	Use parentheses (v1.0)
	v1.4	Access.key/1	Access.key/2 (v1.3)
	v1.4	Behaviour module	@callback module attribute (v1.0)
	v1.4	Enum.uniq/2	Enum.uniq_by/2 (v1.2)
	v1.4	Float.to_char_list/2	:erlang.float_to_list/2 (Erlang/OTP 17)
	v1.4	Float.to_string/2	:erlang.float_to_binary/2 (Erlang/OTP 17)
	v1.4	HashDict module	Map (v1.2)
	v1.4	HashSet module	MapSet (v1.1)
	v1.4	IEx.Helpers.import_file/2	IEx.Helpers.import_file_if_available/1 (v1.3)
	v1.4	Mix.Utils.camelize/1	Macro.camelize/1 (v1.2)
	v1.4	Mix.Utils.underscore/1	Macro.underscore/1 (v1.2)
	v1.4	Multi-letter aliases in OptionParser	Use single-letter aliases (v1.0)
	v1.4	Set module	MapSet (v1.1)
	v1.4	Stream.uniq/2	Stream.uniq_by/2 (v1.2)
	v1.3	\x{X*} inside strings/sigils/charlists	\uXXXX or \u{X*} (v1.1)
	v1.3	Dict module	Keyword (v1.0) or Map (v1.2)
	v1.3	:append_first option in defdelegate/2	Define the function explicitly (v1.0)
	v1.3	Map/dictionary as 2nd argument in Enum.group_by/3	Enum.reduce/3 (v1.0)
	v1.3	Keyword.size/1	length/1 (v1.0)
	v1.3	Map.size/1	map_size/1 (v1.0)
	v1.3	/r option in Regex	/U (v1.1)
	v1.3	Set behaviour	MapSet data structure (v1.1)
	v1.3	String.valid_character?/1	String.valid?/1 (v1.0)
	v1.3	Task.find/2	Use direct message matching (v1.0)
	v1.3	Non-map as 2nd argument in URI.decode_query/2	Use a map (v1.0)
	v1.2	Dict behaviour	Map and Keyword (v1.0)
	v1.1	?\xHEX	0xHEX (v1.0)
	v1.1	Access protocol	Access behaviour (v1.1)
	v1.1	as: true | false in alias/2 and require/2	None

Gradual set-theoretic types
    

Elixir is in the process of incorporating set-theoretic types into the compiler. This document outlines the current stage of our implementation.
The current milestone aims to infer types from patterns and guards and use them to type check programs, enabling the Elixir compiler to find faults and bugs in codebases without requiring changes to existing software. The underlying principles, theory, and roadmap of our work have been outlined in "The Design Principles of the Elixir Type System" by Giuseppe Castagna, Guillaume Duboc, José Valim.

  
    
  
  Supported types


At the moment, Elixir developers will interact with set-theoretic types only through warnings found by the type system. All data types in the language are modelled:
	binary(), integer(), float(), pid(), port(), reference() - these types are indivisible. This means both 1 and 13 get the same integer() type.

	atom() - it represents all atoms and it is divisible. For instance, the atom :foo and :hello_world are also valid (distinct) types.

	map() and structs - maps can be "closed" or "open". Closed maps only allow the specified keys, such as %{key: atom(), value: integer()}. Open maps support any other keys in addition to the ones listed and their definition starts with ..., such as %{..., key: atom(), value: integer()}. Structs are closed maps with the __struct__ key.

	tuple(), list(), and function() - currently they are modelled as indivisible types. The next Elixir versions will also introduce fine-grained types here.



  
    
  
  Set operations


We can compose set-theoretic types by using set operations (hence the name). For example, to say a function returns either atoms or integers, one could write: atom() or integer().
Intersections are available via the and operator, such as atom() and integer(), which in this case it becomes the empty set none(). term() is the union of all types, also known as the "top" type.
Intersections are useful when modelling functions. For example, imagine the following function:
def negate(x) when is_integer(x), do: -x
def negate(x) when is_boolean(x), do: not x
If you give it an integer, it negates it. If you give it a boolean, it negates it.
We can say this function has the type (integer() -> integer()) because it is capable of receiving an integer and returning an integer. In this case, (integer() -> integer()) is a set that represents all functions that can receive an integer and return an integer. Even though this function can receive other arguments and return other values, it is still part of the (integer() -> integer()) set.
This function also has the type (boolean() -> boolean()), because it receives the booleans and returns booleans. Therefore, we can say the overall type of the function is (integer() -> integer()) and (boolean() -> boolean()). The intersection means the function belongs to both sets.
At this point, some may ask, why not a union? As a real-world example, take a t-shirt with green and yellow stripes. We can say the t-shirt belongs to the set of "t-shirts with green color". We can also say the t-shirt belongs to the set of "t-shirts with yellow color". Let's see the difference between unions and intersections:
	(t_shirts_with_green() or t_shirts_with_yellow()) - contains t-shirts with either green or yellow, such as green, green and red, green and yellow, yellow, yellow and red, etc.

	(t_shirts_with_green() and t_shirts_with_yellow()) - contains t-shirts with both green and yellow (and also other colors)


Since the t-shirt has both colors, we say it belongs to the intersection of both sets. The same way that a function that goes from (integer() -> integer()) and (boolean() -> boolean()) is also an intersection. In practice, it does not make sense to define the union of two functions in Elixir, so the compiler will always point to the right direction.
Finally, we can also negate types by using not. For example, to express all atoms, except the atoms :foo and :bar, one can write: atom() and not (:foo or :bar).

  
    
  
  The dynamic() type


Existing Elixir programs do not have type declarations, but we still want to be able to type check them. This is done with the introduction of the dynamic() type.
When Elixir sees the following function:
def negate(x) when is_integer(x), do: -x
def negate(x) when is_boolean(x), do: not x
Elixir type checks it as if the function had the type (dynamic() -> dynamic()). We say dynamic() is a gradual type, which leads us to gradual set-theoretic types.
The simplest way to reason about dynamic() in Elixir is that it is a range of types. If you have a type atom() or integer(), the underlying code needs to work with both atom() or integer(). For example, if you call Integer.to_string(var), and var has type atom() or integer(), the type system will emit a warning, because Integer.to_string/1 does not accept atoms.
However, by intersecting a type with dynamic(), we make the type gradual and therefore only a subset of the type needs to be valid. For instance, if you call Integer.to_string(var), and var has type dynamic() and (atom() or integer()), the type system will not emit a warning, because Integer.to_string/1 works with at least one of the types. For convenience, most programs will write dynamic(atom() or integer()) instead of the intersection. They are equivalent.
Compared to other gradually typed languages, the dynamic() type in Elixir is quite powerful: it restricts our program to certain types, via intersections, while still emitting warnings once it is certain the code will fail. This makes dynamic() an excellent tool for typing existing Elixir code with meaningful warnings.
Once Elixir introduces typed function signatures, typed Elixir programs will behave as a statically typed code, unless the dynamic() type is used. This brings us to one last remark about dynamic types in Elixir: dynamic types are always at the root. For example, when you write a tuple of type {:ok, dynamic()}, Elixir will rewrite it to dynamic({:ok, term()}). While this has the downside that you cannot make part of a tuple/map/list gradual, only the whole tuple/map/list, it comes with the upside that dynamic is always explicitly at the root, making it harder to accidentally sneak dynamic() in a statically typed program.

  
    
  
  Roadmap


The current milestone is to implement type inference and type checking of Elixir programs without changes to the Elixir language. At this stage, we want to collect feedback on the quality of error messages and performance, and therefore the type system has no user facing API.
If the results are satisfactory, the next milestone will include a mechanism for defining typed structs. Elixir programs frequently pattern match on structs, which reveals information about the struct fields, but it knows nothing about their respective types. By propagating types from structs and their fields throughout the program, we will increase the type system’s ability to find errors while further straining our type system implementation. Proposals including the required changes to the language surface will be sent to the community once we reach this stage.
The third milestone is to introduce set-theoretic type signatures for functions. Unfortunately, the existing Erlang Typespecs are not precise enough for set-theoretic types and they will be phased out of the language and have their postprocessing moved into a separate library once this stage concludes.

Library guidelines
    

This document outlines general guidelines for those writing and publishing
Elixir libraries meant to be consumed by other developers.

  
    
  
  Getting started


You can create a new Elixir library by running the mix new command:
$ mix new my_library

The project name is given in the snake_case convention where all letters are lowercase and words are separate with underscores. This is the same convention used by variables, function names and atoms in Elixir. See the Naming Conventions document for more information.
Every project has a mix.exs file, with instructions on how to build, compile, run tests, and so on. Libraries commonly have a lib directory, which includes Elixir source code, and a test directory. A src directory may also exist for Erlang sources.
The mix new command also allows the --sup option to scaffold a new project with a supervision tree out of the box. For more information on running your project, see the official Mix & OTP guide or Mix documentation.

  
    
  
  Publishing


Writing code is only the first of many steps to publish a package. We strongly recommend developers to:
	Choose a versioning schema. Elixir requires versions to be in the format MAJOR.MINOR.PATCH but the meaning of those numbers is up to you. Most projects choose Semantic Versioning.

	Choose a license. The most common licenses in the Elixir community are the MIT License and the Apache License 2.0. The latter is also the one used by Elixir itself.

	Run the code formatter. The code formatter formats your code according to a consistent style shared by your library and the whole community, making it easier for other developers to understand your code and contribute.

	Write tests. Elixir ships with a test-framework named ExUnit. The project generated by mix new includes sample tests and doctests.

	Write documentation. The Elixir community is proud of treating documentation as a first-class citizen and making documentation easily accessible. Libraries contribute to the status quo by providing complete API documentation with examples for their modules, types and functions. See the Writing documentation chapter of the Getting Started guide for more information. Projects like ExDoc can be used to generate HTML and EPUB documents from the documentation. ExDoc also supports "extra pages", like this one that you are reading. Such pages augment the documentation with tutorials, guides, references, and even cheat-sheets.

	Follow best practices. The Elixir project documents a series of anti-patterns that you may want to avoid in your code. The process-related anti-patterns and meta-programming anti-patterns are of special attention to library authors.


Projects are often made available to other developers by publishing a Hex package. Hex also supports private packages for organizations. If ExDoc is configured for the Mix project, publishing a package on Hex will also automatically publish the generated documentation to HexDocs.

  
    
  
  Dependency handling


When your library is published and used as a dependency, its lockfile (usually named mix.lock) is ignored by the host project. Running mix deps.get in the host project attempts to get the latest possible versions of your library’s dependencies, as specified by the requirements in the deps section of your mix.exs. These versions might be greater than those stored in your mix.lock (and hence used in your tests / CI).
On the other hand, contributors of your library, need a deterministic build, which implies the presence of mix.lock in your Version Control System (VCS).
The best practice of handling mix.lock file therefore would be to keep it in VCS, and run two different Continuous Integration (CI) workflows: the usual deterministic one, and another one, that starts with mix deps.unlock --all and always compiles your library and runs tests against latest versions of dependencies. The latter one might be even run nightly or otherwise recurrently to stay notified about any possible issue in regard to dependencies updates.

Naming conventions
    

This document is a reference of the naming conventions in Elixir, from casing to punctuation characters.
The naming convention is, by definition, a subset of the Elixir syntax. A convention aims to
follow and set best practices for language and the community. If instead you want a complete reference into the Elixir syntax, beyond its conventions, see the Syntax reference.

  
    
  
  Casing


Elixir developers must use snake_case when defining variables, function names, module attributes, and the like:
some_map = %{this_is_a_key: "and a value"}
is_map(some_map)
Aliases, commonly used as module names, are an exception as they must be capitalized and written in CamelCase, like OptionParser. For aliases, capital letters are kept in acronyms, like ExUnit.CaptureIO or Mix.SCM.
Atoms can be written either in :snake_case or :CamelCase, although the convention is to use the snake case version throughout Elixir.
Generally speaking, filenames follow the snake_case convention of the module they define. For example, MyApp should be defined inside the my_app.ex file. However, this is only a convention. At the end of the day any filename can be used as they do not affect the compiled code in any way.

  
    
  
  Underscore (_foo)


Elixir relies on underscores in different situations.
For example, a value that is not meant to be used must be assigned to _ or to a variable starting with underscore:
iex> {:ok, _contents} = File.read("README.md")
Function names may also start with an underscore. Such functions are never imported by default:
iex> defmodule Example do
...>   def _wont_be_imported do
...>     :oops
...>   end
...> end

iex> import Example
iex> _wont_be_imported()
** (CompileError) iex:1: undefined function _wont_be_imported/0
Due to this property, Elixir relies on functions starting with underscore to attach compile-time metadata to modules. Such functions are most often in the __foo__ format. For example, every module in Elixir has an __info__/1 function:
iex> String.__info__(:functions)
[at: 2, capitalize: 1, chunk: 2, ...]
Elixir also includes five special forms that follow the double underscore format: __CALLER__/0, __DIR__/0, __ENV__/0and __MODULE__/0 retrieve compile-time information about the current environment, while __STACKTRACE__/0 retrieves the stacktrace for the current exception.

  
    
  
  Trailing bang (foo!)


A trailing bang (exclamation mark) signifies a function or macro where failure cases raise an exception.
Many functions come in pairs, such as File.read/1 and File.read!/1. File.read/1 will return a success or failure tuple, whereas File.read!/1 will return a plain value or else raise an exception:
iex> File.read("file.txt")
{:ok, "file contents"}
iex> File.read("no_such_file.txt")
{:error, :enoent}

iex> File.read!("file.txt")
"file contents"
iex> File.read!("no_such_file.txt")
** (File.Error) could not read file no_such_file.txt: no such file or directory
The version without ! is preferred when you want to handle different outcomes using pattern matching:
case File.read(file) do
  {:ok, body} -> # do something with the `body`
  {:error, reason} -> # handle the error caused by `reason`
end
However, if you expect the outcome to always be successful (for instance, if you expect the file always to exist), the bang variation can be more convenient and will raise a more helpful error message (than a failed pattern match) on failure.
When thinking about failure cases for functions, we are thinking strictly about errors that happen within their domain, such as failing to open a file. Errors that come from invalid argument types, for example, must always raise regardless if the function has a bang or not. The exception is often an ArgumentError or a detailed FunctionClauseError:
iex(1)> File.read(123)
** (FunctionClauseError) no function clause matching in IO.chardata_to_string/1

    The following arguments were given to IO.chardata_to_string/1:

        # 1
        123

    Attempted function clauses (showing 2 out of 2):

        def chardata_to_string(string) when is_binary(string)
        def chardata_to_string(list) when is_list(list)
More examples of paired functions: Base.decode16/2 and Base.decode16!/2, File.cwd/0 and File.cwd!/0.
There are also some non-paired functions, with no non-bang variant. The bang still signifies that it will raise an exception on failure. Example: Protocol.assert_protocol!/1.
In macro code, the bang on alias!/1 and var!/2 signifies that macro hygiene is set aside.

  
    
  
  Trailing question mark (foo?)


Functions that return a boolean are named with a trailing question mark.
Examples: Keyword.keyword?/1, Mix.debug?/0, String.contains?/2
However, functions that return booleans and are valid in guards follow another convention, described next.

  
    
  
  is_ prefix (is_foo)


Type checks and other boolean checks that are allowed in guard clauses are named with an is_ prefix.
Examples: Integer.is_even/1, is_list/1
These functions and macros follow the Erlang convention of an is_ prefix, instead of a trailing question mark, precisely to indicate that they are allowed in guard clauses.
Note that type checks that are not valid in guard clauses do not follow this convention. For example: Keyword.keyword?/1.

  
    
  
  Special names


Some names have specific meaning in Elixir. We detail those cases below.

  
    
  
  length and size


When you see size in a function name, it means the operation runs in constant time (also written as "O(1) time") because the size is stored alongside the data structure.
Examples: map_size/1, tuple_size/1
When you see length, the operation runs in linear time ("O(n) time") because the entire data structure has to be traversed.
Examples: length/1, String.length/1
In other words, functions using the word "size" in its name will take the same amount of time whether the data structure is tiny or huge. Conversely, functions having "length" in its name will take more time as the data structure grows in size.

  
    
  
  get, fetch, fetch!


When you see the functions get, fetch, and fetch! for key-value data structures, you can expect the following behaviours:
	get returns a default value (which itself defaults to nil) if the key is not present, or returns the requested value.
	fetch returns :error if the key is not present, or returns {:ok, value} if it is.
	fetch! raises if the key is not present, or returns the requested value.

Examples: Map.get/2, Map.fetch/2, Map.fetch!/2, Keyword.get/2, Keyword.fetch/2, Keyword.fetch!/2

  
    
  
  compare


The function compare/2 should return :lt if the first term is less than the second, :eq if the two
terms compare as equivalent, or :gt if the first term is greater than the second.
Examples: DateTime.compare/2
Note that this specific convention is important due to the expectations of Enum.sort/2

Operators reference
    

This document is a complete reference of operators in Elixir, how they are parsed, how they can be defined, and how they can be overridden.

  
    
  
  Operator precedence and associativity


The following is a list of all operators that Elixir is capable of parsing, ordered from higher to lower precedence, alongside their associativity:
	Operator	Associativity
	@	Unary
	.	Left
	+ - ! ^ not	Unary
	**	Left
	* /	Left
	+ -	Left
	++ -- +++ --- .. <>	Right
	in not in	Left
	|> <<< >>> <<~ ~>> <~ ~> <~>	Left
	< > <= >=	Left
	== != =~ === !==	Left
	&& &&& and	Left
	|| ||| or	Left
	=	Right
	&, ...	Unary
	=> (valid only inside %{})	Right
	|	Right
	::	Right
	when	Right
	<- \\	Left


  
    
  
  General operators


Elixir provides the following built-in operators:
	+ and - - unary positive/negative
	+, -, *, and / - basic arithmetic operations
	++ and -- - list concatenation and subtraction
	and and 

Patterns and guards
    

Elixir provides pattern matching, which allows us to assert on the shape or extract values from data structures. Patterns are often augmented with guards, which give developers the ability to perform more complex checks, albeit limited.
This document provides a complete reference on patterns and guards, their semantics, where they are allowed, and how to extend them.

  
    
  
  Patterns


Patterns in Elixir are made of variables, literals, and data structure specific syntax. One of the most used constructs to perform pattern matching is the match operator (=):
iex> x = 1
1
iex> 1 = x
1
In the example above, x starts without a value and has 1 assigned to it. Then, we compare the value of x to the literal 1, which succeeds as they are both 1.
Matching x against 2 would raise:
iex> 2 = x
** (MatchError) no match of right hand side value: 1
Patterns are not bidirectional. If you have a variable y that was never assigned to (often called an unbound variable) and you write 1 = y, an error will be raised:
iex> 1 = y
** (CompileError) iex:2: undefined variable "y"
In other words, patterns are allowed only on the left side of =. The right side of = follows the regular evaluation semantics of the language.
Now let's cover the pattern matching rules for each construct and then for each relevant data types.

  
    
  
  Variables


Variables in patterns are always assigned to:
iex> x = 1
1
iex> x = 2
2
iex> x
2
In other words, Elixir supports rebinding. In case you don't want the value of a variable to change, you can use the pin operator (^):
iex> x = 1
1
iex> ^x = 2
** (MatchError) no match of right hand side value: 2
If the same variable appears multiple times in the same pattern, then all of them must be bound to the same value:
iex> {x, x} = {1, 1}
{1, 1}
iex> {x, x} = {1, 2}
** (MatchError) no match of right hand side value: {1, 2}
The underscore variable (_) has a special meaning as it can never be bound to any value. It is especially useful when you don't care about certain value in a pattern:
iex> {_, integer} = {:not_important, 1}
{:not_important, 1}
iex> integer
1
iex> _
** (CompileError) iex:3: invalid use of _

  
    
  
  Literals (numbers and atoms)


Atoms and numbers (integers and floats) can appear in patterns and they are always represented as is. For example, an atom will only match an atom if they are the same atom:
iex> :atom = :atom
:atom
iex> :atom = :another_atom
** (MatchError) no match of right hand side value: :another_atom
Similar rule applies to numbers. Finally, note that numbers in patterns perform strict comparison. In other words, integers to do not match floats:
iex> 1 = 1.0
** (MatchError) no match of right hand side value: 1.0

  
    
  
  Tuples


Tuples may appear in patterns using the curly brackets syntax ({}). A tuple in a pattern will match only tuples of the same size, where each individual tuple element must also match:
iex> {:ok, integer} = {:ok, 13}
{:ok, 13}

# won't match due to different size
iex> {:ok, integer} = {:ok, 11, 13}
** (MatchError) no match of right hand side value: {:ok, 11, 13}

# won't match due to mismatch on first element
iex> {:ok, binary} = {:error, :enoent}
** (MatchError) no match of right hand side value: {:error, :enoent}

  
    
  
  Lists


Lists may appear in patterns using the square brackets syntax ([]). A list in a pattern will match only lists of the same size, where each individual list element must also match:
iex> [:ok, integer] = [:ok, 13]
[:ok, 13]

# won't match due to different size
iex> [:ok, integer] = [:ok, 11, 13]
** (MatchError) no match of right hand side value: [:ok, 11, 13]

# won't match due to mismatch on first element
iex> [:ok, binary] = [:error, :enoent]
** (MatchError) no match of right hand side value: [:error, :enoent]
Opposite to tuples, lists also allow matching on non-empty lists by using the [head | tail] notation, which matches on the head and tail of a list:
iex> [head | tail] = [1, 2, 3]
[1, 2, 3]
iex> head
1
iex> tail
[2, 3]
Multiple elements may prefix the | tail construct:
iex> [first, second | tail] = [1, 2, 3]
[1, 2, 3]
iex> tail
[3]
Note [head | tail] does not match empty lists:
iex> [head | tail] = []
** (MatchError) no match of right hand side value: []
Given charlists are represented as a list of integers, one can also perform prefix matches on charlists using the list concatenation operator (++):
iex> ~c"hello " ++ world = ~c"hello world"
~c"hello world"
iex> world
~c"world"
Which is equivalent to matching on [?h, ?e, ?l, ?l, ?o, ?\s | world]. Suffix matches (hello ++ ~c" world") are not valid patterns.

  
    
  
  Maps


Maps may appear in patterns using the percentage sign followed by the curly brackets syntax (%{}). Opposite to lists and tuples, maps perform a subset match. This means a map pattern will match any other map that has at least all of the keys in the pattern.
Here is an example where all keys match:
iex> %{name: name} = %{name: "meg"}
%{name: "meg"}
iex> name
"meg"
Here is when a subset of the keys match:
iex> %{name: name} = %{name: "meg", age: 23}
%{age: 23, name: "meg"}
iex> name
"meg"
If a key in the pattern is not available in the map, then they won't match:
iex> %{name: name, age: age} = %{name: "meg"}
** (MatchError) no match of right hand side value: %{name: "meg"}
Note that the empty map will match all maps, which is a contrast to tuples and lists, where an empty tuple or an empty list will only match empty tuples and empty lists respectively:
iex> %{} = %{name: "meg"}
%{name: "meg"}
Finally, note map keys in patterns must always be literals or previously bound variables matched with the pin operator.

  
    
  
  Structs


Structs may appear in patterns using the percentage sign, the struct module name or a variable followed by the curly brackets syntax (%{}).
Given the following struct:
defmodule User do
  defstruct [:name]
end
Here is an example where all keys match:
iex> %User{name: name} = %User{name: "meg"}
%User{name: "meg"}
iex> name
"meg"
If an unknown key is given, the compiler will raise an error:
iex> %User{type: type} = %User{name: "meg"}
** (CompileError) iex: unknown key :type for struct User
The struct name can be extracted when putting a variable instead of a module name:
iex> %struct_name{} = %User{name: "meg"}
%User{name: "meg"}
iex> struct_name
User

  
    
  
  Binaries


Binaries may appear in patterns using the double less-than/greater-than syntax (<<>>). A binary in a pattern can match multiple segments at the same time, each with different type, size, and unit:
iex> <<val::unit(8)-size(2)-integer>> = <<123, 56>>
"{8"
iex> val
31544
See the documentation for <<>> for a complete definition of pattern matching for binaries.
Finally, remember that strings in Elixir are UTF-8 encoded binaries. This means that, similar to charlists, prefix matches on strings are also possible with the binary concatenation operator (<>):
iex> "hello " <> world = "hello world"
"hello world"
iex> world
"world"
Suffix matches (hello <> " world") are not valid patterns.

  
    
  
  Guards


Guards are a way to augment pattern matching with more complex checks. They are allowed in a predefined set of constructs where pattern matching is allowed, such as function definitions, case clauses, and others.
Not all expressions are allowed in guard clauses, but only a handful of them. This is a deliberate choice. This way, Elixir (through Erlang) ensures that all guards are predictable (no mutations or other side-effects) and they can be optimized and performed efficiently.

  
    
  
  List of allowed functions and operators


You can find the built-in list of guards in the Kernel module. Here is an overview:
	comparison operators (==, !=, ===, !==,
<, <=, >, >=)
	strictly boolean operators (and, or, not). Note 

Elixir syntax was designed to have a straightforward conversion to an abstract syntax tree (AST). This means the Elixir syntax is mostly uniform with a handful of "syntax sugar" constructs to reduce the noise in common Elixir idioms.
This document covers all of Elixir syntax constructs as a reference and then discuss their exact AST representation.

  
    
  
  Reserved words


These are the reserved words in the Elixir language. They are detailed throughout this guide but summed up here for convenience:
	true, false, nil - used as atoms
	when, and, or, not, in - used as operators
	fn - used for anonymous function definitions
	do, end, catch, rescue, after, else - used in do-end blocks


  
    
  
  Data types



  
    
  
  Numbers


Integers (1234) and floats (123.4) in Elixir are represented as a sequence of digits that may be separated by underscore for readability purposes, such as 1_000_000. Integers never contain a dot (.) in their representation. Floats contain a dot and at least one other digit after the dot. Floats also support the scientific notation, such as 123.4e10 or 123.4E10.

  
    
  
  Atoms


Unquoted atoms start with a colon (:) which must be immediately followed by a Unicode letter or an underscore. The atom may continue using a sequence of Unicode letters, numbers, underscores, and @. Atoms may end in ! or ?. Valid unquoted atoms are: :ok, :ISO8601, and :integer?.
If the colon is immediately followed by a pair of double- or single-quotes surrounding the atom name, the atom is considered quoted. In contrast with an unquoted atom, this one can be made of any Unicode character (not only letters), such as :'🌢 Elixir', :"++olá++", and :"123".
Quoted and unquoted atoms with the same name are considered equivalent, so :atom, :"atom", and :'atom' represent the same atom. The only catch is that the compiler will warn when quotes are used in atoms that do not need to be quoted.
All operators in Elixir are also valid atoms. Valid examples are :foo, :FOO, :foo_42, :foo@bar, and :++. Invalid examples are :@foo (@ is not allowed at start), :123 (numbers are not allowed at start), and :(*) (not a valid operator).
true, false, and nil are reserved words that are represented by the atoms :true, :false and :nil respectively.
To learn more about all Unicode characters allowed in atom, see the Unicode syntax document.

  
    
  
  Strings


Single-line strings in Elixir are written between double-quotes, such as "foo". Any double-quote inside the string must be escaped with \. Strings support Unicode characters and are stored as UTF-8 encoded binaries.
Multi-line strings in Elixir are written with three double-quotes, and can have unescaped quotes within them. The resulting string will end with a newline. The indentation of the last """ is used to strip indentation from the inner string. For example:
iex> test = """
...>     this
...>     is
...>     a
...>     test
...> """
"    this\n    is\n    a\n    test\n"
iex> test = """
...>     This
...>     Is
...>     A
...>     Test
...>     """
"This\nIs\nA\nTest\n"
Strings are always represented as themselves in the AST.

  
    
  
  Charlists


Charlists in Elixir are written in single-quotes, such as 'foo'. Any single-quote inside the string must be escaped with \. Charlists are made of non-negative integers, where each integer represents a Unicode code point.
Multi-line charlists are written with three single-quotes ('''), the same way multi-line strings are.
Charlists are always represented as themselves in the AST.
For more in-depth information, please read the "Charlists" section in the List module.

  
    
  
  Lists, tuples and binaries


Data structures such as lists, tuples, and binaries are marked respectively by the delimiters [...], {...}, and <<...>>. Each element is separated by comma. A trailing comma is also allowed, such as in [1, 2, 3,].

  
    
  
  Maps and keyword lists


Maps use the %{...} notation and each key-value is given by pairs marked with =>, such as %{"hello" => 1, 2 => "world"}.
Both keyword lists (list of two-element tuples where the first element is atom) and maps with atom keys support a keyword notation where the colon character : is moved to the end of the atom. %{hello: "world"} is equivalent to %{:hello => "world"} and [foo: :bar] is equivalent to [{:foo, :bar}]. This notation is a syntax sugar that emits the same AST representation. It will be explained in later sections.

  
    
  
  Structs


Structs built on the map syntax by passing the struct name between % and {. For example, %User{...}.

  
    
  
  Expressions



  
    
  
  Variables


Variables in Elixir must start with an underscore or a Unicode letter that is not in uppercase or titlecase. The variable may continue using a sequence of Unicode letters, numbers, and underscores. Variables may end in ? or !. To learn more about all Unicode characters allowed in variables, see the Unicode syntax document.
Elixir's naming conventions recommend variables to be in snake_case format.

  
    
  
  Non-qualified calls (local calls)


Non-qualified calls, such as add(1, 2), must start with characters and then follow the same rules as variables, which are optionally followed by parentheses, and then arguments.
Parentheses are required for zero-arity calls (i.e. calls without arguments), to avoid ambiguity with variables. If parentheses are used, they must immediately follow the function name without spaces. For example, add (1, 2) is a syntax error, since (1, 2) is treated as an invalid block which is attempted to be given as a single argument to add.
Elixir's naming conventions recommend calls to be in snake_case format.

  
    
  
  Operators


As many programming languages, Elixir also support operators as non-qualified calls with their precedence and associativity rules. Constructs such as =, when, & and @ are simply treated as operators. See the Operators page for a full reference.

  
    
  
  Qualified calls (remote calls)


Qualified calls, such as Math.add(1, 2), must start with characters and then follow the same rules as variables, which are optionally followed by parentheses, and then arguments. Qualified calls also support operators, such as Kernel.+(1, 2). Elixir also allows the function name to be written between double- or single-quotes, allowing any character in between the quotes, such as Math."++add++"(1, 2).
Similar to non-qualified calls, parentheses have different meaning for zero-arity calls (i.e. calls without arguments). If parentheses are used, such as mod.fun(), it means a function call. If parenthesis are skipped, such as map.field, it means accessing a field of a map.
Elixir's naming conventions recommend calls to be in snake_case format.

  
    
  
  Aliases


Aliases are constructs that expand to atoms at compile-time. The alias String expands to the atom :"Elixir.String". Aliases must start with an ASCII uppercase character which may be followed by any ASCII letter, number, or underscore. Non-ASCII characters are not supported in aliases.
Multiple aliases can be joined with ., such as MyApp.String, and it expands to the atom :"Elixir.MyApp.String". The dot is effectively part of the name but it can also be used for composition. If you define alias MyApp.Example, as: Example in your code, then Example will always expand to :"Elixir.MyApp.Example" and Example.String will expand to :"Elixir.MyApp.Example.String".
Elixir's naming conventions recommend aliases to be in CamelCase format.

  
    
  
  Module attributes


Module attributes are module-specific storage and are written as the composition of the unary operator @ with variables and local calls. For example, to write to a module attribute named foo, use @foo "value", and use @foo to read from it. Given module attributes are written using existing constructs, they follow the same rules above defined for operators, variables, and local calls.

  
    
  
  Blocks


Blocks are multiple Elixir expressions separated by newlines or semi-colons. A new block may be created at any moment by using parentheses.

  
    
  
  Left to right arrow


The left to right arrow (->) is used to establish a relationship between left and right, commonly referred as clauses. The left side may have zero, one, or more arguments; the right side is zero, one, or more expressions separated by new line. The -> may appear one or more times between one of the following terminators: do-end, fn-end or (-). When -> is used, only other clauses are allowed between those terminators. Mixing clauses and regular expressions is invalid syntax.
It is seen on case and cond constructs between do and end:
case 1 do
  2 -> 3
  4 -> 5
end

cond do
  true -> false
end
Seen in typespecs between ( and ):
(integer(), boolean() -> integer())
It is also used between fn and end for building anonymous functions:
fn
  x, y -> x + y
end

  
    
  
  Sigils


Sigils start with ~ and are followed by one lowercase letter or by one or more uppercase letters, immediately followed by one of the following pairs:
	( and )
	{ and }
	[ and ]
	< and >
	" and "
	' and '
	| and |
	/ and /

After closing the pair, zero or more ASCII letters and digits can be given as a modifier. Sigils are expressed as non-qualified calls prefixed with sigil_ where the first argument is the sigil contents as a string and the second argument is a list of integers as modifiers:
If the sigil letter is in uppercase, no interpolation is allowed in the sigil, otherwise its contents may be dynamic. Compare the results of the sigils below for more information:
~s/f#{"o"}o/
~S/f#{"o"}o/
Sigils are useful to encode text with their own escaping rules, such as regular expressions, datetimes, and others.

  
    
  
  The Elixir AST


Elixir syntax was designed to have a straightforward conversion to an abstract syntax tree (AST). Elixir's AST is a regular Elixir data structure composed of the following elements:
	atoms - such as :foo
	integers - such as 42
	floats - such as 13.1
	strings - such as "hello"
	lists - such as [1, 2, 3]
	tuples with two elements - such as {"hello", :world}
	tuples with three elements, representing calls or variables, as explained next

The building block of Elixir's AST is a call, such as:
sum(1, 2, 3)
which is represented as a tuple with three elements:
{:sum, meta, [1, 2, 3]}
the first element is an atom (or another tuple), the second element is a list of two-element tuples with metadata (such as line numbers) and the third is a list of arguments.
We can retrieve the AST for any Elixir expression by calling quote:
quote do
  sum()
end
#=> {:sum, [], []}
Variables are also represented using a tuple with three elements and a combination of lists and atoms, for example:
quote do
  sum
end
#=> {:sum, [], Elixir}
You can see that variables are also represented with a tuple, except the third element is an atom expressing the variable context.
Over the course of this section, we will explore many Elixir syntax constructs alongside their AST representations.

  
    
  
  Operators


Operators are treated as non-qualified calls:
quote do
  1 + 2
end
#=> {:+, [], [1, 2]}
Note that . is also an operator. Remote calls use the dot in the AST with two arguments, where the second argument is always an atom:
quote do
  foo.bar(1, 2, 3)
end
#=> {{:., [], [{:foo, [], Elixir}, :bar]}, [], [1, 2, 3]}
Calling anonymous functions uses the dot in the AST with a single argument, mirroring the fact the function name is "missing" from right side of the dot:
quote do
  foo.(1, 2, 3)
end
#=> {{:., [], [{:foo, [], Elixir}]}, [], [1, 2, 3]}

  
    
  
  Aliases


Aliases are represented by an __aliases__ call with each segment separated by a dot as an argument:
quote do
  Foo.Bar.Baz
end
#=> {:__aliases__, [], [:Foo, :Bar, :Baz]}

quote do
  __MODULE__.Bar.Baz
end
#=> {:__aliases__, [], [{:__MODULE__, [], Elixir}, :Bar, :Baz]}
All arguments, except the first, are guaranteed to be atoms.

  
    
  
  Data structures


Remember that lists are literals, so they are represented as themselves in the AST:
quote do
  [1, 2, 3]
end
#=> [1, 2, 3]
Tuples have their own representation, except for two-element tuples, which are represented as themselves:
quote do
  {1, 2}
end
#=> {1, 2}

quote do
  {1, 2, 3}
end
#=> {:{}, [], [1, 2, 3]}
Binaries have a representation similar to tuples, except they are tagged with :<<>> instead of :{}:
quote do
  <<1, 2, 3>>
end
#=> {:<<>>, [], [1, 2, 3]}
The same applies to maps, where pairs are treated as a list of tuples with two elements:
quote do
  %{1 => 2, 3 => 4}
end
#=> {:%{}, [], [{1, 2}, {3, 4}]}

  
    
  
  Blocks


Blocks are represented as a __block__ call with each line as a separate argument:
quote do
  1
  2
  3
end
#=> {:__block__, [], [1, 2, 3]}

quote do 1; 2; 3; end
#=> {:__block__, [], [1, 2, 3]}

  
    
  
  Left to right arrow


The left to right arrow (->) is represented similar to operators except that they are always part of a list, its left side represents a list of arguments and the right side is an expression.
For example, in case and cond:
quote do
  case 1 do
    2 -> 3
    4 -> 5
  end
end
#=> {:case, [], [1, [do: [{:->, [], [[2], 3]}, {:->, [], [[4], 5]}]]]}

quote do
  cond do
    true -> false
  end
end
#=> {:cond, [], [[do: [{:->, [], [[true], false]}]]]}
Between ( and ):
quote do
  (1, 2 -> 3
   4, 5 -> 6)
end
#=> [{:->, [], [[1, 2], 3]}, {:->, [], [[4, 5], 6]}]
Between fn and end:
quote do
  fn
    1, 2 -> 3
    4, 5 -> 6
  end
end
#=> {:fn, [], [{:->, [], [[1, 2], 3]}, {:->, [], [[4, 5], 6]}]}

  
    
  
  Qualified tuples


Qualified tuples (foo.{bar, baz}) are represented by a {:., [], [expr, :{}]} call, where the expr represents the left hand side of the dot, and the arguments represent the elements inside the curly braces. This is used in Elixir to provide multi aliases:
quote do
  Foo.{Bar, Baz}
end
#=> {{:., [], [{:__aliases__, [], [:Foo]}, :{}]}, [], [{:__aliases__, [], [:Bar]}, {:__aliases__, [], [:Baz]}]}

  
    
  
  Optional syntax


All of the constructs above are part of Elixir's syntax and have their own representation as part of the Elixir AST. This section will discuss the remaining constructs that are alternative representations of the constructs above. In other words, the constructs below can be represented in more than one way in your Elixir code and retain AST equivalence. We call this "Optional Syntax".
For a lightweight introduction to Elixir's Optional Syntax, see this document. Below we continue with a more complete reference.

  
    
  
  Integers in other bases and Unicode code points


Elixir allows integers to contain _ to separate digits and provides conveniences to represent integers in other bases:
1_000_000
#=> 1000000

0xABCD
#=> 43981 (Hexadecimal base)

0o01234567
#=> 342391 (Octal base)

0b10101010
#=> 170 (Binary base)

?é
#=> 233 (Unicode code point)
Those constructs exist only at the syntax level. All of the examples above are represented as their underlying integers in the AST.

  
    
  
  Access syntax


The access syntax is represented as a call to Access.get/2:
quote do
  opts[arg]
end
#=> {{:., [], [Access, :get]}, [], [{:opts, [], Elixir}, {:arg, [], Elixir}]}

  
    
  
  Optional parentheses


Elixir provides optional parentheses on local and remote calls with one or more arguments:
quote do
  sum 1, 2, 3
end
#=> {:sum, [], [1, 2, 3]}
The above is treated the same as sum(1, 2, 3) by the parser. You can remove the parentheses on all calls with at least one argument.
You can also skip parentheses on qualified calls, such as Foo.bar 1, 2, 3. Parentheses are required when invoking anonymous functions, such as f.(1, 2, 3).
In practice, developers prefer to add parentheses to most of their calls. They are skipped mainly in Elixir's control-flow constructs, such as defmodule, if, case, etc, and in certain DSLs.

  
    
  
  Keywords


Keywords in Elixir are a list of tuples of two elements, where the first element is an atom. Using the base constructs, they would be represented as:
[{:foo, 1}, {:bar, 2}]
However, Elixir introduces a syntax sugar where the keywords above may be written as follows:
[foo: 1, bar: 2]
Atoms with foreign characters, such as whitespace, must be wrapped in quotes. This rule applies to keywords as well:
[{:"foo bar", 1}, {:"bar baz", 2}] == ["foo bar": 1, "bar baz": 2]
Remember that, because lists and two-element tuples are quoted literals, by definition keywords are also literals (in fact, the only reason tuples with two elements are quoted literals is to support keywords as literals).
In order to be valid keyword syntax, : cannot be preceded by any whitespace (foo : 1 is invalid) and has to be followed by whitespace (foo:1 is invalid).

  
    
  
  Keywords as last arguments


Elixir also supports a syntax where if the last argument of a call is a keyword list then the square brackets can be skipped. This means that the following:
if(condition, do: this, else: that)
is the same as
if(condition, [do: this, else: that])
which in turn is the same as
if(condition, [{:do, this}, {:else, that}])

  
    
  
  do-end blocks


The last syntax convenience are do-end blocks. do-end blocks are equivalent to keywords as the last argument of a function call, where the block contents are wrapped in parentheses. For example:
if true do
  this
else
  that
end
is the same as:
if(true, do: (this), else: (that))
which we have explored in the previous section.
Parentheses are important to support multiple expressions. This:
if true do
  this
  that
end
is the same as:
if(true, do: (
  this
  that
))
Inside do-end blocks you may introduce other keywords, such as else used in the if above. The supported keywords between do-end are static and are:
	after
	catch
	else
	rescue

You can see them being used in constructs such as receive, try, and others.

Typespecs reference
    

Elixir comes with a notation for declaring types and specifications. This document is a
reference into their uses and syntax.
Elixir is a dynamically typed language, and as such, type specifications are never used by the compiler to optimize or modify code. Still, using type specifications is useful because:
	they provide documentation (for example, tools such as ExDoc show type specifications in the documentation)
	they're used by tools such as Dialyzer, that can analyze code with typespecs to find type inconsistencies and possible bugs

Type specifications (most often referred to as typespecs) are defined in different contexts using the following attributes:
	@type
	@opaque
	@typep
	@spec
	@callback
	@macrocallback

In addition, you can use @typedoc to document a custom @type definition.
See the "User-defined types" and "Defining a specification" sub-sections below for more information on defining types and typespecs.

  
    
  
  A simple example


defmodule StringHelpers do
  @typedoc "A word from the dictionary"
  @type word() :: String.t()

  @spec long_word?(word()) :: boolean()
  def long_word?(word) when is_binary(word) do
    String.length(word) > 8
  end
end
In the example above:
	We declare a new type (word()) that is equivalent to the string type (String.t()).

	We describe the type using a @typedoc, which will be included in the generated documentation.

	We specify that the long_word?/1 function takes an argument of type word() and
returns a boolean (boolean()), that is, either true or false.



  
    
  
  Types and their syntax


The syntax Elixir provides for type specifications is similar to the one in Erlang. Most of the built-in types provided in Erlang (for example, pid()) are expressed in the same way: pid() (or simply pid). Parameterized types (such as list(integer)) are supported as well and so are remote types (such as Enum.t()). Integers and atom literals are allowed as types (for example, 1, :atom, or false). All other types are built out of unions of predefined types. Some types can also be declared using their syntactical notation, such as [type] for lists, {type1, type2, ...} for tuples and <<_ * _>> for binaries.
The notation to represent the union of types is the pipe |. For example, the typespec type :: atom() | pid() | tuple() creates a type type that can be either an atom, a pid, or a tuple. This is usually called a sum type in other languages

  
    
  
  Basic types


type ::
      any()                     # the top type, the set of all terms
      | none()                  # the bottom type, contains no terms
      | atom()
      | map()                   # any map
      | pid()                   # process identifier
      | port()                  # port identifier
      | reference()
      | tuple()                 # tuple of any size

                                ## Numbers
      | float()
      | integer()
      | neg_integer()           # ..., -3, -2, -1
      | non_neg_integer()       # 0, 1, 2, 3, ...
      | pos_integer()           # 1, 2, 3, ...

                                                                      ## Lists
      | list(type)                                                    # proper list ([]-terminated)
      | nonempty_list(type)                                           # non-empty proper list
      | maybe_improper_list(content_type, termination_type)           # proper or improper list
      | nonempty_improper_list(content_type, termination_type)        # improper list
      | nonempty_maybe_improper_list(content_type, termination_type)  # non-empty proper or improper list

      | Literals                # Described in section "Literals"
      | BuiltIn                 # Described in section "Built-in types"
      | Remotes                 # Described in section "Remote types"
      | UserDefined             # Described in section "User-defined types"

  
    
  
  Literals


The following literals are also supported in typespecs:
type ::                               ## Atoms
      :atom                           # atoms: :foo, :bar, ...
      | true | false | nil            # special atom literals

                                      ## Bitstrings
      | <<>>                          # empty bitstring
      | <<_::size>>                   # size is 0 or a positive integer
      | <<_::_*unit>>                 # unit is an integer from 1 to 256
      | <<_::size, _::_*unit>>

                                      ## (Anonymous) Functions
      | (-> type)                     # zero-arity, returns type
      | (type1, type2 -> type)        # two-arity, returns type
      | (... -> type)                 # any arity, returns type

                                      ## Integers
      | 1                             # integer
      | 1..10                         # integer from 1 to 10

                                      ## Lists
      | [type]                        # list with any number of type elements
      | []                            # empty list
      | [...]                         # shorthand for nonempty_list(any())
      | [type, ...]                   # shorthand for nonempty_list(type)
      | [key: value_type]             # keyword list with optional key :key of value_type

                                              ## Maps
      | %{}                                   # empty map
      | %{key: value_type}                    # map with required key :key of value_type
      | %{key_type => value_type}             # map with required pairs of key_type and value_type
      | %{required(key_type) => value_type}   # map with required pairs of key_type and value_type
      | %{optional(key_type) => value_type}   # map with optional pairs of key_type and value_type
      | %SomeStruct{}                         # struct with all fields of any type
      | %SomeStruct{key: value_type}          # struct with required key :key of value_type

                                      ## Tuples
      | {}                            # empty tuple
      | {:ok, type}                   # two-element tuple with an atom and any type

  
    
  
  Built-in types


The following types are also provided by Elixir as shortcuts on top of the basic and literal types described above.
	Built-in type	Defined as
	term()	any()
	arity()	0..255
	as_boolean(t)	t
	binary()	<<_::_*8>>
	nonempty_binary()	<<_::8, _::_*8>>
	bitstring()	<<_::_*1>>
	nonempty_bitstring()	<<_::1, _::_*1>>
	boolean()	true | false
	byte()	0..255
	char()	0..0x10FFFF
	charlist()	[char()]
	nonempty_charlist()	[char(), ...]
	fun()	(... -> any)
	function()	fun()
	identifier()	pid() | port() | reference()
	iodata()	iolist() | binary()
	iolist()	maybe_improper_list(byte() | binary() | iolist(), binary() | [])
	keyword()	[{atom(), any()}]
	keyword(t)	[{atom(), t}]
	list()	[any()]
	nonempty_list()	nonempty_list(any())
	maybe_improper_list()	maybe_improper_list(any(), any())
	nonempty_maybe_improper_list()	nonempty_maybe_improper_list(any(), any())
	mfa()	{module(), atom(), arity()}
	module()	atom()
	no_return()	none()
	node()	atom()
	number()	integer() | float()
	struct()	%{:__struct__ => atom(), optional(atom()) => any()}
	timeout()	:infinity | non_neg_integer()

as_boolean(t) exists to signal users that the given value will be treated as a boolean, where nil and false will be evaluated as false and everything else is true. For example, Enum.filter/2 has the following specification: filter(t, (element -> as_boolean(term))) :: list.

  
    
  
  Remote types


Any module is also able to define its own types and the modules in Elixir are no exception. For example, the Range module defines a t/0 type that represents a range: this type can be referred to as Range.t/0. In a similar fashion, a string is String.t/0, and so on.

  
    
  
  Maps


The key types in maps are allowed to overlap, and if they do, the leftmost key takes precedence.
A map value does not belong to this type if it contains a key that is not in the allowed map keys.
If you want to denote that keys that were not previously defined in the map are allowed,
it is common to end a map type with optional(any) => any.
Note that the syntactic representation of map() is %{optional(any) => any}, not %{}. The notation %{} specifies the singleton type for the empty map.

  
    
  
  Keyword Lists


Beyond keyword() and keyword(t), it can be helpful to compose a spec for an expected keyword list.
For example:
@type option :: {:name, String.t} | {:max, pos_integer} | {:min, pos_integer}
@type options :: [option()]
This makes it clear that only these options are allowed, none are required, and order does not matter.
It also allows composition with existing types.
For example:
@type option :: {:my_option, String.t()} | GenServer.option()

@spec start_link([option()]) :: GenServer.on_start()
def start_link(opts) do
  {my_opts, gen_server_opts} = Keyword.split(opts, [:my_option])
  GenServer.start_link(__MODULE__, my_opts, gen_server_opts)
end
The following spec syntaxes are equivalent:
@type options [{:name, String.t} | {:max, pos_integer} | {:min, pos_integer}]

@type options [name: String.t, max: pos_integer, min: pos_integer]

  
    
  
  User-defined types


The @type, @typep, and @opaque module attributes can be used to define new types:
@type type_name :: type
@typep type_name :: type
@opaque type_name :: type
A type defined with @typep is private. An opaque type, defined with @opaque is a type where the internal structure of the type will not be visible, but the type is still public.
Types can be parameterized by defining variables as parameters; these variables can then be used to define the type.
@type dict(key, value) :: [{key, value}]

  
    
  
  Defining a specification


A specification for a function can be defined as follows:
@spec function_name(type1, type2) :: return_type
Guards can be used to restrict type variables given as arguments to the function.
@spec function(arg) :: [arg] when arg: atom
If you want to specify more than one variable, you separate them by a comma.
@spec function(arg1, arg2) :: {arg1, arg2} when arg1: atom, arg2: integer
Type variables with no restriction can also be defined using var.
@spec function(arg) :: [arg] when arg: var
This guard notation only works with @spec, @callback, and @macrocallback.
You can also name your arguments in a typespec using arg_name :: arg_type syntax. This is particularly useful in documentation as a way to differentiate multiple arguments of the same type (or multiple elements of the same type in a type definition):
@spec days_since_epoch(year :: integer, month :: integer, day :: integer) :: integer
@type color :: {red :: integer, green :: integer, blue :: integer}
Specifications can be overloaded, just like ordinary functions.
@spec function(integer) :: atom
@spec function(atom) :: integer

  
    
  
  Behaviours


Behaviours in Elixir (and Erlang) are a way to separate and abstract the generic part of a component (which becomes the behaviour module) from the specific part (which becomes the callback module).
A behaviour module defines a set of functions and macros (referred to as callbacks) that callback modules implementing that behaviour must export. This "interface" identifies the specific part of the component. For example, the GenServer behaviour and functions abstract away all the message-passing (sending and receiving) and error reporting that a "server" process will likely want to implement from the specific parts such as the actions that this server process has to perform.
Say we want to implement a bunch of parsers, each parsing structured data: for example, a JSON parser and a MessagePack parser. Each of these two parsers will behave the same way: both will provide a parse/1 function and an extensions/0 function. The parse/1 function will return an Elixir representation of the structured data, while the extensions/0 function will return a list of file extensions that can be used for each type of data (e.g., .json for JSON files).
We can create a Parser behaviour:
defmodule Parser do
  @doc """
  Parses a string.
  """
  @callback parse(String.t) :: {:ok, term} | {:error, atom}

  @doc """
  Lists all supported file extensions.
  """
  @callback extensions() :: [String.t]
end
As seen in the example above, defining a callback is a matter of defining a specification for that callback, made of:
	the callback name (parse or extensions in the example)
	the arguments that the callback must accept (String.t)
	the expected type of the callback return value

Modules adopting the Parser behaviour will have to implement all the functions defined with the @callback attribute. As you can see, @callback expects a function name but also a function specification like the ones used with the @spec attribute we saw above.

  
    
  
  Implementing behaviours


Implementing a behaviour is straightforward:
defmodule JSONParser do
  @behaviour Parser

  @impl Parser
  def parse(str), do: {:ok, "some json " <> str} # ... parse JSON

  @impl Parser
  def extensions, do: [".json"]
end
defmodule CSVParser do
  @behaviour Parser

  @impl Parser
  def parse(str), do: {:ok, "some csv " <> str} # ... parse CSV

  @impl Parser
  def extensions, do: [".csv"]
end
If a module adopting a given behaviour doesn't implement one of the callbacks required by that behaviour, a compile-time warning will be generated.
Furthermore, with @impl you can also make sure that you are implementing the correct callbacks from the given behaviour in an explicit manner. For example, the following parser implements both parse and extensions. However, thanks to a typo, BADParser is implementing parse/0 instead of parse/1.
defmodule BADParser do
  @behaviour Parser

  @impl Parser
  def parse, do: {:ok, "something bad"}

  @impl Parser
  def extensions, do: ["bad"]
end
This code generates a warning letting you know that you are mistakenly implementing parse/0 instead of parse/1.
You can read more about @impl in the module documentation.

  
    
  
  Using behaviours


Behaviours are useful because you can pass modules around as arguments and you can then call back to any of the functions specified in the behaviour. For example, we can have a function that receives a filename, several parsers, and parses the file based on its extension:
@spec parse_path(Path.t(), [module()]) :: {:ok, term} | {:error, atom}
def parse_path(filename, parsers) do
  with {:ok, ext} <- parse_extension(filename),
       {:ok, parser} <- find_parser(ext, parsers),
       {:ok, contents} <- File.read(filename) do
    parser.parse(contents)
  end
end

defp parse_extension(filename) do
  if ext = Path.extname(filename) do
    {:ok, ext}
  else
    {:error, :no_extension}
  end
end

defp find_parser(ext, parsers) do
  if parser = Enum.find(parsers, fn parser -> ext in parser.extensions() end) do
    {:ok, parser}
  else
    {:error, :no_matching_parser}
  end
end
You could also invoke any parser directly: CSVParser.parse(...).
Note you don't need to define a behaviour in order to dynamically dispatch on a module, but those features often go hand in hand.

  
    
  
  Optional callbacks


Optional callbacks are callbacks that callback modules may implement if they want to, but are not required to. Usually, behaviour modules know if they should call those callbacks based on configuration, or they check if the callbacks are defined with function_exported?/3 or macro_exported?/3.
Optional callbacks can be defined through the @optional_callbacks module attribute, which has to be a keyword list with function or macro name as key and arity as value. For example:
defmodule MyBehaviour do
  @callback vital_fun() :: any
  @callback non_vital_fun() :: any
  @macrocallback non_vital_macro(arg :: any) :: Macro.t
  @optional_callbacks non_vital_fun: 0, non_vital_macro: 1
end
One example of optional callback in Elixir's standard library is GenServer.format_status/1.

  
    
  
  Inspecting behaviours


The @callback and @optional_callbacks attributes are used to create a behaviour_info/1 function available on the defining module. This function can be used to retrieve the callbacks and optional callbacks defined by that module.
For example, for the MyBehaviour module defined in "Optional callbacks" above:
MyBehaviour.behaviour_info(:callbacks)
#=> [vital_fun: 0, "MACRO-non_vital_macro": 2, non_vital_fun: 0]
MyBehaviour.behaviour_info(:optional_callbacks)
#=> ["MACRO-non_vital_macro": 2, non_vital_fun: 0]
When using iex, the IEx.Helpers.b/1 helper is also available.

  
    
  
  Pitfalls


There are some known pitfalls when using typespecs, they are documented next.

  
    
  
  The string() type


Elixir discourages the use of the string() type. The string() type refers to Erlang strings, which are known as "charlists" in Elixir. They do not refer to Elixir strings, which are UTF-8 encoded binaries. To avoid confusion, if you attempt to use the type string(), Elixir will emit a warning. You should use charlist(), nonempty_charlist(), binary() or String.t() accordingly, or any of the several literal representations for these types.
Note that String.t() and binary() are equivalent to analysis tools. Although, for those reading the documentation, String.t() implies it is a UTF-8 encoded binary.

  
    
  
  Functions which raise an error


Typespecs do not need to indicate that a function can raise an error; any function can fail any time if given invalid input.
In the past, the Elixir standard library sometimes used no_return() to indicate this, but these usages have been removed.
The no_return() type also should not be used for functions which do return but whose purpose is a "side effect", such as IO.puts/1.
In these cases, the expected return type is :ok.
Instead, no_return() should be used as the return type for functions which can never return a value.
This includes functions which loop forever calling receive, or which exist specifically to raise an error, or which shut down the VM.

Unicode syntax
    

Elixir supports Unicode throughout the language. This document is a complete reference of how
Elixir supports Unicode in its syntax.
Strings ("olá") and charlists ('olá') support Unicode since Elixir v1.0. Strings are UTF-8 encoded. Charlists are lists of Unicode code points. In such cases, the contents are kept as written by developers, without any transformation.
Elixir also supports Unicode in variables, atoms, and calls since Elixir v1.5. The focus of this document is to provide a high-level introduction to how Elixir allows Unicode in its syntax. We also provide technical documentation describing how Elixir complies with the Unicode specification.
To check the Unicode version of your current Elixir installation, run String.Unicode.version().

  
    
  
  Introduction


Elixir allows Unicode characters in its variables, atoms, and calls. However, the Unicode characters must still obey the rules of the language syntax. In particular, variables and calls cannot start with an uppercase letter. From now on, we will refer to those terms as identifiers.
The characters allowed in identifiers are the ones specified by Unicode. Generally speaking, it is restricted to characters typically used by the writing system of human languages still in activity. In particular, it excludes symbols such as emojis, alternate numeric representations, musical notes, and the like.
Elixir imposes many restrictions on identifiers for security purposes. For example, the word "josé" can be written in two ways in Unicode: as the combination of the characters j o s é and as a combination of the characters j o s e ́, where the accent is its own character. The former is called NFC form and the latter is the NFD form. Elixir normalizes all characters to be the in the NFC form.
Elixir also disallows mixed-scripts in most scenarios. For example, it is not possible to name a variable аdmin, where а is in Cyrillic and the remaining characters are in Latin. Doing so will raise the following error:
** (SyntaxError) invalid mixed-script identifier found: аdmin

Mixed-script identifiers are not supported for security reasons. 'аdmin' is made of the following scripts:

  \u0430 а {Cyrillic}
  \u0064 d {Latin}
  \u006D m {Latin}
  \u0069 i {Latin}
  \u006E n {Latin}

Make sure all characters in the identifier resolve to a single script or a highly
restrictive script. See https://hexdocs.pm/elixir/unicode-syntax.html for more information.
The character must either be all in Cyrillic or all in Latin. The only mixed-scripts that Elixir allows, according to the Highly Restrictive Unicode recommendations, are:
	Latin and Han with Bopomofo
	Latin and Japanese
	Latin and Korean

Finally, Elixir will also warn on confusable identifiers in the same file. For example, Elixir will emit a warning if you use both variables а (Cyrillic) and а (Latin) in your code.
That's the overall introduction of how Unicode is used in Elixir identifiers. In a nutshell, its goal is to support different writing systems in use today while keeping the Elixir language itself clear and secure.
For the technical details, see the next sections that cover the technical Unicode requirements.

  
    
  
  Unicode Annex #31


Elixir implements the requirements outlined in the Unicode Annex #31, version 15.0.

  
    
  
  R1. Default Identifiers


The general Elixir identifier rule is specified as:
<Identifier> := <Start> <Continue>* <Ending>?
where <Start> uses the same categories as the spec but normalizes them to the NFC form (see R4):
characters derived from the Unicode General Category of uppercase letters, lowercase letters, titlecase letters, modifier letters, other letters, letter numbers, plus Other_ID_Start, minus Pattern_Syntax and Pattern_White_Space code points
In set notation: [\p{L}\p{Nl}\p{Other_ID_Start}-\p{Pattern_Syntax}-\p{Pattern_White_Space}].

and <Continue> uses the same categories as the spec but normalizes them to the NFC form (see R4):
ID_Start characters, plus characters having the Unicode General Category of nonspacing marks, spacing combining marks, decimal number, connector punctuation, plus Other_ID_Continue, minus Pattern_Syntax and Pattern_White_Space code points.
In set notation: [\p{ID_Start}\p{Mn}\p{Mc}\p{Nd}\p{Pc}\p{Other_ID_Continue}-\p{Pattern_Syntax}-\p{Pattern_White_Space}].

<Ending> is an addition specific to Elixir that includes only the code points ? (003F) and ! (0021).
The spec also provides a <Medial> set, but Elixir does not include any character on this set. Therefore, the identifier rule has been simplified to consider this.
Elixir does not allow the use of ZWJ or ZWNJ in identifiers and therefore does not implement R1a. Bidirectional control characters are also not supported. R1b is guaranteed for backwards compatibility purposes.
Atoms
Unicode atoms in Elixir follow the identifier rule above with the following modifications:
	<Start> additionally includes the code point _ (005F)
	<Continue> additionally includes the code point @ (0040)

Note atoms can also be quoted, which allows any characters, such as :"hello elixir". All Elixir operators are also valid atoms, such as :+, :@, :|>, and others. The full description of valid atoms is available in the "Atoms" section in the syntax reference.
Variables, local calls, and remote calls
Variables in Elixir follow the identifier rule above with the following modifications:
	<Start> additionally includes the code point _ (005F)
	<Start> additionally excludes Lu (letter uppercase) and Lt (letter titlecase) characters

In set notation: [\u{005F}\p{Ll}\p{Lm}\p{Lo}\p{Nl}\p{Other_ID_Start}-\p{Pattern_Syntax}-\p{Pattern_White_Space}].
Aliases
Aliases in Elixir only allow ASCII characters, starting in uppercase, and no punctuation characters.

  
    
  
  R3. Pattern_White_Space and Pattern_Syntax Characters


Elixir supports only code points \t (0009), \n (000A), \r (000D) and \s (0020) as whitespace and therefore does not follow requirement R3. R3 requires a wider variety of whitespace and syntax characters to be supported.

  
    
  
  R4. Equivalent Normalized Identifiers


Identifiers in Elixir are case sensitive.
Elixir normalizes all atoms and variables to NFC form. Quoted-atoms and strings can, however, be in any form and are not verified by the parser.
In other words, the atom :josé can only be written with the code points 006A 006F 0073 00E9 or 006A 006F 0073 0065 0301, but Elixir will rewrite it to the former (from Elixir 1.14). On the other hand, :"josé" may be written as 006A 006F 0073 00E9 or 006A 006F 0073 0065 0301 and its form will be retained, since it is written between quotes.
Choosing requirement R4 automatically excludes requirements R5, R6, and R7.

  
    
  
  Unicode Technical Standard #39


Elixir conforms to the clauses outlined in the Unicode Technical Standard #39 on Security, version 15.0.

  
    
  
  C1. General Security Profile for Identifiers


Elixir will not allow tokenization of identifiers with codepoints in \p{Identifier_Status=Restricted}.
An implementation following the General Security Profile does not permit any characters in \p{Identifier_Status=Restricted}, ...

For instance, the 'HANGUL FILLER' (ㅤ) character, which is often invisible, is an uncommon codepoint and will trigger this warning.
See the note below about additional normalizations, which can perform automatic replacement of some Restricted identifiers.

  
    
  
  C2. Confusable detection


Elixir will warn on identifiers that look the same, but aren't. Examples: in а = a = 1, the two 'a' characters are Cyrillic and Latin, and could be confused for each other; in 力 = カ = 1, both are Japanese, but different codepoints, in different scripts of that writing system. Confusable identifiers can lead to hard-to-catch bugs (say, due to copy-pasted code) and can be unsafe, so we will warn about identifiers within a single file that could be confused with each other.
We use the means described in Section 4, 'Confusable Detection', with one noted modification
Alternatively, it shall declare that it uses a modification, and provide a precise list of character mappings that are added to or removed from the provided ones.

Elixir will not warn on confusability for identifiers made up exclusively of characters in a-z, A-Z, 0-9, and _. This is because ASCII identifiers have existed for so long that the programming community has had their own means of dealing with confusability between identifiers like l,1 or O,0 (for instance, fonts designed for programming usually make it easy to differentiate between those characters).

  
    
  
  C3. Mixed Script Detection


Elixir will not allow tokenization of mixed-script identifiers unless the mixing is one of the exceptions defined in UTS 39 5.2, 'Highly Restrictive'. We use the means described in Section 5.1, Mixed-Script Detection, to determine if script mixing is occurring, with the modification documented in the section 'Additional Normalizations', below.
Examples: Elixir allows an identifiers like 幻ㄒㄧㄤ, even though it includes characters from multiple 'scripts', because those scripts all 'resolve' to Japanese when applying the resolution rules from UTS 39 5.1. It also allows an atom like :Tシャツ, the Japanese word for 't-shirt', which incorporates a Latin capital T, because {Latn, Jpan} is one of the allowed script mixing in the definition of 'Highly Restrictive' in UTS 39 5.2, and it 'covers' the string.
However, Elixir would prevent tokenization in code like if аdmin, do: :ok, else: :err, where the scriptset for the 'a' character is {Cyrillic} but all other characters have scriptsets of {Latin}. The scriptsets fail to resolve, and the scriptsets from the definition of 'Highly Restrictive' in UTS 39 5.2 do not cover the string either, so a descriptive error is shown.

  
    
  
  C4, C5 (inapplicable)


'C4 - Restriction Level detection' conformance is not claimed and does not apply to identifiers in code; rather, it applies to classifying the level of safety of a given arbitrary string into one of 5 restriction levels.
'C5 - Mixed number detection' conformance is inapplicable as Elixir does not support Unicode numbers.

  
    
  
  Addition normalizations and documented UTS 39 modifications


As of Elixir 1.14, some codepoints in \p{Identifier_Status=Restricted} are normalized to other, unrestricted codepoints.
Initially this is only done to translate MICRO SIGN µ to Greek lowercase mu, μ.
This is not a modification of UTS39 clauses C1 (General Security Profile) or C2 (Confusability Detection); however, it is a documented modification of C3, 'Mixed-Script detection'.
Mixed-script detection is modified by these normalizations to the extent that the normalized codepoint is given the union of scriptsets from both characters.
	For instance, in the example of MICRO => MU, Micro was a 'Common'-script character -- the same script given to the '_' underscore codepoint -- and thus the normalized character's scriptset will be {Greek, Common}. 'Common' intersects with all non-empty scriptsets, and thus the normalized character can be used in tokens written in any script without causing script mixing.

	The code points normalized in this fashion are those that are in use in the community, and judged not likely to cause issues with unsafe script mixing. For instance, the MICRO or MU codepoint may be used in an atom or variable dealing with microseconds.

Kernel 
    



      
Kernel is Elixir's default environment.
It mainly consists of:
	basic language primitives, such as arithmetic operators, spawning of processes,
data type handling, and others
	macros for control-flow and defining new functionality (modules, functions, and the like)
	guard checks for augmenting pattern matching

You can invoke Kernel functions and macros anywhere in Elixir code
without the use of the Kernel. prefix since they have all been
automatically imported. For example, in IEx, you can call:
iex> is_number(13)
true
If you don't want to import a function or macro from Kernel, use the :except
option and then list the function/macro by arity:
import Kernel, except: [if: 2, unless: 2]
See import/2 for more information on importing.
Elixir also has special forms that are always imported and
cannot be skipped. These are described in Kernel.SpecialForms.

  
    
  
  The standard library


Kernel provides the basic capabilities the Elixir standard library
is built on top of. It is recommended to explore the standard library
for advanced functionality. Here are the main groups of modules in the
standard library (this list is not a complete reference, see the
documentation sidebar for all entries).

  
    
  
  Built-in types


The following modules handle Elixir built-in data types:
	Atom - literal constants with a name (true, false, and nil are atoms)
	Float - numbers with floating point precision
	Function - a reference to code chunk, created with the fn/1 special form
	Integer - whole numbers (not fractions)
	List - collections of a variable number of elements (linked lists)
	Map - collections of key-value pairs
	Process - light-weight threads of execution
	Port - mechanisms to interact with the external world
	Tuple - collections of a fixed number of elements

There are two data types without an accompanying module:
	Bitstring - a sequence of bits, created with <<>>/1.
When the number of bits is divisible by 8, they are called binaries and can
be manipulated with Erlang's :binary module
	Reference - a unique value in the runtime system, created with make_ref/0


  
    
  
  Data types


Elixir also provides other data types that are built on top of the types
listed above. Some of them are:
	Date - year-month-day structs in a given calendar
	DateTime - date and time with time zone in a given calendar
	Exception - data raised from errors and unexpected scenarios
	MapSet - unordered collections of unique elements
	NaiveDateTime - date and time without time zone in a given calendar
	Keyword - lists of two-element tuples, often representing optional values
	Range - inclusive ranges between two integers
	Regex - regular expressions
	String - UTF-8 encoded binaries representing characters
	Time - hour:minute:second structs in a given calendar
	URI - representation of URIs that identify resources
	Version - representation of versions and requirements


  
    
  
  System modules


Modules that interface with the underlying system, such as:
	IO - handles input and output
	File - interacts with the underlying file system
	Path - manipulates file system paths
	System - reads and writes system information


  
    
  
  Protocols


Protocols add polymorphic dispatch to Elixir. They are contracts
implementable by data types. See Protocol for more information on
protocols. Elixir provides the following protocols in the standard library:
	Collectable - collects data into a data type
	Enumerable - handles collections in Elixir. The Enum module
provides eager functions for working with collections, the Stream
module provides lazy functions
	Inspect - converts data types into their programming language
representation
	List.Chars - converts data types to their outside world
representation as charlists (non-programming based)
	String.Chars - converts data types to their outside world
representation as strings (non-programming based)


  
    
  
  Process-based and application-centric functionality


The following modules build on top of processes to provide concurrency,
fault-tolerance, and more.
	Agent - a process that encapsulates mutable state
	Application - functions for starting, stopping and configuring
applications
	GenServer - a generic client-server API
	Registry - a key-value process-based storage
	Supervisor - a process that is responsible for starting,
supervising and shutting down other processes
	Task - a process that performs computations
	Task.Supervisor - a supervisor for managing tasks exclusively


  
    
  
  Supporting documents


Under the "Pages" section in sidebar you will find tutorials, guides,
and reference documents that outline Elixir semantics and behaviors
in more detail. Those are:
	Compatibility and deprecations - lists
compatibility between every Elixir version and Erlang/OTP, release schema;
lists all deprecated functions, when they were deprecated and alternatives
	Library guidelines - general guidelines, anti-patterns,
and rules for those writing libraries
	Naming conventions - naming conventions for Elixir code
	Operators reference - lists all Elixir operators and their precedences
	Patterns and guards - an introduction to patterns,
guards, and extensions
	Syntax reference - the language syntax reference
	Typespecs reference- types and function specifications, including list of types
	Unicode syntax - outlines Elixir support for Unicode


  
    
  
  Guards


This module includes the built-in guards used by Elixir developers.
They are a predefined set of functions and macros that augment pattern
matching, typically invoked after the when operator. For example:
def drive(%User{age: age}) when age >= 16 do
  ...
end
The clause above will only be invoked if the user's age is more than
or equal to 16. Guards also support joining multiple conditions with
and and or. The whole guard is true if all guard expressions will
evaluate to true. A more complete introduction to guards is available
in the Patterns and guards page.

  
    
  
  Truthy and falsy values


Besides the booleans true and false, Elixir has the
concept of a "truthy" or "falsy" value.
	 a value is truthy when it is neither false nor nil
	 a value is falsy when it is either false or nil

Elixir has functions, like and/2, that only work with
booleans, but also functions that work with these
truthy/falsy values, like 

Kernel.SpecialForms 
    



      
Special forms are the basic building blocks of Elixir, and therefore
cannot be overridden by the developer.
The Kernel.SpecialForms module consists solely of macros that can be
invoked anywhere in Elixir code without the use of the
Kernel.SpecialForms. prefix. This is possible because they all have
been automatically imported, in the same fashion as the functions and
macros from the Kernel module.
These building blocks are defined in this module. Some of these special forms are lexical (such as
alias/2 and case/2). The macros {}/1 and <<>>/1 are also special
forms used to define tuple and binary data structures respectively.
This module also documents macros that return information about Elixir's
compilation environment, such as (__ENV__/0, __MODULE__/0, __DIR__/0,
__STACKTRACE__/0, and __CALLER__/0).
Additionally, it documents two special forms, __block__/1 and
__aliases__/1, which are not intended to be called directly by the
developer but they appear in quoted contents since they are essential
in Elixir's constructs.

Atoms are constants whose values are their own name.
They are often useful to enumerate over distinct values, such as:
iex> :apple
:apple
iex> :orange
:orange
iex> :watermelon
:watermelon
Atoms are equal if their names are equal.
iex> :apple == :apple
true
iex> :apple == :orange
false
Often they are used to express the state of an operation, by using
values such as :ok and :error.
The booleans true and false are also atoms:
iex> true == :true
true
iex> is_atom(false)
true
iex> is_boolean(:false)
true
Elixir allows you to skip the leading : for the atoms false, true,
and nil.
Atoms must be composed of Unicode characters such as letters, numbers,
underscore, and @. If the keyword has a character that does not
belong to the category above, such as spaces, you can wrap it in
quotes:
iex> :"this is an atom with spaces"
:"this is an atom with spaces"

This module provides data encoding and decoding functions
according to RFC 4648.
This document defines the commonly used base 16, base 32, and base
64 encoding schemes.

  
    
  
  Base 16 alphabet


	Value	Encoding	Value	Encoding	Value	Encoding	Value	Encoding
	0	0	4	4	8	8	12	C
	1	1	5	5	9	9	13	D
	2	2	6	6	10	A	14	E
	3	3	7	7	11	B	15	F


  
    
  
  Base 32 alphabet


	Value	Encoding	Value	Encoding	Value	Encoding	Value	Encoding
	0	A	9	J	18	S	27	3
	1	B	10	K	19	T	28	4
	2	C	11	L	20	U	29	5
	3	D	12	M	21	V	30	6
	4	E	13	N	22	W	31	7
	5	F	14	O	23	X		
	6	G	15	P	24	Y	(pad)	=
	7	H	16	Q	25	Z		
	8	I	17	R	26	2		


  
    
  
  Base 32 (extended hex) alphabet


	Value	Encoding	Value	Encoding	Value	Encoding	Value	Encoding
	0	0	9	9	18	I	27	R
	1	1	10	A	19	J	28	S
	2	2	11	B	20	K	29	T
	3	3	12	C	21	L	30	U
	4	4	13	D	22	M	31	V
	5	5	14	E	23	N		
	6	6	15	F	24	O	(pad)	=
	7	7	16	G	25	P		
	8	8	17	H	26	Q		


  
    
  
  Base 64 alphabet


	Value	Encoding	Value	Encoding	Value	Encoding	Value	Encoding
	0	A	17	R	34	i	51	z
	1	B	18	S	35	j	52	0
	2	C	19	T	36	k	53	1
	3	D	20	U	37	l	54	2
	4	E	21	V	38	m	55	3
	5	F	22	W	39	n	56	4
	6	G	23	X	40	o	57	5
	7	H	24	Y	41	p	58	6
	8	I	25	Z	42	q	59	7
	9	J	26	a	43	r	60	8
	10	K	27	b	44	s	61	9
	11	L	28	c	45	t	62	+
	12	M	29	d	46	u	63	/
	13	N	30	e	47	v		
	14	O	31	f	48	w	(pad)	=
	15	P	32	g	49	x		
	16	Q	33	h	50	y		


  
    
  
  Base 64 (URL and filename safe) alphabet


	Value	Encoding	Value	Encoding	Value	Encoding	Value	Encoding
	0	A	17	R	34	i	51	z
	1	B	18	S	35	j	52	0
	2	C	19	T	36	k	53	1
	3	D	20	U	37	l	54	2
	4	E	21	V	38	m	55	3
	5	F	22	W	39	n	56	4
	6	G	23	X	40	o	57	5
	7	H	24	Y	41	p	58	6
	8	I	25	Z	42	q	59	7
	9	J	26	a	43	r	60	8
	10	K	27	b	44	s	61	9
	11	L	28	c	45	t	62	-
	12	M	29	d	46	u	63	_
	13	N	30	e	47	v		
	14	O	31	f	48	w	(pad)	=
	15	P	32	g	49	x		
	16	Q	33	h	50	y		

Bitwise 
    



      
A set of functions that perform calculations on bits.
All bitwise functions work only on integers, otherwise an
ArithmeticError is raised. The functions band/2,
bor/2, bsl/2, and bsr/2 also have operators,
respectively: 

Date 
    



      
A Date struct and functions.
The Date struct contains the fields year, month, day and calendar.
New dates can be built with the new/3 function or using the
~D (see sigil_D/2) sigil:
iex> ~D[2000-01-01]
~D[2000-01-01]
Both new/3 and sigil return a struct where the date fields can
be accessed directly:
iex> date = ~D[2000-01-01]
iex> date.year
2000
iex> date.month
1
The functions on this module work with the Date struct as well
as any struct that contains the same fields as the Date struct,
such as NaiveDateTime and DateTime. Such functions expect
Calendar.date/0 in their typespecs (instead of t/0).
Developers should avoid creating the Date structs directly
and instead rely on the functions provided by this module as well
as the ones in third-party calendar libraries.

  
    
  
  Comparing dates


Comparisons in Elixir using ==/2, >/2, </2 and similar are structural
and based on the Date struct fields. For proper comparison between
dates, use the compare/2 function. The existence of the compare/2
function in this module also allows using Enum.min/2 and Enum.max/2
functions to get the minimum and maximum date of an Enum. For example:
iex>  Enum.min([~D[2017-03-31], ~D[2017-04-01]], Date)
~D[2017-03-31]

  
    
  
  Using epochs


The add/2, diff/2 and shift/2 functions can be used for computing dates
or retrieving the number of days between instants. For example, if there
is an interest in computing the number of days from the Unix epoch
(1970-01-01):
iex> Date.diff(~D[2010-04-17], ~D[1970-01-01])
14716

iex> Date.add(~D[1970-01-01], 14716)
~D[2010-04-17]

iex> Date.shift(~D[1970-01-01], year: 40, month: 3, week: 2, day: 2)
~D[2010-04-17]
Those functions are optimized to deal with common epochs, such
as the Unix Epoch above or the Gregorian Epoch (0000-01-01).

A datetime implementation with a time zone.
This datetime can be seen as a snapshot of a date and time
at a given time zone. For such purposes, it also includes both
UTC and Standard offsets, as well as the zone abbreviation
field used exclusively for formatting purposes. Note future
datetimes are not necessarily guaranteed to exist, as time
zones may change any time in the future due to geopolitical
reasons. See the "Datetimes as snapshots" section for more
information.
Remember, comparisons in Elixir using ==/2, >/2, </2 and friends
are structural and based on the DateTime struct fields. For proper
comparison between datetimes, use the compare/2 function. The
existence of the compare/2 function in this module also allows
using Enum.min/2 and Enum.max/2 functions to get the minimum and
maximum datetime of an Enum. For example:
iex> Enum.min([~U[2022-01-12 00:01:00.00Z], ~U[2021-01-12 00:01:00.00Z]], DateTime)
~U[2021-01-12 00:01:00.00Z]
Developers should avoid creating the DateTime struct directly
and instead rely on the functions provided by this module as
well as the ones in third-party calendar libraries.

  
    
  
  Time zone database


Many functions in this module require a time zone database.
By default, it uses the default time zone database returned by
Calendar.get_time_zone_database/0, which defaults to
Calendar.UTCOnlyTimeZoneDatabase which only handles "Etc/UTC"
datetimes and returns {:error, :utc_only_time_zone_database}
for any other time zone.
Other time zone databases can also be configured. Here are some
available options and libraries:
	time_zone_info
	tz
	tzdata
	zoneinfo -
recommended for embedded devices

To use them, first make sure it is added as a dependency in mix.exs.
It can then be configured either via configuration:
config :elixir, :time_zone_database, Tz.TimeZoneDatabase
or by calling Calendar.put_time_zone_database/1:
Calendar.put_time_zone_database(Tz.TimeZoneDatabase)
See the proper names in the library installation instructions.

  
    
  
  Datetimes as snapshots


In the first section, we described datetimes as a "snapshot of
a date and time at a given time zone". To understand precisely
what we mean, let's see an example.
Imagine someone in Poland who wants to schedule a meeting with someone
in Brazil in the next year. The meeting will happen at 2:30 AM
in the Polish time zone. At what time will the meeting happen in
Brazil?
You can consult the time zone database today, one year before,
using the API in this module and it will give you an answer that
is valid right now. However, this answer may not be valid in the
future. Why? Because both Brazil and Poland may change their timezone
rules, ultimately affecting the result. For example, a country may
choose to enter or abandon "Daylight Saving Time", which is a
process where we adjust the clock one hour forward or one hour
back once per year. Whenever the rules change, the exact instant
that 2:30 AM in Polish time will be in Brazil may change.
In other words, whenever working with future DateTimes, there is
no guarantee the results you get will always be correct, until
the event actually happens. Therefore, when you ask for a future
time, the answers you get are a snapshot that reflects the current
state of the time zone rules. For datetimes in the past, this is
not a problem, because time zone rules do not change for past
events.
To make matters worse, it may be that 2:30 AM in Polish time
does not actually even exist or it is ambiguous. If a certain
time zone observes "Daylight Saving Time", they will move their
clock forward once a year. When this happens, there is a whole
hour that does not exist. Then, when they move the clock back,
there is a certain hour that will happen twice. So if you want to
schedule a meeting when this shift back happens, you would need to
explicitly say which occurrence of 2:30 AM you mean: the one in
"Summer Time", which occurs before the shift, or the one
in "Standard Time", which occurs after it. Applications that are
date and time sensitive need to take these scenarios into account
and correctly communicate them to users.
The good news is: Elixir contains all of the building blocks
necessary to tackle those problems. The default timezone database
used by Elixir, Calendar.UTCOnlyTimeZoneDatabase, only works
with UTC, which does not observe those issues. Once you bring
a proper time zone database, the functions in this module will
query the database and return the relevant information. For
example, look at how DateTime.new/4 returns different results
based on the scenarios described in this section.

  
    
  
  Converting between timezones


Bearing in mind the cautions above, and assuming you've brought in a full
timezone database, here are some examples of common shifts between time
zones.
# Local time to UTC
new_york = DateTime.from_naive!(~N[2023-06-26T09:30:00], "America/New_York")
#=> #DateTime<2023-06-26 09:30:00-04:00 EDT America/New_York>

utc = DateTime.shift_zone!(new_york, "Etc/UTC")
#=> ~U[2023-06-26 13:30:00Z]

# UTC to local time
DateTime.shift_zone!(utc, "Europe/Paris")
#=> #DateTime<2023-06-26 15:30:00+02:00 CEST Europe/Paris>

Struct and functions for handling durations.
A Duration struct represents a collection of time scale units,
allowing for manipulation and calculation of durations.
Date and time scale units are represented as integers, allowing for
both positive and negative values.
Microseconds are represented using a tuple {microsecond, precision}.
This ensures compatibility with other calendar types implementing time,
such as Time, DateTime, and NaiveDateTime.

  
    
  
  Shifting


The most common use of durations in Elixir's standard library is to
"shift" the calendar types.
iex> Date.shift(~D[2016-01-03], month: 2)
~D[2016-03-03]
In the example above, Date.shift/2 automatically converts the units
into a Duration struct, although one can also be given directly:
iex> Date.shift(~D[2016-01-03], Duration.new!(month: 2))
~D[2016-03-03]
It is important to note that shifting is not an arithmetic operation.
For example, adding date + 1 month + 1 month does not yield the same
result as date + 2 months. Let's see an example:
iex> ~D[2016-01-31] |> Date.shift(month: 1) |> Date.shift(month: 1)
~D[2016-03-29]

iex> ~D[2016-01-31] |> Date.shift(month: 2)
~D[2016-03-31]
As you can see above, the results differ, which explains why operations
with durations are called "shift" rather than "add". This happens because,
once we add one month to 2016-01-31, we get 2016-02-29. Then adding
one extra month gives us 2016-03-29 instead of 2016-03-31.
In particular, when applying durations to Calendar.ISO types:
	larger units (such as years and months) are applied before
smaller ones (such as weeks, hours, days, and so on)

	units are collapsed into months (:year and :month),
seconds (:week, :day, :hour, :minute, :second)
and microseconds (:microsecond) before they are applied

	1 year is equivalent to 12 months, 1 week is equivalent to 7 days.
Therefore, 4 weeks are not equivalent to 1 month

	in case of non-existing dates, the results are rounded down to the
nearest valid date


As the shift/2 functions are calendar aware, they are guaranteed to return
valid date/times, considering leap years as well as DST in applicable time zones.

  
    
  
  Intervals


Durations in Elixir can be combined with stream operations to build intervals.
For example, to retrieve the next three Wednesdays starting from 17th April, 2024:
iex> ~D[2024-04-17] |> Stream.iterate(&Date.shift(&1, week: 1)) |> Enum.take(3)
[~D[2024-04-17], ~D[2024-04-24], ~D[2024-05-01]]
However, once again, it is important to remember that shifting a duration is not
arithmetic, so you may want to use the functions in this module depending on what
you to achieve. Compare the results of both examples below:
# Adding one month after the other
iex> date = ~D[2016-01-31]
iex> duration = Duration.new!(month: 1)
iex> stream = Stream.iterate(date, fn prev_date -> Date.shift(prev_date, duration) end)
iex> Enum.take(stream, 3)
[~D[2016-01-31], ~D[2016-02-29], ~D[2016-03-29]]

# Multiplying durations by an index
iex> date = ~D[2016-01-31]
iex> duration = Duration.new!(month: 1)
iex> stream = Stream.from_index(fn i -> Date.shift(date, Duration.multiply(duration, i)) end)
iex> Enum.take(stream, 3)
[~D[2016-01-31], ~D[2016-02-29], ~D[2016-03-31]]
The second example consistently points to the last day of the month,
as it performs operations on the duration, rather than shifting date
after date.

Functions for dealing with throw/catch/exit and exceptions.
This module also defines the behaviour required by custom
exceptions. To define your own, see defexception/1.

  
    
  
  Formatting functions


Several functions in this module help format exceptions.
Some of these functions expect the stacktrace as argument.
The stacktrace is typically available inside catch and
rescue by using the __STACKTRACE__/0 variable.
Do not rely on the particular format returned by the
functions in this module. They may be changed in future releases
in order to better suit Elixir's tool chain. In other words,
by using the functions in this module it is guaranteed you will
format exceptions as in the current Elixir version being used.

Functions for working with floating-point numbers.
For mathematical operations on top of floating-points,
see Erlang's :math module.

  
    
  
  Kernel functions


There are functions related to floating-point numbers on the Kernel module
too. Here is a list of them:
	Kernel.round/1: rounds a number to the nearest integer.
	Kernel.trunc/1: returns the integer part of a number.


  
    
  
  Known issues


There are some very well known problems with floating-point numbers
and arithmetic due to the fact most decimal fractions cannot be
represented by a floating-point binary and most operations are not exact,
but operate on approximations. Those issues are not specific
to Elixir, they are a property of floating point representation itself.
For example, the numbers 0.1 and 0.01 are two of them, what means the result
of squaring 0.1 does not give 0.01 neither the closest representable. Here is
what happens in this case:
	The closest representable number to 0.1 is 0.1000000014
	The closest representable number to 0.01 is 0.0099999997
	Doing 0.1 * 0.1 should return 0.01, but because 0.1 is actually 0.1000000014,
the result is 0.010000000000000002, and because this is not the closest
representable number to 0.01, you'll get the wrong result for this operation

There are also other known problems like flooring or rounding numbers. See
round/2 and floor/2 for more details about them.
To learn more about floating-point arithmetic visit:
	0.30000000000000004.com
	What Every Programmer Should Know About Floating-Point Arithmetic

A set of functions for working with functions.
Anonymous functions are typically created by using fn:
iex> add = fn a, b -> a + b end
iex> add.(1, 2)
3
Anonymous functions can also have multiple clauses. All clauses
should expect the same number of arguments:
iex> negate = fn
...>   true -> false
...>   false -> true
...> end
iex> negate.(false)
true

  
    
  
  The capture operator


It is also possible to capture public module functions and pass them
around as if they were anonymous functions by using the capture
operator 

Functions for working with integers.
Some functions that work on integers are found in Kernel:
	Kernel.abs/1
	Kernel.div/2
	Kernel.max/2
	Kernel.min/2
	Kernel.rem/2

Provides functions to deal with modules during compilation time.
It allows a developer to dynamically add, delete and register
attributes, attach documentation and so forth.
After a module is compiled, using many of the functions in
this module will raise errors, since it is out of their scope
to inspect runtime data. Most of the runtime data can be inspected
via the __info__/1 function attached to
each compiled module.

  
    
  
  Module attributes


Each module can be decorated with one or more attributes. The following ones
are currently defined by Elixir:

  
    
  
  @after_compile


A hook that will be invoked right after the current module is compiled.
Accepts a module or a {module, function_name}. See the "Compile callbacks"
section below.

  
    
  
  @after_verify (since v1.14.0)


A hook that will be invoked right after the current module is verified for
undefined functions, deprecations, etc. Accepts a module or a {module, function_name}.
See the "Compile callbacks" section below.

  
    
  
  @before_compile


A hook that will be invoked before the module is compiled.
Accepts a module or a {module, function_or_macro_name} tuple.
See the "Compile callbacks" section below.

  
    
  
  @behaviour


Note the British spelling!
Behaviours can be referenced by modules to ensure they implement
required specific function signatures defined by @callback.
For example, you could specify a URI.Parser behaviour as follows:
defmodule URI.Parser do
  @doc "Defines a default port"
  @callback default_port() :: integer

  @doc "Parses the given URL"
  @callback parse(uri_info :: URI.t()) :: URI.t()
end
And then a module may use it as:
defmodule URI.HTTP do
  @behaviour URI.Parser
  def default_port(), do: 80
  def parse(info), do: info
end
If the behaviour changes or URI.HTTP does not implement
one of the callbacks, a warning will be raised.
For detailed documentation, see the
behaviour typespec documentation.

  
    
  
  @impl (since v1.5.0)


To aid in the correct implementation of behaviours, you may optionally declare
@impl for implemented callbacks of a behaviour. This makes callbacks
explicit and can help you to catch errors in your code. The compiler will warn
in these cases:
	if you mark a function with @impl when that function is not a callback.

	if you don't mark a function with @impl when other functions are marked
with @impl. If you mark one function with @impl, you must mark all
other callbacks for that behaviour as @impl.


@impl works on a per-context basis. If you generate a function through a macro
and mark it with @impl, that won't affect the module where that function is
generated in.
@impl also helps with maintainability by making it clear to other developers
that the function is implementing a callback.
Using @impl, the example above can be rewritten as:
defmodule URI.HTTP do
  @behaviour URI.Parser

  @impl true
  def default_port(), do: 80

  @impl true
  def parse(info), do: info
end
You may pass either false, true, or a specific behaviour to @impl.
defmodule Foo do
  @behaviour Bar
  @behaviour Baz

  # Will warn if neither Bar nor Baz specify a callback named bar/0.
  @impl true
  def bar(), do: :ok

  # Will warn if Baz does not specify a callback named baz/0.
  @impl Baz
  def baz(), do: :ok
end
The code is now more readable, as it is now clear which functions are
part of your API and which ones are callback implementations. To reinforce this
idea, @impl true automatically marks the function as @doc false, disabling
documentation unless @doc is explicitly set.

  
    
  
  @compile


Defines options for module compilation. This is used to configure
both Elixir and Erlang compilers, as any other compilation pass
added by external tools. For example:
defmodule MyModule do
  @compile {:inline, my_fun: 1}

  def my_fun(arg) do
    to_string(arg)
  end
end
Multiple uses of @compile will accumulate instead of overriding
previous ones. See the "Compile options" section below.

  
    
  
  @deprecated (since v1.6.0)


Provides the deprecation reason for a function. For example:
defmodule Keyword do
  @deprecated "Use Kernel.length/1 instead"
  def size(keyword) do
    length(keyword)
  end
end
The Mix compiler automatically looks for calls to deprecated modules
and emit warnings during compilation.
Using the @deprecated attribute will also be reflected in the
documentation of the given function and macro. You can choose between
the @deprecated attribute and the documentation metadata to provide
hard-deprecations (with warnings) and soft-deprecations (without warnings):
This is a soft-deprecation as it simply annotates the documentation
as deprecated:
@doc deprecated: "Use Kernel.length/1 instead"
def size(keyword)
This is a hard-deprecation as it emits warnings and annotates the
documentation as deprecated:
@deprecated "Use Kernel.length/1 instead"
def size(keyword)
Currently @deprecated only supports functions and macros. However
you can use the :deprecated key in the annotation metadata to
annotate the docs of modules, types and callbacks too.
We recommend using this feature with care, especially library authors.
Deprecating code always pushes the burden towards library users. We
also recommend for deprecated functionality to be maintained for long
periods of time, even after deprecation, giving developers plenty of
time to update (except for cases where keeping the deprecated API is
undesired, such as in the presence of security issues).

  
    
  
  @doc and @typedoc


Provides documentation for the entity that follows the attribute.
@doc is to be used with a function, macro, callback, or
macrocallback, while @typedoc with a type (public or opaque).
Accepts one of these:
	a string (often a heredoc)
	false, which will make the entity invisible to documentation-extraction
tools like ExDoc
	a keyword list, since Elixir 1.7.0

For example:
defmodule MyModule do
  @typedoc "This type"
  @typedoc since: "1.1.0"
  @type t :: term

  @doc "Hello world"
  @doc since: "1.1.0"
  def hello do
    "world"
  end

  @doc """
  Sums `a` to `b`.
  """
  def sum(a, b) do
    a + b
  end
end
As can be seen in the example above, since Elixir 1.7.0 @doc and @typedoc
also accept a keyword list that serves as a way to provide arbitrary metadata
about the entity. Tools like ExDoc and
IEx may use this information to display annotations. A common use
case is the :since key, which may be used to annotate in which version the
function was introduced.
As illustrated in the example, it is possible to use these attributes
more than once before an entity. However, the compiler will warn if
used twice with binaries as that replaces the documentation text from
the preceding use. Multiple uses with keyword lists will merge the
lists into one.
Note that since the compiler also defines some additional metadata,
there are a few reserved keys that will be ignored and warned if used.
Currently these are: :opaque and :defaults.
Once this module is compiled, this information becomes available via
the Code.fetch_docs/1 function.

  
    
  
  @dialyzer


Defines warnings to request or suppress when using :dialyzer.
Accepts an atom, a tuple, or a list of atoms and tuples. For example:
defmodule MyModule do
  @dialyzer {:nowarn_function, [my_fun: 1]}

  def my_fun(arg) do
    M.not_a_function(arg)
  end
end
For the list of supported warnings, see :dialyzer module.
Multiple uses of @dialyzer will accumulate instead of overriding
previous ones.

  
    
  
  @external_resource


Specifies an external resource for the current module.
Sometimes a module embeds information from an external file. This
attribute allows the module to annotate which external resources
have been used.
Tools may use this information to ensure the module is recompiled
in case any of the external resources change, see for example:
mix compile.elixir.
The specified file path provided is interpreted as relative to
the folder containing the project's mix.exs, which is the
current working directory, not the file where @external_resource
is declared.
If the external resource does not exist, the module still has
a dependency on it, causing the module to be recompiled as soon
as the file is added.

  
    
  
  @file


Changes the filename used in stacktraces for the function or macro that
follows the attribute, such as:
defmodule MyModule do
  @doc "Hello world"
  @file "hello.ex"
  def hello do
    "world"
  end
end
Note that this is only valid for exceptions/diagnostics that come from the
definition inner scope (which includes its patterns and guards). For example:
defmodule MyModule do # <---- module definition
  @file "hello.ex"
  defp unused(a) do # <---- function definition
    "world" # <---- function scope
  end

  @file "bye.ex"
  def unused(_), do: true
end
If you run this code with the second "unused" definition commented, you will
see that hello.ex is used as the stacktrace when reporting warnings, but if
you uncomment it you'll see that the error will not mention bye.ex, because
it's a module-level error rather than an expression-level error.

  
    
  
  @moduledoc


Provides documentation for the current module.
defmodule MyModule do
  @moduledoc """
  A very useful module.
  """
  @moduledoc authors: ["Alice", "Bob"]
end
Accepts a string (often a heredoc) or false where @moduledoc false
will make the module invisible to documentation extraction tools like
ExDoc.
Similarly to @doc also accepts a keyword list to provide metadata
about the module. For more details, see the documentation of @doc
above.
Once this module is compiled, this information becomes available via
the Code.fetch_docs/1 function.

  
    
  
  @nifs (since v1.16.0)


A list of functions and their arities which will be overridden
by a native implementation (NIF).
defmodule MyLibrary.MyModule do
  @nifs [foo: 1, bar: 2]

  def foo(arg1), do: :erlang.nif_error(:not_loaded)
  def bar(arg1, arg2), do: :erlang.nif_error(:not_loaded)
end
See the Erlang documentation for more information:
https://www.erlang.org/doc/man/erl_nif

  
    
  
  @on_definition


A hook that will be invoked when each function or macro in the current
module is defined. Useful when annotating functions.
Accepts a module or a {module, function_name} tuple. The function
must take 6 arguments:
	the module environment
	the kind of the function/macro: :def, :defp, :defmacro, or :defmacrop
	the function/macro name
	the list of quoted arguments
	the list of quoted guards
	the quoted function body

If the function/macro being defined has multiple clauses, the hook will
be called for each clause.
Unlike other hooks, @on_definition will only invoke functions and
never macros. This is to avoid @on_definition callbacks from
redefining functions that have just been defined in favor of more
explicit approaches.
When just a module is provided, the function is assumed to be
__on_definition__/6.
Example
defmodule Hooks do
  def on_def(_env, kind, name, args, guards, body) do
    IO.puts("Defining #{kind} named #{name} with args:")
    IO.inspect(args)
    IO.puts("and guards")
    IO.inspect(guards)
    IO.puts("and body")
    IO.puts(Macro.to_string(body))
  end
end

defmodule MyModule do
  @on_definition {Hooks, :on_def}

  def hello(arg) when is_binary(arg) or is_list(arg) do
    "Hello" <> to_string(arg)
  end

  def hello(_) do
    :ok
  end
end

  
    
  
  @on_load


A hook that will be invoked whenever the module is loaded.
Accepts the function name (as an atom) of a function in the current module.
The function must have an arity of 0 (no arguments). If the function does
not return :ok, the loading of the module will be aborted.
For example:
defmodule MyModule do
  @on_load :load_check

  def load_check do
    if some_condition() do
      :ok
    else
      :abort
    end
  end

  def some_condition do
    false
  end
end

  
    
  
  @vsn


Specify the module version. Accepts any valid Elixir value, for example:
defmodule MyModule do
  @vsn "1.0"
end

  
    
  
  Struct attributes


	@derive - derives an implementation for the given protocol for the
struct defined in the current module

	@enforce_keys - ensures the given keys are always set when building
the struct defined in the current module


See defstruct/1 for more information on building and using structs.

  
    
  
  Typespec attributes


The following attributes are part of typespecs and are also built-in in
Elixir:
	@type - defines a type to be used in @spec
	@typep - defines a private type to be used in @spec
	@opaque - defines an opaque type to be used in @spec
	@spec - provides a specification for a function
	@callback - provides a specification for a behaviour callback
	@macrocallback - provides a specification for a macro behaviour callback
	@optional_callbacks - specifies which behaviour callbacks and macro
behaviour callbacks are optional
	@impl - declares an implementation of a callback function or macro

For detailed documentation, see the typespec documentation.

  
    
  
  Custom attributes


In addition to the built-in attributes outlined above, custom attributes may
also be added. Custom attributes are expressed using the @/1 operator followed
by a valid variable name. The value given to the custom attribute must be a valid
Elixir value:
defmodule MyModule do
  @custom_attr [some: "stuff"]
end
For more advanced options available when defining custom attributes, see
register_attribute/3.

  
    
  
  Compile callbacks


There are three compilation callbacks, invoked in this order:
@before_compile, @after_compile, and @after_verify.
They are described next.

  
    
  
  @before_compile


A hook that will be invoked before the module is compiled. This is
often used to change how the current module is being compiled.
Accepts a module or a {module, function_or_macro_name} tuple. The
function/macro must take one argument: the module environment. If
it's a macro, its returned value will be injected at the end of the
module definition before the compilation starts.
When just a module is provided, the function/macro is assumed to be
__before_compile__/1.
Callbacks will run in the order they are registered. Any overridable
definition will be made concrete before the first callback runs.
A definition may be made overridable again in another before compile
callback and it will be made concrete one last time after all callbacks
run.
Note: the callback function/macro must be placed in a separate module
(because when the callback is invoked, the current module does not yet exist).
Example
defmodule A do
  defmacro __before_compile__(_env) do
    quote do
      def hello, do: "world"
    end
  end
end

defmodule B do
  @before_compile A
end

B.hello()
#=> "world"

  
    
  
  @after_compile


A hook that will be invoked right after the current module is compiled.
Accepts a module or a {module, function_name} tuple. The function
must take two arguments: the module environment and its bytecode.
When just a module is provided, the function is assumed to be
__after_compile__/2.
Callbacks will run in the order they are registered.
Module functions expecting not yet compiled modules (such as definitions_in/1)
are still available at the time @after_compile is invoked.
Example
defmodule MyModule do
  @after_compile __MODULE__

  def __after_compile__(env, _bytecode) do
    IO.inspect(env)
  end
end

  
    
  
  @after_verify


A hook that will be invoked right after the current module is verified for
undefined functions, deprecations, etc. A module is always verified after
it is compiled. In Mix projects, a module is also verified when any of its
runtime dependencies change. Therefore this is useful to perform verification
of the current module while avoiding compile-time dependencies. Given the
callback is invoked under different scenarios, Elixir provides no guarantees
of when in the compilation cycle nor in which process the callback runs.
Accepts a module or a {module, function_name} tuple. The function
must take one argument: the module name. When just a module is provided,
the function is assumed to be __after_verify__/1.
Callbacks will run in the order they are registered.
Module functions expecting not yet compiled modules are no longer available
at the time @after_verify is invoked.
Example
defmodule MyModule do
  @after_verify __MODULE__

  def __after_verify__(module) do
    IO.inspect(module)
    :ok
  end
end

  
    
  
  Compile options


The @compile attribute accepts different options that are used by both
Elixir and Erlang compilers. Some of the common use cases are documented
below:
	@compile :debug_info - includes :debug_info regardless of the
corresponding setting in Code.get_compiler_option/1

	@compile {:debug_info, false} - disables :debug_info regardless
of the corresponding setting in Code.get_compiler_option/1. Note
disabling :debug_info is not recommended as it removes the ability
of the Elixir compiler and other tools to static analyse the code.
If you want to remove the :debug_info while deploying, tools like
mix release already do such by default.

	@compile {:inline, some_fun: 2, other_fun: 3} - inlines the given
name/arity pairs. Inlining is applied locally, calls from another
module are not affected by this option

	@compile {:autoload, false} - disables automatic loading of
modules after compilation. Instead, the module will be loaded after
it is dispatched to

	@compile {:no_warn_undefined, Mod} or
@compile {:no_warn_undefined, {Mod, fun, arity}} - does not warn if
the given module or the given Mod.fun/arity are not defined

A NaiveDateTime struct (without a time zone) and functions.
The NaiveDateTime struct contains the fields year, month, day, hour,
minute, second, microsecond and calendar. New naive datetimes can be
built with the new/2 and new/8 functions or using the
~N (see sigil_N/2) sigil:
iex> ~N[2000-01-01 23:00:07]
~N[2000-01-01 23:00:07]
The date and time fields in the struct can be accessed directly:
iex> naive = ~N[2000-01-01 23:00:07]
iex> naive.year
2000
iex> naive.second
7
We call them "naive" because this datetime representation does not
have a time zone. This means the datetime may not actually exist in
certain areas in the world even though it is valid.
For example, when daylight saving changes are applied by a region,
the clock typically moves forward or backward by one hour. This means
certain datetimes never occur or may occur more than once. Since
NaiveDateTime is not validated against a time zone, such errors
would go unnoticed.
Developers should avoid creating the NaiveDateTime structs directly
and instead, rely on the functions provided by this module as well
as the ones in third-party calendar libraries.

  
    
  
  Comparing naive date times


Comparisons in Elixir using ==/2, >/2, </2 and similar are structural
and based on the NaiveDate